

# noventa y tres **PROGRAMAS EN BASIC**

**PARA LA  
RESOLUCION DE PROBLEMAS  
tomo 1**

**FP - BUP - COU - UNIVERSIDAD**



**ANALISIS MATEMATICO  
Y  
CALCULO DE PROBABILIDADES Y ESTADISTICA**

**COMPATIBLES CON CUALQUIER MICRO**

**J.R. VIZMANOS**

# **PROGRAMAS EN BASIC**

**PARA LA  
RESOLUCION DE PROBLEMAS**

**FP - BUP - COU - UNIVERSIDAD**

**ANALISIS MATEMATICO  
CALCULO DE PROBABILIDADES  
ESTADISTICA**

**j.r. vizmanos**  
*Catedrático*



**PROGRAMAS EN BASIC PARA LA RESOLUCION DE PROBLEMAS  
ANALISIS MATEMATICO Y ESTADISTICA**

---

Prohibida la reproducción total  
o parcial de esta obra, por cualquier  
medio, sin autorización escrita  
del autor.

---

Reservados los derechos de edición,  
reproducción y adaptación a todos los  
países.

---

Es propiedad del autor.

Queda hecho el depósito que marca la Ley.

I.S.B.N.: 84-398-2143-3

Dep. Legal M-32.171-1984

---

Imprime FARESO

Paseo de la Dirección, 5  
28039 MADRID

---

**PEDIDOS AL AUTOR:**

c/ Príncipe de Vergara, 130, 4ªA.

28002 MADRID

Tel: 2 62 12 47

2 79 70 54

---

*A mi esposa, Marisa,  
ya que sin su comprensión y estímulo,  
este libro no hubiera sido posible,  
y a mis hijos,  
Marisa y José Ramón.*

## **agradecimientos**

Deseo expresar mi más profundo agradecimiento a Dario Crespo Gazapo, Catedrático de Matemáticas y gran amigo mio, que ha tenido la extraordinaria paciencia de revisar, con todo rigor el libro y que ha sabido, además de detectar erratas, proporcionar muy buenas ideas sobre algunos puntos concretos.

Al mismo tiempo, quiero agradecer a la Empresa INVESTRONICA, distribuidora en España de los microordenadores Spectrum, por autorizar la reproducción en la cubierta, de una fotografía del citado microordenador.

El autor





## prólogo

En la actualidad, son ya muchas las personas que poseen un microordenador personal en sus casas, pero generalmente, se limitan en la mayoría de las ocasiones, a utilizar el micro únicamente para la ejecución de juegos, y pocas cosas más; ya que cuando tratan de realizar sus propios programas, se encuentran, muchas veces, con grandes dificultades, produciéndoles un sentimiento de incapacidad, que con frecuencia, les lleva a dejar a un lado al microordenador.

Ahora bien, de la misma forma que todos estamos familiarizados a conducir un vehículo, ver la televisión, oír un transistor, etc, y no por ello, conocemos con profundidad como funciona el motor de un coche, o como es posible que lleguen hasta nuestra casa imágenes que se están produciendo a miles de kilómetros, o que complicados mecanismos hacen que una cajita pequeña, nos permitan oír voces que se emiten a cientos de kilómetros, etc, pues bien, de la misma forma, debe ser posible utilizar nuestro microordenador, sin tener grandes conocimientos de informática, sin más que servirnos de los programas que otras personas realizan para nosotros.

Es esta una de las ideas iniciales que nos indujo a la realización de este libro, junto con la consideración de que la producción en España de software es muy reducida y practicamente se restringe a programas de contabilidad, cálculo de estructuras, etc, o programas aislados de Matemáticas que generalmente, están traducidos del inglés, y que en muchas ocasiones son de dudosa utilidad.

Por ello y por nuestra relación con el mundo de las Matemáticas nos ha parecido muy interesante aplicar el micro a multitud de situaciones concretas, con las que se enfrenta un estudiante en la soledad de su cuarto.

Así por ejemplo, calcular:  $\sqrt[7]{2 + 3i}$  o bien hallar el valor de  $\int_0^1 e^x \sin x \, dx$ , o hallar los parámetros estadísticos que caracterizan a una distribución unidimensional de datos agrupados, etc.

El alumno realiza el ejercicio pero, en muchas ocasiones, no sabe si está bien, cuando tiene a su lado un "trasto" (microordenador) que convenientemente programado le puede servir de "profesor particular" guiándole en sus éxitos y mostrándole sus errores. Además de esta manera, puede familiarizarse con procedimientos que, a pesar de muchos, han de implantarse con carácter generalizado en la enseñanza.

Si además el usuario es de aquellos que empiezan a ser capaces de realizar sus propios programas, estos listados le pueden servir de guía y ejemplo, que con certeza, pronto será capaz de mejorarlos y hacerlos más suyos.

Así pues, este libro se ha iniciado con el Análisis Matemático seguido del Cálculo de Probabilidades y la Estadística, desarrollando sistemáticamente aquellas cuestiones que, con más frecuencia, se utilizan y que se hacen más susceptibles de ser tratadas con los microordenadores personales..

Ya desde estas páginas, anunciamos la próxima aparición de otro libro dedicado al Álgebra y la Geometría, con una "filosofía" totalmente análoga a este.

Por último nos queda pedir disculpas por los errores tipográficos que se hayan podido "escapar", así como los de programación, aceptando de antemano, con mucho gusto, todas las sugerencias que se nos hagan en este sentido.

El autor



## advertencias de interés

Los programas de este libro han sido probados y ejecutados en un microordenador NEW BRAIN, pero todos ellos pueden ser realizados, en cuanto a su extensión, en cualquier micro.

Hemos procurado utilizar un BASIC común a todos los micros, con el fin de que estos programas puedan ser ejecutados en cualquier microordenador de los existentes actualmente en el mercado, pero es posible, que en algunos programas sea necesario, introducir algunas modificaciones con arreglo al BASIC concreto de su micro.

Por ello vamos a exponer a continuación algunas de estas ideas.

- 1.- LET: Hoy en día es muy frecuente en casi todos los microordenadores, poder hacer asignación de variables en la forma:

```
120 A = B*C
```

pero todavía existen algunos micros, para los que es necesario escribir:

```
120 LET A = B*C
```

- 2.- ON GOTO : Si por ejemplo en una línea figura:

```
50 ON L GOTO 100,200,300
```

quiere decir que, según L sea igual a 1, 2 ó 3, se transferirá la ejecución a las líneas 100, 200 ó 300. Como algunos micros no permiten esta instrucción de bifurcación múltiple, se puede cambiar por:

```
50 IF L=1 THEN GOTO 100
```

```
52 IF L=2 THEN GOTO 200
```

```
54 IF L=3 THEN GOTO 300
```

- 3.- IF THEN : Algunos ordenadores solo admiten después de IF THEN, poner GOTO n° de línea, en cambio algunos permiten hacer:

```
60 IF B=1 THEN PRINT A
```

o también

```
80 IF A=3 THEN L=A*B
```

En estos programas se ha utilizado en escasas ocasiones esta última posibilidad, si su aparato no lo admite, modifíquelo con arreglo a lo expuesto.

- 4.- La función RND, no se utiliza del mismo modo en todo los micros, por ello repase con cuidado el manual del suyo.
- 5.- Se ha procurado casi siempre, incluir en cada línea una única instrucción, pero en algunas ocasiones hemos puesto:

100 PRINT: A=0

Si su ordenador no lo permite, modifíquelo del siguiente modo:

100 PRINT

102 A=0

- 6.- En algunas ejecuciones se han obtenido resultados como:  
1.5436783E-9, o bien -3.45243546E-12, evidentemente estos números son aproximaciones al número cero. Así pues, como tal hemos de tomar estos valores.
- 7.- Se ha procurado que los programas sean cortos y sencillos, ahora bien se podrían mejorar desde el punto de vista informático pero probablemente, perderían claridad desde el punto de vista matemático.
- 8.- Intencionadamente, no se ha incluido ningún programa sobre gráficos, dado que las instrucciones sobre gráficos, difieren mucho de unos ordenadores a otros.
- 9.- En general, no se ha incluido ninguna sentencia sobre el "borrado de la pantalla", para que sea el usuario el que en cada caso utilice esa posibilidad a su gusto.
- 10.- Si lo desea puede omitir todas las líneas que empiezan por REM ya que únicamente sirven para proporcionar información sobre el programa, pero el micro no las lee y además ocupan memoria.
- 11.- En caso de que, tras escribir cuidadosamente el programa y al ejecutar este, les proporcione mensaje de error, o resultados sensiblemente distintos, revise nuevamente, línea a línea, carácter a carácter, su listado y el del libro, revise las advertencias anteriores, tenga cuidado en no confundir cero con O, 1 con I, 2 con Z, 5 con S, y U con V., después de todo esto el programa deberá funcionar. Si a pesar de todo, el programa no "funciona", le aconsejamos que: lo deje de momento, para continuar con ello al cabo de unas horas, o bien lo revise con otra persona, o lo ponga en manos de alguien que sea capaz de detectar el fallo, pues generalmente será muy fácil, pero a veces, pueden ser extraordinariamente sutiles. Únicamente con mucha experiencia, habilidad y conocimiento profundo del BASIC de su micro será posible detectar estas sutilezas.

Ahora bien, esperamos que estas últimas advertencias no le ocurran prácticamente nunca, y deseamos que pueda Vd disfrutar de su micro sacándole todo el partido posible.

**PROGRAMAS  
DE**

**ANALISIS  
MATEMATICO**





# INTERES SIMPLE

## introducción:

Este programa sirve para calcular el capital final CF, en que se convierte un capital inicial CI, colocado al interés simple, siendo R el tanto por ciento anual, y T el número de años.

La expresión del capital final a interés simple anual es la siguiente:

$$CF = CI (1 + R T/100)$$

Este programa permite hallar, igualmente, el capital final colocado a interés simple mensual o diario, con arreglo a las siguientes expresiones:

$$CF = CI (1 + R M/1200) \quad (M = \text{n}^\circ \text{de meses})$$

$$CF = CI (1 + R D/36000) \quad (D = \text{n}^\circ \text{de días})$$

## listado del programa:

```
10 REM *****
20 REM
30 REM             INTERES     SIMPLE
40 REM
50 REM             ANUAL      MENSUAL      Y      DIARIO
60 REM
70 REM *****
100 PRINT
110 PRINT "ESTE PROGRAMA LE PERMITE CALCULAR : "
120 PRINT
130 PRINT
140 PRINT " 1.- INTERES SIMPLE ANUAL "
150 PRINT " 2.- INTERES SIMPLE MENSUAL "
160 PRINT " 3.- INTERES SIMPLE DIARIO "
170 PRINT
180 PRINT
190 PRINT "SEGUN LA OPCION DESEADA PULSE EL NUMERO QUE LE PRECEDE";
200 INPUT P
210 ON P GOTO 219,319,419
219 REM - CALCULO DEL INTERES SIMPLE ANUAL -
220 PRINT "CAPITAL INICIAL CI = ";
230 INPUT CI
240 PRINT "TANTO POR CIENTO ANUAL R = ";
250 INPUT R
260 PRINT "NUMERO DE AÑOS T = ";
```

```

270 INPUT T
280 CF= CI*(1+R*T/100)
290 PRINT
300 PRINT "EL CAPITAL FINAL AL CABO DE";T;"AÑOS ES : CF =" ;CF
310 GOTO 520
319 REM - CALCULO DEL INTERES SIMPLE MENSUAL -
320 PRINT "CAPITAL INICIAL CI =" ;
330 INPUT CI
340 PRINT "TANTO POR CIENTO ANUAL R =" ;
350 INPUT R
360 PRINT "NUMERO DE MESES M =" ;
370 INPUT M
380 CF= CI*(1+R*M/1200)
390 PRINT
400 PRINT "EL CAPITAL FINAL AL CABO DE";M;"MESES ES : CF =" ;CF
410 GOTO 520
419 REM - CALCULO DEL INTERES SIMPLE DIARIO -
420 PRINT "CAPITAL INICIAL CI =" ;
430 INPUT CI
440 PRINT "TANTO POR CIENTO ANUAL R =" ;
450 INPUT R
460 PRINT "NUMERO DE DIAS D =" ;
470 INPUT D
480 CF= CI*(1+R*D/36000)
490 PRINT
500 PRINT "EL CAPITAL FINAL AL CABO DE";D;"DIAS ES : CF =" ;CF
510 GOTO 520
520 REM - PARA REALIZAR MAS CALCULOS -
530 PRINT
540 PRINT
550 PRINT
560 PRINT
570 PRINT "DESEA CONTINUAR REALIZANDO OTRAS OPCIONES ? S/N";
580 INPUT C$
590 IF C$="N" THEN GOTO 610
600 GOTO 100
610 END

```

## ejercicios resueltos:

**1.** Un padre al nacer su hijo mayor ingresa 100000 ptas en un Banco al interés simple anual y al 12% . ¿Cuál será el capital formado cuando el hijo tenga 25 años?.

### SOLUCION :

ESTE PROGRAMA LE PERMITE CALCULAR :

- 1.- INTERES SIMPLE ANUAL
- 2.- INTERES SIMPLE MENSUAL
- 3.- INTERES SIMPLE DIARIO

SEGUN LA OPCION DESEADA PULSE EL NUMERO QUE LE PRECEDE? 1  
 CAPITAL INICIAL CI =? 100000  
 TANTO POR CIENTO ANUAL R =? 12  
 NUMERO DE AÑOS T =? 25

EL CAPITAL FINAL AL CABO DE 25 AÑOS ES : CF = 400000



**2.** Un comerciante impone por espacio de 65 meses, un millón de ptas en un Banco a un 7% de interés simple anual. ¿Qué cantidad de ptas tendrá al cabo de los 65 meses?.

**SOLUCION :**

ESTE PROGRAMA LE PERMITE CALCULAR :

- 1.- INTERES SIMPLE ANUAL
- 2.- INTERES SIMPLE MENSUAL
- 3.- INTERES SIMPLE DIARIO

SEGUN LA OPCION DESEADA PULSE EL NUMERO QUE LE PRECEDE? 2

CAPITAL INICIAL CI =? 1000000

TANTO POR CIENTO ANUAL R =? 7

NUMERO DE MESES M =? 65

EL CAPITAL FINAL AL CABO DE 65 MESES ES : CF = 1379166.7

**3.** Durante 20000 días una persona coloca un capital de 20000 ptas al 8.5% de interés simple anual. ¿Cuál será el capital final?.

**SOLUCION :**

ESTE PROGRAMA LE PERMITE CALCULAR :

- 1.- INTERES SIMPLE ANUAL
- 2.- INTERES SIMPLE MENSUAL
- 3.- INTERES SIMPLE DIARIO

SEGUN LA OPCION DESEADA PULSE EL NUMERO QUE LE PRECEDE? 3

CAPITAL INICIAL CI =? 20000

TANTO POR CIENTO ANUAL R =? 8.5

NUMERO DE DIAS D =? 20000

EL CAPITAL FINAL AL CABO DE 20000 DIAS ES : CF = 114444.44

# INTERES COMPUESTO

## introducción:

Este programa sirve para estudiar el interés compuesto, permitiendo realizar las siguientes opciones:

- 1.- Cálculo del capital final CF
- 2.- Cálculo del capital inicial CI
- 3.- Cálculo del tanto por ciento anual R
- 4.- Cálculo del número de años T.

La expresión del capital final es la siguiente:

$$CF = CI(1 + R/100)^T$$

De esta expresión se deducen todas las demás para realizar los cálculos de las otras opciones.

## listado del programa:

```
10 REM *****
20 REM
30 REM
40 REM          I N T E R E S      C O M P U E S T O
50 REM
60 REM
70 REM *****
100 PRINT
110 PRINT "ESTE PROGRAMA LE PERMITE CALCULAR : "
120 PRINT
130 PRINT
140 PRINT " 1.- EL CAPITAL FINAL"
150 PRINT " 2.- EL CAPITAL INICIAL"
160 PRINT " 3.- EL TANTO POR CIENTO ANUAL"
170 PRINT " 4.- EL NUMERO DE AÑOS"
180 PRINT
190 PRINT "SEGUN LA OPCION DESEADA PULSE EL NUMERO QUE LE PRECEDE";
200 INPUT P
205 PRINT
210 ON P GOTO 219,319,419,519
219 REM - CALCULO DEL CAPITAL FINAL -
220 PRINT "CAPITAL INICIAL CI = ";
230 INPUT CI
```

```

240 PRINT "TANTO POR CIENTO ANUAL R =";
250 INPUT R
260 PRINT "NUMERO DE AÑOS T =";
270 INPUT T
280 CF= CI*(1+R/100)^T
290 PRINT
300 PRINT "EL CAPITAL FINAL AL CABO DE";T;"AÑOS ES : CF =";CF
310 GOTO 630
319 REM - CALCULO DEL CAPITAL INICIAL -
320 PRINT "CAPITAL FINAL CF =";
330 INPUT CF
340 PRINT "TANTO POR CIENTO ANUAL R =";
350 INPUT R
360 PRINT "NUMERO DE AÑOS T =";
370 INPUT T
380 CI= CF/((1+R/100)^T)
390 PRINT
400 PRINT "EL CAPITAL INICIAL ES CI =";CI
410 GOTO 630
419 REM - CALCULO DEL TANTO POR CIENTO ANUAL -
420 PRINT "CAPITAL INICIAL CI =";
430 INPUT CI
440 PRINT "CAPITAL FINAL CF =";
450 INPUT CF
460 PRINT "NUMERO DE AÑOS T =";
470 INPUT T
480 R=(-1+(CF/CI)^(1/T))*100
490 PRINT
500 PRINT "EL TANTO POR CIENTO ANUAL ES : R =";R;"%"
510 GOTO 630
519 REM - CALCULO DE NUMERO DE AÑOS -
520 PRINT "CAPITAL INICIAL CI =";
530 INPUT CI
540 PRINT "CAPITAL FINAL CF =";
550 INPUT CF
560 PRINT "TANTO POR CIENTO ANUAL R =";
570 INPUT R
580 PRINT
590 T=(LOG(CF)-LOG(CI))/LOG(1+R/100)
600 PRINT
610 PRINT "EL NUMERO DE AÑOS ES T =";T
620 PRINT
630 REM - PARA REALIZAR MAS CALCULOS -
640 PRINT
650 PRINT
660 PRINT
670 PRINT
680 PRINT "DESEA CONTINUAR REALIZANDO OTRAS OPCIONES ? S/N";
690 INPUT C$
700 IF C$="N" THEN GOTO 720
710 GOTO 100
720 END

```

## ejercicios resueltos:

1. Se calcula que un bosque tiene  $24000 \text{ m}^3$  de madera y que aumenta en un 3,5% anual al interés compuesto. ¿cuántos años habrá de transcurrir para que tenga  $36265,64 \text{ m}^3$  de madera?.

**SOLUCION :**

ESTE PROGRAMA LE PERMITE CALCULAR :

- 1.- EL CAPITAL FINAL
- 2.- EL CAPITAL INICIAL
- 3.- EL TANTO POR CIENTO ANUAL
- 4.- EL NUMERO DE AÑOS

SEGUN LA OPCION DESEADA PULSE EL NUMERO QUE LE PRECEDE? 4

CAPITAL INICIAL CI =? 24000

CAPITAL FINAL CF =? 36265.64

TANTO POR CIENTO ANUAL R =? 3.5

EL NUMERO DE AÑOS ES T = 11.999994

**2.** ¿Qué capital inicial, prestado al 2% de interés compuesto durante 6 años, se convertirá en 750 ptas?.

**SOLUCION :**

CAPITAL FINAL CF =? 750

TANTO POR CIENTO ANUAL R =? 2

NUMERO DE AÑOS T =? 6

EL CAPITAL INICIAL ES CI = 665.97854

**3.** ¿A que tanto por ciento anual al interés compuesto se han de colocar 300000 ptas para que al cabo de 3 años se hayan convertido en 357306 ptas.

**SOLUCION :**

CAPITAL INICIAL CI =? 300000

CAPITAL FINAL CF =? 357306

NUMERO DE AÑOS T =? 3

EL TANTO POR CIENTO ANUAL ES : R = 6.0001187 %

# TABLAS DE INTERES COMPUESTO

## introducción:

Este programa sirve para imprimir tablas que proporcionen el valor a que se convierte una peseta colocada al R% anual durante T años. Por tanto es el valor de la expresión:  $(1 + R)^T$ .

En la línea 150 se pondrá el tanto por ciento anual deseado y en la línea 190 el periodo de años para el que se desea realizar el cálculo, en nuestro caso lo hemos hecho para los primeros 50 años.

## listado del programa:

```
10 REM *****
20 REM
30 REM          T A B L A          D E          ( 1 + R ) ^ T          P A R A          E L
40 REM
50 REM          C A L C U L O          D E L          I N T E R E S          C O M P U E S T O
60 REM
70 REM *****
100 PRINT
110 PRINT "          VALOR DE (1+R)^T"
120 PRINT
130 PRINT "*****"
140 PRINT
150 PRINT "NUM. DE AÑOS", "4%", "6%", "8%", "10%"
160 PRINT
170 PRINT "*****"
180 PRINT
190 FOR T=1 TO 50
200 A=(1+.04)^T
210 B=(1+.06)^T
220 C=(1+.08)^T
230 D=(1+.10)^T
240 PRINT T,A,B,C,D
250 NEXT T
260 END
```

1. Construir una tabla que exprese en que se convierte una peseta al 4%, 6%, 8% y 10% de interés compuesto durante los primeros cincuenta años.

**SOLUCION :**

VALOR DE  $(1+R)^T$

\*\*\*\*\*

AÑOS	4%	6%	8%	10%
1	1.04000000	1.06000000	1.08000000	1.10000000
2	1.08160000	1.12360000	1.16640000	1.21000000
3	1.12486400	1.19101600	1.25971200	1.33100000
4	1.16985856	1.26247696	1.36048896	1.46410000
5	1.21665290	1.33822558	1.46932808	1.61051000
6	1.26531902	1.41851911	1.58687432	1.77156100
7	1.31593178	1.50363026	1.71382427	1.94871710
8	1.36856905	1.59384808	1.85093021	2.14358881
9	1.42331181	1.68947896	1.99900463	2.35794769
10	1.48024429	1.79084770	2.15892500	2.59374246
11	1.53945406	1.89829856	2.33163900	2.85311671
12	1.60103222	2.01219647	2.51817012	3.13842838
13	1.66507351	2.13292826	2.71962373	3.45227122
14	1.73167645	2.26090396	2.93719363	3.79749834
15	1.80094351	2.39655819	3.17216912	4.17724818
16	1.87298125	2.54035169	3.42594265	4.59497299
17	1.94790050	2.69277279	3.70001806	5.05447029
18	2.02581652	2.85433916	3.99601951	5.55991732
19	2.10684918	3.02559951	4.31570107	6.11590906
20	2.19112314	3.20713548	4.66095715	6.72749996
21	2.27876807	3.39956360	5.03383372	7.40024996
22	2.36991879	3.60353742	5.43654042	8.14027495
23	2.46471554	3.81974967	5.87146366	8.95430245
24	2.56330417	4.04893465	6.34118075	9.84973270
25	2.66583633	4.29187073	6.84847521	10.83470597
26	2.77246979	4.54938297	7.39635323	11.91817657
27	2.88336858	4.82234595	7.98806149	13.10999422
28	2.99870332	5.11168670	8.62710641	14.42099365
29	3.11865145	5.41838791	9.31727492	15.86309301
30	3.24339751	5.74349118	10.06265691	17.44940232
31	3.37313341	6.08810065	10.86766947	19.19434255
32	3.50805875	6.45338669	11.73708302	21.11377681
33	3.64838110	6.84058989	12.67604966	23.22515449
34	3.79431635	7.25102529	13.69013364	25.54766995
35	3.94608900	7.68608680	14.78534433	28.10243694
36	4.10393256	8.14725201	15.96817188	30.91268064
37	4.26808986	8.63608713	17.24562563	34.00394870
38	4.43881346	9.15425236	18.62527568	37.40434358
39	4.61636600	9.70350751	20.11529774	41.14477795
40	4.80102064	10.28571795	21.72452156	45.25925574
41	4.99306146	10.90286104	23.46248328	49.78518133
42	5.19278392	11.55703269	25.33948195	54.76369947
43	5.40049528	12.25045466	27.36664050	60.24006941
44	5.61651509	12.98548194	29.55597175	66.26407637
45	5.84117569	13.76461085	31.92044948	72.89048399
46	6.07482272	14.59048751	34.47408545	80.17953242
47	6.31781563	15.46591676	37.23201229	88.19748566
48	6.57052826	16.39387177	40.21057327	97.01723424
49	6.83334939	17.37750407	43.42741913	106.7189577
50	7.10668336	18.42015432	46.90161267	117.3908534

# ANUALIDADES DE AMORTIZACION

## introducción:

El presente programa sirve para calcular la anualidad de amortización que es una cantidad fija que se paga al final de cada año, para extinguir con estas anualidades y sus intereses (compuesto) el importe de una deuda contraída y los intereses por ella producidos.

El programa permite calcular las siguientes opciones:

- 1.- La anualidad de amortización A
- 2.- La deuda contraída D
- 3.- El numero de años T

La expresión que nos da la anualidad de amortización es la siguiente:

$$A = \frac{D R (1 + R)^T}{(1 + R)^T - 1}$$

De esta expresión se deducen las demás para efectuar los cálculos de las otras opciones.

## listado del programa:

```
10 REM *****
20 REM
30 REM
40 REM          A N U A L I D A D E S      D E      A M O R T I Z A C I O N
50 REM
60 REM
70 REM *****
100 PRINT
110 PRINT "ESTE PROGRAMA LE PERMITE CALCULAR : "
120 PRINT
130 PRINT
140 PRINT " 1.- LA ANUALIDAD DE AMORTIZACION"
150 PRINT " 2.- LA DEUDA CONTRAIDA"
160 PRINT " 3.- EL NUMERO DE AÑOS"
170 PRINT
180 PRINT
190 PRINT "SEGUN LA OPCION DESEADA PULSE EL NUMERO QUE LE PRECEDE";
200 INPUT P
```

```

210 ON P GOTO 219,319,419
219 REM - CALCULO DE LA ANUALIDAD DE AMORTIZACION -
220 PRINT "DEUDA CONTRAIDA D =";
230 INPUT D
240 PRINT "TANTO POR CIENTO ANUAL R =";
250 INPUT R
260 PRINT "NUMERO DE AÑOS T =";
270 INPUT T
280 A=(D*R/100*(1+R/100)^T)/((1+R/100)^T -1)
290 PRINT
300 PRINT "LA ANUALIDAD NECESARIA PARA CANCELAR LA DEUDA CONTRAIDA ES : A =";A
310 GOTO 530
319 REM - CALCULO DE LA DEUDA CONTRAIDA -
320 PRINT "ANUALIDAD DE AMORTIZACION A =";
330 INPUT A
340 PRINT "TANTO POR CIENTO ANUAL R =";
350 INPUT R
360 PRINT "NUMERO DE AÑOS T =";
370 INPUT T
380 D=(A*((1+R/100)^T -1))/((R/100)*(1+R/100)^T)
390 PRINT
400 PRINT "LA DEUDA CONTRAIDA ES D =";D
410 GOTO 530
419 REM - CALCULO DE NUMERO DE AÑOS -
420 PRINT "ANUALIDAD A PAGAR A =";
430 INPUT A
440 PRINT "DEUDA CONTRAIDA D =";
450 INPUT D
460 PRINT "TANTO POR CIENTO ANUAL R =";
470 INPUT R
480 PRINT
490 T=(LOG(A)-LOG(A-D*R/100))/LOG(1+R/100)
500 PRINT
510 PRINT "EL NUMERO DE AÑOS ES T =";T
520 GOTO 530
530 REM - PARA CONTINUAR -
540 PRINT
550 PRINT
560 PRINT
570 PRINT
580 PRINT "DESEA CONTINUAR REALIZANDO OTRAS OPCIONES ? S/N";
590 INPUT C$
600 IF C$="N" THEN GOTO 620
610 GOTO 100
620 END

```

## ejercicios resueltos:

1. ¿Cuál será la anualidad que habra que pagar durante dos años para amortizar una deuda de 93633,39 ptas junto con sus intereses al 4,5% anual?

### S O L U C I O N :

```

SEGUN LA OPCION DESEADA PULSE EL NUMERO QUE LE PRECEDE? 1
DEUDA CONTRAIDA D =? 93633.39
TANTO POR CIENTO ANUAL R =? 4.5
NUMERO DE AÑOS T =? 2

```

LA ANUALIDAD NECESARIA PARA CANCELAR LA DEUDA CONTRAIDA ES : A = 50000.001



**2.** ¿Cuál será la deuda contraída por una persona que se compromete a pagar una anualidad de 367,2 ptas durante tres años, sabiendo que la deuda se incrementa con sus intereses al 5% anual?.

**S O L U C I O N :**

SEGUN LA OPCION DESEADA PULSE EL NUMERO QUE LE PRECEDE? 2

ANUALIDAD DE AMORTIZACION A =? 367.2

TANTO POR CIENTO ANUAL R =? 5

NUMERO DE AÑOS T =? 3

LA DEUDA CONTRAIDA ES D = 999.97668

**3.** ¿En cuantos años se amortizará una deuda de 50000 ptas al 6% anual sabiendo que la anualidad que se paga al final de cada año es de 14429,5 ptas?.

**S O L U C I O N :**

SEGUN LA OPCION DESEADA PULSE EL NUMERO QUE LE PRECEDE? 3

ANUALIDAD A PAGAR A =? 14429.5

DEUDA CONTRAIDA D =? 50000

TANTO POR CIENTO ANUAL R =? 6

EL NUMERO DE AÑOS ES T = 4.0000233

# ANUALIDADES DE CAPITALIZACION

## introducción:

El presente programa sirve para calcular las anualidades de capitalización que son unas cantidades fijas que se pagan al principio de cada año para formar junto con sus intereses ( compuesto) un capital al cabo de un cierto número de años.

El programa permite calcular las siguientes opciones:

- 1.- La anualidad de capitalización A
- 2.- El capital C formado al cabo de T años
- 3.- El número de años T.

La expresión que nos da el capital final formado es la siguiente:

$$C = \frac{A (1 + R) ((1 + R)^T - 1)}{R}$$

De esta expresión se deducen las demás para efectuar los cálculos de las otras opciones.

## listado del programa:

```
10 REM *****
20 REM
30 REM
40 REM      A N U A L I D A D E S      D E      C A P I T A L I Z A C I O N
50 REM
60 REM
70 REM *****
100 PRINT
110 PRINT "ESTE PROGRAMA LE PERMITE CALCULAR : "
120 PRINT
130 PRINT
140 PRINT " 1.- LA ANUALIDAD DE CAPITALIZACION"
150 PRINT " 2.- EL CAPITAL FORMADO AL CABO DE T ANOS"
160 PRINT " 3.- EL NUMERO DE AÑOS"
170 PRINT
180 PRINT
190 PRINT "SEGUN LA OPCION DESEADA PULSE EL NUMERO QUE LE PRECEDE";
200 INPUT P
```

```

210 ON P GOTO 219,319,419
219 REM - CALCULO DE LA ANUALIDAD DE CAPITALIZACION -
220 PRINT "CAPITAL QUE SE DESEA FORMAR C =";
230 INPUT C
240 PRINT "TANTO POR CIENTO ANUAL R =";
250 INPUT R
260 PRINT "NUMERO DE AÑOS T =";
270 INPUT T
280 A=(C*R/100)/(((1+R/100)*((1+R/100)^T-1))
290 PRINT
300 PRINT "LA ANUALIDAD NECESARIA PARA FORMAR EL CAPITAL";C;"ES : A =";A
310 GOTO 530
319 REM - CALCULO DEL CAPITAL FORMADO AL CABO DE T AÑOS -
320 PRINT "ANUALIDAD DE CAPITALIZACION A =";
330 INPUT A
340 PRINT "TANTO POR CIENTO ANUAL R =";
350 INPUT R
360 PRINT "NUMERO DE AÑOS T =";
370 INPUT T
380 C=A*((1+R/100)*((1+R/100)^T-1))*100/R
390 PRINT
400 PRINT "EL CAPITAL FORMADO AL CABO DE";T;"AÑOS ES : C =";C
410 GOTO 530
419 REM - CALCULO DE NUMERO DE AÑOS -
420 PRINT "ANUALIDAD A PAGAR A =";
430 INPUT A
440 PRINT "CAPITAL FORMADO AL CABO DE T AÑOS C =";
450 INPUT C
460 PRINT "TANTO POR CIENTO ANUAL R =";
470 INPUT R
480 PRINT
487 X=(C*R/100)/(A*(1+R/100))
490 T=LOG(1+X)/LOG(1+R/100)
500 PRINT
510 PRINT "EL NUMERO DE AÑOS ES T =";T
520 GOTO 530
530 REM - PARA CONTINUAR -
540 PRINT
550 PRINT
560 PRINT
570 PRINT
580 PRINT "DESEA CONTINUAR REALIZANDO OTRAS OPCIONES ? S/N";
590 INPUT C$
600 IF C$="N" THEN GOTO 620
610 GOTO 100
620 END

```

## ejercicios resueltos:

1. ¿Cuál será la anualidad de capitalización que habrá de pagarse para que al cabo de 10 años, se forme junto con los intereses al 12% anual un capital de 1000000 de ptas?

### S O L U C I O N :

SEGUN LA OPCION DESEADA PULSE EL NUMERO QUE LE PRECEDE? 1  
CAPITAL QUE SE DESEA FORMAR C =? 1000000

TANTO POR CIENTO ANUAL  $R = ?$  12  
NUMERO DE AÑOS  $T = ?$  10

LA ANUALIDAD NECESARIA PARA FORMAR EL CAPITAL 1000000 ES :  $A = 50878.718$

**2.** Una persona a los 40 años decide ahorrar al comienzo de cada año 25000 ptas colocadas al 10% anual. ¿Qué capital tendrá al cumplir los 70 años?.

**S O L U C I O N :**

SEGUN LA OPCION DESEADA. PULSE EL NUMERO QUE LE PRECEDE? 2  
ANUALIDAD DE CAPITALIZACION  $A = ?$  25000  
TANTO POR CIENTO ANUAL  $R = ?$  10  
NUMERO DE AÑOS  $T = ?$  30

EL CAPITAL FORMADO AL CABO DE 30 AÑOS ES :  $C = 4523585.6$

**3.** ¿Cuántos años habrán de transcurrir, para que se forme un capital de 1197878,8 teniendo en cuenta que se abona una anualidad de capitalización de 30000 ptas, y que el tanto por ciento anual es de 11,5 ?

**S O L U C I O N :**

SEGUN LA OPCION DESEADA PULSE EL NUMERO QUE LE PRECEDE? 3  
ANUALIDAD A PAGAR  $A = ?$  30000  
CAPITAL FORMADO AL CABO DE  $T$  AÑOS  $C = ?$  1197878.8  
TANTO POR CIENTO ANUAL  $R = ?$  11.5

EL NUMERO DE AÑOS ES  $T = 15$

# SUCESIONES ARITMETICAS

## introducción:

Este programa sirve para estudiar sucesiones (progresiones) aritméticas. El programa permite calcular:

- 1.- El término n-esimo AN
- 2.- La razón de la progresión D
- 3.- El término primero A1
- 4.- El número de términos N
- 5.- Interpolar H medios aritméticos entre dos términos dados
- 6.- La suma de N términos consecutivos S

La expresión que permite hallar el término n-esimo en función del primero, el número de términos y la razón, es la siguiente:

$$AN = A1 + (N - 1) D$$

De esta expresión se deduce las otras para realizar los cálculos de las otras opciones.

## listado del programa:

```
10 REM *****
20 REM
30 REM          P R O G R E S I O N E S          A R I T M E T I C A S
40 REM
50 REM *****
100 PRINT
110 PRINT "EL PRESENTE PROGRAMA LE PERMITE CALCULAR LO SIGUIENTE : "
120 PRINT
130 PRINT "1.- CALCULO DEL TERMINO N-ESIMO"
140 PRINT "2.- CALCULO DE LA RAZON"
150 PRINT "3.- CALCULO DEL TERMINO PRIMERO"
160 PRINT "4.- CALCULO DEL NUMERO DE TERMINOS"
170 PRINT "5.- INTERPOLAR H MEDIOS ARITMETICOS ENTRE DOS TERMINOS DADOS"
180 PRINT "6.- CALCULO DE LA SUMA DE N TERMINOS CONSECUTIVOS"
190 PRINT
200 PRINT
210 PRINT "SEGUN LA OPCION DESEADA PULSE EL NUMERO QUE LE PRECEDE" ;
220 INPUT P
230 PRINT
```

```
240 ON P GOTO 249,349,449,549,649,789
249 REM - CALCULO DEL TERMINO N-ESIMO -
250 PRINT "TERMINO PRIMERO A1 =";
260 INPUT A1
270 PRINT "NUMERO DE TERMINOS N =";
280 INPUT N
290 PRINT "RAZON O DIFERENCIA D =";
300 INPUT D
310 AN=A1+(N-1)*D
320 PRINT
330 PRINT "EL TERMINO N-ESIMO DE LA PROGRESION ES: AN =";AN
340 GOTO 890
349 REM - CALCULO DE LA RAZON -
350 PRINT "TERMINO PRIMERO A1 =";
360 INPUT A1
370 PRINT "TERMINO N-ESIMO AN =";
380 INPUT AN
390 PRINT "NUMERO DE TERMINOS N =";
400 INPUT N
410 D=(AN-A1)/(N-1)
420 PRINT
430 PRINT "LA RAZON O DIFERENCIA DE LA PROGRESION ES: D =";D
440 GOTO 890
449 REM - CALCULO DEL TERMINO PRIMERO -
450 PRINT "TERMINO N-ESIMO AN =";
460 INPUT AN
470 PRINT "RAZON O DIFERENCIA D =";
480 INPUT D
490 PRINT "NUMERO DE TERMINOS N =";
500 INPUT N
510 A1=AN-(N-1)*D
530 PRINT "EL PRIMER TERMINO DE LA PROGRESION ES : A1 =";A1
540 GOTO 890
549 REM - CALCULO DEL NUMERO DE TERMINOS -
550 PRINT "TERMINO PRIMERO A1 =";
560 INPUT A1
570 PRINT "TERMINO N-ESIMO AN =";
580 INPUT AN
590 PRINT "RAZON O DIFERENCIA D =";
600 INPUT D
610 N=1+(AN-A1)/D
620 PRINT
630 PRINT "EL NUMERO DE TERMINOS DE LA PROGRESION ES : N =";N
640 GOTO 890
649 REM - CALCULO DEL NUMERO DE TERMINOS -
650 PRINT "CUANTOS MEDIOS DESEA INTERPOLAR ";
660 INPUT H
670 PRINT "INTRODUZCA EL PRIMERO Y EL ULTIMO TERMINO ENTRE LOS QUE QUIERE INTERPOLAR";H;"MEDIOS ARITMETICOS";
680 INPUT P,U
690 D=(U-P)/(H+1)
700 PRINT
710 PRINT "LA RAZON DE LA PROGRESION ES : D =";D
720 PRINT
730 PRINT "LOS MEDIOS ARITMETICOS INTERPOLADOS SON : "
740 FOR I=1 TO H
750 T=P+I*D
760 PRINT T,
770 NEXT I
780 GOTO 890
```

```

789 REM - CALCULO DE LA SUMA DE N TERMINOS CONSECUTIVOS -
790 PRINT "CUANTOS TERMINOS CONSECUTIVOS DESEA SUMAR: N = ";
800 INPUT N
810 PRINT "TERMINO PRIMERO A1 = ";
820 INPUT A1
830 PRINT "RAZON O DIFERENCIA D = ";
840 INPUT D
850 S=N*(A1+(N-1)*(D/2))
860 PRINT
870 PRINT "LA SUMA DE LOS";N;"TERMINOS CONSECUTIVOS DE LA PROGRESION ES: S =";S
880 GOTO 890
890 REM - PARA REALIZAR MAS CALCULOS -
900 PRINT
910 PRINT
920 PRINT
930 PRINT
940 PRINT "DESEA CONTINUAR REALIZANDO OTRAS OPCIONES ? S/N";
950 INPUT C$
960 IF C$="N" THEN GOTO 980
970 GOTO 100
980 END

```

## ejercicios resueltos:

1. Interpoliar 32 medios aritméticos entre -7 y 125.

### SOLUCION :

EL PRESENTE PROGRAMA LE PERMITE CALCULAR LO SIGUIENTE :

- 1.- CALCULO DEL TERMINO N-ESIMO
- 2.- CALCULO DE LA RAZON
- 3.- CALCULO DEL TERMINO PRIMERO
- 5.- INTERPOLAR H MEDIOS ARITMETICOS ENTRE DOS TERMINOS DADOS
- 6.- CALCULO DE LA SUMA DE N TERMINOS CONSECUTIVOS

SEGUN LA OPCION DESEADA PULSE EL NUMERO QUE LE PRECEDE? 5

CUANTOS MEDIOS DESEA INTERPOLAR ? 32

INTRODUZCA EL PRIMERO Y EL ULTIMO TERMINO ENTRE LOS QUE QUIERE INTERPOLAR 32 MEDIOS ARITMETICOS?-7 125

LA RAZON DE LA PROGRESION ES : D = 4

LOS MEDIOS ARITMETICOS INTERPOLADOS SON :

-3	1	5	9	13	17	21	25	29	33
37	41	45	49	53	57	61	65	69	73
77	81	85	89	93	97	101	105	109	113
117	121								

2. El primer término de una progresión aritmética es 7 y el término 20 es -31. Hallar la razón de la progresión.

### SOLUCION :

SEGUN LA OPCION DESEADA PULSE EL NUMERO QUE LE PRECEDE? 2

TERMINO PRIMERO A1 =? 7

TERMINO N-ESIMO AN =?-31

NUMERO DE TERMINOS  $N = ?$  20

LA RAZON O DIFERENCIA DE LA PROGRESION ES:  $D = -2$

**3.** Sabiendo que el primer término de una progresión aritmética es 1003 y que la razón es 59, calcular la suma de los 16 primeros términos.

**S O L U C I O N :**

SEGUN LA OPCION DESEADA PULSE EL NUMERO QUE LE PRECEDE? 6

CUANTOS TERMINOS CONSECUTIVOS DESEA SUMAR:  $N = ?$  16

TERMINO PRIMERO  $A_1 = ?$  1003

RAZON O DIFERENCIA  $D = ?$  59

LA SUMA DE LOS 16 TERMINOS CONSECUTIVOS DE LA PROGRESION ES:  $S = 23128$

**4.** El primer término de una progresión aritmética es 1003 y la razón es 59, hallar el término 16.

**S O L U C I O N :**

SEGUN LA OPCION DESEADA PULSE EL NUMERO QUE LE PRECEDE? 1

TERMINO PRIMERO  $A_1 = ?$  1003

NUMERO DE TERMINOS  $N = ?$  16

RAZON O DIFERENCIA  $D = ?$  59

EL TERMINO N-ESIMO DE LA PROGRESION ES:  $A_N = 1888$



# SUCESIONES GEOMETRICAS

## introducción:

Este programa sirve para operar con sucesiones (progresiones) geométricas. El programa permite calcular:

- 1.- El término n-esimo  $A_N$
- 2.- La razón de la progresión  $R$
- 3.- El término primero  $A_1$
- 4.- El número de términos  $N$
- 5.- Interpoliar  $H$  medios geométricos entre dos términos dados.
- 6.- La suma de  $N$  términos consecutivos ( $|R| \geq 1$ )
- 7.- La suma de los infinitos términos ( $|R| < 1$ )
- 8.- El producto de  $N$  términos consecutivos.

La expresión que permite hallar el término n-esimo en función del primero, el número de términos y la razón es la siguiente:

$$A_N = A_1 R^{(N-1)}$$

De esta expresión se deducen las otras para realizar el cálculo de las otras opciones.

## listado del programa:

```
10 REM *****
20 REM
30 REM          P R O G R E S I O N E S          G E O M E T R I C A S
40 REM
50 REM *****
100 PRINT
110 PRINT "EL PRESENTE PROGRAMA LE PERMITE CALCULAR LO SIGUIENTE : "
120 PRINT
130 PRINT "1.- CALCULO DEL TERMINO N-ESIMO"
140 PRINT "2.- CALCULO DE LA RAZON"
150 PRINT "3.- CALCULO DEL TERMINO PRIMERO"
160 PRINT "4.- CALCULO DEL NUMERO DE TERMINOS"
170 PRINT "5.- INTERPOLAR H MEDIOS GEOMETRICOS ENTRE DOS TERMINOS DADOS"
180 PRINT "6.- CALCULO DE LA SUMA DE N TERMINOS CONSECUTIVOS ABS(R)>=1"
190 PRINT "7.- CALCULO DE LA SUMA DE LOS INFINITOS TERMINOS (-1<R<1)"
200 PRINT "8.- CALCULO DEL PRODUCTO DE N TERMINOS CONSECUTIVOS"
202 PRINT
204 PRINT
```

```

210 PRINT "SEGUN LA OPCION DESEADA PULSE EL NUMERO QUE LE PRECEDE" ;
220 INPUT P
230 PRINT
240 ON P GOTO 249,349,449,549,649,789,889,969
249 REM - CALCULO DEL TERMINO N-ESIMO -
250 PRINT "TERMINO PRIMERO A1 =";
260 INPUT A1
270 PRINT "NUMERO DE TERMINOS N =";
280 INPUT N
290 PRINT "RAZON R =";
300 INPUT R
310 AN=A1*R^(N-1)
320 PRINT
330 PRINT "EL TERMINO N-ESIMO DE LA PROGRESION ES: AN =";AN
340 GOTO 1070
349 REM - CALCULO DE LA RAZON -
350 PRINT "TERMINO PRIMERO A1 =";
360 INPUT A1
370 PRINT "TERMINO N-ESIMO AN =";
380 INPUT AN
390 PRINT "NUMERO DE TERMINOS N =";
400 INPUT N
410 R=(AN/A1)^(1/(N-1))
420 PRINT
430 PRINT "LA RAZON O DIFERENCIA DE LA PROGRESION ES: R =";R
440 GOTO 1070
449 REM - CALCULO DEL TERMINO PRIMERO -
450 PRINT "TERMINO N-ESIMO AN =";
460 INPUT AN
470 PRINT "RAZON R =";
480 INPUT R
490 PRINT "NUMERO DE TERMINOS N =";
500 INPUT N
510 A1=AN/(R^(N-1))
530 PRINT "EL PRIMER TERMINO DE LA PROGRESION ES : A1 =";A1
540 GOTO 1070
549 REM - CALCULO DEL NUMERO DE TERMINOS -
550 PRINT "TERMINO PRIMERO A1 =";
560 INPUT A1
570 PRINT "TERMINO N-ESIMO AN =";
580 INPUT AN
590 PRINT "RAZON O DIFERENCIA R =";
600 INPUT R
610 N=1+(LOG(AN)-LOG(A1))/LOG(R)
620 PRINT
630 PRINT "EL NUMERO DE TERMINOS DE LA PROGRESION ES : N =";N
640 GOTO 1070
649 REM - INTERPOLAR H MEDIOS GEOMETRICOS ENTRE DOS TERMINOS DADOS -
650 PRINT "CUANTOS MEDIOS DESEA INTERPOLAR ";
660 INPUT H
670 PRINT "INTRODUZCA EL PRIMERO Y EL ULTIMO TERMINO ENTRE LOS QUE QUIERE INTERPOLAR";H;"MEDIOS GEOMETRICOS";
680 INPUT P,U
690 R=(U/P)^(1/(H+1))
700 PRINT
710 PRINT "LA RAZON DE LA PROGRESION ES : R =";R
720 PRINT
730 PRINT "LOS MEDIOS GEOMETRICOS INTERPOLADOS SON : "
740 FOR I=1 TO H
750 T=P*R^I
760 PRINT T,

```

```

770 NEXT I
780 GOTO 1070
789 REM - CALCULO DE LA SUMA DE N TERMINOS CONSECUTIVOS (ABS(R)>=1)
790 PRINT "CUANTOS TERMINOS CONSECUTIVOS DESEA SUMAR: N =";
800 INPUT N
810 PRINT "TERMINO PRIMERO A1 =";
820 INPUT A1
830 PRINT "RAZON R =";
840 INPUT R
850 S=A1*((R^N)-1)/(R-1)
860 PRINT
870 PRINT "LA SUMA DE LOS";N;"TERMINOS CONSECUTIVOS DE LA PROGRESION ES: S =";S
880 GOTO 1070
889 REM - CALCULO DE LA SUMA DE LOS INFINITOS TERMINOS (-1<R<1) -
890 PRINT "TERMINO PRIMERO A1 =";
900 INPUT A1
910 PRINT "RAZON R =";
920 INPUT R
930 S=A1/(1-R)
940 PRINT
950 PRINT "LA SUMA DE LOS INFINITOS TERMINOS DE LA PROGRESION ES: S =";S
960 GOTO 1070
969 REM - CALCULO DEL PRODUCTO DE N TERMINOS CONSECUTIVOS-
970 PRINT "TERMINO PRIMERO A1 =";
980 INPUT A1
990 PRINT "TERMINO N-ESIMO AN =";
1000 INPUT AN
1010 PRINT "NUMERO DE TERMINOS N =";
1020 INPUT N
1030 Y=(A1*AN)^(N/2)
1040 PRINT
1050 PRINT "EL PRODUCTO DE";N;"TERMINOS CONSECUTIVOS ES : P =";Y
1060 GOTO 1070
1070 REM - PARA REALIZAR MAS CALCULOS -
1080 PRINT
1090 PRINT
1100 PRINT
1110 PRINT
1120 PRINT "DESEA CONTINUAR REALIZANDO OTRAS OPCIONES ? S/N";
1130 INPUT C#
1140 IF C#="N" THEN GOTO 1160
1150 GOTO 100
1160 END

```

## ejercicios resueltos:

### 1. Interpolar 5 medios geométricos entre 7 y 5103

#### SOLUCION :

EL PRESENTE PROGRAMA LE PERMITE CALCULAR LO SIGUIENTE :

- 1.- CALCULO DEL TERMINO N-ESIMO
- 2.- CALCULO DE LA RAZON
- 4.- CALCULO DEL NUMERO DE TERMINOS
- 5.- INTERPOLAR H MEDIOS GEOMETRICOS ENTRE DOS TERMINOS DADOS
- 6.- CALCULO DE LA SUMA DE N TERMINOS CONSECUTIVOS  $ABS(R) \geq 1$
- 7.- CALCULO DE LA SUMA DE LOS INFINITOS TERMINOS  $(-1 < R < 1)$
- 8.- CALCULO DEL PRODUCTO DE N TERMINOS CONSECUTIVOS

SEGUN LA OPCION DESEADA PULSE EL NUMERO QUE LE PRECEDE? 5  
 CUANTOS MEDIOS DESEA INTERPOLAR ? 5  
 INTRODUZCA EL PRIMERO Y EL ULTIMO TERMINO ENTRE LOS QUE QUIERE INTERPOLAR 5 MEDIOS GEOMETRICOS? 7      5103

LA RAZON DE LA PROGRESION ES :  $R = 3$

LOS MEDIOS GEOMETRICOS INTERPOLADOS SON :

21      63      189      567      1701

**2.** En una progresión geométrica el primer término es 0,03125 y el término  $n$ -ésimo es 512, sabiendo que la razón es 2, hallar el número de términos.

**SOLUCION :**

SEGUN LA OPCION DESEADA PULSE EL NUMERO QUE LE PRECEDE? 4  
 TERMINO PRIMERO     $A_1 = ?$  .03125  
 TERMINO N-ESIMO     $A_N = ?$  512  
 RAZON O DIFERENCIA  $R = ?$  2

EL NUMERO DE TERMINOS DE LA PROGRESION ES :  $N = 15$

**3.** Hallar el producto de los 10 primeros términos de una progresión geométrica sabiendo que el primero es 3 y el décimo es 1536.

**SOLUCION :**

SEGUN LA OPCION DESEADA PULSE EL NUMERO QUE LE PRECEDE? 8  
 TERMINO PRIMERO     $A_1 = ?$  3  
 TERMINO N-ESIMO     $A_N = ?$  1536  
 NUMERO DE TERMINOS  $N = ?$  10

EL PRODUCTO DE 10 TERMINOS CONSECUTIVOS ES :  $P = 2.077602E+18$

**4.** Hallar la suma de los infinitos términos de una progresión geométrica, sabiendo que el primer término es 6 y la razón  $1/3$ .

**SOLUCION :**

SEGUN LA OPCION DESEADA PULSE EL NUMERO QUE LE PRECEDE? 7  
 TERMINO PRIMERO     $A_1 = ?$  6  
 RAZON     $R = ?$  .33333333

LA SUMA DE LOS INFINITOS TERMINOS DE LA PROGRESION ES:  $S = 9$

# FORMACION DE LOS TERMINOS DE UNA SUCESION

## introducción:

Este programa sirve para formar los términos de una sucesión de números reales.

En la línea 160 se indica los términos que se desean formar, también se podría formarlos de 10 en 10 etc, sin más que añadir STEP 10. En la línea 170 se escribe la expresión del término general de la sucesión.

## listado del programa:

```
10 REM *****
20 REM
30 REM          S U C E S I O N E S      D E      N U M E R O S      R E A L E S
40 REM
50 REM          F O R M A C I O N  D E  T E R M I N O S  C O N O C I D O  E L  T E R M I N O  G E N E R A L
60 REM
70 REM *****
100 PRINT
110 PRINT "CUANTOS TERMINOS DESEA FORMAR ";
120 INPUT P
130 PRINT
140 PRINT "INDICE", " T E R M I N O "
150 PRINT "*****", "*****"
160 FOR N=1 TO P
170 AN=3*N/(N*N+1)
180 PRINT " N=";N,"      A";N;"=";AN
190 NEXT N
200 END
```

## ejercicios resueltos:

1. Formar los 60 primeros términos de la sucesión de números reales que tiene por término general: 
$$a_n = \frac{3n}{n^2 + 1}$$

**S O L U C I O N :**

CUANTOS TERMINOS DESEA FORMAR ? 60

INDICE	TERMINO
*****	*****
N= 1	A 1 = 1.5
N= 2	A 2 = 1.2
N= 3	A 3 = .9
N= 4	A 4 = .70588235
N= 5	A 5 = .57692308
N= 6	A 6 = .48648649
N= 7	A 7 = .42
N= 8	A 8 = .36923077
N= 9	A 9 = .32926829
N= 10	A 10 = .2970297
N= 11	A 11 = .2704918
N= 12	A 12 = .24827586
N= 13	A 13 = .22941176
N= 14	A 14 = .21319797
N= 15	A 15 = .19911504
N= 16	A 16 = .18677043
N= 17	A 17 = .17586207
N= 18	A 18 = .16615385
N= 19	A 19 = .15745856
N= 20	A 20 = .14962594
N= 21	A 21 = .14253394
N= 22	A 22 = .13608247
N= 23	A 23 = .13018868
N= 24	A 24 = .12478336
N= 25	A 25 = .11980831
N= 26	A 26 = .11521418
N= 27	A 27 = .1109589
N= 28	A 28 = .10700637
N= 29	A 29 = .10332542
N= 30	A 30 = .099889012
N= 31	A 31 = .096673597
N= 32	A 32 = .093658537
N= 33	A 33 = .090825688
N= 34	A 34 = .088159032
N= 35	A 35 = .085644372
N= 36	A 36 = .083269082
N= 37	A 37 = .081021898
N= 38	A 38 = .078892734
N= 39	A 39 = .076872536
N= 40	A 40 = .074953154
N= 41	A 41 = .073127229
N= 42	A 42 = .071388102
N= 43	A 43 = .06972973
N= 44	A 44 = .068146618
N= 45	A 45 = .066633761
N= 46	A 46 = .065186585
N= 47	A 47 = .063800905
N= 48	A 48 = .062472885
N= 49	A 49 = .061199001
N= 50	A 50 = .05997601
N= 51	A 51 = .058800922
N= 52	A 52 = .05767098
N= 53	A 53 = .05658363
N= 54	A 54 = .05553651
N= 55	A 55 = .054527429
N= 56	A 56 = .053554351
N= 57	A 57 = .052615385
N= 58	A 58 = .051708767
N= 59	A 59 = .050832855
N= 60	A 60 = .049986115

# SUCESION DEL NUMERO e

## introducción:

Este breve programa es un caso particular del programa anterior, sirve para formar términos de la sucesión de números reales que tiene por límite el número e.

La expresión del término general es:

$$a_n = \left(1 + \frac{1}{n}\right)^n$$

## listado del programa:

```
10 REM *****
20 REM
30 REM      S U C E S I O N E S      D E      N U M E R O S      R E A L E S
40 REM
50 REM      FORMACION DE TERMINOS DE LA SUCESION DEL NUMERO E
60 REM
70 REM *****
100 PRINT
110 PRINT "CUANTOS TERMINOS DESEA FORMAR ";
120 INPUT P
130 PRINT
140 PRINT "INDICE", " T E R M I N O "
150 PRINT "*****", "*****"
160 FOR N=1 TO P
170 AN=(1+1/N)^N
180 PRINT " N=";N,"      A";N;"=";AN
190 NEXT N
200 END
```

## ejercicios resueltos:

1. Formar los 40 primeros términos de la sucesión del número e.

S O L U C I O N :

CUANTOS TERMINOS DESEA FORMAR ? 40

INDICE	T E R M I N O
*****	*****
N= 1	A 1 = 2
N= 2	A 2 = 2.25
N= 3	A 3 = 2.3703704
N= 4	A 4 = 2.4414062
N= 5	A 5 = 2.48832
N= 6	A 6 = 2.5216264
N= 7	A 7 = 2.5464997
N= 8	A 8 = 2.5657845
N= 9	A 9 = 2.5811748
N= 10	A 10 = 2.5937425
N= 11	A 11 = 2.604199
N= 12	A 12 = 2.6130353
N= 13	A 13 = 2.6206009
N= 14	A 14 = 2.6271516
N= 15	A 15 = 2.6328787
N= 16	A 16 = 2.6379285
N= 17	A 17 = 2.6424144
N= 18	A 18 = 2.6464258
N= 19	A 19 = 2.6500343
N= 20	A 20 = 2.6532977
N= 21	A 21 = 2.6562632
N= 22	A 22 = 2.6589699
N= 23	A 23 = 2.6614501
N= 24	A 24 = 2.6637313
N= 25	A 25 = 2.6658363
N= 26	A 26 = 2.667785
N= 27	A 27 = 2.669594
N= 28	A 28 = 2.6712779
N= 29	A 29 = 2.6728492
N= 30	A 30 = 2.6743188
N= 31	A 31 = 2.6756963
N= 32	A 32 = 2.6769901
N= 33	A 33 = 2.6782077
N= 34	A 34 = 2.6793554
N= 35	A 35 = 2.6804393
N= 36	A 36 = 2.6814644
N= 37	A 37 = 2.6824355
N= 38	A 38 = 2.6833566
N= 39	A 39 = 2.6842316
N= 40	A 40 = 2.6850638



# OPERACIONES CON SUCESIONES

## introducción:

Este programa permite realizar las operaciones de suma, diferencia producto y cociente de dos sucesiones de números reales. Para ello es preciso introducir en las líneas 250 y 260 los términos generales de las dos sucesiones.

En la línea 240 se indica el número de términos que se quiere formar.

Cuando la sucesión  $b_n$  tiende a cero o bien algunos de sus términos son ceros, la sucesión cociente  $a_n/b_n$  no está definida.

## listado del programa:

```
10 REM *****
20 REM
30 REM          O P E R A C I O N E S      C O N      S U C E S I O N E S
40 REM
50 REM                      D E      N U M E R O S      R E A L E S
60 REM
70 REM *****
100 PRINT
110 PRINT "EL PRESENTE PROGRAMA LE PERMITE CALCULAR"
120 PRINT
130 PRINT
140 PRINT "1.- SUMA DE DOS SUCESIONES CONOCIDOS SUS TERMINOS GENERALES"
150 PRINT "2.- DIFERENCIA DE DOS SUCESIONES CONOCIDOS SUS TERMINOS GENERALES"
160 PRINT "3.- PRODUCTO DE DOS SUCESIONES CONOCIDOS SUS TERMINOS GENERALES"
170 PRINT "4.- COCIENTE DE DOS SUCESIONES CONOCIDOS SUS TERMINOS GENERALES"
180 PRINT
190 PRINT "CUANTOS TERMINOS QUIERE OBTENER";
200 INPUT P
210 PRINT
220 PRINT "INDICE", "      SUMA"      , " DIFERENC" , " PRODUCTO" , " COCIENTE"
230 PRINT "*****", "*****", "*****", "*****", "*****"
240 FOR N=1 TO P
250 AN=
260 BN=
```

```

270 SN=AN+BN
280 DN=AN-BN
290 PN=AN*BN
300 CN=AN/BN
310 PRINT N,SN,DN,PN,CN
320 NEXT N
330 END

```

## ejercicios resueltos:

### 1. Dadas las sucesiones de términos generales:

$a_n = \frac{n^2}{3n+2}$ ,  $b_n = \frac{2n^2}{n^2+1}$  formar los cuarenta primeros términos de las sucesiones siguientes:

$$a_n + b_n, \quad a_n - b_n, \quad a_n \cdot b_n, \quad a_n/b_n$$

### **SOLUCION :**

EL PRESENTE PROGRAMA LE PERMITE CALCULAR

- 1.- SUMA DE DOS SUCESIONES CONOCIDOS SUS TERMINOS GENERALES
- 2.- DIFERENCIA DE DOS SUCESIONES CONOCIDOS SUS TERMINOS GENERALES
- 3.- PRODUCTO DE DOS SUCESIONES CONOCIDOS SUS TERMINOS GENERALES
- 4.- COCIENTE DE DOS SUCESIONES CONOCIDOS SUS TERMINOS GENERALES

CUANTOS TERMINOS QUIERE OBTENER? 40

INDICE	SUMA	DIFERENC	PRODUCTO	COCIENTE
*****				
1	1.20000000	- .80000000	.20000000	.20000000
2	2.10000000	- 1.10000000	.80000000	.31250000
3	2.61818182	- .98181818	1.47272727	.45454545
4	3.02521008	- .73949580	2.15126050	.60714286
5	3.39366516	- .45248869	2.82805430	.76470588
6	3.74594595	- .14594595	3.50270270	.92500000
7	4.09043478	.17043478	4.17565217	1.08695652
8	4.43076923	.49230769	4.84733728	1.25000000
9	4.76871320	.81749369	5.51808242	1.41379310
10	5.10519802	1.14480198	6.18811881	1.57812500
11	5.44074941	1.47353630	6.85761124	1.74285714
12	5.77568058	1.80326679	7.52667877	1.90789474
13	6.11018651	2.13371593	8.19540889	2.07317073
14	6.44439317	2.46469774	8.86386710	2.23863636
15	6.77838448	2.79608360	9.53210318	2.40425532
16	7.11221790	3.12778210	10.20015564	2.57000000
17	7.44593364	3.45972674	10.86805465	2.73584906
18	7.77956044	3.79186813	11.53582418	2.90178571
19	8.11311921	4.12416893	12.20348347	3.06779661
20	8.44662537	4.45660043	12.87104818	3.23387097
21	8.78009050	4.78914027	13.53853115	3.40000000
22	9.11352335	5.12177077	14.20594300	3.56617647
23	9.44693064	5.45447781	14.87329259	3.73239437

---

24	9.78031758	5.78724999	15.54058738	3.89864865
25	10.11368823	6.12007800	16.20783370	4.06493506
26	10.44704579	6.45295421	16.87503693	4.23125000
27	10.78039280	6.78587226	17.54220168	4.39759036
28	11.11373130	7.11882684	18.20933195	4.56395349
29	11.44706290	7.45181350	18.87643118	4.73033708
30	11.78038894	7.78482845	19.54350239	4.89673913
31	12.11371047	8.11786848	20.21054820	5.06315789
32	12.44702837	8.45093081	20.87757093	5.22959184
33	12.78034335	8.78401308	21.54457262	5.39603960
34	13.11365601	9.11711322	22.21155508	5.56250000
35	13.44696681	9.45022945	22.87851992	5.72897196
36	13.78027616	9.78336020	23.54546856	5.89545455
37	14.11358439	10.11650410	24.21240230	6.06194690
38	14.44689178	10.44965995	24.87932228	6.22844828
39	14.78019855	10.78282667	25.54622953	6.39495798
40	15.11350488	11.11600332	26.21312499	6.56147541

# SUMA DE TERMINOS DE UNA SUCESION

## introducción:

Este programa sirve para calcular la suma de  $p$  términos consecutivos de una sucesión de números reales. En la línea 210 se introduce la expresión del término general de la sucesión a sumar. En caso de que queramos sumar a partir del término 200, en lugar del término primero, modificaremos la línea 200 escribiendo  $N = 200$ .

El programa imprime el índice de cada término, el término y la suma de los términos acumulados hasta el correspondiente al índice.

## listado del programa:

```
10 REM *****
20 REM
30 REM      S U M A      D E      P      T E R M I N O S      C O N S E C U T I V O S
40 REM
50 REM                      D E      U N A      S U C E S I O N      D E      N U M E R O S      R E A L E S
60 REM
70 REM *****
100 PRINT
110 PRINT "INTRODUZCA EL TERMINO GENERAL DE LA SUCESION EN LA LINEA 210"
120 PRINT
130 PRINT "CUANTOS TERMINOS DE LA SUCESION DESEA SUMAR";
140 INPUT P
150 PRINT
160 PRINT
170 PRINT "INDICE", "TERMINO AN", "SUMA SN"
180 PRINT "*****", "*****", "*****"
190 SN=0
200 N=1
210 AN=
220 SN=SN+AN
230 PRINT N, AN, SN
240 N=N+1
250 IF N<=P THEN GOTO 210
260 END
```

## ejercicios resueltos:

1. Calcular la suma de los 30 primeros términos de la sucesión que tiene por término general:

$$a_n = \left(1 + \frac{1}{n}\right)^n$$

### S O L U C I O N :

INTRODUZCA EL TERMINO GENERAL DE LA SUCESION EN LA LINEA 210

CUANTOS TERMINOS DE LA SUCESION DESEA SUMAR? 30

INDICE	TERMINO AN	SUMA	SN
*****			
1	2.000000000	2.00000000	
2	2.250000000	4.25000000	
3	2.370370370	6.62037037	
4	2.441406250	9.06177662	
5	2.488320000	11.55009662	
6	2.521626372	14.07172299	
7	2.546499698	16.61822269	
8	2.565784514	19.18400720	
9	2.581174790	21.76518199	
10	2.593742463	24.35892446	
11	2.604199010	26.96312347	
12	2.613035288	29.57615875	
13	2.620600891	32.19675964	
14	2.627151554	34.82391120	
15	2.632878717	37.45678992	
16	2.637928498	40.09471841	
17	2.642414375	42.73713279	
18	2.646425819	45.38355861	
19	2.650034323	48.03359293	
20	2.653297707	50.68689064	
21	2.656263212	53.34315385	
22	2.658969856	56.00212370	
23	2.661450125	58.66357383	
24	2.663731263	61.32730509	
25	2.665836334	63.99314143	
26	2.667784969	66.66092640	
27	2.669593980	69.33052038	
28	2.671277852	72.00179823	
29	2.672849153	74.67464738	
30	2.674318785	77.34896617	

2. Calcular la suma de los cuadrados de los 30 primeros números naturales, es decir:

$$S = 1^2 + 2^2 + 3^2 + \dots + 30^2$$

### S O L U C I O N :

INTRODUZCA EL TERMINO GENERAL DE LA SUCESION EN LA LINEA 210

CUANTOS TERMINOS DE LA SUCESION DESEA SUMAR? 30

INDICE	TERMINO AN	SUMA	SN
*****			
1	1.0		1.0
2	4.0		5.0
3	9.0		14.0
4	16.0		30.0
5	25.0		55.0
6	36.0		91.0
7	49.0		140.0
8	64.0		204.0
9	81.0		285.0
10	100.0		385.0
11	121.0		506.0
12	144.0		650.0
13	169.0		819.0
14	196.0		1015.0
15	225.0		1240.0
16	256.0		1496.0
17	289.0		1785.0
18	324.0		2109.0
19	361.0		2470.0
20	400.0		2870.0
21	441.0		3311.0
22	484.0		3795.0
23	529.0		4324.0
24	576.0		4900.0
25	625.0		5525.0
26	676.0		6201.0
27	729.0		6930.0
28	784.0		7714.0
29	841.0		8555.0
30	900.0		9455.0

**3.** Calcular la suma de los cubos de los 50 primeros números naturales es decir:

$$S = 1^3 + 2^3 + 3^3 + \dots + 50^3$$

**S O L U C I O N :**

INTRODUZCA EL TERMINO GENERAL DE LA SUCESSION EN LA LINEA 210

CUANTOS TERMINOS DE LA SUCESSION DESEA SUMAR? 50

INDICE	TERMINO AN	SUMA	SN
*****			
1	1.0		1.0
2	8.0		9.0
3	27.0		36.0
4	64.0		100.0
5	125.0		225.0
6	216.0		441.0
7	343.0		784.0
8	512.0		1296.0
9	729.0		2025.0
10	1000.0		3025.0

---

11	1331.0	4356.0
12	1728.0	6084.0
13	2197.0	8281.0
14	2744.0	11025.0
15	3375.0	14400.0
16	4096.0	18496.0
17	4913.0	23409.0
18	5832.0	29241.0
19	6859.0	36100.0
20	8000.0	44100.0
21	9261.0	53361.0
22	10648.0	64009.0
23	12167.0	76176.0
24	13824.0	90000.0
25	15625.0	105625.0
26	17576.0	123201.0
27	19683.0	142884.0
28	21952.0	164836.0
29	24389.0	189225.0
30	27000.0	216225.0
31	29791.0	246016.0
32	32768.0	278784.0
33	35937.0	314721.0
34	39304.0	354025.0
35	42875.0	396900.0
36	46656.0	443556.0
37	50653.0	494209.0
38	54872.0	549081.0
39	59319.0	608400.0
40	64000.0	672400.0
41	68921.0	741321.0
42	74088.0	815409.0
43	79507.0	894916.0
44	85184.0	980100.0
45	91125.0	1071225.0
46	97336.0	1168561.0
47	103823.0	1272384.0
48	110592.0	1382976.0
49	117649.0	1500625.0
50	125000.0	1625625.0

# SUCESION DE FIBONACCI

## introducción:

Este breve programa sirve para formar los términos de la sucesión de Fibonacci. Esta sucesión de números reales se define de la siguiente forma:  $a_1 = 1$ ,  $a_2 = 1$ ,  $a_n = a_{n-1} + a_{n-2}$

Vemos que los dos primeros términos han de estar separados y el resto siguen la ley de recurrencia descrita en la relación:  $a_n = a_{n-1} + a_{n-2}$

Este programa es igualmente válido, salvo en los dos primeros términos, para cualquier otra sucesión cuya ley de recurrencia sea la misma.

## listado del programa:

```
10 REM *****
20 REM
30 REM          S U C E S I O N   D E   F I B O N A C C I
40 REM
50 REM *****
100 PRINT
110 PRINT "CUANTOS TERMINOS DE LA SUCESION DESEA FORMAR";
120 INPUT P
130 PRINT
140 N=1
150 A=1
160 PRINT "INDICE", "TERMINO"
170 PRINT "*****", "*****"
180 PRINT N,A
190 N=2
200 B=1
210 PRINT N,B
220 N=N+1
230 C=B+A
240 PRINT N,C
250 A=B
260 B=C
270 IF N<P THEN GOTO 220
280 END
```



**1.** Formar los 35 primeros términos de la sucesión de FIBONACCI.

**S O L U C I O N :**

CUANTOS TERMINOS DE LA SUCESION DESEA FORMAR? 35

INDICE	TERMINO
*****	*****
1	1
2	1
3	2
4	3
5	5
6	8
7	13
8	21
9	34
10	55
11	89
12	144
13	233
14	377
15	610
16	987
17	1597
18	2584
19	4181
20	6765
21	10946
22	17711
23	28657
24	46368
25	75025
26	121393
27	196418
28	317811
29	514229
30	832040
31	1346269
32	2178309
33	3524578
34	5702887
35	9227465

# OPERACIONES CON NUMEROS COMPLEJOS EN FORMA BINOMICA

## introducción:

Este programa sirve para sumar, restar, multiplicar y dividir números complejos escritos en forma binómica. En las líneas 140 y 160 se introducen cada uno de los dos números complejos, escribiendo primero la parte real separada por una coma de la parte imaginaria.

## listado del programa:

```
10 REM *****
20 REM
30 REM   OPERACIONES   CON   NUMEROS   COMPLEJOS
40 REM
50 REM                               EN   FORMA   BINOMICA
60 REM
70 REM *****
100 PRINT
110 PRINT "INTRODUZCA LA PARTE REAL Y LA IMAGINARIA SEPARADAS POR UNA COMA"
120 PRINT
129 REM -ENTRADA DE DATOS -
130 PRINT "COMPLEJO Z1 = ";
140 INPUT X1,Y1
150 PRINT "COMPLEJO Z2 = ";
160 INPUT X2,Y2
170 PRINT
180 PRINT
190 PRINT
199 REM - SUMA DE COMPLEJOS -
200 XS=X1+X2
210 YS=Y1+Y2
220 PRINT "COMPLEJO SUMA           : Z1+Z2 =";XS;
230 IF YS>=0 THEN PRINT "+";
240 PRINT YS;"i"
250 PRINT
259 REM - DIFERENCIA DE COMPLEJOS -
260 XR=X1-X2
270 YR=Y1-Y2
280 PRINT "COMPLEJO DIFERENCIA : Z1-Z2 =";XR;
290 IF YR>=0 THEN PRINT "+";
300 PRINT YR;"i"
```

```

310 PRINT
319 REM - PRODUCTO DE COMPLEJOS -
320 XP=X1*X2-Y1*Y2
330 YP=X1*Y2+Y1*X2
340 PRINT "COMPLEJO PRODUCTO : Z1*Z2 =" ;XP;
350 IF YP>=0 THEN PRINT "+";
360 PRINT YP;"i"
370 PRINT
379 REM - COCIENTE DE COMPLEJOS -
380 IF X2=0 AND Y2=0 THEN GOTO 460
390 M=X2*X2+Y2*Y2
400 XC=(X1*X2+Y1*Y2)/M
410 YC=(-X1*Y2+Y1*X2)/M
420 PRINT "COMPLEJO COCIENTE : Z1/Z2 =" ;XC;
430 IF YC>=0 THEN PRINT "+";
440 PRINT YC;"i"
450 STOP
460 PRINT "NO ES POSIBLE LA DIVISION POR EL COMPLEJO NULO"
470 END

```

## ejercicios resueltos:

1. Dados los complejos:  $z_1 = 12 - 8i$  y  $z_2 = 3 - 7i$ , calcular:  
 $z_1 + z_2$ ,  $z_1 - z_2$ ,  $z_1 \cdot z_2$ ,  $z_1/z_2$

### SOLUCION :

INTRODUZCA LA PARTE REAL Y LA IMAGINARIA SEPARADAS POR UNA COMA

```

COMPLEJO Z1 =? 12      -8
COMPLEJO Z2 =? 3       -7

```

COMPLEJO SUMA :  $Z1+Z2 = 15 -15 i$

COMPLEJO DIFERENCIA :  $Z1-Z2 = 9 -1 i$

COMPLEJO PRODUCTO :  $Z1*Z2 = -20 -108 i$

COMPLEJO COCIENTE :  $Z1/Z2 = 1.5862069 + 1.0344828 i$

2. Dados los complejos:  $z_1 = 7 + 13i$  y  $z_2 = 0 + 0i$ , calcular:  
 $z_1 + z_2$ ,  $z_1 - z_2$ ,  $z_1 \cdot z_2$ ,  $z_1/z_2$

### SOLUCION :

INTRODUZCA LA PARTE REAL Y LA IMAGINARIA SEPARADAS POR UNA COMA

```

COMPLEJO Z1 =? 7      13
COMPLEJO Z2 =? 0       0

```

COMPLEJO SUMA :  $z_1 + z_2 = 7 + 13 i$

COMPLEJO DIFERENCIA :  $z_1 - z_2 = 7 + 13 i$

COMPLEJO PRODUCTO :  $z_1 * z_2 = 0 + 0 i$

NO ES POSIBLE LA DIVISION POR EL COMPLEJO NULO

**3.** Dados los complejos:  $z_1 = 6 + 7 i$  y  $z_2 = -5 + 17 i$ , calcular:  $z_1 + z_2$ ,  $z_1 - z_2$ ,  $z_1 \cdot z_2$ ,  $z_1 / z_2$ .

**S O L U C I O N :**

INTRODUZCA LA PARTE REAL Y LA IMAGINARIA SEPARADAS POR UNA COMA

COMPLEJO Z1 =? 6 7  
COMPLEJO Z2 =? -5 17

COMPLEJO SUMA :  $z_1 + z_2 = 1 + 24 i$

COMPLEJO DIFERENCIA :  $z_1 - z_2 = 11 - 10 i$

COMPLEJO PRODUCTO :  $z_1 * z_2 = -149 + 67 i$

COMPLEJO COCIENTE :  $z_1 / z_2 = .28343949 - .43630573 i$

# CONVERSION DE COORDENADAS CARTESIANAS A POLARES Y VICEVERSA

## introducción:

Este programa sirve para pasar un número complejo expresado en coordenadas cartesianas  $(x,y)$  a coordenadas polares  $(r,w)$  y reciprocamente. Donde  $r$  es el módulo del número complejo y  $w$  es su argumento.

## listado del programa:

```
10 REM *****
20 REM
30 REM             CONVERSION DE COORDENADAS CARTESIANAS A POLARES
40 REM
50 REM             Y RECIPROCAMENTE
60 REM
70 REM *****
80 PRINT
90 PRINT
100 PRINT
110 PRINT "DE POLARES A CARTESIANAS PULSE 1"
120 PRINT "DE CARTESIANAS A POLARES PULSE 2"
130 INPUT N
140 IF N=1 THEN GOTO 330
150 PRINT
160 PRINT
169 REM - DE CARTESIANAS A POLARES -
170 PRINT "INTRODUZCA LAS COORDENADAS CARTESIANAS SEPARADAS POR UNA COMA" ;
180 INPUT X,Y
190 R=SQR(X*X+Y*Y)
200 IF X=0 AND Y=0 GOTO 240
210 IF X=0 THEN X=(.1)^30
220 IF X>0 AND Y=0 GOTO 240
230 W=Y/X
240 V=0
250 V=ATN(W)
260 IF X<0 THEN V=V+PI
270 IF V<0 THEN V=V+2*PI
280 PRINT
290 PRINT "LAS COORDENADAS POLARES SON:"
300 PRINT "MODULO      =";R
310 PRINT "ARGUMENTO   =";V;"radianes"
320 GOTO 80
```

```

329 REM - DE POLARES A CARTESIANAS -
330 PRINT "INTRODUZCA LAS COORDENADAS POLARES"
340 PRINT "MODULO      R =";
350 INPUT R
360 PRINT "ARGUMENTO  W =";
370 INPUT W
380 X=R*COS(W)
390 Y=R*SIN(W)
400 PRINT
410 PRINT "LAS COORDENADAS CARTESIANAS SON : (";X;",";Y;")"
420 GOTO 80
430 END

```

## ejercicios resueltos:

**1.** Pasar a coordenadas cartesianas los siguientes números complejos expresados en forma polar:  $z = 3_\pi$  ;  $z = 4_{\pi/2}$

### S O L U C I O N :

```

DE POLARES A CARTESIANAS PULSE 1
DE CARTESIANAS A POLARES PULSE 2
? 1
INTRODUZCA LAS COORDENADAS POLARES
MODULO      R =? 3
ARGUMENTO  W =? 3.1415927

```

LAS COORDENADAS CARTESIANAS SON : (-3 , -1.3824576E-07 ) = (-3, 0)

Obsérvese que -1.3824566E-07 es una aproximación de cero.

```

DE POLARES A CARTESIANAS PULSE 1
DE CARTESIANAS A POLARES PULSE 2
? 1
INTRODUZCA LAS COORDENADAS POLARES
MODULO      R =? 4
ARGUMENTO  W =? 1.5707963

```

LAS COORDENADAS CARTESIANAS SON : ( 1.0734733E-07 , 4 ) = (0, 4)

**2.** Pasar a coordenadas polares los siguientes números complejos expresados en forma cartesiana:  $z = (5, 5)$ ;  $z = (-7, 5)$

### S O L U C I O N :

INTRODUZCA LAS COORDENADAS CARTESIANAS SEPARADAS POR UNA COMA? 5

5

```

LAS COORDENADAS POLARES SON:
MODULO      = 7.0710678
ARGUMENTO = .78539816 radianes

```

DE POLARES A CARTESIANAS PULSE 1  
DE CARTESIANAS A POLARES PULSE 2  
? 2

INTRODUZCA LAS COORDENADAS CARTESIANAS SEPARADAS POR UNA COMA?-7

5

LAS COORDENADAS POLARES SON:

MODULO = 8.6023253

ARGUMENTO = 2.5213432 radianes

# POTENCIACION Y RADICACION DE NUMEROS COMPLEJOS EN FORMA BINOMICA

## introducción:

Este programa permite calcular la potencia de un número complejo en forma binómica y la radicación de un número complejo en forma binómica. El cálculo se realiza en polares para lo cual en la línea 590 se inicia una subrutina para la conversión de cartesianas a polares

## listado del programa:

```
10 REM *****
20 REM
30 REM          P O T E N C I A C I O N          Y          R A D I C A C I O N
40 REM
50 REM          D E          N U M E R O S          C O M P L E J O S
60 REM
70 REM          E N          F O R M A          B I N O M I C A
80 REM
90 REM *****
100 PRINT
110 PRINT "EL PRESENTE PROGRAMA LE PERMITE CALCULAR : "
120 PRINT
130 PRINT
140 PRINT "1.-POTENCIA DE UN NUMERO COMPLEJO ESCRITO EN FORMA BINOMICA"
150 PRINT "2.-RADICACION DE UN NUMERO COMPLEJO ESCRITO EN FORMA BINOMICA"
160 PRINT
170 PRINT "SEGUN LA OPCION DESEADA PULSE EL NUMERO QUE LE PRECEDE"
180 INPUT P
190 PRINT
200 IF P=2 THEN GOTO 359
209 REM - POTENCIA DE UN NUMERO COMPLEJO EN FORMA BINOMICA -
210 PRINT "PARTE REAL E IMAGINARIA DEL NUMERO COMPLEJO SEPARADAS POR UNA COMA";
220 INPUT X,Y
230 PRINT "EXPONENTE DE LA POTENCIA DESEADA N =";
240 INPUT N
250 GOSUB 590
260 R1=R^N
270 V1=V*N
280 X1=R1*COS(V1)
290 Y1=R1*SIN(V1)
300 PRINT
310 PRINT
```



```

320 PRINT "LA POTENCIA PEDIDA ES =";X1;
330 IF Y1>=0 THEN PRINT "+";
340 PRINT Y1;"i"
350 GOTO 700
359 REM - RADICACION DE UN NUMERO COMPLEJO EN FORMA BINOMICA -
360 PRINT "PARTE REAL E IMAGINARIA DEL NUMERO COMPLEJO SEPARADAS POR UNA COMA";
370 INPUT X,Y
380 PRINT "INDICE DE LA RAIZ DESEADA N =";
390 INPUT N
400 GOSUB 590
410 R2=R^(1/N)
420 V2=V/N
430 PRINT
440 PRINT
450 PRINT "ORDEN DE LA RAIZ", "VALOR DE LA RAIZ"
460 PRINT
470 PRINT "*****", "*****"
480 PRINT
490 FOR J=1 TO N
500 V3=V2+2*PI*(J-1)/N
510 X=R2*COS(V3)
520 Y=R2*SIN(V3)
530 PRINT " ";J;
540 PRINT " ";INT(1000*X+.5)/1000;
550 IF Y>=0 THEN PRINT "+";
560 PRINT INT(1000*Y+.5)/1000;"i"
570 NEXT J
580 GOTO 700
590 REM - SUBROUTINA DE CONVERSION DE CARTESIANAS A POLARES -
600 R=SQR(X*X+Y*Y)
610 IF X=0 AND Y=0 GOTO 650
620 IF X=0 THEN X=(.1)^30
625 IF Y=0 THEN Y=(.1)^30
630 IF X>0 AND Y=0 GOTO 650
640 W=Y/X
650 V=0
660 V=ATN(W)
670 IF X<0 THEN V=V+PI
680 IF V<0 THEN V=V+2*PI
690 RETURN
700 REM - PARA REALIZAR MAS CALCULOS -
710 PRINT
720 PRINT
730 PRINT
740 PRINT "DESEA CONTINUAR REALIZANDO OTRA OPCION ? S/N";
750 INPUT C$
760 IF C$="N" THEN 780
770 GOTO 170
780 END

```

## ejercicios resueltos:

1. Calcule las siguientes potencias:  $(1+i)^5$ ;  $(-5-3i)^7$ ;  $(-2+7i)^6$ ;  $(1-i)^{18}$ .

**SOLUCION :**

EL PRESENTE PROGRAMA LE PERMITE CALCULAR :

- 1.-POTENCIA DE UN NUMERO COMPLEJO ESCRITO EN FORMA BINOMICA
- 2.-RADICACION DE UN NUMERO COMPLEJO ESCRITO EN FORMA BINOMICA

SEGUN LA OPCION DESEADA PULSE EL NUMERO QUE LE PRECEDE

? 1

PARTE REAL E IMAGINARIA DEL NUMERO COMPLEJO SEPARADAS POR UNA COMA? 1 1  
EXPONENTE DE LA POTENCIA DESEADA N=? 5

LA POTENCIA PEDIDA ES  $= -4 -4 i$

DESEA CONTINUAR REALIZANDO OTRA OPCION ? S/N?S

SEGUN LA OPCION DESEADA PULSE EL NUMERO QUE LE PRECEDE

? 1

PARTE REAL E IMAGINARIA DEL NUMERO COMPLEJO SEPARADAS POR UNA COMA? -5 -3  
EXPONENTE DE LA POTENCIA DESEADA N=? 7

LA POTENCIA PEDIDA ES  $= 183640 + 137112 i$

DESEA CONTINUAR REALIZANDO OTRA OPCION ? S/N?S

SEGUN LA OPCION DESEADA PULSE EL NUMERO QUE LE PRECEDE

? 1

PARTE REAL E IMAGINARIA DEL NUMERO COMPLEJO SEPARADAS POR UNA COMA? -2 7  
EXPONENTE DE LA POTENCIA DESEADA N=? 6

LA POTENCIA PEDIDA ES  $= 14715 -148148 i$

DESEA CONTINUAR REALIZANDO OTRA OPCION ? S/N?S

SEGUN LA OPCION DESEADA PULSE EL NUMERO QUE LE PRECEDE

? 1

PARTE REAL E IMAGINARIA DEL NUMERO COMPLEJO SEPARADAS POR UNA COMA? 1 -1  
EXPONENTE DE LA POTENCIA DESEADA N=? 18

LA POTENCIA PEDIDA ES  $= 2.2470422E-06 -512 i = -512 i$

## 2. Calcular la raiz cuadrada del complejo: $z = 5 + 5 i$

### SOLUCION :

PARTE REAL E IMAGINARIA DEL NUMERO COMPLEJO SEPARADAS POR UNA COMA? 5 5  
INDICE DE LA RAIZ DESEADA N=? 2

ORDEN DE LA RAIZ

VALOR DE LA RAIZ

\*\*\*\*\*

\*\*\*\*\*

1 2.457 + 1.018 i  
2 -2.457 -1.018 i

3. Resolver las siguientes ecuaciones:  $z^8 - 64 = 0$ ;  $z^3 - 1 = 0$

**SOLUCION :**

? 2

PARTE REAL E IMAGINARIA DEL NUMERO COMPLEJO SEPARADAS POR UNA COMA? 64 0  
INDICE DE LA RAZ DESEADA N=? 8

ORDEN DE LA RAZ	VALOR DE LA RAZ
*****	*****
1	1.682 + 0 i
2	1.189 + 1.189 i
3	0 + 1.682 i
4	-1.189 + 1.189 i
5	1.682 + 0 i
6	-1.189 - 1.189 i
7	0 - 1.682 i
8	1.189 - 1.189 i

DESEA CONTINUAR REALIZANDO OTRA OPCION ? S/N?S

SEGUN LA OPCION DESEADA PULSE EL NUMERO QUE LE PRECEDE

? 2

PARTE REAL E IMAGINARIA DEL NUMERO COMPLEJO SEPARADAS POR UNA COMA? 1 0  
INDICE DE LA RAZ DESEADA N=? 3

ORDEN DE LA RAZ	VALOR DE LA RAZ
*****	*****
1	1 + 0 i
2	-.5 + .866 i
3	-.5 - .866 i

# OPERACIONES CON NUMEROS COMPLEJOS EN FORMA POLAR

## introducción:

Este programa permite calcular:

- 1.- El producto de dos números complejos en forma polar.
  - 2.- El cociente de dos números complejos en forma polar.
  - 3.- La potencia n-esima de cada uno de los dos complejos en forma polar
  - 4.- Las raices n-esimas de cada uno de los dos complejos en forma polar.
- Es preciso introducir los dos complejos en coordenadas polares.  
En caso de que los argumentos estén en radianes, basta con modificar la línea 790 del siguiente modo:

$$790 \text{ W7} = \text{W6} + 2 * \text{PI} * (\text{J} - 1) / \text{N}$$

y sustituir todas las palabras GRADOS que figuran en el programa por la palabra RADIANS.

## listado del programa:

```
10 REM *****
20 REM
30 REM   O P E R A C I O N E S   C O N   N U M E R O S   C O M P L E J O S
40 REM
50 REM                               E N   F O R M A   P O L A R
60 REM
70 REM *****
100 PRINT
110 PRINT "EL PRESENTE PROGRAMA LE PERMITE CALCULAR : "
120 PRINT
130 PRINT
140 PRINT "1.-PRODUCTO DE DOS NUMEROS COMPLEJOS EN FORMA POLAR"
150 PRINT "2.-COCIENTE DE DOS NUMEROS COMPLEJOS EN FORMA POLAR"
160 PRINT "3.-POTENCIA N-ESIMA DE CADA UNO DE LOS DOS COMPLEJOS EN FORMA POLAR"
170 PRINT "4.-RAICES N-ESIMAS DE CADA UNO DE LOS DOS COMPLEJOS EN FORMA POLAR"
180 PRINT
190 PRINT
200 PRINT
209 REM - ENTRADA DE DATOS -
210 PRINT "MODULO DEL COMPLEJO Z1,          R1 =";
```

```

220 INPUT R1
230 PRINT "ARGUMENTO DEL COMPLEJO Z1,   W1 = ";
240 INPUT W1
250 PRINT
260 PRINT "MODULO DEL COMPLEJO Z2,      R2 = ";
270 INPUT R2
280 PRINT "ARGUMENTO DEL COMPLEJO Z2,   W2 = ";
290 INPUT W2
300 PRINT
310 PRINT "EXPONENTE - INDICE PARA LA POTENCIACION Y LA RADICACION   N = ";
320 INPUT N
330 PRINT
339 REM - PRODUCTO DE NUMEROS COMPLEJOS -
340 RM=R1*R2
350 WM=W1+W2
360 PRINT "EL COMPLEJO PRODUCTO ES   Z1*Z2 = MODULO";RM;" , ARGUMENTO";WM;" GRADOS"
370 PRINT
379 REM - COCIENTE DE NUMEROS COMPLEJOS -
390 RC=R1/R2
400 WC=W1-W2
410 PRINT "EL COMPLEJO COCIENTE ES   Z1/Z2 = MODULO";RC;" , ARGUMENTO";WC;" GRADOS"
420 PRINT
429 REM - POTENCIA N-ESIMA DE Z1 -
430 R3=R1^N
440 W3=W1*N
450 PRINT "POTENCIA N-ESIMA DE Z1 = (Z1)^N = MODULO";R3;" , ARGUMENTO";W3;" GRADOS"
460 PRINT
469 REM - POTENCIA N-ESIMA DE Z2 -
470 R4=R2^N
480 W4=W2*N
490 PRINT "POTENCIA N-ESIMA DE Z2 = (Z2)^N = MODULO";R4;" , ARGUMENTO";W4;" GRADOS"
500 PRINT
509 REM - RAICES N-ESIMAS DE Z1 -
510 R=R1
520 W=W1
530 GOSUB 700
540 PRINT
550 R=R2
560 W=W2
570 GOSUB 700
580 END
700 R6=R^(1/N)
710 W6=W/N
720 PRINT
730 PRINT
740 PRINT "ORDEN DE LA RAIZ", "          VALOR DE LA RAIZ          "
750 PRINT
760 PRINT "*****", "*****"
770 PRINT
780 FOR J=1 TO N
790 W7=W6+360*(J-1)/N
800 PRINT "          ";J,
810 PRINT "MODULO";R6;" , ARGUMENTO";W7;" GRADOS"
820 NEXT J
830 RETURN

```

## ejercicios resueltos:

1. Dados los complejos:  $z_1 = 64_{120^\circ}$  y  $z_2 = 1_{180^\circ}$ , se pide, calcular:
- El complejo producto  $z_1 \cdot z_2$
  - El complejo cociente  $z_1/z_2$
  - Las potencias de índice 6 de  $z_1$  y de  $z_2$
  - Las raíces sextas de  $z_1$  y de  $z_2$

### SOLUCION :

MODULO DEL COMPLEJO Z1,            R1 =? 64  
 ARGUMENTO DEL COMPLEJO Z1,    W1 =? 120

MODULO DEL COMPLEJO Z2,            R2 =? 1  
 ARGUMENTO DEL COMPLEJO Z2,    W2 =? 180

EXPONENTE - INDICE PARA LA POTENCIACION Y LA RADICACION    N =? 6

EL COMPLEJO PRODUCTO ES     $Z1 \cdot Z2$  = MODULO 64 , ARGUMENTO 300 GRADOS

EL COMPLEJO COCIENTE ES     $Z1/Z2$  = MODULO 64 , ARGUMENTO -60 GRADOS

POTENCIA N-ESIMA DE Z1 =  $(Z1)^N$  = MODULO 6.8719477E+10 , ARGUMENTO 720 GRADOS

POTENCIA N-ESIMA DE Z2 =  $(Z2)^N$  = MODULO 1 , ARGUMENTO 1080 GRADOS

el argumento principal será:  $0^\circ$

ORDEN DE LA RAIZ

VALOR DE LA RAIZ

\*\*\*\*\*

\*\*\*\*\*

1	MODULO 2 , ARGUMENTO 20 GRADOS
2	MODULO 2 , ARGUMENTO 80 GRADOS
3	MODULO 2 , ARGUMENTO 140 GRADOS
4	MODULO 2 , ARGUMENTO 200 GRADOS
5	MODULO 2 , ARGUMENTO 260 GRADOS
6	MODULO 2 , ARGUMENTO 320 GRADOS

ORDEN DE LA RAIZ

VALOR DE LA RAIZ

\*\*\*\*\*

\*\*\*\*\*

1	MODULO 1 , ARGUMENTO 30 GRADOS
2	MODULO 1 , ARGUMENTO 90 GRADOS
3	MODULO 1 , ARGUMENTO 150 GRADOS
4	MODULO 1 , ARGUMENTO 210 GRADOS
5	MODULO 1 , ARGUMENTO 270 GRADOS
6	MODULO 1 , ARGUMENTO 330 GRADOS

2. Dados los complejos:  $z_1 = 3_{30^\circ}$  y  $z_2 = 4_{45^\circ}$ , calcular:
- El complejo producto  $z_1 \cdot z_2$
  - El complejo cociente  $z_1/z_2$

- c) Las potencias de índice 5 de  $z_1$  y de  $z_2$   
 d) Las raíces quintas de  $z_1$  y de  $z_2$

**SOLUCION :**

MODULO DEL COMPLEJO Z1, R1 =? 3  
 ARGUMENTO DEL COMPLEJO Z1, W1 =? 30

MODULO DEL COMPLEJO Z2, R2 =? 4  
 ARGUMENTO DEL COMPLEJO Z2, W2 =? 45

EXPONENTE - INDICE PARA LA POTENCIACION Y LA RADICACION N =? 5

EL COMPLEJO PRODUCTO ES  $Z1 \cdot Z2$  = MODULO 12 , ARGUMENTO 75 GRADOS

EL COMPLEJO COCIENTE ES  $Z1/Z2$  = MODULO .75 , ARGUMENTO -15 GRADOS

POTENCIA N-ESIMA DE Z1 =  $(Z1)^N$  = MODULO 243 , ARGUMENTO 150 GRADOS

POTENCIA N-ESIMA DE Z2 =  $(Z2)^N$  = MODULO 1024 , ARGUMENTO 225 GRADOS

ORDEN DE LA RAIZ

VALOR DE LA RAIZ

\*\*\*\*\*

\*\*\*\*\*

1	MODULO 1.2457309 , ARGUMENTO 6 GRADOS
2	MODULO 1.2457309 , ARGUMENTO 78 GRADOS
3	MODULO 1.2457309 , ARGUMENTO 150 GRADOS
4	MODULO 1.2457309 , ARGUMENTO 222 GRADOS
5	MODULO 1.2457309 , ARGUMENTO 294 GRADOS

ORDEN DE LA RAIZ

VALOR DE LA RAIZ

\*\*\*\*\*

\*\*\*\*\*

1	MODULO 1.3195079 , ARGUMENTO 9 GRADOS
2	MODULO 1.3195079 , ARGUMENTO 81 GRADOS
3	MODULO 1.3195079 , ARGUMENTO 153 GRADOS
4	MODULO 1.3195079 , ARGUMENTO 225 GRADOS
5	MODULO 1.3195079 , ARGUMENTO 297 GRADOS

**3.** Dados los complejos:  $z_1 = 8_{0^\circ}$  y  $z_2 = 27_{90^\circ}$  , calcular:

- a) El complejo producto  $z_1 \cdot z_2$   
 b) El complejo cociente  $z_1/z_2$   
 c) Las potencias de índice 3 de  $z_1$  y de  $z_2$   
 d) Las raíces cúbicas de  $z_1$  y de  $z_2$

**SOLUCION :**

MODULO DEL COMPLEJO Z1, R1 =? 8  
 ARGUMENTO DEL COMPLEJO Z1, W1 =? 0

MODULO DEL COMPLEJO Z2, R2 =? 27  
 ARGUMENTO DEL COMPLEJO Z2, W2 =? 90

EXPONENTE - INDICE PARA LA POTENCIACION Y LA RADICACION  $N = 3$

EL COMPLEJO PRODUCTO ES  $Z1 \cdot Z2 =$  MODULO 216 , ARGUMENTO 90 GRADOS

EL COMPLEJO COCIENTE ES  $Z1/Z2 =$  MODULO .2962963 , ARGUMENTO -90 GRADOS

POTENCIA N-ESIMA DE  $Z1 = (Z1)^N =$  MODULO 512 , ARGUMENTO 0 GRADOS

POTENCIA N-ESIMA DE  $Z2 = (Z2)^N =$  MODULO 19683 , ARGUMENTO 270 GRADOS

ORDEN DE LA RAIZ

VALOR DE LA RAIZ

\*\*\*\*\*

\*\*\*\*\*

1

MODULO 2 , ARGUMENTO 0 GRADOS

2

MODULO 2 , ARGUMENTO 120 GRADOS

3

MODULO 2 , ARGUMENTO 240 GRADOS

ORDEN DE LA RAIZ

VALOR DE LA RAIZ

\*\*\*\*\*

\*\*\*\*\*

1

MODULO 3 , ARGUMENTO 30 GRADOS

2

MODULO 3 , ARGUMENTO 150 GRADOS

3

MODULO 3 , ARGUMENTO 270 GRADOS



# ECUACION ALGEBRAICA DE SEGUNDO GRADO

## introducción:

Este programa sirve para calcular las raíces reales y complejas de ecuaciones algebraicas de 2° grado. Para su ejecución, basta con introducir los coeficientes de  $x^2$ , de  $x$ , y el término independiente.

## listado del programa:

```
10 REM *****
20 REM
30 REM           RESOLUCION           DE           ECUACIONES
40 REM
50 REM           ALGEBRAICAS           DE           SEGUNDO           GRADO
60 REM
70 REM *****
100 PRINT
109 REM - ENTRADA DE DATOS -
110 PRINT "COEFICIENTE DE X^2 .... A = ";
120 INPUT A
130 PRINT "COEFICIENTE DE X .... B = ";
140 INPUT B
150 PRINT "TERMINO INDEPENDIENTE C = ";
160 INPUT C
170 D=B^2-4*A*C
180 IF D<0 THEN GOTO 260
190 X1=(-B+SQR(D))/(2*A)
200 X2=(-B-SQR(D))/(2*A)
210 PRINT
220 PRINT "LAS RAICES DE LA ECUACION SON REALES : "
230 PRINT "X1 = "; X1
240 PRINT "X2 = "; X2
250 STOP
260 R=-B/(2*A)
270 Y=SQR(-D)/(2*A)
280 PRINT
290 PRINT "LAS RAICES DE LA ECUACION SON COMPLEJAS : "
300 PRINT "X1 = "; R ; "+" ; Y ; "i"
310 PRINT "X2 = "; R ; "-" ; Y ; "i"
320 END
```

## ejercicios resueltos:

1. Resolver la ecuación:  $x^2 + 2x - 3 = 0$

### SOLUCION :

COEFICIENTE DE  $x^2$  .... A = ? 1  
 COEFICIENTE DE X .... B = ? 2  
 TERMINO INDEPENDIENTE C = ? -3

LAS RAICES DE LA ECUACION SON REALES :

$x_1 = 1$   
 $x_2 = -3$

2. Resolver la ecuacion:  $x^2 - 8x + 16 = 0$

### SOLUCION :

COEFICIENTE DE  $x^2$  .... A = ? 1  
 COEFICIENTE DE X .... B = ? -8  
 TERMINO INDEPENDIENTE C = ? 16

LAS RAICES DE LA ECUACION SON REALES :

$x_1 = 4$   
 $x_2 = 4$

3. Resolver la ecuación:  $x^2 - 6x + 13 = 0$

### SOLUCION :

COEFICIENTE DE  $x^2$  .... A = ? 1  
 COEFICIENTE DE X .... B = ? -6  
 TERMINO INDEPENDIENTE C = ? 13

LAS RAICES DE LA ECUACION SON COMPLEJAS :

$x_1 = 3 + 2i$   
 $x_2 = 3 - 2i$

4. Resolver la ecuación:  $2x^2 + 3x + 4 = 0$

### SOLUCION :

COEFICIENTE DE  $x^2$  .... A = ? 2  
 COEFICIENTE DE X .... B = ? 3  
 TERMINO INDEPENDIENTE C = ? 4

LAS RAICES DE LA ECUACION SON COMPLEJAS :

$x_1 = -.75 + 1.1989579i$   
 $x_2 = -.75 - 1.1989579i$

# BUSQUEDA DE RAICES: METODO DE BISECCION

## introducción:

Este programa sirve para la búsqueda de raíces reales de funciones. El método de la bisección está basado en el teorema de Bolzano y es un procedimiento iterativo que consiste en ir haciendo cada vez menor el intervalo que contiene a una raíz. La iteración se para cuando el intervalo que contiene a la raíz es menor que el error máximo deseado prefijado D.

La función se introduce en la línea 270 en la forma: 270 Y = F(x) a continuación se introducen los límites del intervalo en el que se trata de ver si existe una raíz para la función.

La línea 370 sirve para ver como se realiza la convergencia hacia la raíz en las sucesivas iteraciones. En caso de que unicamente se quiera la raíz, se puede omitir la línea 370.

## listado del programa:

```
10 REM *****
20 REM
30 REM      B U S Q U E D A      D E      R A I C E S      R E A L E S      D E
40 REM
50 REM      F U N C I O N E S      (METODO DE LA BISECCION)
60 REM
70 REM *****
100 PRINT
109 REM - ENTRADA DE DATOS -
110 PRINT "DEFINA LA FUNCION QUE DESEA ESTUDIAR EN LA LINEA 270"
120 PRINT "DEL-SIGUIENTE MODO : 270 Y = ....."
130 PRINT "A CONTINUACION PULSAR <CONT>"
140 STOP
150 PRINT "LIMITES DEL INTERVALO : INFERIOR Y SUPERIOR" ;
160 INPUT X0,X1
170 PRINT "ERROR MAXIMO DESEADO";
180 INPUT D
190 PRINT "NUMERO MAXIMO DE ITERACCIONES A REALIZAR";
200 INPUT I
210 GOSUB 290
```

```

219 REM - SALIDA DE RESULTADOS -
220 PRINT
230 PRINT "UNA RAIZ ES X ="; X ; "CON UN ERROR MENOR QUE"; D
240 PRINT "EL VALOR CORRESPONDIENTE PARA Y ES "; Y
250 PRINT "EL NUMERO DE ITERACIONES REALIZADO ES "; J
260 GOTO 450
270 Y=
280 RETURN
290 J=0
300 X=X0
310 GOSUB 270
320 Y0=Y
330 X=X1
340 GOSUB 270
350 X=(X0+X1)/2
360 GOSUB 270
369 REM - SI SE DESEA VER LA CONVERGENCIA HACIA LA RAIZ INCLUIR LINEA 370 -
370 PRINT J,X
380 J=J+1
390 IF J>=I THEN 440
400 IF Y*Y0 = 0 THEN GOTO 440
410 IF Y*Y0 < 0 THEN X1=X
420 IF Y*Y0 > 0 THEN X0=X
430 IF ABS(X0-X1)>D THEN GOTO 300
440 RETURN
450 END

```

## ejercicios resueltos:

1. Calcular las raíces de la ecuación:  $y = x^3 + 3x^2 - 4x - 12$

### SOLUCION :

Estudiaremos la función en los siguientes intervalos: (0,5); (-6, 2.5); (-2.5, 0) y (5, 10)

DEFINA LA FUNCION QUE DESEA ESTUDIAR EN LA LINEA 270  
 DEL SIGUIENTE MODO: 270 Y = .....  
 A CONTINUACION PULSAR <CONT>  
 270 Y= X\*X\*X+3\*X\*X-4\*X-12

#### 1ª Ejecución:

LIMITES DEL INTERVALO : INFERIOR Y SUPERIOR? 0 5  
 ERROR MAXIMO DESEADO? 1E-06  
 NUMERO MAXIMO DE ITERACIONES A REALIZAR? 100

0	2.5
1	1.25
2	1.875
3	2.1875
4	2.03125
5	1.953125
6	1.9921875
7	2.0117187
8	2.0019531
9	1.9970703
10	1.9995117
11	2.0007324

12	2.0001221
13	1.9998169
14	1.9999695
15	2.0000458
16	2.0000076
17	1.9999886
18	1.9999981
19	2.0000029
20	2.0000005
21	1.9999993
22	1.9999999

UNA RAIZ ES  $X = 1.9999999$  CON UN ERROR MENOR QUE  $1E-06$   
 EL VALOR CORRESPONDIENTE PARA Y ES  $= -2.3841858E-06$   
 EL NUMERO DE ITERACCIONES REALIZADO ES = 23

### 2ª Ejecución:

LIMITES DEL INTERVALO : INFERIOR Y SUPERIOR? -6 -2.5  
 ERROR MAXIMO DESEADO? .00001  
 NUMERO MAXIMO DE ITERACCIONES A REALIZAR? 100

0	-4.25
1	-3.375
2	-2.9375
3	-3.15625
4	-3.046875
5	-2.9921875
6	-3.0195312
7	-3.0058594
8	-2.9990234
9	-3.0024414
10	-3.0007324
11	-2.9998779
12	-3.0003052
13	-3.0000916
14	-2.9999847
15	-3.0000381
16	-3.0000114
17	-2.9999981
18	-3.0000048

UNA RAIZ ES  $X = -3.0000048$  CON UN ERROR MENOR QUE .00001  
 EL VALOR CORRESPONDIENTE PARA Y ES  $= -.000023842091$   
 EL NUMERO DE ITERACCIONES REALIZADO ES = 19

### 3ª Ejecución:

LIMITES DEL INTERVALO : INFERIOR Y SUPERIOR? -2.5 0  
 ERROR MAXIMO DESEADO?  $1E-08$   
 NUMERO MAXIMO DE ITERACCIONES A REALIZAR? 100

UNA RAIZ ES  $X = -2$  CON UN ERROR MENOR QUE  $1E-08$   
 EL VALOR CORRESPONDIENTE PARA Y ES  $= 7.4505806E-09$   
 EL NUMERO DE ITERACCIONES REALIZADO ES = 28

**4ª Ejecución:**

LIMITES DEL INTERVALO : INFERIOR Y SUPERIOR? 5 10  
 ERROR MAXIMO DESEADO? 1E-06  
 NUMERO MAXIMO DE ITERACCIONES A REALIZAR? 100  
 UNA RAIZ ES  $x = 9.9999994$  CON UN ERROR MENOR QUE 1E-06  
 EL VALOR CORRESPONDIENTE PARA Y ES = 1247.9999  
 EL NUMERO DE ITERACCIONES REALIZADO ES = 23

En las dos primeras ejecuciones hemos visto como convergen los valores hacia la raíz. En la tercera ejecución hemos quitado la instrucción 370 para obtener unicamente la raíz.

En estos casos como existe una raíz en los intervalos dados, la búsqueda converge hacia la raíz proporcionandonos una aproximación de ella. En cambio en la cuarta ejecución, como el intervalo (5, 10) no contiene ninguna raíz, la iteracción acaba en uno de los extremos del intervalo produciendo se aparentemente un error, que se detecta facilmente al ver que el valor de la y correspondiente no tiende a cero como en los otros casos.

**2.** Calcular una raíz de la función:  $y = x - 2 \cos x$  en el intervalo  $(0, \pi/2)$  con un error de  $10^{-5}$ .

**SOLUCION :**

DEFINA LA FUNCION QUE DESEA ESTUDIAR EN LA LINEA 270  
 DEL SIGUIENTE MODO : 270 Y = .....  
 A CONTINUACION PULSAR <CONT>  
 270 Y = X-2\*COS(X)

LIMITES DEL INTERVALO : INFERIOR Y SUPERIOR? 0 1.57  
 ERROR MAXIMO DESEADO? .00001  
 NUMERO MAXIMO DE ITERACCIONES A REALIZAR? 100

0	.785
1	1.1775
2	.98125
3	1.079375
4	1.0303125
5	1.0057813
6	1.0180469
7	1.0241797
8	1.0272461
9	1.0287793
10	1.0295459
11	1.0299292
12	1.0297375
13	1.0298334
14	1.0298813
15	1.0298573
16	1.0298693
17	1.0298633

UNA RAIZ ES  $x = 1.0298633$  CON UN ERROR MENOR QUE .00001  
 EL VALOR CORRESPONDIENTE PARA Y ES = -8.712057E-06  
 EL NUMERO DE ITERACCIONES REALIZADO ES = 18

**3.** Calcular una raíz de la función:  $y = e^x + x$  en el intervalo  $(-1,0)$  con un error de  $10^{-5}$ .

**SOLUCION :**

DEFINA LA FUNCION QUE DESEA ESTUDIAR EN LA LINEA 270  
DEL SIGUIENTE MODO : 270 Y = .....  
A CONTINUACION PULSAR <CONT>  
270 Y=EXP(X)+X

LIMITES DEL INTERVALO : INFERIOR Y SUPERIOR? -1 0  
ERROR MAXIMO DESEADO? .00001  
NUMERO MAXIMO DE ITERACCIONES A REALIZAR? 60

0	-.5
1	-.75
2	-.625
3	-.5625
4	-.59375
5	-.578125
6	-.5703125
7	-.56640625
8	-.56835938
9	-.56738281
10	-.56689453
11	-.56713867
12	-.56726074
13	-.56719971
14	-.56716919
15	-.56715393
16	-.5671463

UNA RAZ ES X = -.5671463 CON UN ERROR MENOR QUE .00001  
EL VALOR CORRESPONDIENTE PARA Y ES = -4.7184221E-06  
EL NUMERO DE ITERACCIONES REALIZADO ES = 17

# BUSQUEDA DE RAICES: METODO DE NEWTON

## introducción:

Este programa sirve para la búsqueda de raíces reales de funciones. El método de Newton es un procedimiento iterativo de forma que en cada nueva iteración obtenemos un valor más aproximado del verdadero valor de la raíz. Es preciso introducir un valor inicial o valor de conjetura.

La función se introduce en la línea 310 en la forma: 310 Y = F(X) y en la línea 320 se introduce la derivada de la función, en la forma siguiente: 320 DY = F'(X).

La línea 360 sirve para ver como se realiza la convergencia hacia la raíz en las sucesivas iteraciones. En caso de que unicamente se desee la raíz, se debe omitir la línea 360.

## listado del programa:

```
10 REM *****
20 REM
30 REM      B U S Q U E D A   D E   R A I C E S   R E A L E S   D E
40 REM
50 REM      F U N C I O N E S       (M E T O D O   D E   N E W T O N)
60 REM
70 REM *****
100 PRINT
109 REM - ENTRADA DE DATOS -
110 PRINT "DEFINA LA FUNCION QUE DESEA ESTUDIAR EN LA LINEA 310"
120 PRINT "DEL SIGUIENTE MODO : 310 Y =....."
130 PRINT "A CONTINUACION PULSAR <CONT>"
140 STOP
150 PRINT "DEFINA LA FUNCION DERIVADA DE Y=F(X) EN LA LINEA 320"
160 PRINT "DEL SIGUIENTE MODO : 320 DY =....."
170 PRINT "A CONTINUACION PULSAR <CONT>"
180 STOP
190 PRINT "INTRODUZCA UN VALOR INICIAL";
200 INPUT XO
210 PRINT "ERROR MAXIMO DESEADO";
220 INPUT D
```



```

230 PRINT "NUMERO MAXIMO DE ITERACCIONES A REALIZAR";
240 INPUT I
250 GOSUB 340
259 REM - SALIDA DE RESULTADOS -
260 PRINT
270 PRINT "UNA RAIZ ES X ="; X ; "CON UN ERROR MENOR QUE"; D
280 PRINT "EL VALOR CORRESPONDIENTE PARA Y ES ="; Y
290 PRINT "EL NUMERO DE ITERACCIONES REALIZADO ES ="; J
300 GOTO 440
310 Y=
320 DY=
330 RETURN
340 J=0
350 X=X0
359 REM - SI SE DESEA VER LA CONVERGENCIA HAY QUE INCLUIR LA INSTRUCCION 360 -
360 PRINT J,X
370 GOSUB 310
380 X0=X0-Y/DY
390 J=J+1
400 IF J >= I THEN GOTO 430
410 IF ABS(Y/DY)>D THEN GOTO 350
420 X=X0
430 RETURN
440 END

```

## ejercicios resueltos:

1. Calcular una raíz de la función:  $y = e^x - \cos x$

### SOLUCION :

DEFINA LA FUNCION QUE DESEA ESTUDIAR EN LA LINEA 310  
 DEL SIGUIENTE MODO : 310 Y =.....  
 A CONTINUACION PULSAR <CONT>  
 310 Y=EXP(X)-COS(X)

DEFINA LA FUNCION DERIVADA DE  $Y=F(X)$  EN LA LINEA 320  
 DEL SIGUIENTE MODO : 320 DY =.....  
 A CONTINUACION PULSAR <CONT>  
 320 DY=EXP(X)+SIN(X)

INTRODUZCA UN VALOR INICIAL? 1  
 ERROR MAXIMO DESEADO? .00001  
 NUMERO MAXIMO DE ITERACCIONES A REALIZAR? 50

0	1
1	.38816552
2	.092030346
3	.0073463293
4	.00005331583
5	2.7509124E-09

UNA RAIZ ES  $X = -1.6401812E-10$  CON UN ERROR MENOR QUE .00001  
 EL VALOR CORRESPONDIENTE PARA Y ES = 2.9149305E-09  
 EL NUMERO DE ITERACCIONES REALIZADO ES = 6

2. Hallar una raíz de la función:  $y = x(x-3)(x-5)$

### SOLUCION :

DEFINA LA FUNCION QUE DESEA ESTUDIAR EN LA LINEA 310  
 DEL SIGUIENTE MODO : 310 Y =.....  
 A CONTINUACION PULSAR <CONT>  
 310 Y=X\*(X-3)\*(X+5)

DEFINA LA FUNCION DERIVADA DE Y=F(X) EN LA LINEA 320  
 DEL SIGUIENTE MODO : 320 DY =.....  
 A CONTINUACION PULSAR <CONT>  
 320 DY = 3\*X\*X+4\*X-15

INTRODUZCA UN VALOR INICIAL? 2  
 ERROR MAXIMO DESEADO? 1E-08  
 NUMERO MAXIMO DE ITERACCIONES A REALIZAR? 100

0	2
1	4.8
2	3.6451718
3	3.129703
4	3.0070404
5	3.0000226
6	3

UNA RAIZ ES  $x = 3$  CON UN ERROR MENOR QUE  $1E-08$   
 EL VALOR CORRESPONDIENTE PARA Y ES =  $5.5879354E-09$   
 EL NUMERO DE ITERACCIONES REALIZADO ES = 7

### 3. Hallar una raiz de la función: $y = x - \text{sen } x$

#### **SOLUCION :**

INTRODUZCA UN VALOR INICIAL? 1  
 ERROR MAXIMO DESEADO? .0001  
 NUMERO MAXIMO DE ITERACCIONES A REALIZAR? 40

0	1
1	.65514507
2	.43359037
3	.2881484
4	.19183231
5	.12780967
6	.085183233
7	.056781953
8	.037852598
9	.02523446
10	.016822795
11	.011215126
12	.0074767224
13	.0049845038
14	.0033230309
15	.0022153889
16	.0014768584
17	.00098466714
18	.00065648657
19	.00043741155
20	.00029172183

UNA RAIZ ES  $x = .00019396122$  CON UN ERROR MENOR QUE .0001  
 EL VALOR CORRESPONDIENTE PARA Y ES =  $4.1602277E-12$   
 EL NUMERO DE ITERACCIONES REALIZADO ES = 21

# BUSQUEDA DE RAICES: METODO DE LA SECANTE

## introducción:

Este programa sirve para la búsqueda de raíces reales de funciones. El método de la secante es un procedimiento iterativo, para el cuál se precisa introducir dos valores iniciales o de conjetura, el error máximo deseado, y el número máximo de iteracciones a realizar.

En la línea 270 se introduce la función, del siguiente modo:

270 Y = F(X).

Si se desea ver la convergencia hacia la raíz se debe incluir la instrucción 380. Por el contrario si unicamente se quiere el resultado del valor aproximado de la raíz se debe omitir la instrucción 380.

## listado del programa:

```
10 REM *****
20 REM
30 REM      BUSQUEDA DE RAICES REALES DE
40 REM
50 REM      FUNCIONES      (METODO DE LA SECANTE)
60 REM
70 REM *****
100 PRINT
109 REM - ENTRADA DE DATOS -
110 PRINT "DEFINA LA FUNCION QUE DESEA ESTUDIAR EN LA LINEA 270"
120 PRINT "DEL SIGUIENTE MODO : 270 Y = ....."
130 PRINT "A CONTINUACION PULSAR <CONT>"
140 STOP
150 PRINT "INTRODUZCA DOS VALORES INICIALES";
160 INPUT X0,X1
170 PRINT "ERROR MAXIMO DESEADO";
180 INPUT D
190 PRINT "NUMERO MAXIMO DE ITERACCIONES A REALIZAR";
200 INPUT I
210 GOSUB 270
219 REM - SALIDA DE RESULTADOS -
220 PRINT
```

```

230 PRINT "UNA RAIZ ES X ="; X ; "CON UN ERROR MENOR QUE"; D
240 PRINT "EL VALOR CORRESPONDIENTE PARA Y ES ="; Y
250 PRINT "EL NUMERO DE ITERACCIONES REALIZADO ES ="; J
260 GOTO 460
270 Y=
280 RETURN
290 J=0
300 X=X0
310 GOSUB 270
320 Y0=Y
330 X=X1
340 GOSUB 270
350 Y1=Y
360 IF Y1=Y0 THEN Y1=Y1+.0001
370 X=(X0*Y1-Y0*X1)/(Y1-Y0)
379 REM - SI SE DESEA VER LA CONVERGENCIA HACIA LA RAIZ INCLUIR LINEA 380
380 PRINT J,X
390 J=J+1
400 IF J>=I THEN GOTO 450
410 IF ABS(X0-X1)<D THEN GOTO 450
420 X0=X1
430 X1=X
440 GOTO 300
450 RETURN
460 END

```

## ejercicios resueltos:

1. Calcular una raíz de la función :  $y = x^3 - 10x + 8$

### SOLUCION :

DEFINA LA FUNCION QUE DESEA ESTUDIAR EN LA LINEA 270  
 DEL SIGUIENTE MODO : 270 Y = .....  
 A CONTINUACION PULSAR <CONT>  
 270 Y=X\*X\*X-10\*X+8

INTRODUZCA DOS VALORES INICIALES? 1            6  
 ERROR MAXIMO DESEADO? .00001  
 NUMERO MAXIMO DE ITERACCIONES A REALIZAR? 100

0	1.030303
1	1.0666814
2	.84984948
3	.86578992
4	.86464651
5	.86464088
6	.86464089

UNA RAIZ ES X = .86464089 CON UN ERROR MENOR QUE .00001  
 EL VALOR CORRESPONDIENTE PARA Y ES = 1.6996637E-08  
 EL NUMERO DE ITERACCIONES REALIZADO ES = 7

2. Calcular una raíz de la función:  $y = e^x - \cotg x$

### SOLUCION :

DEFINA LA FUNCION QUE DESEA ESTUDIAR EN LA LINEA 270  
DEL SIGUIENTE MODO : 270 Y = .....  
A CONTINUACION PULSAR <CONT>  
270 Y=EXP(X)-(COS(X)/SIN(X))

INTRODUZCA DOS VALORES INICIALES? 1        4  
ERROR MAXIMO DESEADO? .00001  
NUMERO MAXIMO DE ITERACCIONES A REALIZAR? 100  
0        .87942748  
1        .78477501  
2        .49669796  
3        .53842123  
4        .5316511  
5        .53138898  
6        .53139086  
7        .53139086

UNA RAIZ ES X = .53139086 CON UN ERROR MENOR QUE .00001  
EL VALOR CORRESPONDIENTE PARA Y ES = 2.7939677E-09  
EL NUMERO DE ITERACCIONES REALIZADO ES = 8.

# BUSQUEDA DE RAICES: METODO DE LA REGULA FALSI

## introducción:

Este programa sirve para la búsqueda de raíces reales de funciones. El método de la regla falsi o la regla de falsa posición es un procedimiento iterativo, tal que en cada iteración se obtiene un valor más aproximado de la raíz. Al igual que el método de la secante, en este caso, se precisa introducir dos valores iniciales a partir de los cuales se inicia la iteración, también es preciso dar el error máximo deseado para la obtención de la raíz, así como el número máximo de iteraciones a realizar.

La función se introduce en la línea 270 del siguiente modo:

270 Y = F(X).

Si se desea ver como se realiza la convergencia hacia la raíz en las sucesivas iteraciones se incluirá la instrucción 410, en caso contrario se omitirá dicha instrucción.

## listado del programa:

```
10 REM *****
20 REM
30 REM      BUSQUEDA      DE      RAICES      REALES      DE
40 REM
50 REM      FUNCIONES      (METODO DE LA REGULA FALSI)
60 REM
70 REM *****
100 PRINT
109 REM - ENTRADA DE DATOS -
110 PRINT "DEFINA LA FUNCION QUE DESEA ESTUDIAR EN LA LINEA 270"
120 PRINT "DEL SIGUIENTE MODO : 270 Y = ....."
130 PRINT "A CONTINUACION PULSAR <CONT>"
140 STOP
150 PRINT "INTRODUZCA DOS VALORES INICIALES";
160 INPUT X0,X1
170 PRINT "ERROR MAXIMO DESEADO";
180 INPUT D
190 PRINT "NUMERO MAXIMO DE ITERACIONES A REALIZAR";
200 INPUT I
210 GOSUB 290
```

```

219 REM - SALIDA DE RESULTADOS -
220 PRINT
230 PRINT "UNA RAIZ ES X =";X;"CON UN ERROR MENOR QUE";D
240 PRINT "EL VALOR CORRESPONDIENTE PARA Y ES = ";Y
250 PRINT "EL NUMERO DE ITERACCIONES REALIZADAS ES ="; J
260 GOTO 570
270 Y=
280 RETURN
290 J=0
300 IF X0<X1 THEN GOTO 340
310 X=X0
320 X0=X1
330 X1=X
340 X=X0
350 GOSUB 270
360 Y0=Y
370 X=X1
380 GOSUB 270
390 Y1=Y
400 X=(X0*Y1-Y0*X1)/(Y1-Y0)
409 REM - SI SE DESEA VER LA CONVERGENCIA HACIA LA RAIZ INCLUIR LINEA 410 -
410 PRINT J,X
420 J=J+1
430 IF J>=I THEN 560
440 IF ABS(X-X1)<D THEN 560
450 GOSUB 270
460 IF Y*Y0>0 THEN GOTO 520
470 IF Y*Y1=0 THEN 560
480 X1=X
490 Y1=Y
500 Y0=.5*Y0
510 GOTO 400
520 X0=X
530 Y0=Y
540 Y1=.5*Y1
550 GOTO 400
560 RETURN
570 END

```

## ejercicios resueltos:

1. Calcular una raíz de la función:  $y = e^x + x$

### SOLUCION :

```

DEFINA LA FUNCION QUE DESEA ESTUDIAR EN LA LINEA 270
DEL SIGUIENTE MODO : 270 Y = .....
A CONTINUACION PULSAR <CONT>
270 Y=EXP(X)+X

```

```

INTRODUZCA DOS VALORES INICIALES?-1      2
ERROR MAXIMO DESEADO? .00001
NUMERO MAXIMO DE ITERACCIONES A REALIZAR? 10

```

```

0      -.81076457
1      -.60735051
2      -.53966461
3      -.57895455
4      -.5608617
5      -.57020197
6      -.56559764

```

```

7      -.56791297
8      -.56675791
9      -.56733591

```

UNA RAIZ ES X = -.56733591 CON UN ERROR MENOR QUE .00001  
 EL VALOR CORRESPONDIENTE PARA Y ES = .00060399232  
 EL NUMERO DE ITERACCIONES REALIZADAS ES = 10

INTRODUZCA DOS VALORES INICIALES? -1      2  
 ERROR MAXIMO DESEADO? .00001  
 NUMERO MAXIMO DE ITERACCIONES A REALIZAR? 100

```

0      -.81076457
1      -.60735051
2      -.53966461
3      -.57895455
4      -.5608617
5      -.57020197
6      -.56559764
7      -.56791297
8      -.56675791
9      -.56733591
10     -.56704698
11     -.56719145
12     -.56711921
13     -.56715533
14     -.56713727
15     -.5671463

```

UNA RAIZ ES X = -.5671463 CON UN ERROR MENOR QUE .00001  
 EL VALOR CORRESPONDIENTE PARA Y ES = 9.433561E-06  
 EL NUMERO DE ITERACCIONES REALIZADAS ES = 16

## 2. Calcular una raiz de la función: $y = 2x^4 + 3x^2 - 5x - 7$

### SOLUCION :

DEFINA LA FUNCION QUE DESEA ESTUDIAR EN LA LINEA 270  
 DEL SIGUIENTE MODO : 270 Y = .....  
 A CONTINUACION PULSAR <CONT>  
 270 Y=2\*XXXXXX+3\*XX-5\*X-7

INTRODUZCA DOS VALORES INICIALES? 2      7  
 ERROR MAXIMO DESEADO? .00001  
 NUMERO MAXIMO DE ITERACCIONES A REALIZAR? 10

```

0      1.9723361
1      1.9204232
2      1.8291026
3      1.6881374
4      1.5210734
5      1.4101946
6      1.4216704
7      1.4149256
8      1.4180014
9      1.416407

```



UNA RAIZ ES  $X = 1.416407$  CON UN ERROR MENOR QUE .00001  
 EL VALOR CORRESPONDIENTE PARA Y ES = .028230861  
 EL NUMERO DE ITERACCIONES REALIZADAS ES = 10

INTRODUZCA DOS VALORES INICIALES? 2 7  
 ERROR MAXIMO DESEADO? .00001  
 NUMERO MAXIMO DE ITERACCIONES A REALIZAR? 100

0	1.9723361
1	1.9204232
2	1.8291026
3	1.6881374
4	1.5210734
5	1.4101946
6	1.4216704
7	1.4149256
8	1.4180014
9	1.416407
10	1.4171912
11	1.4167964
12	1.4169932
13	1.4168947
14	1.4169439
15	1.4169193
16	1.4169316
17	1.4169255

UNA RAIZ ES  $X = 1.4169255$  CON UN ERROR MENOR QUE .00001  
 EL VALOR CORRESPONDIENTE PARA Y ES = .00010776334  
 EL NUMERO DE ITERACCIONES REALIZADAS ES = 18

### 3. Calcular una raiz de la función: $y = x - 3 \text{ sen } x$

#### SOLUCION :

DEFINA LA FUNCION QUE DESEA ESTUDIAR EN LA LINEA 270  
 DEL SIGUIENTE MODO : 270 Y = .....  
 A CONTINUACION PULSAR <CONT>  
 270 Y=X-3\*SIN(X)

INTRODUZCA DOS VALORES INICIALES? 1 5  
 ERROR MAXIMO DESEADO? 1E-06  
 NUMERO MAXIMO DE ITERACCIONES A REALIZAR? 10

0	1.6486045
1	2.5005062
2	2.0636869
3	2.3350529
4	2.2349181
5	2.2950812
6	2.269185
7	2.2832108
8	2.2765535
9	2.2799782

UNA RAIZ ES  $X = 2.2799782$  CON UN ERROR MENOR QUE 1E-06  
 EL VALOR CORRESPONDIENTE PARA Y ES = -.0068084879  
 EL NUMERO DE ITERACCIONES REALIZADAS ES = 10

# BUSQUEDA DE RAICES: METODO DE AITKEN

## introducción:

Este programa sirve para la búsqueda de raíces reales de funciones. El método de Aitken es un procedimiento iterativo, para el que se requiere la introducción de un valor inicial o valor de conjetura, el error máximo deseado para la raíz aproximada, el número máximo de iteraciones a realizar y el factor de convergencia, (un valor normal es  $F = -1$ ).

La función se introduce en la línea 300 en la forma siguiente:

300 Y = F(X).

Si se desea ver como se realiza la convergencia hacia la raíz es preciso incluir las líneas 390 y 430, en caso contrario, cuando unicamente se desea el valor de la raíz, basta con omitir las dos líneas anteriormente dichas.

## listado del programa:

```
10 REM *****
20 REM
30 REM      B U S Q U E D A      D E      R A I C E S      R E A L E S      D E
40 REM
50 REM      F U N C I O N E S      (M E T O D O      D E      A I T K E N)
60 REM
70 REM *****
100 PRINT
109 REM - ENTRADA DE DATOS -
110 PRINT "DEFINA LA FUNCION QUE DESEA ESTUDIAR EN LA LINEA 300"
120 PRINT "DEL SIGUIENTE MODO : 300 Y =....."
130 PRINT "A CONTINUACION PULSAR <CONT>"
140 STOP
150 PRINT "INTRODUZCA UN VALOR INICIAL";
160 INPUT XO
170 PRINT "ERROR MAXIMO DESEADO";
180 INPUT D
190 PRINT "NUMERO MAXIMO DE ITERACCIONES A REALIZAR";
200 INPUT I
210 PRINT "FACTOR DE CONVERGENCIA";
220 INPUT F
```

```

230 GOSUB 320
239 REM - SALIDA DE RESULTADOS -
240 PRINT
250 PRINT "UNA RAZA ES X ="; X ; "CON UN ERROR MENOR QUE"; D
260 GOSUB 300
270 PRINT "EL VALOR CORRESPONDIENTE PARA Y ES "; Y
280 PRINT "EL NUMERO DE ITERACIONES REALIZADO ES "; J
290 GOTO 560
300 Y=
310 RETURN
320 J=0
330 X=X0
340 GOSUB 300
350 Y=X+Y*F
360 IF J>0 THEN GOTO 420
370 X1=Y
380 X=X1
389 REM -SI SE DESEA VER LA CONVERGENCIA HACIA LA RAZA INCLUIR LINEA 390 Y 430-
390 PRINT J,X
400 J=J+1
410 GOTO 340
420 X2=Y
430 PRINT J,X
440 J=J+1
450 K=X2-2*X1+X0
460 C=(X1-X2)*(X1-X2)/K
470 C=(X1-X2)*(X1-X2)/K
480 X2=X2-C
490 IF J>I THEN GOTO 550
500 IF ABS(C)<D THEN GOTO 550
510 X0=X1
520 X1=X2
530 X=X1
540 GOTO 340
550 RETURN
560 END

```

## ejercicios resueltos:

**1.** Calcular una raíz de la ecuación:  $y = e^x + x$

### SOLUCION :

Haremos tres ejecuciones del mismo programa variando únicamente el factor de convergencia F, para  $F = 0,1$ ;  $F = 1$ ;  $F = 2$ . En cambio, el valor inicial, el error máximo deseado y el número máximo de iteraciones permanecen constantes en las tres ejecuciones.

```

DEFINA LA FUNCION QUE DESEA ESTUDIAR EN LA LINEA 300
DEL SIGUIENTE MODO : 300 Y =.....
A CONTINUACION PULSAR <CONT>
300 Y=EXP(X)+X

```

```

INTRODUZCA UN VALOR INICIAL?-1
ERROR MAXIMO DESEADO? .00001
NUMERO MAXIMO DE ITERACIONES A REALIZAR? 100

```

FACTOR DE CONVERGENCIA? .1

0	-1.0632121
1	-1.0632121
2	-.53400396
3	-.52872696
4	-.56944213

UNA RAIZ ES  $X = -.56944213$  CON UN ERROR MENOR QUE .00001  
 EL VALOR CORRESPONDIENTE PARA Y ES  $= -.003601106$   
 EL NUMERO DE ITERACIONES REALIZADO ES = 5

INTRODUZCA UN VALOR INICIAL? -1

ERROR MAXIMO DESEADO? .00001

NUMERO MAXIMO DE ITERACIONES A REALIZAR? 100

FACTOR DE CONVERGENCIA? 1

0	-1.6321206
1	-1.6321206
2	-.50331433
3	-.39213667
4	-.57505349
5	-.58833073
6	-.56615025
7	-.56447626
8	-.56726773
9	-.56747738
10	-.56712769

UNA RAIZ ES  $X = -.56712769$  CON UN ERROR MENOR QUE .00001  
 EL VALOR CORRESPONDIENTE PARA Y ES  $= .000024442058$   
 EL NUMERO DE ITERACIONES REALIZADO ES = 11

INTRODUZCA UN VALOR INICIAL? -1

ERROR MAXIMO DESEADO? .00001

NUMERO MAXIMO DE ITERACIONES A REALIZAR? 100

FACTOR DE CONVERGENCIA? 2

0	-2.2642411
1	-2.2642411
2	-.47706669
3	-.13506262
4	-.5801036
5	-.62469022
6	-.5652613
7	-.55871011
8	-.56741568
9	-.56836221
10	-.56710385
11	-.56696676
12	-.567149

UNA RAIZ ES  $X = -.567149$  CON UN ERROR MENOR QUE .00001  
 EL VALOR CORRESPONDIENTE PARA Y ES  $= -8.9494179E-06$   
 EL NUMERO DE ITERACIONES REALIZADO ES = 13

**2.** Calcular una raíz de la ecuación:  $y = x - 2 \text{ sen } x$

**SOLUCION :**

Para esta ejecución hemos omitido las líneas 390 y 430.

DEFINA LA FUNCION QUE DESEA ESTUDIAR EN LA LINEA 300  
DEL SIGUIENTE MODO : 300 Y =.....  
A CONTINUACION PULSAR <CONT>  
300 Y=X-2\*SIN(X)

INTRODUZCA UN VALOR INICIAL? 1  
ERROR MAXIMO DESEADO? .00001  
NUMERO MAXIMO DE ITERACCIONES A REALIZAR? 100  
FACTOR DE CONVERGENCIA? 2

UNA RAIZ ES X = .000013428417 CON UN ERROR MENOR QUE .00001  
EL VALOR CORRESPONDIENTE PARA Y ES = -.000013428417  
EL NUMERO DE ITERACCIONES REALIZADO ES = 11

**3.** Calcular una raiz de la ecuación:  $y = x(x + 3)(x - 2)(x - 5)$

**SOLUCION :**

DEFINA LA FUNCION QUE DESEA ESTUDIAR EN LA LINEA 300  
DEL SIGUIENTE MODO : 300 Y =.....  
A CONTINUACION PULSAR <CONT>  
300 Y=X\*(X+3)\*(X-2)\*(X-5)

INTRODUZCA UN VALOR INICIAL? 3  
ERROR MAXIMO DESEADO? .00001  
NUMERO MAXIMO DE ITERACCIONES A REALIZAR? 1000  
FACTOR DE CONVERGENCIA? 2

UNA RAIZ ES X = -3 CON UN ERROR MENOR QUE .00001  
EL VALOR CORRESPONDIENTE PARA Y ES = 5.4202975E-06  
EL NUMERO DE ITERACCIONES REALIZADO ES = 465

# TABULACION DE UNA FUNCION REAL DE VARIABLE REAL

## introducción:

Este breve programa sirve para realizar una tabulación de cualquier función de la forma:  $Y = F(X)$ . Para ello es preciso introducir la función en la línea 250 mediante su definición, del siguiente modo:

```
250 DEF FN F(X) = .....
```

Además se ha de introducir en la línea 160 los límites del intervalo en el que se desea la tabulación y a continuación el valor del incremento deseado en la línea 190.

A veces puede interesar redondear los valores de  $x$  y de  $y$  para no trabajar con tantos decimales.

## listado del programa:

```
10 REM *****
20 REM
30 REM          TABULACION      DE      UNA      FUNCION
40 REM
50 REM          REAL      DE      VARIABLE      REAL
60 REM
70 REM *****
100 PRINT
110 PRINT "DEFINA LA FUNCION QUE DESEA TABULAR EN LA LINEA 250"
120 PRINT "DEL SIGUIENTE MODO : 250 DEF FN F(X)= ..... "
130 PRINT "A CONTINUACION PULSAR <CONT>"
140 PRINT:STOP
150 PRINT "INTRODUZCA LOS EXTREMOS DEL INTERVALO A TABULAR";
160 INPUT A,B
170 PRINT
180 PRINT "INTRODUZCA EL VALOR DEL INCREMENTO";
190 INPUT H
200 PRINT
210 PRINT
220 PRINT "VALORES DE X","VALORES DE Y=F(X)"
230 PRINT "*****"
240 FOR X=A TO B STEP H
250 DEF FN F(X)=
260 PRINT X,FN F(X)
270 NEXT X
280 END
```

## ejercicios resueltos:

1. Estudiar la función:  $y = \frac{\text{sen } x}{|x|}$  en el intervalo [1, 4] con un incremento de 0.25.

### SOLUCION :

DEFINA LA FUNCION QUE DESEA TABULAR EN LA LINEA 250  
DEL SIGUIENTE MODO : 250 DEF FN F(X)= .....  
A CONTINUACION PULSAR <CONT>

INTRODUZCA LOS EXTREMOS DEL INTERVALO A TABULAR? 1 4

INTRODUZCA EL VALOR DEL INCREMENTO? .25

VALORES DE X	VALORES DE Y=F(X)
*****	
1	.84147098
1.25	.75918770
1.5	.66499666
1.75	.56227768
2	.45464871
2.25	.34581031
2.5	.23938886
2.75	.13878582
3	.04704000
3.25	-.03329081
3.5	-.10022378
3.75	-.15241635
4	-.18920062

2. Estudiar la función:  $y = x^x$  en el intervalo [0, 3] con un incremento de 0.2.

### SOLUCION :

DEFINA LA FUNCION QUE DESEA TABULAR EN LA LINEA 250  
DEL SIGUIENTE MODO : 250 DEF FN F(X)= .....  
A CONTINUACION PULSAR <CONT>

INTRODUZCA LOS EXTREMOS DEL INTERVALO A TABULAR? 0 3

INTRODUZCA EL VALOR DEL INCREMENTO? .2

VALORES DE X	VALORES DE Y=F(X)
*****	
0.0	1.0000
.2	.7248
.4	.6931
.6	.7360
.8	.8365
1.0	1.0000
1.2	1.2446
1.4	1.6017
1.6	2.1213

1.8	2.8807
2.0	4.0000
2.2	5.6667
2.4	8.1754
2.6	11.9931
2.8	17.8666
3.0	27.0000

3. Estudiar la función:  $y = \frac{\ln(x+1)}{\sin^3 x}$  en el intervalo  $[1, \pi]$  con un incremento de 0.1.

**S O L U C I O N :**

DEFINA LA FUNCION QUE DESEA TABULAR EN LA LINEA 250  
DEL SIGUIENTE MODO : 250 DEF FN F(X)= .....  
A CONTINUACION PULSAR <CONT>

INTRODUZCA LOS EXTREMOS DEL INTERVALO A TABULAR? 1 3.14159

INTRODUZCA EL VALOR DEL INCREMENTO? .1

VALORES DE X	VALORES DE Y=F(X)
1	1.16334
1.1	1.04817
1.2	.97381
1.3	.93103
1.4	.91482
1.5	.92321
1.6	.95673
1.7	1.01851
1.8	1.11482
1.9	1.25645
2	1.46126
2.1	1.75901
2.2	2.20091
2.3	2.87922
2.4	3.97097
2.5	5.84436
2.6	9.35055
2.7	16.76008
2.8	35.51348
2.9	99.37988
3	493.27567
3.1	19626.77069



# OPERACIONES CON FUNCIONES REALES DE VARIABLE REAL

## introducción:

Este programa sirve para calcular la suma, diferencia, producto y cociente de dos funciones dadas en forma explícita. Para ello hemos de introducir las funciones, mediante su definición del siguiente modo:

```
270 DEF FN F(X) = .....
```

```
280 DEF FN G(X) = .....
```

En el caso de que la función  $G(X)$  se anule en el intervalo  $[A, B]$  el cociente no será posible produciéndose un error aritmético.

## listado del programa:

```
10 REM *****
20 REM
30 REM          O P E R A C I O N E S          C O N          F U N C I O N E S
40 REM
50 REM          R E A L E S          D E          V A R I A B L E          R E A L
60 REM
70 REM *****
100 PRINT
110 PRINT "DEFINA LAS FUNCIONES QUE DESEA OPERAR EN LAS LINEAS 270,280"
120 PRINT "DEL SIGUIENTE MODO Y PULSAR <CONT> DESPUES"
130 PRINT
140 PRINT "270 DEF FNF(X)= ....."
150 PRINT "280 DEF FNG(X)= ....."
160 PRINT
170 STOP
180 PRINT
190 PRINT "INTRODUZCA LOS EXTREMOS DEL INTERVALO A TABULAR";
200 INPUT A,B
210 PRINT "INTRODUZCA EL VALOR DEL INCREMENTO";
220 INPUT H
230 PRINT
240 PRINT "VALORES DE X","FUNC.SUMA","FUNC.DIFERENC","FUNC.PRODUC","FUNC.COCIENT
"
250 PRINT "*****
*****"
260 FOR X=A TO B STEP H
```

```

270 DEF FNF(X)=
280 DEF FNG(X)=
290 FS=FNF(X)+FNG(X)
300 FD=FNF(X)-FNG(X)
310 FP=FNF(X)*FNG(X)
320 FC=FNF(X)/FNG(X)
330 PRINT X,FS,FD,FP,FC
340 NEXT X
350 END

```

## ejercicios resueltos:

1. Dadas las funciones:  $F(x) = 3x + 5$  y  $G(x) = \frac{x^2 + 3x - 7}{2x - 1}$

Hallar las funciones:  $F(x) + G(x)$ ;  $F(x) - G(x)$ ;  $F(x) \cdot G(x)$ ;  $F(x)/G(x)$  en el intervalo  $[1, 3]$  con un incremento de 0.25.

### SOLUCION :

DEFINA LAS FUNCIONES QUE DESEA OPERAR EN LAS LINEAS 270,280 DEL SIGUIENTE MODO Y PULSAR <CONT> DESPUES

```

270 DEF FNF(X)=3*X+5
280 DEF FNG(X)=(X*X+3*X-7)/(2*X-1)

```

INTRODUZCA LOS EXTREMOS DEL INTERVALO A TABULAR? 1 3  
INTRODUZCA EL VALOR DEL INCREMENTO? .25

VAL.DE X	FUNC.SUMA	FUNC.DIFERENC	FUNC.PRODUC	FUNC.COCIENT
*****	*****	*****	*****	*****
1	5.0000	11.0000	- 24.0000	- 2.6667
1.25	7.6250	9.8750	- 9.8437	- 7.7778
1.5	9.3750	9.6250	- 1.1875	- 78.0000
1.75	10.7750	9.7250	5.3813	19.5238
2	12.0000	10.0000	11.0000	11.0000
2.25	13.1250	10.3750	16.1563	8.5455
2.5	14.1875	10.8125	21.0937	7.4074
2.75	15.2083	11.2917	25.9479	6.7660
3	16.2000	11.8000	30.8000	6.3636

2. Dadas las funciones:  $F(x) = \frac{x^2}{2 + \sin x}$  y  $G(x) = \ln(x+3)$

Hallar las funciones siguientes:

$F(x) + G(x)$ ;  $F(x) - G(x)$ ;  $F(x) \cdot G(x)$ ;  $F(x)/G(x)$ .

en el intervalo  $[1, 3]$  con un incremento de 0.1.

### SOLUCION :

DEFINA LAS FUNCIONES QUE DESEA OPERAR EN LAS LINEAS 270,280 DEL SIGUIENTE MODO Y PULSAR <CONT> DESPUES

```

270 DEF FNF(X)=X*X/(2+SIN(X))
280 DEF FNG(X)=LOG(X+3)

```

INTRODUZCA LOS EXTREMOS DEL INTERVALO A TABULAR? 1 3  
INTRODUZCA EL VALOR DEL INCREMENTO? .1

VALORES DE X	FUNC.SUMA	FUNC.DIFERENC	FUNC.PRODUC	FUNC.COCIENT
*****	*****	*****	*****	*****
1.00000	1.73822	- 1.03436	.48788	.25386
1.10000	1.82950	- .99248	.59051	.29661
1.20000	1.92621	- .94396	.70481	.34223
1.30000	2.02888	- .88835	.83179	.39096
1.40000	2.13812	- .82509	.97270	.44311
1.50000	2.25470	- .75345	1.12900	.49906
1.60000	2.37951	- .67260	1.30242	.55926
1.70000	2.51358	- .58155	1.49497	.62422
1.80000	2.65811	- .47912	1.70900	.69456
1.90000	2.81450	- .36397	1.94724	.77098
2.00000	2.98434	- .23454	2.21282	.85427
2.10000	3.16947	- .08901	2.50940	.94537
2.20000	3.37200	.07468	2.84120	1.04530
2.30000	3.59435	.25894	3.21308	1.15527
2.40000	3.83930	.46650	3.63065	1.27662
2.50000	4.11001	.70051	4.10036	1.41092
2.60000	4.41010	.96457	4.62965	1.55990
2.70000	4.74370	1.26277	5.22703	1.72554
2.80000	5.11548	1.59976	5.90222	1.91006
2.90000	5.53068	1.98077	6.66623	2.11596
3.00000	5.99517	2.41165	7.53150	2.34597

3. Dadas las funciones:  $F(x) = \sqrt[3]{\frac{2x+1}{x^2+1}}$  y  $G(x) = \cos(x^4)$

Hallar las funciones siguientes:

$F(x) + G(x)$ ;  $F(x) - G(x)$ ;  $F(x) \cdot G(x)$ ;  $F(x)/G(x)$

en el intervalo  $[1, 3]$  con un incremento de 0.1.

### SOLUCION :

DEFINIR LAS FUNCIONES QUE DESEA OPERAR EN LAS LINEAS 270,280  
DEL SIGUIENTE MODO Y PULSAR <CONT> DESPUES

```
270 DEF FNF(X)=((2*X+1)/(X*X+1))^(1/3)
280 DEF FNG(X)=COS(X*X*X*X)
```

INTRODUZCA LOS EXTREMOS DEL INTERVALO A TABULAR? 1 3  
INTRODUZCA EL VALOR DEL INCREMENTO? .1

VALORES DE X	FUNC.SUMA	FUNC.DIFERENC	FUNC.PRODUC	FUNC.COCIENT
*****	*****	*****	*****	*****
1.00000	1.68502	.60441	.61849	2.11866
1.10000	1.23782	1.02483	.12048	10.62333
1.20000	.63506	1.59882	- .53824	- 2.31786
1.30000	.14248	2.06153	- 1.05740	- 1.14849
1.40000	.32200	1.85167	- .83125	- 1.42100
1.50000	1.41467	.72866	.36758	3.12437
1.60000	2.02031	.09299	1.01826	1.09650
1.70000	.56415	1.51969	- .49780	- 2.18078
1.80000	.54939	1.50518	- .49080	- 2.15125
1.90000	1.90705	.12006	.90560	1.13437
2.00000	.04234	1.95766	- .95766	- 1.04421
2.10000	1.81305	.16073	.81533	1.19456
2.20000	.83830	1.11015	- .13242	- 7.16750
2.30000	.00383	1.92019	- .92177	- 1.00400
2.40000	.76051	1.13995	- .18028	- 5.00857
2.50000	1.14479	.73294	.19334	4.55929
2.60000	.78391	1.07192	- .13362	- 6.44369
2.70000	- .04826	1.88299	- .88583	- .95003
2.80000	1.11031	.70407	.18427	4.46626
2.90000	.85512	.93963	- .03792	-21.23815
3.00000	1.66459	.11122	.68962	1.14320

# COMPOSICION DE FUNCIONES REALES DE VARIABLE REAL

## introducción:

Este sencillo programa sirve para realizar la composición de dos funciones. Para ello es preciso introducir las dos funciones mediante su definición del siguiente modo:

```
260 DEF FN F(X) = .....
```

```
270 DEF FN G(X) = .....
```

Como siempre que se pretende tabular una función hemos de dar el intervalo en el que se pretende estudiar y el incremento de la variable X.

## listado del programa:

```
10 REM *****
20 REM
30 REM          C O M P O S I C I O N          D E          F U N C I O N E S
40 REM
50 REM          R E A L E S          D E          V A R I A B L E          R E A L
60 REM
70 REM *****
100 PRINT
110 PRINT "DEFINA LAS FUNCIONES QUE DESEA COMPONER EN LAS LINEAS 260,270"
120 PRINT "DEL SIGUIENTE MODO Y PULSAR <CONT> AL FINALIZAR."
130 PRINT
140 PRINT:PRINT "260 DEF FN F(X)=....."
150 PRINT:PRINT "270 DEF FN G(X)=....."
160 PRINT:STOP
170 PRINT
180 PRINT "EXTREMOS DEL INTERVALO EN QUE SE DESEA ESTUDIAR LA F.COMPUESTA";
190 INPUT A,B
200 PRINT "INTRODUZCA EL VALOR DEL INCREMENTO";
210 INPUT C
220 PRINT
230 PRINT "X","F(X)","G(X)","H(X)=G(F(X))"
240 PRINT "*****"
250 FOR X=A TO B STEP C
260 DEF FN F(X)=
270 DEF FN G(X)=
280 HX=FNG(FN F(X))
290 PRINT X,FN F(X),FN G(X),HX
300 NEXT X
310 END
```

## ejercicios resueltos:

1. Dadas las funciones:  $F(x) = \sin^3 x$ , y  $G(x) = \ln(x + 2)$ , estudiar la función compuesta  $H(x) = G(F(x))$  en el intervalo  $[1, 3]$  con un incremento de 0.05.

### SOLUCION :

DEFINA LAS FUNCIONES QUE DESEA COMPONER EN LAS LINEAS 260,270 DEL SIGUIENTE MODO Y PULSAR <CONT> AL FINALIZAR.

```
260 DEF FN F(X)=(SIN(X))^3
270 DEF FN G(X)=LOG(X+2)
```

EXTREMOS DEL INTERVALO EN QUE SE DESEA ESTUDIAR LA F.COMPUSTA? 1  
INTRODUZCA EL VALOR DEL INCREMENTO? .05

3

X	F(X)	G(X)	H(X)=G(F(X))
1	.59582324	1.0986123	.95390371
1.05	.65266923	1.1151416	.97556639
1.1	.70784194	1.1314021	.99615199
1.15	.76045833	1.1474025	1.0153967
1.2	.80965943	1.1631508	1.0330633
1.25	.85462879	1.178655	1.0489418
1.3	.89461018	1.1939225	1.0629504
1.35	.92892383	1.2089603	1.0746351
1.4	.95698124	1.2237754	1.0841689
1.45	.97829789	1.2383742	1.091352
1.5	.99250377	1.252763	1.0961104
1.55	.99935143	1.2669476	1.0983961
1.6	.99872135	1.2809338	1.098186
1.65	.99062454	1.2947272	1.0954822
1.7	.97520228	1.3083328	1.090312
1.75	.95272308	1.3217558	1.0827278
1.8	.92357684	1.3350011	1.0728078
1.85	.88826637	1.3480731	1.0606564
1.9	.84739645	1.3609766	1.0464051
1.95	.80166079	1.3737156	1.0302124
2	.75182694	1.3862944	1.012265
2.05	.69871975	1.3987169	.99277749
2.1	.64320355	1.410987	.97199165
2.15	.58616357	1.4231083	.95017553
2.2	.52848696	1.4350845	.92762108
2.25	.47104388	1.446919	.90464068
2.3	.41466897	1.458615	.8815622
2.35	.36014378	1.4701758	.85872254
2.4	.30818042	1.4816045	.83645952
2.45	.25940685	1.4929041	.81510232
2.5	.21435411	1.5040774	.79496076
2.55	.17344584	1.5151272	.77631385
2.6	.13699019	1.5260563	.75939839
2.65	.10517446	1.5368672	.74439834
2.7	.078062457	1.5475625	.73143595
2.75	.055594691	1.5581446	.72056519
2.8	.037591385	1.5686159	.71176842
2.85	.02375822	1.5789787	.70495629
2.9	.013694689	1.5892352	.69997119
2.95	.0069048744	1.5993876	.69659367
3	.00281038	1.6094379	.69455139

**2.** Dadas las funciones:  $F(x) = x + 2$ , y  $G(x) = 5x$ , estudiar la función compuesta  $H(x) = G(F(x))$  en el intervalo  $[1, 3]$  con un incremento de 0.2.

**SOLUCION :**

DEFINA LAS FUNCIONES QUE DESEA COMPONER EN LAS LINEAS 260,270 DEL SIGUIENTE MODO Y PULSAR <CONT> AL FINALIZAR.

```
260 DEF FN F(X)= X+2
270 DEF FN G(X)= 5X
```

EXTREMOS DEL INTERVALO EN QUE SE DESEA ESTUDIAR LA F.COMPUSTA? 1  
INTRODUZCA EL VALOR DEL INCREMENTO? .2

3

X	F(X)	G(X)	H(X)=G(F(X))
1	3	5	15
1.2	3.2	6	16
1.4	3.4	7	17
1.6	3.6	8	18
1.8	3.8	9	19
2	4	10	20
2.2	4.2	11	21
2.4	4.4	12	22
2.6	4.6	13	23
2.8	4.8	14	24
3	5	15	25

# FUNCION EXPONENCIAL

## introducción:

Este programa sirve para estudiar la función exponencial de la forma  $y = a^x$ . Para ello el programa solicita la introducción de la base  $a$  ( $a > 0$ ) posteriormente se fijan los límites del intervalo y el incremento de la variable  $x$ .

El programa imprime el estudio de la función  $y = a^x$  y de la función  $y = (\frac{1}{a})^x$ , con el fin de poder ver las propiedades de las funciones exponenciales según que la base sea mayor o menor que la unidad.

## listado del programa:

```
10 REM *****
20 REM
30 REM           E S T U D I O   D E   L A
40 REM
50 REM           F U N C I O N   E X P O N E N C I A L
60 REM
70 REM *****
100 PRINT
110 PRINT "INTRODUZCA LA BASE DESEADA";
120 INPUT A
130 PRINT
140 PRINT "EXTREMOS DEL INTERVALO EN QUE DESEA ESTUDIAR LA FUNCION";
150 INPUT M,N
160 PRINT
170 PRINT "VALOR DEL INCREMENTO";
180 INPUT C
190 PRINT
200 PRINT
210 PRINT "X","A^X","(1/A)^X"
220 PRINT "*****"
230 FOR X=M TO N STEP C
240 F=A^X
250 G=(1/A)^X
260 PRINT X,F,G
270 NEXT X
280 END
```

1. Estudiar la función:  $y = 2^x$  e  $y = (\frac{1}{2})^x$  en el intervalo  $[0, 3]$  con un incremento de 0.25.


**SOLUCION :**

INTRODUZCA LA BASE DESEADA? 2

EXTREMOS DEL INTERVALO EN QUE DESEA ESTUDIAR LA FUNCION? 0 3

VALOR DEL INCREMENTO? .25

X	A^X	(1/A)^X
*****		
0.00	1.00000000	1.00000000
.25	1.18920711	.840896415
.50	1.41421356	.707106781
.75	1.68179283	.594603557
1.00	2.00000000	.500000000
1.25	2.37841423	.420448208
1.50	2.82842713	.353553391
1.75	3.36358566	.297301779
2.00	4.00000000	.250000000
2.25	4.75682846	.210224104
2.50	5.65685425	.176776695
2.75	6.72717133	.148650889
3.00	8.00000000	.125000000



2. Estudiar las funciones:  $y = 1.5^x$  y  $y = (\frac{1}{1.5})^x$  en el intervalo  $[-6, 6]$  con un incremento de 0.5.

**SOLUCION :**

INTRODUZCA LA BASE DESEADA? 1.5

EXTREMOS DEL INTERVALO EN QUE DESEA ESTUDIAR LA FUNCION? -6 6

VALOR DEL INCREMENTO? .5

X	A^X	(1/A)^X
*****		
-6.00	.08779150	11.39062500
-5.50	.10752218	9.30040637
-5.00	.13168724	7.59375000
-4.50	.16128328	6.20027091
-4.00	.19753086	5.06250000
-3.50	.24192491	4.13351394
-3.00	.29629630	3.37500000
-2.50	.36288737	2.75567596
-2.00	.44444444	2.25000000
-1.50	.54433105	1.83711731
-1.00	.66666667	1.50000000
-.50	.81649658	1.22474487
0.00	1.00000000	1.00000000
.50	1.22474487	.81649658
1.00	1.50000000	.66666667
1.50	1.83711731	.54433105
2.00	2.25000000	.44444444
2.50	2.75567596	.36288737
3.00	3.37500000	.29629630
3.50	4.13351394	.24192491
4.00	5.06250000	.19753086
4.50	6.20027091	.16128328
5.00	7.59375000	.13168724
5.50	9.30040637	.10752218
6.00	11.39062500	.08779150



# CAMBIO DE BASE LOGARITMICA

## introducción:

Los microordenadores tienen como función de librería la función logaritmo neperiano, que se escribe  $\text{LOG}(x)$ . Este programa sirve para calcular logaritmos en cualquier base  $a$ , ( $a > 0$ ), sin más que tener en cuenta la siguiente expresión:

$$\log_a x = \frac{\ln x}{\ln a}$$

## listado del programa:

```
10 REM *****
20 REM
30 REM             C A M B I O             D E
40 REM
50 REM             B A S E             L O G A R I T M I C A
60 REM
70 REM *****
100 PRINT
110 PRINT "INTRODUZCA LA BASE EN LA QUE DESEA TRABAJAR";
120 INPUT A
130 PRINT
140 PRINT "VALOR DE X PARA EL QUE DESEA CALCULAR SU LOGARITMO";
150 INPUT X
160 PRINT
170 LX =LOG(X)/LOG(A)
180 PRINT "EL LOGARITMO EN BASE";A;"DE";X;"ES = ";LX
190 PRINT
200 PRINT
210 GOTO 140
220 END
```

## ejercicios resueltos:

1. Calcular:  $\log_{12} 7$ ,  $\log_{12} 256$ ,  $\log_{12} 3209$

S O L U C I O N :

INTRODUZCA LA BASE DESEADA EN LA QUE DESEA TRABAJAR? 12  
VALOR DE X PARA EL QUE DESEA CALCULAR SU LOGARITMO? 7

EL LOGARITMO EN BASE 12 DE 7 ES = .78309185

VALOR DE X PARA EL QUE DESEA CALCULAR SU LOGARITMO? 256

EL LOGARITMO EN BASE 12 DE 256 ES = 2.2315436

VALOR DE X PARA EL QUE DESEA CALCULAR SU LOGARITMO? 3209

EL LOGARITMO EN BASE 12 DE 3209 ES = 3.2491018

# FUNCION LOGARITMICA

## introducción:

Este breve programa sirve para estudiar la función logarítmica de base  $a$ , ( $a > 0$ ). También hemos de dar el intervalo en el que se desea estudiar y el incremento de la variable  $x$ .

El programa imprime el estudio de la función:  $y = \log_a x$  y el de la función:  $y = \log_{(1/a)} x$ , con el fin de poder ver las propiedades de las funciones logarítmicas según que la base sea mayor o menor que la unidad.

## listado del programa:

```
10 REM *****
20 REM
30 REM           E S T U D I O       D E       L A
40 REM
50 REM           F U N C I O N       L O G A R I T M I C A
60 REM
70 REM *****
100 PRINT
110 PRINT "INTRODUZCA LA BASE DESEADA";
120 INPUT A
130 PRINT
140 PRINT "EXTREMOS DEL INTERVALO EN QUE DESEA ESTUDIAR LA FUNCION";
150 INPUT M,N
160 PRINT
170 PRINT "VALOR DEL INCREMENTO";
180 INPUT C
190 PRINT
200 PRINT
210 PRINT "X","LOGA(X)","LOG(1/A)(X)"
220 PRINT "*****"
230 FOR X=M TO N STEP C
240 F=LOG(X)/LOG(A)
250 G=LOG(X)/LOG(1/A)
260 PRINT X,F,G
270 NEXT X
280 END
```

## ejercicios resueltos:

1. Estudiar las funciones:  $y = \log_{10} x$ , e  $y = \log_{0.1} x$  en el intervalo  $[1, 15]$  con un incremento igual a 1.

### SOLUCION :

INTRODUZCA LA BASE DESEADA? 10

EXTREMOS DEL INTERVALO EN QUE DESEA ESTUDIAR LA FUNCION? 1

15

VALOR DEL INCREMENTO? 1

X	LOGA(X)	LOG(1/A)(X)
*****		
1.00	0.00000000	0.00000000
2.00	.30103000	-.30103000
3.00	.47712125	-.47712125
4.00	.60205999	-.60205999
5.00	.69897000	-.69897000
6.00	.77815125	-.77815125
7.00	.84509804	-.84509804
8.00	.90308999	-.90308999
9.00	.95424251	-.95424251
10.00	1.00000000	-1.00000000
11.00	1.04139269	-1.04139269
12.00	1.07918125	-1.07918125
13.00	1.11394335	-1.11394335
14.00	1.14612804	-1.14612804
15.00	1.17609126	-1.17609126

2. Estudiar las funciones:  $y = \log_2 x$ , e  $y = \log_{0.5} x$  en el intervalo  $[0.5; 3.9]$  con un increment igual a 0.2.


### SOLUCION :

INTRODUZCA LA BASE DESEADA? 2

EXTREMOS DEL INTERVALO EN QUE DESEA ESTUDIAR LA FUNCION? .5

VALOR DEL INCREMENTO? .2

X	LOGA(X)	LOG(1/A)(X)
*****		
.50	-1.00000000	1.00000000
.70	-.51457317	.51457317
.90	-.15200309	.15200309
1.10	.13750352	-.13750352
1.30	.37851162	-.37851162
1.50	.58496250	-.58496250
1.70	.76553475	-.76553475
1.90	.92599942	-.92599942
2.10	1.07038933	-1.07038933
2.30	1.20163386	-1.20163386
2.50	1.32192809	-1.32192809
2.70	1.43295941	-1.43295941
2.90	1.53605290	-1.53605290
3.10	1.63226822	-1.63226822
3.30	1.72246602	-1.72246602
3.50	1.80735492	-1.80735492
3.70	1.88752527	-1.88752527
3.90	1.96347412	-1.96347412



# FUNCIONES TRIGONOMETRICAS DIRECTAS

## introducción:

Este programa permite estudiar las funciones trigonométricas directas:  $y = \text{sen } x$ ,  $y = \text{cos } x$ ,  $y = \text{tag } x$ .

Una vez seleccionada la función que desea estudiar, se ha de introducir los límites del intervalo así como el incremento de la variable  $x$ .

Los ángulos están dados en radianes, en el supuesto de que se desea se trabajar con ángulos expresados en grados, minutos y segundos, bastaría con convertirlos previamente a radianes.

## listado del programa:

```
10 REM *****
20 REM
30 REM          ESTUDIO          DE          LAS          FUNCIONES
40 REM
50 REM          TRIGONOMETRICAS          DIRECTAS
60 REM
70 REM *****
100 PRINT
110 R$="*****"
120 PRINT "ESTE PROGRAMA LE PERMITE ESTUDIAR LAS SIGUIENTES FUNCIONES : "
130 PRINT
140 PRINT
150 PRINT "1.- FUNCION Y = SEN (X) "
160 PRINT "2.- FUNCION Y = COS (X) "
170 PRINT "3.- FUNCION Y = TAG (X) "
180 PRINT:PRINT
190 PRINT "SEGUN LA OPCION DESEADA PULSE EL NUMERO QUE LE PRECEDE" ;
200 INPUT P
210 IF P=1 THEN 240
220 IF P=2 THEN 320
230 IF P=3 THEN 400
240 GOSUB 479
249 REM - ESTUDIO DE LA FUNCION Y = SEN (X) -
250 PRINT "VALORES DE X", "VALORES DE Y=SEN(X) "
260 PRINT R$
270 FOR X=A TO B STEP H
280 Y = SIN(X)
290 PRINT X,Y
```

```

300 NEXT X
310 GOTO 180
320 GOSUB 479
329 REM - ESTUDIO DE LA FUNCION Y = COS (X) -
330 PRINT "VALORES DE X", "VALORES DE Y=cos(X)"
340 PRINT R$
350 FOR X=A TO B STEP H
360 Y = COS(X)
370 PRINT X,Y
380 NEXT X
390 GOTO 180
400 GOSUB 479
409 REM - ESTUDIO DE LA FUNCION Y = TAG (X) -
410 PRINT "VALORES DE X", "VALORES DE Y=TAG(X)"
420 PRINT R$
430 FOR X=A TO B STEP H
440 Y=SIN(X)/COS(X)
450 PRINT X,Y
460 NEXT X
470 GOTO 180
479 REM - SUBROUTINA DE ENTRADA DE DATOS -
480 PRINT "INTRODUZCA LOS EXTREMOS DEL INTERVALO A TABULAR" ;
490 INPUT A,B
500 PRINT "INTRODUZCA EL VALOR DEL INCREMENTO";
510 INPUT H
520 PRINT
530 PRINT
540 RETURN
550 END

```

## ejercicios resueltos:

1. Estudiar la función:  $y = \tan x$  en el entorno del punto  $\pi/2$ .

### **SOLUCION :**

Obsérvese la discontinuidad de la función  $y = \tan x$  en el punto  $x = \pi/2$ .

SEGUN LA OPCION DESEADA PULSE EL NUMERO QUE LE PRECEDE? 3  
 INTRODUZCA LOS EXTREMOS DEL INTERVALO A TABULAR? 1.569 1.573  
 INTRODUZCA EL VALOR DEL INCREMENTO? .0001

VALORES DE X	VALORES DE Y=TAG(X)
1.5690	556.69
1.5691	589.51
1.5692	626.44
1.5693	668.30
1.5694	716.16
1.5695	771.41
1.5696	835.89
1.5697	912.14
1.5698	1003.69
1.5699	1115.66
1.5700	1255.77
1.5701	1436.11
1.5702	1676.93
1.5703	2014.80
1.5704	2523.18
1.5705	3374.66

1.5706	5093.57
1.5707	10381.43
1.5708	- 272174.11
1.5709	- 9645.58
1.5710	- 4909.80
1.5711	- 3293.00
1.5712	- 2477.24
1.5713	- 1985.41
1.5714	- 1656.52
1.5715	- 1421.11
1.5716	- 1244.28
1.5717	- 1106.59
1.5718	- 996.34
1.5719	- 906.06
1.5720	- 830.79
1.5721	- 767.06
1.5722	- 712.42
1.5723	- 665.04
1.5724	- 623.57
1.5725	- 586.97
1.5726	- 554.42
1.5727	- 525.30
1.5728	- 499.08
1.5729	- 475.36

**2.** Estudiar la función:  $y = \sin x$  en el primer cuadrante, con un incremento de 0.05.

**SOLUCION :**

SEGUN LA OPCION DESEADA PULSE EL NUMERO QUE LE PRECEDE? 1


INTRODUZCA LOS EXTREMOS DEL INTERVALO A TABULAR? 0 1.5707963

INTRODUZCA EL VALOR DEL INCREMENTO? .05

VALORES DE X VALORES DE Y=SEN(X)

\*\*\*\*\*

0	0.00000000
.05	.049979169
.1	.099833417
.15	.149438132
.2	.198669331
.25	.247403959
.3	.295520207
.35	.342897807
.4	.389418342
.45	.434965534
.5	.479425539
.55	.522687229
.6	.564642473
.65	.605186406
.7	.644217687
.75	.681638760
.8	.717356091
.85	.751280405
.9	.783326910
.95	.813415505
1	.841470985
1.05	.867423226
1.1	.891207360
1.15	.912763940



1.2	.932039086
1.25	.948984619
1.3	.963558185
1.35	.975723358
1.4	.985449730
1.45	.992712991
1.5	.997494987
1.55	.999783764


**3.** Estudiar la función:  $y = \cos x$  en el segundo cuadrante, con un incremento de 0.05.

**SOLUCION :**

SEGUN LA OPCION DESEADA PULSE EL NUMERO QUE LE PRECEDE? 2  
 INTRODUZCA LOS EXTREMOS DEL INTERVALO A TABULAR? 1.5707963  
 INTRODUZCA EL VALOR DEL INCREMENTO? .05

3.1415927

VALORES DE X	VALORES DE Y=cos(X)
*****	*****
1.5707963	.000000027
1.6207963	-.049979143
1.6707963	-.099833390
1.7207963	-.149438106
1.7707963	-.198669305
1.8207963	-.247403934
1.8707963	-.295520181
1.9207963	-.342897782
1.9707963	-.389418318
2.0207963	-.434965510
2.0707963	-.479425515
2.1207963	-.522687206
2.1707963	-.564642452
2.2207963	-.605186385
2.2707963	-.644217667
2.3207963	-.681638741
2.3707963	-.717356073
2.4207963	-.751280388
2.4707963	-.783326893
2.5207963	-.813415490
2.5707963	-.841470971
2.6207963	-.867423213
2.6707963	-.891207348
2.7207963	-.912763930
2.7707963	-.932039077
2.8207963	-.948984611
2.8707963	-.963558179
2.9207963	-.975723352
2.9707963	-.985449725
3.0207963	-.992712988
3.0707963	-.997494985
3.1207963	-.999783764





# FUNCIONES TRIGONOMETRICAS INVERSAS

## introducción:

Este programa sirve para estudiar las funciones trigonométricas inversas:  $y = \operatorname{cosec} x$ ,  $y = \sec x$ ,  $y = \cotg x$ .

Una vez seleccionada la función que desea estudiar, se ha de introducir los límites del intervalo, así como el incremento de la variable  $x$ .

Los ángulos han de darse en radianes.

## listado del programa:

```
10 REM *****
20 REM
30 REM          ESTUDIO      DE      LAS      FUNCIONES
40 REM
50 REM          TRIGONOMETRICAS      INVERSAS
60 REM
70 REM *****
100 PRINT
110 R$="*****"
120 PRINT "ESTE PROGRAMA LE PERMITE ESTUDIAR LAS SIGUIENTES FUNCIONES : "
130 PRINT
140 PRINT
150 PRINT "1.- FUNCION Y = COSEC (X) "
160 PRINT "2.- FUNCION Y = SEC (X) "
170 PRINT "3.- FUNCION Y = COTAG (X) "
180 PRINT:PRINT
190 PRINT "SEGUN LA OPCION DESEADA PULSE EL NUMERO QUE LE PRECEDE" ;
200 INPUT P
210 IF P=1 THEN 240
220 IF P=2 THEN 320
230 IF P=3 THEN 400
240 GOSUB 479
249 REM - ESTUDIO DE LA FUNCION Y = COSEC (X) -
250 PRINT "VALORES DE X", "VALORES DE Y=COSEC (X) "
260 PRINT R$
270 FOR X=A TO B STEP H
280 Y = 1/SIN(X)
290 PRINT X,Y
300 NEXT X
```

```

310 GOTO 180
320 GOSUB 479
329 REM - ESTUDIO DE LA FUNCION Y = SEC (X) -
330 PRINT "VALORES DE X", "VALORES DE Y=SEC(X)"
340 PRINT R$
350 FOR X=A TO B STEP H
360 Y=1/COS(X)
370 PRINT X,Y
380 NEXT X
390 GOTO 180
400 GOSUB 479
409 REM - ESTUDIO DE LA FUNCION Y = COTAG (X) -
410 PRINT "VALORES DE X", "VALORES DE Y=COTAG(X)"
420 PRINT R$
430 FOR X=A TO B STEP H
440 Y=COS(X)/SIN(X)
450 PRINT X,Y
460 NEXT X
470 GOTO 180
479 REM - SUBROUTINA DE ENTRADA DE DATOS -
480 PRINT "INTRODUZCA LOS EXTREMOS DEL INTERVALO A TABULAR" ;
490 INPUT A,B
500 PRINT "INTRODUZCA EL VALOR DEL INCREMENTO";
510 INPUT H
520 PRINT
530 PRINT
540 RETURN
550 END

```

## ejercicios resueltos:

1. Estudiar la función:  $y = \operatorname{cosec} x$  en el intervalo  $[1; 1.95]$  con un incremento de 0.05.

### SOLUCION :


ESTE PROGRAMA LE PERMITE ESTUDIAR LAS SIGUIENTES FUNCIONES :

- 1.- FUNCION  $Y = \operatorname{COSEC} (X)$
- 2.- FUNCION  $Y = \operatorname{SEC} (X)$
- 3.- FUNCION  $Y = \operatorname{COTAG} (X)$

SEGUN LA OPCION DESEADA PULSE EL NUMERO QUE LE PRECEDE? 1  
 INTRODUZCA LOS EXTREMOS DEL INTERVALO A TABULAR? 1 2  
 INTRODUZCA EL VALOR DEL INCREMENTO? .05

VALORES DE X	VALORES DE Y=COSEC (X)
1	1.188395
1.05	1.152840
1.1	1.122073
1.15	1.095574
1.2	1.072916
1.25	1.053758
1.3	1.037820
1.35	1.024881
1.4	1.014765
1.45	1.007340

1.5	1.002511
1.55	1.000216
1.6	1.000427
1.65	1.003145
1.7	1.008405
1.75	1.016275
1.8	1.026855
1.85	1.040285
1.9	1.056747
1.95	1.076473




2. Estudiar la función:  $y = \sec x$  en el intervalo  $[0, 1]$  con un incremento de 0.05.

**SOLUCION :**

SEGUN LA OPCION DESEADA PULSE EL NUMERO QUE LE PRECEDE? 2  
INTRODUZCA LOS EXTREMOS DEL INTERVALO A TABULAR? 0 1  
INTRODUZCA EL VALOR DEL INCREMENTO? .05

VALORES DE X	VALORES DE Y=SEC(X)
0	1.000000
.05	1.001251
.1	1.005021
.15	1.011356
.2	1.020339
.25	1.032085
.3	1.046752
.35	1.064540
.4	1.085704
.45	1.110559
.5	1.139494
.55	1.172987
.6	1.211628
.65	1.256149
.7	1.307459
.75	1.366701
.8	1.435324
.85	1.515190
.9	1.608726
.95	1.719149
1	1.850816



3. Estudiar la función:  $y = \cotg x$  en el intervalo  $[1.5; 1.64]$  con un incremento igual a 0.005.


**SOLUCION :**

SEGUN LA OPCION DESEADA PULSE EL NUMERO QUE LE PRECEDE? 3  
INTRODUZCA LOS EXTREMOS DEL INTERVALO A TABULAR? 1.5 1.64  
INTRODUZCA EL VALOR DEL INCREMENTO? .005

VALORES DE X VALORES DE  $Y=\cotag(X)$

\*\*\*\*\*

1.5000		.070915
1.5050		.065891
1.5100		.060871
1.5150		.055854
1.5200		.050840
1.5250		.045828
1.5300		.040819
1.5350		.035812
1.5400		.030806
1.5450		.025802
1.5500		.020799
1.5550		.015798
1.5600		.010797
1.5650		.005796
1.5700		.000796
1.5750	-	.004204
1.5800	-	.009204
1.5850	-	.014205
1.5900	-	.019206
1.5950	-	.024208
1.6000	-	.029212
1.6050	-	.034217
1.6100	-	.039224
1.6150	-	.044232
1.6200	-	.049243
1.6250	-	.054257
1.6300	-	.059273
1.6350	-	.064292
1.6400	-	.069314



# RAZONES TRIGONOMETRICAS DE UN ANGULO DADO

## introducción:

Este programa sirve para calcular las razones trigonométricas directas e inversas de un ángulo dado. El ángulo puede venir en radianes o en grados, minutos y segundos

## listado del programa:

```
10 REM *****
20 REM
30 REM          R A Z O N E S      T R I G O N O M E T R I C A S
40 REM
50 REM          D E      U N      A N G U L O      D A D O
60 REM
70 REM *****
100 PRINT
110 PRINT "SI EL ANGULO VIENE DADO EN RADIANES, PULSE 1 "
120 PRINT "SI EL ANGULO VIENE DADO EN GRADOS,MINUTOS Y SEGUNDOS, PULSE 2"
130 INPUT P
140 PRINT
150 PRINT
160 IF P=1 THEN GOTO 280
169 REM - PASO DE GRADOS, MINUTOS Y SEGUNDOS A RADIANES -
170 PRINT "INTRODUZCA LOS GRADOS, MINUTOS Y SEGUNDOS DEL ANGULO DADO"
180 PRINT "GRADOS      =";
190 INPUT G
200 PRINT "MINUTOS      =";
210 INPUT M
220 PRINT "SEGUNDOS      =";
230 INPUT S
240 A=G+M/60+S/3600
250 B=INT(A/360)
260 R=A*.01745329-B*2*PI
270 GOTO 300
280 PRINT "INTRODUZCA EL ANGULO EN RADIANES";
290 INPUT R
300 F1=SIN(R)
310 F2=COS(R)
320 F3=F1/F2
330 F4=1/F1
340 F5=1/F2
```

```

350 F6=1/F3
360 PRINT
370 PRINT
379 REM - SALIDA DE RESULTADOS -
380 PRINT "SEN(X)  =";F1
390 PRINT "COS(X)  =";F2
400 PRINT "TAG(X)  =";F3
410 PRINT "COSEC(X)=";F4
420 PRINT "SEC (X) =";F5
430 PRINT "COTAG(X)=";F6
440 END

```

## ejercicios resueltos:

1. Calcular las razones trigonométricas de un ángulo igual a 2.54 radianes.

### SOLUCION :

SI EL ANGULO VIENE DADO EN Radianes, PULSE 1  
 SI EL ANGULO VIENE DADO EN GRADOS,MINUTOS Y SEGUNDOS, PULSE 2  
 ? 1

INTRODUZCA EL ANGULO EN Radianes? 2.54

```

SEN(X)  = .56595623
COS(X)  = -.82443529
TAG(X)  = -.68647744
COSEC(X)= 1.7669211
SEC (X) = -1.2129515
COTAG(X)=-1.4567121

```

2. Calcular las razones trigonométricas del ángulo: 35° 27' 45"

### SOLUCION :

SI EL ANGULO VIENE DADO EN Radianes, PULSE 1  
 SI EL ANGULO VIENE DADO EN GRADOS,MINUTOS Y SEGUNDOS, PULSE 2  
 ? 2

INTRODUZCA LOS GRADOS, MINUTOS Y SEGUNDOS DEL ANGULO DADO

```

GRADOS   =? 35
MINUTOS  =? 27
SEGUNDOS =? 45

```

```

SEN(X)  = .58016992
COS(X)  = .81449547
TAG(X)  = .71230589
COSEC(X)= 1.723633
SEC (X) = 1.2277539
COTAG(X)= 1.4038912

```

# FUNCIONES TRIGONOMETRICAS RECIPROCAS

## introducción:

Este programa sirve para calcular las funciones recíprocas de las trigonométricas. Permite estudiar las siguientes funciones:

- |                                 |                                   |
|---------------------------------|-----------------------------------|
| 1. $y = \arcsin x$              | 4. $y = \operatorname{arccsc} x$  |
| 2. $y = \arccos x$              | 5. $y = \operatorname{arcsec} x$  |
| 3. $y = \operatorname{arctg} x$ | 6. $y = \operatorname{arccotg} x$ |

Los arcos obtenidos vienen dados en radianes.

Las funciones:  $y = \arcsin x$  e  $y = \arccos x$  solo están definidas en el intervalo  $[-1, 1]$

Las funciones:  $y = \operatorname{arccsc} x$  e  $y = \operatorname{arcsec} x$  están definidas en  $\mathbb{R} - (-1, 1)$ .

Las funciones:  $y = \operatorname{arctg} x$  e  $y = \operatorname{arccotg} x$  están definidos para todo  $x$  real.

## listado del programa:

```
10 REM *****
20 REM
30 REM          E  S  T  U  D  I  O          D  E          L  A  S          F  U  N  C  I  O  N  E  S
40 REM
50 REM          T  R  I  G  O  N  O  M  E  T  R  I  C  A  S          R  E  C  I  P  R  O  C  A  S
60 REM
70 REM *****
100 PRINT
110 R$="*****"
120 PRINT "ESTE PROGRAMA LE PERMITE ESTUDIAR LAS SIGUIENTES FUNCIONES : "
130 PRINT
140 PRINT
150 PRINT "1.- FUNCION Y = ARC SEN (X)"
160 PRINT "2.- FUNCION Y = ARC COS (X)"
170 PRINT "3.- FUNCION Y = ARC TAG (X)"
180 PRINT "4.- FUNCION Y = ARC COSEC (X)"
190 PRINT "5.- FUNCION Y = ARC SEC (X)"
200 PRINT "6.- FUNCION Y = ARC COTAG (X)"
210 PRINT
220 PRINT
```

```
230 PRINT "SEGUN LA OPCION DESEADA PULSE EL NUMERO QUE LE PRECEDE" ;
240 INPUT P
250 ON P GOTO 259,339,419,499,579,659
259 REM - ESTUDIO DE LA FUNCION  $Y = \text{ARC SEN}(X)$  -
260 GOSUB 739
270 PRINT "VALORES DE X", "VALORES DE  $Y=\text{ARC SEN}(X)$ "
280 PRINT R$
290 FOR X=A TO B STEP H
300 Y=ASN(X)
310 PRINT X,,Y
320 NEXT X
330 GOTO 210
339 REM - ESTUDIO DE LA FUNCION  $Y = \text{ARC COS}(X)$  -
340 GOSUB 739
350 PRINT "VALORES DE X", "VALORES DE  $Y=\text{ARC COS}(X)$ "
360 PRINT R$
370 FOR X=A TO B STEP H
380 Y=PI/2-ASN(X)
390 PRINT X,,Y
400 NEXT X
410 GOTO 210
419 REM - ESTUDIO DE LA FUNCION  $Y = \text{ARC TAG}(X)$  -
420 GOSUB 739
430 PRINT "VALORES DE X", "VALORES DE  $Y=\text{ARC TAG}(X)$ "
440 PRINT R$
450 FOR X=A TO B STEP H
460 Y=ATN(X)
470 PRINT X,,Y
480 NEXT X
490 GOTO 210
499 REM - ESTUDIO DE LA FUNCION  $Y = \text{ARC COSEC}(X)$  -
500 GOSUB 739
510 PRINT "VALORES DE X", "VALORES DE  $Y=\text{ARC COSEC}(X)$ "
520 PRINT R$
530 FOR X=A TO B STEP H
540 Y=ASN(1/X)
550 PRINT X,,Y
560 NEXT X
570 GOTO 210
579 REM - ESTUDIO DE LA FUNCION  $Y = \text{ARC SEC}(X)$  -
580 GOSUB 739
590 PRINT "VALORES DE X", "VALORES DE  $Y=\text{ARC SEC}(X)$ "
600 PRINT R$
610 FOR X=A TO B STEP H
620 Y=PI/2-ASN(1/X)
630 PRINT X,,Y
640 NEXT X
650 GOTO 210
659 REM - ESTUDIO DE LA FUNCION  $Y = \text{ARC COTAG}(X)$  -
660 GOSUB 739
670 PRINT "VALORES DE X", "VALORES DE  $Y=\text{ARC COTAG}(X)$ "
680 PRINT R$
690 FOR X=A TO B STEP H
700 Y=ATN(1/X)
710 PRINT X,,Y
720 NEXT X
730 GOTO 210
739 REM - SUBROUTINA DE ENTRADA DE DATOS -
740 PRINT "INTRODUZCA LOS EXTREMOS DEL INTERVALO A TABULAR";
750 INPUT A,B
760 PRINT "INTRODUZCA EL VALOR DEL INCREMENTO";
770 INPUT H
```



```

780 PRINT
790 PRINT
800 RETURN
810 END

```

## ejercicios resueltos:

1. Estudiar la función:  $y = \arcsin x$  en el intervalo  $[0, 1]$  con un incremento de 0.05.


### SOLUCION :

ESTE PROGRAMA LE PERMITE ESTUDIAR LAS SIGUIENTES FUNCIONES :

- 1.- FUNCION  $Y = \arcsin (X)$
- 2.- FUNCION  $Y = \arccos (X)$
- 3.- FUNCION  $Y = \arctan (X)$
- 4.- FUNCION  $Y = \operatorname{arccsc} (X)$
- 5.- FUNCION  $Y = \operatorname{arcsec} (X)$
- 6.- FUNCION  $Y = \operatorname{arccot} (X)$

SEGUN LA OPCION DESEADA PULSE EL NUMERO QUE LE PRECEDE? 1  
 INTRODUZCA LOS EXTREMOS DEL INTERVALO A TABULAR? 0  
 INTRODUZCA EL VALOR DEL INCREMENTO? .05

VALORES DE X	VALORES DE $Y = \arcsin (X)$
*****	*****
0	0
.05	.050020857
.1	.10016742
.15	.15056827
.2	.20135792
.25	.25268026
.3	.30469265
.35	.3575711
.4	.41151685
.45	.46676534
.5	.52359878
.55	.58236424
.6	.64350111
.65	.70758444
.7	.7753975
.75	.84806208
.8	.92729522
.85	1.0159853
.9	1.1197695
.95	1.2532359
1	1.5707963




2. Estudiar la función:  $y = \operatorname{arccsc} x$  en el intervalo  $[1, 2]$  con un incremento de 0.1.

### SOLUCION :

SEGUN LA OPCION DESEADA PULSE EL NUMERO QUE LE PRECEDE? 4  
 INTRODUZCA LOS EXTREMOS DEL INTERVALO A TABULAR? 1 2  
 INTRODUZCA EL VALOR DEL INCREMENTO? .1

VALORES DE X	VALORES DE Y=ARC COSEC (X)
*****	*****
1	1.5707963
1.1	1.1410967
1.2	.98511078
1.3	.87763642
1.4	.79560295
1.5	.72972766
1.6	.67513153
1.7	.62887493
1.8	.58903097
1.9	.55426183
2	.52359878




**3.** Estudiar la función:  $y = \arccos x$  en el intervalo  $[-1; -0.05]$  con un incremento de 0.05.

### S O L U C I O N :

SEGUN LA OPCION DESEADA PULSE EL NUMERO QUE LE PRECEDE? 2  
 INTRODUZCA LOS EXTREMOS DEL INTERVALO A TABULAR? -1 -.05  
 INTRODUZCA EL VALOR DEL INCREMENTO? .05

VALORES DE X	VALORES DE Y=ARC COS (X)
*****	*****
-1	3.1415927
-.95	2.8240322
-.9	2.6905658
-.85	2.5867816
-.8	2.4980915
-.75	2.4188584
-.7	2.3461938
-.65	2.2783808
-.6	2.2142974
-.55	2.1531606
-.5	2.0943951
-.45	2.0375617
-.4	1.9823132
-.35	1.9283674
-.3	1.875489
-.25	1.8234766
-.2	1.7721542
-.15	1.7213646
-.1	1.6709637
-.05	1.6208172



# VALOR DE LAS FUNCIONES TRIGONOMETRICAS RECIPROCAS

## introducción:

Este programa sirve para calcular las funciones recíprocas de las funciones trigonométricas para un valor dado. Los resultados se expresan en radianes.

## listado del programa:

```
10 REM *****
20 REM
30 REM          VALOR DE LAS FUNCIONES TRIG.
40 REM
50 REM          RECIPROCAS DE UN X DADO
60 REM
70 REM *****

100 N$="NO EXISTE NINGUN ANGULO"
105 R$="RADIANES"
110 PRINT:PRINT:PRINT
120 PRINT "INTRODUZCA EL VALOR DE X";
130 INPUT X
140 IF ABS(X)<1 AND X<>0 THEN GOTO 190
150 IF ABS(X)>1 THEN GOTO 300
160 IF X=1 THEN GOTO 410
170 IF X=-1 THEN GOTO 480
180 IF X=0 THEN 550
190 A=ASN(X)
200 B=1.5707963-ASN(X)
210 C=ATN(X)
220 G=ATN(1/X)
230 PRINT "ARC SEN(";X;")   =";A;R$
240 PRINT "ARC COS(";X;")   =";B;R$
250 PRINT "ARC TAG(";X;")   =";C;R$
260 PRINT "ARC COSEC(";X;") =";N$
270 PRINT "ARC SEC(";X;")   =";N$
280 PRINT "ARC COTAG(";X;") =";G;R$
290 GOTO 110
300 D=ASN(1/X)
310 F=1.5707963-ASN(1/X)
320 G=ATN(1/X)
330 C=ATN(X)
```

```

340 PRINT "ARC SEN(";X;")   =";N$
350 PRINT "ARC COS(";X;")   =";N$
360 PRINT "ARC TAG(";X;")   =";C;R$
370 PRINT "ARC COSEC(";X;") =";D;R$
380 PRINT "ARC SEC  (";X;") =";F;R$
390 PRINT "ARC COTAG(";X;") =";G;R$
400 GOTO 110
405 PI= 3.141592654
410 PRINT "ARC SEN (1)  =";PI/2 ;R$
420 PRINT "ARC COS(1)   = 0 ";R$
430 PRINT "ARC TAG(1)   =";PI/4 ;R$
440 PRINT "ARC COSEC(1) =";PI/2;R$
450 PRINT "ARC SEC (1)  = 0 ";R$
460 PRINT "ARC COTAG(1) =";PI/4;R$
470 GOTO 110
480 PRINT "ARC SEN (-1) =";-PI/2;R$
490 PRINT "ARC COS(-1)  =";PI;R$
500 PRINT "ARC TAG(-1)  =";-PI/4 ;R$
510 PRINT "ARC COSEC(-1) =";-PI/2;R$
520 PRINT "ARC SEC (-1) =";PI;R$
530 PRINT "ARC COTAG(-1) =";3*PI/4;R$
540 GOTO 110
550 PRINT "ARC SEN (0)  =";0;R$
560 PRINT "ARC COS(0)   =";PI/2;R$
570 PRINT "ARC TAG(0)   =";0;R$
580 PRINT "ARC COSEC(0) =";N$
590 PRINT "ARC SEC (0)  =";N$
600 PRINT "ARC COTAG(0) =";PI/2;R$
610 GOTO 110
620 END

```

## ejercicios resueltos:

**1.** Calcular el valor de las funciones trigonométricas recíprocas para  $x = 1.54$  y para  $x = 0$ .

### SOLUCION :

```

INTRODUZCA EL VALOR DE X? 1.54
ARC SEN( 1.54 )   =NO EXISTE NINGUN ANGULO
ARC COS( 1.54 )   =NO EXISTE NINGUN ANGULO
ARC TAG( 1.54 )   = .99487773 RADIANTES
ARC COSEC( 1.54 ) = .70673027 RADIANTES
ARC SEC ( 1.54 )  = .86406603 RADIANTES
ARC COTAG( 1.54 ) = .5759186 RADIANTES

```

```

INTRODUZCA EL VALOR DE X? 0
ARC SEN (0)  = 0 RADIANTES
ARC COS(0)   = 1.5707963 RADIANTES
ARC TAG(0)   = 0 RADIANTES
ARC COSEC(0) =NO EXISTE NINGUN ANGULO
ARC SEC (0)  =NO EXISTE NINGUN ANGULO
ARC COTAG(0) = 1.5707963 RADIANTES

```

## 2. Calcular el valor de las funciones trigonométricas recíprocas para

$x = 1$        $y$        $x = -1$ .

### SOLUCION :

INTRODUZCA EL VALOR DE X? 1

ARC SEN (1) = 1.5707963 RADIANS

ARC COS(1) = 0 RADIANS

ARC TAG(1) = .78539816 RADIANS

ARC COSEC(1) = 1.5707963 RADIANS

ARC SEC (1) = 0 RADIANS

ARC COTAG(1) = .78539816 RADIANS

INTRODUZCA EL VALOR DE X?-1

ARC SEN (-1) =-1.5707963 RADIANS

ARC COS(-1) = 3.1415927 RADIANS

ARC TAG(-1) =-.78539816 RADIANS

ARC COSEC(-1) =-1.5707963 RADIANS

ARC SEC (-1) = 3.1415927 RADIANS

ARC COTAG(-1) = 2.3561945 RADIANS

## 3. Calcular el valor de las funciones trigonométricas recíprocas para

$x = -0.234$        $y$        $x = 0.785$

### SOLUCION :

INTRODUZCA EL VALOR DE X?-.234

ARC SEN(-.234 ) =-.23618988 RADIANS

ARC COS(-.234 ) = 1.8069862 RADIANS

ARC TAG(-.234 ) =-.22986408 RADIANS

ARC COSEC(-.234 ) =NO EXISTE NINGUN ANGULO

ARC SEC(-.234 ) =NO EXISTE NINGUN ANGULO

ARC COTAG(-.234 ) =-1.3409322 RADIANS

INTRODUZCA EL VALOR DE X? .785

ARC SEN( .785 ) = .90269613 RADIANS

ARC COS( .785 ) = .66810017 RADIANS

ARC TAG( .785 ) = .66552744 RADIANS

ARC COSEC( .785 ) =NO EXISTE NINGUN ANGULO

ARC SEC( .785 ) =NO EXISTE NINGUN ANGULO

ARC COTAG( .785 ) = .90526888 RADIANS

# FUNCIONES HIPERBOLICAS DIRECTAS

## introducción:

Este programa permite estudiar las funciones hiperbólicas directas. El cálculo se realiza a partir de las definiciones de las funciones, del siguiente modo:

$$\begin{aligned}\operatorname{sen} h x &= \frac{e^x - e^{-x}}{2} & \cos h x &= \frac{e^x + e^{-x}}{2} \\ \operatorname{tg} h x &= \frac{e^x - e^{-x}}{e^x + e^{-x}}\end{aligned}$$

Es preciso introducir los límites del intervalo así como el incremento de la variable  $x$ .

## listado del programa:

```
10 REM *****
20 REM
30 REM          ESTUDIO          DE          LAS          FUNCIONES
40 REM
50 REM          HIPERBOLICAS          DIRECTAS
60 REM
70 REM *****
100 PRINT
110 R$="*****"
120 PRINT "ESTE PROGRAMA LE PERMITE ESTUDIAR LAS SIGUIENTES FUNCIONES : "
130 PRINT
140 PRINT
150 PRINT "1.- FUNCION Y = SENH (X) "
160 PRINT "2.- FUNCION Y = COSH (X) "
170 PRINT "3.- FUNCION Y = TAGH (X) "
180 PRINT:PRINT
190 PRINT "SEGUN LA OPCION DESEADA PULSE EL NUMERO QUE LE PRECEDE" ;
200 INPUT P
210 IF P=1 THEN 240
220 IF P=2 THEN 320
230 IF P=3 THEN 400
240 GOSUB 479
249 REM - ESTUDIO DE LA FUNCION Y = SENH (X) -
250 PRINT "VALORES DE X", "VALORES DE Y=SENH(X)"
260 PRINT R$
```

```

270 FOR X=A TO B STEP H
280 Y = (EXP(X)-EXP(-X))/2
290 PRINT X,Y
300 NEXT X
310 GOTO 180
320 GOSUB 479
329 REM - ESTUDIO DE LA FUNCION Y = COSH (X) -
330 PRINT "VALORES DE X", "VALORES DE Y=COSH(X)"
340 PRINT R$
350 FOR X=A TO B STEP H
360 Y = (EXP(X)+EXP(-X))/2
370 PRINT X,Y
380 NEXT X
390 GOTO 180
400 GOSUB 479
409 REM - ESTUDIO DE LA FUNCION Y = TAGH (X) -
410 PRINT "VALORES DE X", "VALORES DE Y=TAGH(X)"
420 PRINT R$
430 FOR X=A TO B STEP H
440 Y=(EXP(X)-EXP(-X))/(EXP(X)+EXP(-X))
450 PRINT X,Y
460 NEXT X
470 GOTO 180
479 REM - SUBROUTINA DE ENTRADA DE DATOS -
480 PRINT "INTRODUZCA LOS EXTREMOS DEL INTERVALO A TABULAR" ;
490 INPUT A,B
500 PRINT "INTRODUZCA EL VALOR DEL INCREMENTO";
510 INPUT H
520 PRINT
530 PRINT
540 RETURN
550 END

```

## ejercicios resueltos:

1. Estudiar la función:  $y = \cosh x$  en el intervalo  $[-1, 1]$  con un incremento de 0.05.

### SOLUCION :

ESTE PROGRAMA LE PERMITE ESTUDIAR LAS SIGUIENTES FUNCIONES :

- 1.- FUNCION Y = SENH (X)
- 2.- FUNCION Y = COSH (X)
- 3.- FUNCION Y = TAGH (X)

SEGUN LA OPCION DESEADA PULSE EL NUMERO QUE LE PRECEDE? 2  
 INTRODUZCA LOS EXTREMOS DEL INTERVALO A TABULAR?-1 1  
 INTRODUZCA EL VALOR DEL INCREMENTO? .05

VALORES DE X	VALORES DE Y=COSH(X)
*****	*****
-1	1.5430806
-.75	1.4862253
-.9	1.4330864
-.85	1.3835309

-.8	1.3374349
-.75	1.2946833
-.7	1.255169
-.65	1.2187933
-.6	1.1854652
-.55	1.1551014
-.5	1.127626
-.45	1.1029702
-.4	1.0810724
-.35	1.0618778
-.3	1.0453385
-.25	1.0314131
-.2	1.0200668
-.15	1.0112711
-.1	1.0050042
-.05	1.0012503
3.6379788E-12	1
.05	1.0012503
.1	1.0050042
.15	1.0112711
.2	1.0200668
.25	1.0314131
.3	1.0453385
.35	1.0618778
.4	1.0810724
.45	1.1029702
.5	1.127626
.55	1.1551014
.6	1.1854652
.65	1.2187933
.7	1.255169
.75	1.2946833
.8	1.3374349
.85	1.3835309
.9	1.4330864
.95	1.4862253
1	1.5430806

**2.** Estudiar la función:  $y = \operatorname{sen} h x$  en el intervalo  $[-3, -2]$  con un incremento de 0.05.

**SOLUCION :**

SEGUN LA OPCION DESEADA PULSE EL NUMERO QUE LE PRECEDE? 1  
 INTRODUCZA LOS EXTREMOS DEL INTERVALO A TABULAR? -3 -2  
 INTRODUCZA EL VALOR DEL INCREMENTO? .05

VALORES DE X	VALORES DE Y=SENH(X)
*****	*****
-3	-10.017875
-2.95	-9.526807
-2.9	-9.0595611
-2.85	-8.6149688
-2.8	-8.1919184
-2.75	-7.789352
-2.7	-7.4062631
-2.65	-7.0416937
-2.6	-6.6947322
-2.55	-6.3645111



-2.5	-6.0502045
-2.45	-5.7510266
-2.4	-5.4662292
-2.35	-5.1951003
-2.3	-4.9369618
-2.25	-4.6911683
-2.2	-4.4571052
-2.15	-4.2341871
-2.1	-4.0218567
-2.05	-3.8195831
-2	-3.6268604

**3.** Estudiar la función:  $y = \operatorname{tag} h x$  en el intervalo  $[2, 3]$  con un incremento de 0.1.

**S O L U C I O N :**

SEGUN LA OPCION DESEADA PULSE EL NUMERO QUE LE PRECEDE? 3  
 INTRODUZCA LOS EXTREMOS DEL INTERVALO A TABULAR? 2 3  
 INTRODUZCA EL VALOR DEL INCREMENTO? .1

VALORES DE X VALORES DE Y=TAGH(X)

\*\*\*\*\*

2	.96402758
2.1	.97045194
2.2	.97574313
2.3	.9800964
2.4	.98367486
2.5	.9866143
2.6	.9890274
2.7	.99100745
2.8	.99263152
2.9	.99396317
3	.99505475

# FUNCIONES HIPERBOLICAS INVERSAS

## introducción:

Este programa sirve para estudiar las funciones hiperbólicas inversas:  $y = \operatorname{cosech} h x$ ,  $y = \operatorname{sech} h x$ ,  $y = \operatorname{cotg} h x$ .

El calculo se realiza directamente a partir de las definiciones de las funciones hiperbólicas directas.

Es necesario introducir los límites del intervalo así como el incremento de la variable  $x$ .

## listado del programa:

```
10 REM *****
20 REM
30 REM          ESTUDIO      DE      LAS      FUNCIONES
40 REM
50 REM          HIPERBOLICAS      INVERSAS
60 REM
70 REM *****
100 PRINT
110 R$="*****"
120 PRINT "ESTE PROGRAMA LE PERMITE ESTUDIAR LAS SIGUIENTES FUNCIONES : "
130 PRINT
140 PRINT
150 PRINT "1.- FUNCION Y = COSECH (X)"
160 PRINT "2.- FUNCION Y = SECH (X)"
170 PRINT "3.- FUNCION Y = COTAGH (X)"
180 PRINT:PRINT
190 PRINT "SEGUN LA OPCION DESEADA PULSE EL NUMERO QUE LE PRECEDE" ;
200 INPUT P
210 IF P=1 THEN 240
220 IF P=2 THEN 320
230 IF P=3 THEN 400
240 GOSUB 479
249 REM - ESTUDIO DE LA FUNCION Y = COSECH (X) -
250 PRINT "VALORES DE X", "VALORES DE Y=COSECH (X)"
260 PRINT R$
270 FOR X=A TO B STEP H
280 Y = 2/(EXP(X)-EXP(-X))
290 PRINT X,Y
```

```

300 NEXT X
310 GOTO 180
320 GOSUB 479
329 REM - ESTUDIO DE LA FUNCION Y = SECH (X) -
330 PRINT "VALORES DE X", "VALORES DE Y=SECH (X)"
340 PRINT R$
350 FOR X=A TO B STEP H
360 Y=2/(EXP(X)+EXP(-X))
370 PRINT X,Y
380 NEXT X
390 GOTO 180
400 GOSUB 479
409 REM - ESTUDIO DE LA FUNCION Y = COTAGH (X) -
410 PRINT "VALORES DE X", "VALORES DE Y=COTAGH (X)"
420 PRINT R$
430 FOR X=A TO B STEP H
440 Y=(EXP(X)+EXP(-X))/(EXP(X)-EXP(-X))
450 PRINT X,Y
460 NEXT X
470 GOTO 180
479 REM - SUBROUTINA DE ENTRADA DE DATOS -
480 PRINT "INTRODUZCA LOS EXTREMOS DEL INTERVALO A TABULAR" ;
490 INPUT A,B
500 PRINT "INTRODUZCA EL VALOR DEL INCREMENTO";
510 INPUT H
520 PRINT
530 PRINT
540 RETURN
550 END

```

## ejercicios resueltos:

1. Estudiar la función:  $y = \operatorname{cosec} h x$  en el intervalo  $[1, 2]$  con un incremento de 0.05.

### SOLUCION :

ESTE PROGRAMA LE PERMITE ESTUDIAR LAS SIGUIENTES FUNCIONES :

- 1.- FUNCION Y = COSECH (X)
- 2.- FUNCION Y = SECH (X)
- 3.- FUNCION Y = COTAGH (X)

SEGUN LA OPCION DESEADA PULSE EL NUMERO QUE LE PRECEDE? 1  
 INTRODUZCA LOS EXTREMOS DEL INTERVALO A TABULAR? 1 2  
 INTRODUZCA EL VALOR DEL INCREMENTO? .05

VALORES DE X	VALORES DE Y=COSECH (X)
*****	*****
1	.85091813
1.05	.79753931
1.1	.74870055
1.15	.70383969
1.2	.66248798
1.25	.62425126
1.3	.58879554

1.35	.55583575
1.4	.52512693
1.45	.49645725
1.5	.46964244
1.55	.44452128
1.6	.42095197
1.65	.39880916
1.7	.37798153
1.75	.35836971
1.8	.33988469
1.85	.32244634
1.9	.3059823
1.95	.29042694
2	.27572056

**2.** Estudiar la función:  $y = \operatorname{sech} x$  en el intervalo  $[-1; 1.1]$  con un incremento de 0.1.

**S O L U C I O N :**

SEGUN LA OPCION DESEADA PULSE EL NUMERO QUE LE PRECEDE? 2  
 INTRODUZCA LOS EXTREMOS DEL INTERVALO A TABULAR? -1 1.1  
 INTRODUZCA EL VALOR DEL INCREMENTO? .1

VALORES DE X	VALORES DE Y=SECH (X)
*****	
-1	.64805427
.9	.69779464
-.8	.74769992
-.7	.79670546
-.6	.84355069
-.5	.88681888
-.4	.92500745
-.3	.95662791
-.2	.980328
-.1	.99502075
-5.4569682E-12	1
.1	.99502075
.2	.980328
.3	.95662791
.4	.92500745
.5	.88681888
.6	.84355069
.7	.79670546
.8	.74769992
.9	.69779464
1	.64805427
1.1	.59933406

# RAZONES HIPERBOLICAS DE UN ANGULO DADO

## introducción:

Este programa sirve para calcular las razones hiperbólicas directas e inversas de un ángulo dado.

## listado del programa:

```
10 REM *****
20 REM
30 REM           R A Z O N E S       H I P E R B O L I C A S
40 REM
50 REM           D E       U N       A N G U L O       D A D O
60 REM
70 REM *****
100 PRINT
110 PRINT
120 PRINT "INTRODUZCA EL VALOR DE X";
130 INPUT X
140 A=EXP(X)-EXP(-X)
150 B=EXP(X)+EXP(-X)
160 F1=A/2
170 F2=B/2
180 F3=A/B
190 F4=1/F1
200 F5=1/F2
210 F6=1/F3
220 PRINT
230 PRINT
239 REM -SALIDA DE RESULTADOS -
240 PRINT "SENH(";X;") =";F1
250 PRINT "COSH(";X;") =";F2
260 PRINT "TAGH(";X;") =";F3
270 PRINT "COSECH(";X;") =";F4
280 PRINT "SECH(";X;") =";F5
290 PRINT "COTAGH(";X;") =";F6
300 GOTO 100
310 END
```

## ejercicios resueltos:

**1.** Calcular las razones hiperbólicas directas e inversas de  $x = 2.4$  y de  $x = -6.78$ .

### **SOLUCION :**

INTRODUZCA EL VALOR DE X

INTRODUZCA EL VALOR DE X? 2.4

$\text{SENH}(2.4) = 5.4662292$   
 $\text{COSH}(2.4) = 5.5569472$   
 $\text{TAGH}(2.4) = .98367486$   
 $\text{COSECH}(2.4) = .18294147$   
 $\text{SECH}(2.4) = .17995492$   
 $\text{COTAGH}(2.4) = 1.0165961$

INTRODUZCA EL VALOR DE X? -6.78

$\text{SENH}(-6.78) = -440.03379$   
 $\text{COSH}(-6.78) = 440.03493$   
 $\text{TAGH}(-6.78) = -.99999742$   
 $\text{COSECH}(-6.78) = -.0022725527$   
 $\text{SECH}(-6.78) = .0022725469$   
 $\text{COTAGH}(-6.78) = -1.0000026$

**2.** Calcular las razones hiperbólicas directas e inversas de  $x = 2.6$  y de  $x = -0.25$ .

### **SOLUCION :**

INTRODUZCA EL VALOR DE X? 2.6

$\text{SENH}(2.6) = 6.6947322$   
 $\text{COSH}(2.6) = 6.7690058$   
 $\text{TAGH}(2.6) = .9890274$   
 $\text{COSECH}(2.6) = .14937117$   
 $\text{SECH}(2.6) = .14773218$   
 $\text{COTAGH}(2.6) = 1.0110943$

INTRODUZCA EL VALOR DE X? -.25

$\text{SENH}(-.25) = -.25261232$   
 $\text{COSH}(-.25) = 1.0314131$   
 $\text{TAGH}(-.25) = -.24491866$   
 $\text{COSECH}(-.25) = -3.9586352$   
 $\text{SECH}(-.25) = .96954363$   
 $\text{COTAGH}(-.25) = -4.0829882$

# FUNCIONES HIPERBOLICAS RECIPROCAS

## introducción:

Este programa sirve para estudiar las funciones hiperbólicas recíprocas de las funciones hiperbólicas directas, es decir:

$y = \operatorname{arc} \operatorname{sen} h x$                        $y = \operatorname{arc} \operatorname{cos} h x$                        $y = \operatorname{arc} \operatorname{tg} h x$

La función  $y = \operatorname{arc} \operatorname{sen} h x$  está definida para todo valor de  $x$ .

La función  $y = \operatorname{arc} \operatorname{cos} h x$  está definida en el abierto  $(1, \infty)$ .

La función  $y = \operatorname{arc} \operatorname{tag} h x$  está definida en el abierto  $(-1, 1)$ .

Como siempre hemos de introducir los límites del intervalo a estudiar, así como el incremento de la variable  $x$ .

## listado del programa:

```
10 REM *****
20 REM
30 REM          ESTUDIO  DE    LAS  FUNCIONES
40 REM
50 REM          HIPERBOLICAS    RECIPROCAS
60 REM
70 REM *****
100 PRINT
110 R$="*****"
120 PRINT "ESTE PROGRAMA LE PERMITE ESTUDIAR LAS SIGUIENTES FUNCIONES"
130 PRINT
140 PRINT
150 PRINT "1.- FUNCION  Y = ARC SENH (X)"
160 PRINT "2.- FUNCION  Y = ARC COSH (X)"
170 PRINT "3.- FUNCION  Y = ARC TAGH (X)"
180 PRINT:PRINT
190 PRINT "SEGUN LA OPCION DESEADA PULSE EL NUMERO QUE LE PRECEDE";
200 INPUT P
210 IF P=1 THEN 240
220 IF P=2 THEN 320
230 IF P=3 THEN 400
240 GOSUB 479
249 REM - ESTUDIO DE LA FUNCION  Y = ARC SENH (X) -
250 PRINT "VALORES DE X", "VALORES DE Y = ARC SENH (X)"
260 PRINT R$
```

```

270 FOR X=A TO B STEP H
280 Y=LOG(X+SQR(X*X+1))
290 PRINT X,Y
300 NEXT X
310 GOTO 180
320 GOSUB 479
329 REM - ESTUDIO DE LA FUNCION Y = ARC COSH (X) -
330 PRINT "VALORES DE X", "VALORES DE Y = ARC COSH (X)"
340 PRINT R$
350 FOR X=A TO B STEP H
360 Y=LOG(X+SQR(X*X-1))
370 PRINT X,Y
380 NEXT X
390 GOTO 180
400 GOSUB 479
409 REM - ESTUDIO DE LA FUNCION Y = ARC TAGH (X) -
410 PRINT "VALORES DE X", "VALORES DE Y = ARC TAGH (X)"
420 PRINT R$
430 FOR X=A TO B STEP H
440 Y=.5*LOG((1+X)/(1-X))
450 PRINT X,Y
460 NEXT X
470 GOTO 180
479 REM - SURUTINA DE ENTRADA DE DATOS -
480 PRINT "INTRODUZCA LOS EXTREMOS DEL INTERVALO A TABULAR";
490 INPUT A,B
500 PRINT "INTRODUZCA EL VALOR DEL INCREMENTO";
510 INPUT H
520 PRINT
530 PRINT
540 RETURN
550 END

```

## ejercicios resueltos:

1. Estudiar la función:  $y = \text{arc sen } h x$  en el intervalo  $[-1, 1.1]$  con un incremento igual a 0.05.

### SOLUCION :

ESTE PROGRAMA LE PERMITE ESTUDIAR LAS SIGUIENTES FUNCIONES

- 1.- FUNCION  $Y = \text{ARC SENH } (X)$
- 2.- FUNCION  $Y = \text{ARC COSH } (X)$
- 3.- FUNCION  $Y = \text{ARC TAGH } (X)$

SEGUN LA OPCION DESEADA PULSE EL NUMERO QUE LE PRECEDE? 1  
 INTRODUZCA LOS EXTREMOS DEL INTERVALO A TABULAR?-1 1.1  
 INTRODUZCA EL VALOR DEL INCREMENTO? .05

```

VALORES DE X      VALORES DE Y = ARC SENH (X)
*****
-1                -.88137359
-.95              -.8455727
-.9               -.80886694
-.85              -.77123743

```



-.8	-.73266826
-.75	-.69314718
-.7	-.65266657
-.65	-.61122431
-.6	-.5688249
-.55	-.52548045
-.5	-.48121183
-.45	-.43604967
-.4	-.39003532
-.35	-.34322156
-.3	-.29567305
-.25	-.24746646
-.2	-.19869011
-.15	-.14944312
-.1	-.099834079
-.05	-.04997919
3.6379788E-12	0
.05	.04997919
.1	.099834079
.15	.14944312
.2	.19869011
.25	.24746646
.3	.29567305
.35	.34322156
.4	.39003532
.45	.43604967
.5	.48121182
.55	.52548045
.6	.5688249
.65	.61122431
.7	.65266657
.75	.69314718
.8	.73266826
.85	.77123743
.9	.80886694
.95	.8455727
1	.88137359
1.05	.91629073
1.1	.95034693

**2.** Estudiar la función:  $y = \text{arc tag } h \ x$  en el intervalo  $[-.9, .9]$  con un incremento igual a 0.1.

**SOLUCION :**

SEGUN LA OPCION DESEADA PULSE EL NUMERO QUE LE PRECEDE? 3  
 INTRODUZCA LOS EXTREMOS DEL INTERVALO A TABULAR? -.9 .9  
 INTRODUZCA EL VALOR DEL INCREMENTO? .1

VALORES DE X	VALORES DE Y = ARC TAGH (X)
*****	
-.9	-1.4722195
-.8	-1.0986123
-.7	-.86730053
-.6	-.69314718
-.5	-.54930614
-.4	-.42364893
-.3	-.3095196
-.2	-.20273255

-.1	-.10033535
-4.5474735E-12	-1.4551915E-11
.1	.10033535
.2	.20273255
.3	.3095196
.4	.42364893
.5	.54930614
.6	.69314718
.7	.86730053
.8	1.0986123
.9	1.4722195

**3.** Estudiar la función:  $y = \text{arc cosh } x$  en el intervalo  $[1.1, 3.1]$  con un incremento igual a 0.2.

### **S O L U C I O N :**

SEGUN LA OPCION DESEADA PULSE EL NUMERO QUE LE PRECEDE? 2  
 INTRODUZCA LOS EXTREMOS DEL INTERVALO A TABULAR? 1.1      3.1  
 INTRODUZCA EL VALOR DEL INCREMENTO? .2

VALORES DE X	VALORES DE Y = ARC COSH (X)
*****	
1.1	.44356825
1.3	.75643291
1.5	.96242365
1.7	1.123231
1.9	1.2571958
2.1	1.3728591
2.3	1.4750448
2.5	1.5667992
2.7	1.6501935
2.9	1.7267109
3.1	1.7974566

# VALOR NUMERICO DE LA PRIMERA DERIVADA DE UNA FUNCION EN UN PUNTO

## introducción:

Este programa permite hallar el valor aproximado que toma la derivada de una función en un punto. La función a derivar se ha de definir en la línea 170, del siguiente modo: 170 DEF FN F(X) = ..... El cálculo de la derivada se realiza con arreglo a la definición de derivada.

## listado del programa:

```
10 REM *****
20 REM
30 REM          D E R I V A D A      D E      U N A      F U N C I O N
40 REM
50 REM                      E N      U N      P U N T O
60 REM
70 REM *****
100 PRINT
109 REM - ENTRADA DE DATOS -
110 PRINT "DEFINA LA FUNCION A DERIVAR EN LA LINEA 170"
120 PRINT "DEL MODO SIGUIENTE : 170 DEF FN F(X)=....."
130 PRINT "A CONTINUACION PULSA <CONT>"
140 STOP
150 PRINT "PUNTO EN EL QUE QUIERE ESTUDIAR LA DERIVADA XO =";
160 INPUT XO
170 DEF FN F(X)=
180 H=0
190 FOR J=1 TO 8
200 H1=H
210 X=XO+.5^J
220 H=(FN F(X)- FN F(XO))/(X-XO)
230 NEXT J
240 D=2*H-H1
250 PRINT
259 REM -SALIDA DEL RESULTADO -
260 PRINT "EL VALOR APROXIMADO DE LA DERIVADA EN EL PUNTO XO=";XO;"ES D(XO)=";D
270 PRINT
280 PRINT
290 PRINT
300 GOTO 150
310 END
```

## ejercicios resueltos:

**1.** Hallar la derivada de la función:  $y = x^x + \operatorname{sen} x \cdot e^x - \operatorname{Ln}(x+3)$   
en los puntos: 1, 1/2, 2.

### SOLUCION :

DEFINA LA FUNCION A DERIVAR EN LA LINEA 170  
DEL MODO SIGUIENTE : 170 DEF FN F(X)=.....  
A CONTINUACION PULSA <CONT>  
170 DEF FN F(X)=X^X+SIN(X)\*EXP(X)-LOG(X+3)

PUNTO EN EL QUE QUIERE ESTUDIAR LA DERIVADA X0=? 1  
EL VALOR APROXIMADO DE LA DERIVADA EN EL PUNTO X0= 1 ES D(X0)= 4.5060422

PUNTO EN EL QUE QUIERE ESTUDIAR LA DERIVADA X0=? .5  
EL VALOR APROXIMADO DE LA DERIVADA EN EL PUNTO X0= .5 ES D(X0)= 2.1685925

PUNTO EN EL QUE QUIERE ESTUDIAR LA DERIVADA X0=? 2  
EL VALOR APROXIMADO DE LA DERIVADA EN EL PUNTO X0= 2 ES D(X0)= 10.21646

**2.** Hallar la derivada de la función:  $y = e^x + \cos x$  en los puntos  
1, -2, 0.

### SOLUCION :

DEFINA LA FUNCION A DERIVAR EN LA LINEA 170  
DEL MODO SIGUIENTE : 170 DEF FN F(X)=.....  
A CONTINUACION PULSA <CONT>  
170 DEF FN F(X)=EXP(X)+COS(X)

PUNTO EN EL QUE QUIERE ESTUDIAR LA DERIVADA X0=? 1  
EL VALOR APROXIMADO DE LA DERIVADA EN EL PUNTO X0= 1 ES D(X0)= 1.8767927

PUNTO EN EL QUE QUIERE ESTUDIAR LA DERIVADA X0=? -2  
EL VALOR APROXIMADO DE LA DERIVADA EN EL PUNTO X0=-2 ES D(X0)= 1.0446367

PUNTO EN EL QUE QUIERE ESTUDIAR LA DERIVADA X0=? 0  
EL VALOR APROXIMADO DE LA DERIVADA EN EL PUNTO X0= 0 ES D(X0)= .99999493

# ECUACIONES DE LA TANGENTE Y LA NORMAL A UNA CURVA, CRECIMIENTO Y DECRECIMIENTO DE UNA FUNCION EN UN PUNTO

## introducción:

Este programa permite obtener las ecuaciones de la recta tangente y normal a una curva en un punto. Además estudia, a partir del coeficiente angular de la recta tangente, el crecimiento o decrecimiento de la función en el punto considerado.

Es importante destacar que todo el cálculo es aproximado, y tener mucho cuidado en los puntos de discontinuidad de la función.

La función se define en la línea 170 del siguiente modo:

170 DEF FN F(X) = .....

## listado del programa:

```
10 REM *****
20 REM
30 REM      ECUACIONES DE LA TANGENTE Y LA
40 REM
50 REM      NORMAL A UNA CURVA EN UN PUNTO
60 REM
70 REM      CRECIMIENTO Y DECRECIMIENTO
80 REM
90 REM *****
100 PRINT
109 REM - ENTRADA DE DATOS -
110 PRINT "DEFINA LA FUNCION PARA LA QUE QUIERE HALLAR LA TANGENTE EN LA LINEA 1
170"
120 PRINT "DEL MODO SIGUIENTE : 170 DEF FN F(X)=....."
130 PRINT "A CONTINUACION PULSE <CONT>"
140 STOP
150 PRINT "PUNTO EN EL QUE QUIERE HALLAR LA TANGENTE Y LA NORMAL XO =";
160 INPUT XO
170 DEF FN F(X)=
180 H=0
190 FOR J=1 TO 8
200 H1=H
210 X=XO+.5^J
220 H=(FN F(X) - FN F(XO))/(X-XO)
230 NEXT J
```

```

240 D=2*H-H1
250 Y0=FN F(X0)
259 REM - REDONDEO DEL VALOR DE LA DERIVADA -
260 D1=INT(D*1000+.5)/1000
270 PRINT
280 PRINT "LA ECUACION DE LA TANGENTE A LA CURVA EN EL PUNTO X0 =" ; X0 ; "ES : "
290 PRINT "Y =" ; D1 ; "* X +" ; Y0-D1*X0
300 PRINT
310 IF D1=0 THEN GOTO 350
320 PRINT "LA ECUACION DE LA NORMAL A LA CURVA EN EL PUNTO X0 =" ; X0 ; "ES : "
330 PRINT "Y =" ; -1/D1 ; "* X +" ; Y0+X0/D1
340 GOTO 390
350 PRINT
360 PRINT "LA ECUACION DE LA NORMAL A LA CURVA EN EL PUNTO X0 =" ; X0 ; "ES : "
370 PRINT "X =" ; X0
380 REM
390 PRINT
400 IF D1>0 THEN 450
410 IF D1=0 THEN 470
420 PRINT
430 PRINT "LA FUNCION ES EstrictAMENTE DECRECIENTE EN EL PUNTO X0 =" ; X0
440 STOP
450 PRINT "LA FUNCION ES EstrictAMENTE CRECIENTE EN EL PUNTO X0 =" ; X0
460 STOP
470 PRINT "LA FUNCION TIENE UN PUNTO ESTACIONARIO EN EL PUNTO X0 =" ; X0
480 END

```

## ejercicios resueltos:

1. Hallar las ecuaciones de la tangente y la normal a la curva:

$y = x^2 - \cos x + \ln x$ , en el punto  $x_0 = 1$ . Decir si la función es creciente o decreciente en dicho punto. Análogo en los puntos  $x_0 = 2$ ,  $x_0 = 10$ .

### SOLUCION :

DEFINA LA FUNCION PARA LA QUE QUIERE HALLAR LA TANGENTE EN LA LINEA 170 DEL MODO SIGUIENTE: 170 DEF FN F(X)=.....  
A CONTINUACION PULSE <CONT>  
170 DEF FN F(X) = X\*X-COS(X)+LOG(X)

PUNTO EN EL QUE QUIERE HALLAR LA TANGENTE Y LA NORMAL X0 =? 1

LA ECUACION DE LA TANGENTE A LA CURVA EN EL PUNTO X0 = 1 ES :  
Y = 3.841 \* X + -3.3813023

LA ECUACION DE LA NORMAL A LA CURVA EN EL PUNTO X0 = 1 ES :  
Y = -.26034887 \* X + .72004656

LA FUNCION ES EstrictAMENTE CRECIENTE EN EL PUNTO X0 = 1

PUNTO EN EL QUE QUIERE HALLAR LA TANGENTE Y LA NORMAL X0 =? 2

LA ECUACION DE LA TANGENTE A LA CURVA EN EL PUNTO X0 = 2 ES :  
Y = 5.409 \* X + -5.708706

LA ECUACION DE LA NORMAL A LA CURVA EN EL PUNTO  $x_0 = 2$  ES :  
 $Y = -.18487706 * X + 5.4790481$

LA FUNCION ES Estrictamente creciente en el punto  $x_0 = 2$

PUNTO EN EL QUE QUIERE HALLAR LA TANGENTE Y LA NORMAL  $x_0 = ?$  10

LA ECUACION DE LA TANGENTE A LA CURVA EN EL PUNTO  $x_0 = 10$  ES :  
 $Y = 19.556 * X + -92.418343$

LA ECUACION DE LA NORMAL A LA CURVA EN EL PUNTO  $x_0 = 10$  ES :  
 $Y = -.051135201 * X + 103.65301$

LA FUNCION ES Estrictamente creciente en el punto  $x_0 = 10$

**2.** Hallar las ecuaciones de la tangente y la normal a la curva:  
 $y = e^x - \sin x$ , en los puntos: 1, 0, -2. Estudiar el crecimiento de la función en dichos puntos.

### SOLUCION :

DEFINA LA FUNCION PARA LA QUE QUIERE HALLAR LA TANGENTE EN LA LINEA 170  
 DEL MODO SIGUIENTE : 170 DEF FN F(X)=.....  
 A CONTINUACION PULSE <CONT>  
 170 DEF FN F(X)=EXP(X)-SIN(X)

PUNTO EN EL QUE QUIERE HALLAR LA TANGENTE Y LA NORMAL  $x_0 = ?$  1

LA ECUACION DE LA TANGENTE A LA CURVA EN EL PUNTO  $x_0 = 1$  ES :  
 $Y = 2.178 * X + -.30118916$

LA ECUACION DE LA NORMAL A LA CURVA EN EL PUNTO  $x_0 = 1$  ES :  
 $Y = -.45913682 * X + 2.3359477$

LA FUNCION ES Estrictamente creciente en el punto  $x_0 = 1$

PUNTO EN EL QUE QUIERE HALLAR LA TANGENTE Y LA NORMAL  $x_0 = ?$  0

LA ECUACION DE LA TANGENTE A LA CURVA EN EL PUNTO  $x_0 = 0$  ES :  
 $Y = 0 * X + 1$

LA ECUACION DE LA NORMAL A LA CURVA EN EL PUNTO  $x_0 = 0$  ES :  
 $X = 0$

LA FUNCION TIENE UN PUNTO ESTACIONARIO EN EL PUNTO  $x_0 = 0$

PUNTO EN EL QUE QUIERE HALLAR LA TANGENTE Y LA NORMAL  $X_0 = -2$

LA ECUACION DE LA TANGENTE A LA CURVA EN EL PUNTO  $X_0 = -2$  ES :  
 $Y = .551 * X + 2.1466327$

LA ECUACION DE LA NORMAL A LA CURVA EN EL PUNTO  $X_0 = -2$  ES :  
 $Y = -1.814882 * X + -2.5851314$

LA FUNCION ES ERICTAMENTE CRECIENTE EN EL PUNTO  $X_0 = -2$

### 3. Hallar las ecuaciones de la tangente y la normal a la curva:

$y = x^2 - 6x + 8$ , en los puntos: 0, 1, 3. Estudiar el crecimiento de la función en dichos puntos.

#### **SOLUCION :**

DEFINA LA FUNCION PARA LA QUE QUIERE HALLAR LA TANGENTE EN LA LINEA 170  
 DEL MODO SIGUIENTE : 170 DEF FN F(X)=.....  
 A CONTINUACION PULSE <CONT>  
 170 DEF FN F(X)=X\*X-6\*X+8

PUNTO EN EL QUE QUIERE HALLAR LA TANGENTE Y LA NORMAL  $X_0 = 0$

LA ECUACION DE LA TANGENTE A LA CURVA EN EL PUNTO  $X_0 = 0$  ES :  
 $Y = -6 * X + 8$

LA ECUACION DE LA NORMAL A LA CURVA EN EL PUNTO  $X_0 = 0$  ES :  
 $Y = .16666667 * X + 8$

LA FUNCION ES ERICTAMENTE DECRECIENTE EN EL PUNTO  $X_0 = 0$

PUNTO EN EL QUE QUIERE HALLAR LA TANGENTE Y LA NORMAL  $X_0 = 1$

LA ECUACION DE LA TANGENTE A LA CURVA EN EL PUNTO  $X_0 = 1$  ES :  
 $Y = -4 * X + 7$

LA ECUACION DE LA NORMAL A LA CURVA EN EL PUNTO  $X_0 = 1$  ES :  
 $Y = .25 * X + 2.75$

LA FUNCION ES ERICTAMENTE DECRECIENTE EN EL PUNTO  $X_0 = 1$

PUNTO EN EL QUE QUIERE HALLAR LA TANGENTE Y LA NORMAL  $X_0 = 3$

LA ECUACION DE LA TANGENTE A LA CURVA EN EL PUNTO  $X_0 = 3$  ES :  
 $Y = 0 * X + -1$

LA ECUACION DE LA NORMAL A LA CURVA EN EL PUNTO  $X_0 = 3$  ES :  
 $X = 3$

LA FUNCION TIENE UN PUNTO ESTACIONARIO EN EL PUNTO  $X_0 = 3$



# VALOR NUMERICO DE UN POLINOMIO Y SUS DERIVADAS EN UN PUNTO

## introducción:

Este es un interesante programa que permite obtener el valor numérico de un polinomio en un punto, así como el valor numérico de sus derivadas en ese punto.

El polinomio se introduce escribiendo los coeficientes  $C(J)$  del polinomio en la línea 160.

## listado del programa:

```
10 REM *****
20 REM
30 REM      VALOR      NUMERICO      DE      UN      POLINOMIO
40 REM
50 REM      Y      DE      SUS      DERIVADAS      EN      UN      PUNTO
60 REM
70 REM *****
100 PRINT
109 REM - ENTRADA DE DATOS -
110 PRINT "GRADO DEL POLINOMIO";
120 INPUT N
130 DIM C(N+1),D(2*N)
140 FOR J=0 TO N
150 PRINT "COEFICIENTE DE X^";N-J;
160 INPUT C(J)
170 NEXT J
180 PRINT
190 PRINT "PUNTO EN EL QUE DESEA ESTUDIAR LA FUNCION Y SUS DERIVADAS XO =";
200 INPUT XO
210 PRINT
220 FOR J=0 TO N
230 D(J)=C(N-J)
240 NEXT J
250 F=1
260 FOR J=1 TO N
270 FOR I=1 TO N
280 L=N+J-I-1
290 D(L)=D(L)+D(L+1)*XO
300 NEXT I
310 D(L)=F*D(L)
320 F=F*J
```

```

330 NEXT J
340 D(N)=F*D(N)
349 REM - SALIDA DE RESULTADOS -
350 PRINT "EL POLINOMIO EN X=";X0;"ES =";D(0)
360 FOR J=1 TO N
370 PRINT "LA DERIVADA DE ORDEN";J;"EN X=";X0;"ES =";D(J)
380 NEXT J
390 PRINT
400 PRINT "DESEA CALCULAR LA DERIVADA EN OTRO PUNTO S/N?";
410 INPUT C$
420 IF C$="N" THEN GOTO 440
430 GOTO 180
440 END

```

## ejercicios resueltos:

1. Calcular el valor numérico del polinomio siguiente:

$$x^6 - 2x^5 + 3x^4 + 4x^3 - 5x^2 + 6x + 7$$

asi como el de sus derivadas en los puntos -1, 5, -0.5.

### SOLUCION :

```

GRADO DEL POLINOMIO? 6
COEFICIENTE DE X^ 6 ? 1
COEFICIENTE DE X^ 5 ? -2
COEFICIENTE DE X^ 4 ? 3
COEFICIENTE DE X^ 3 ? 4
COEFICIENTE DE X^ 2 ? -5
COEFICIENTE DE X^ 1 ? 6
COEFICIENTE DE X^ 0 ? 7

```

PUNTO EN EL QUE DESEA ESTUDIAR LA FUNCION Y SUS DERIVADAS X0 =? -1

```

EL POLINOMIO EN X=-1 ES = -2
LA DERIVADA DE ORDEN 1 EN X=-1 ES = 0
LA DERIVADA DE ORDEN 2 EN X=-1 ES = 72
LA DERIVADA DE ORDEN 3 EN X=-1 ES = -288
LA DERIVADA DE ORDEN 4 EN X=-1 ES = 672
LA DERIVADA DE ORDEN 5 EN X=-1 ES = -960
LA DERIVADA DE ORDEN 6 EN X=-1 ES = 720

```

DESEA CALCULAR LA DERIVADA EN OTRO PUNTO S/N?S  
PUNTO EN EL QUE DESEA ESTUDIAR LA FUNCION Y SUS DERIVADAS X0 =? 5

```

EL POLINOMIO EN X= 5 ES = 11662
LA DERIVADA DE ORDEN 1 EN X= 5 ES = 14256
LA DERIVADA DE ORDEN 2 EN X= 5 ES = 14760
LA DERIVADA DE ORDEN 3 EN X= 5 ES = 12384
LA DERIVADA DE ORDEN 4 EN X= 5 ES = 7872
LA DERIVADA DE ORDEN 5 EN X= 5 ES = 3360
LA DERIVADA DE ORDEN 6 EN X= 5 ES = 720

```

DESEA CALCULAR LA DERIVADA EN OTRO PUNTO S/N?S

PUNTO EN EL QUE DESEA ESTUDIAR LA FUNCION Y SUS DERIVADAS X0 =? -0.5

```

EL POLINOMIO EN X=-0.5 ES = 2.515625
LA DERIVADA DE ORDEN 1 EN X=-0.5 ES = 11.6875

```

LA DERIVADA DE ORDEN 2 EN X=-.5 ES =-6.125  
 LA DERIVADA DE ORDEN 3 EN X=-.5 ES =-57  
 LA DERIVADA DE ORDEN 4 EN X=-.5 ES = 282  
 LA DERIVADA DE ORDEN 5 EN X=-.5 ES =-600  
 LA DERIVADA DE ORDEN 6 EN X=-.5 ES = 720

## 2. Calcular el valor numérico del polinomio siguiente:

$$2x^3 - x^2 + 4$$

asi como el de sus derivadas en los puntos: -7, 4, 3.

### S O L U C I O N :

GRADO DEL POLINOMIO? 3  
 COEFICIENTE DE X^ 3 ? 2  
 COEFICIENTE DE X^ 2 ? -1  
 COEFICIENTE DE X^ 1 ? 0  
 COEFICIENTE DE X^ 0 ? 4

PUNTO EN EL QUE DESEA ESTUDIAR LA FUNCION Y SUS DERIVADAS XO =? -7

EL POLINOMIO EN X=-7 ES =-731  
 LA DERIVADA DE ORDEN 1 EN X=-7 ES = 308  
 LA DERIVADA DE ORDEN 2 EN X=-7 ES =-86  
 LA DERIVADA DE ORDEN 3 EN X=-7 ES = 12

DESEA CALCULAR LA DERIVADA EN OTRO PUNTO S/N? S

PUNTO EN EL QUE DESEA ESTUDIAR LA FUNCION Y SUS DERIVADAS XO =? 4

EL POLINOMIO EN X= 4 ES = 116  
 LA DERIVADA DE ORDEN 1 EN X= 4 ES = 88  
 LA DERIVADA DE ORDEN 2 EN X= 4 ES = 46  
 LA DERIVADA DE ORDEN 3 EN X= 4 ES = 12

DESEA CALCULAR LA DERIVADA EN OTRO PUNTO S/N ? S

PUNTO EN EL QUE DESEA ESTUDIAR LA FUNCION Y SUS DERIVADAS XO =? 3

EL POLINOMIO EN X= 3 ES = 49  
 LA DERIVADA DE ORDEN 1 EN X= 3 ES = 48  
 LA DERIVADA DE ORDEN 2 EN X= 3 ES = 34  
 LA DERIVADA DE ORDEN 3 EN X= 3 ES = 12

## 3. Calcular el valor numérico del polinomio siguiente:

$$-3x^4 + 2x^3 + 8x^2 + x + 2$$

asi como el de sus derivadas en los puntos: -1, 0, 3.

### S O L U C I O N :

GRADO DEL POLINOMIO? 4  
 COEFICIENTE DE X^ 4 ? -3  
 COEFICIENTE DE X^ 3 ? 2  
 COEFICIENTE DE X^ 2 ? 8  
 COEFICIENTE DE X^ 1 ? 1  
 COEFICIENTE DE X^ 0 ? 2

PUNTO EN EL QUE DESEA ESTUDIAR LA FUNCION Y SUS DERIVADAS  $x_0 = -1$

EL POLINOMIO EN  $x = -1$  ES = 4  
LA DERIVADA DE ORDEN 1 EN  $x = -1$  ES = 3  
LA DERIVADA DE ORDEN 2 EN  $x = -1$  ES = -32  
LA DERIVADA DE ORDEN 3 EN  $x = -1$  ES = 84  
LA DERIVADA DE ORDEN 4 EN  $x = -1$  ES = -72

DESEA CALCULAR LA DERIVADA EN OTRO PUNTO S/N?S

PUNTO EN EL QUE DESEA ESTUDIAR LA FUNCION Y SUS DERIVADAS  $x_0 = 0$

EL POLINOMIO EN  $x = 0$  ES = 2  
LA DERIVADA DE ORDEN 1 EN  $x = 0$  ES = 1  
LA DERIVADA DE ORDEN 2 EN  $x = 0$  ES = 16  
LA DERIVADA DE ORDEN 3 EN  $x = 0$  ES = 12  
LA DERIVADA DE ORDEN 4 EN  $x = 0$  ES = -72

DESEA CALCULAR LA DERIVADA EN OTRO PUNTO S/N?S

PUNTO EN EL QUE DESEA ESTUDIAR LA FUNCION Y SUS DERIVADAS  $x_0 = 3$

EL POLINOMIO EN  $x = 3$  ES = -112  
LA DERIVADA DE ORDEN 1 EN  $x = 3$  ES = -221  
LA DERIVADA DE ORDEN 2 EN  $x = 3$  ES = -272  
LA DERIVADA DE ORDEN 3 EN  $x = 3$  ES = -204  
LA DERIVADA DE ORDEN 4 EN  $x = 3$  ES = -72

# APROXIMACION DE LA FUNCION $Y = \text{SEN}(X)$

## introducción:

Este programa permite obtener una aproximación de la función  $y = \text{sen } x$  mediante desarrollo en serie. Al mismo tiempo calcula el valor del  $\text{sen } x$  con arreglo a la función de librería  $\text{SIN}(X)$  que posee el microordenador y por último calcula la diferencia entre las dos funciones.

## listado del programa:

```
10 REM *****
20 REM
30 REM      A P R O X I M A C I O N      D E      L A      F U N C I O N
40 REM
50 REM      Y=SEN(X)      P O R      D E S A R R O L L O      E N      S E R I E
60 REM
70 REM *****
100 PRINT
110 PRINT "      X      serie: SIN(X)      func: SIN(X)      DIFERENCIA"
120 PRINT "*****"
130 PRINT
140 PI= 3.141592653589793
150 X0=1
160 FOR X=0 TO 2*PI STEP .1
170 X1=ABS(X)
180 X1=X1-2*PI*INT(X1/(2*PI))
190 IF X1>PI THEN X0=-1
200 IF X1>PI THEN X1=X1-PI
210 IF X1>PI/2 THEN X1=PI-X1
220 X2=X1*X1
230 Y=X0*X1*Y
239 REM - COEFICIENTES DEL DESARROLLO DE LA SERIE SEN(X) -
240 C(1)=1
250 C(2)=-.16666666666671334
260 C(3)= .0083333333333809067
270 C(4)=-.000198412715551283
280 C(5)= .0000027557589750762
290 C(6)=-.00000002507059876207
300 C(7)= .000000000164105986683
309 REM - SUMACION DE LOS TERMINOS DE LA SERIE -
```

```

310 N=7
320 Y=0
330 FOR J=N TO 1 STEP -1
340 Y=Y*X2+C(J)
350 NEXT J
360 Y=X0*X1*Y
370 PRINT X,Y,SIN(X),Y-SIN(X)
380 NEXT X
390 END

```


## ejercicios resueltos:

1. Aproximar la función  $y = \sin x$  mediante desarrollo en serie entre 0 y  $2\pi$ , con un incremento de 0,1.

### SOLUCION :

X	serie: SIN(X)	func: SIN(X)	DIFERENCIA
*****			
0.0	.0000000000	.0000000000	0
.1	.0998334166	.0998334166	-1.8189894E-12
.2	.1986693308	.1986693308	-3.6379788E-12
.3	.2955202067	.2955202067	-5.4569682E-12
.4	.3894183423	.3894183423	-2.910383E-11
.5	.4794255386	.4794255386	-7.2759576E-12
.6	.5646424734	.5646424734	2.0918378E-11
.7	.6442176872	.6442176872	4.456524E-11
.8	.7173560909	.7173560909	-2.7284841E-11
.9	.7833269096	.7833269096	2.2737368E-11
1.0	.8414709849	.8414709849	-5.9117156E-11
1.1	.8912073601	.8912073602	-6.5483619E-11
1.2	.9320390860	.9320390860	5.6388672E-11
1.3	.9635581855	.9635581854	1.2278178E-10
1.4	.9854497301	.9854497299	1.3278623E-10
1.5	.9974949868	.9974949867	1.4370016E-10
1.6	.9995736033	.9995736031	2.582965E-10
1.7	.9916648104	.9916648102	2.1282176E-10
1.8	.9738476307	.9738476308	-6.2755134E-11
1.9	.9463000874	.9463000873	1.5916157E-10
2.0	.9092974265	.9092974264	6.9121597E-11
2.1	.8632093662	.8632093661	1.1550583E-10
2.2	.8084964032	.8084964032	4.638423E-11
2.3	.7457052114	.7457052112	1.9736035E-10
2.4	.6754631797	.6754631796	4.5474735E-11
2.5	.5984721431	.5984721428	2.6557245E-10
2.6	.5155013706	.5155013706	4.638423E-11
2.7	.4273798789	.4273798786	2.5374902E-10
2.8	.3349881487	.3349881486	1.1004886E-10
2.9	.2392493276	.2392493273	2.8830982E-10
3.0	.1411200063	.1411200062	1.0822987E-10
3.1	.0415806606	.0415806607	-7.0031092E-11
3.2	-.0583741454	-.0583741455	1.1732482E-10
3.3	-.1577456961	-.1577456961	-7.0031092E-11
3.4	-.2555411041	-.2555411042	1.0913936E-10
3.5	-.3507832298	-.3507832297	-4.0017767E-11
3.6	-.4425204454	-.4425204455	1.0913936E-10
3.7	-.5298361429	-.5298361429	-1.8189894E-11
3.8	-.6118578929	-.6118578930	1.1550583E-10

3.9	-.6877661611	-.6877661610	-9.8225428E-11
4.0	-.7568024971	-.7568024971	2.2737368E-11
4.1	-.8182771127	-.8182771126	-4.5474735E-12
4.2	-.8715757738	-.8715757739	1.0822987E-10
4.3	-.9161659379	-.9161659379	-5.0022209E-11
4.4	-.9516020748	-.9516020748	-5.4569682E-12
4.5	-.9775301183	-.9775301182	-1.1186785E-10
4.6	-.9936910041	-.9936910040	-1.1186785E-10
4.7	-.9999232580	-.9999232577	-2.8921932E-10
4.8	-.9961646087	-.9961646086	-1.1368684E-10
4.9	-.9824526120	-.9824526120	-7.1850081E-11
5.0	-.9589242737	-.9589242735	-2.0190782E-10
5.1	-.9258146809	-.9258146809	-1.546141E-11
5.2	-.8834546540	-.8834546538	-1.8644641E-10
5.3	-.8322674401	-.8322674400	-9.7315933E-11
5.4	-.7727644851	-.7727644849	-2.1645974E-10
5.5	-.7055403228	-.7055403226	-1.2187229E-10
5.6	-.6312666347	-.6312666345	-2.6830094E-10
5.7	-.5506855391	-.5506855390	-1.2551027E-10
5.8	-.4646021757	-.4646021753	-3.610694E-10
5.9	-.3738766608	-.3738766607	-1.5188562E-10
6.0	-.2794154939	-.2794154936	-3.8107828E-10
6.1	-.1821624998	-.1821624996	-2.1464075E-10
6.2	-.0830893982	-.0830893978	-3.8835424E-10



# APROXIMACION DE LA FUNCION $Y = \cos(X)$

## introducción:

Este programa permite obtener una aproximación de la función:  
 $y = \cos x$  mediante desarrollo en serie. Al mismo tiempo calcula el valor del  $\cos x$ , con arreglo a la función de librería  $\cos(X)$  que posee el microordenador, por último calcula la diferencia entre las dos funciones.

## listado del programa:

```
10 REM *****
20 REM
30 REM      A P R O X I M A C I O N      D E      L A      F U N C I O N
40 REM
50 REM      Y=COS(X)      P O R      D E S A R R O L L O      E N      S E R I E
60 REM
70 REM *****
100 PRINT
110 PRINT "          X          serie: COS(X)          func: COS(X)          DIFERENCIA"
120 PRINT "*****"
130 PRINT
140 PI= 3.141592653589793
150 X0=1
160 FOR X=0 TO PI STEP .1
170 X1=ABS(X)
180 X1=X1-2*PI*INT(X1/(2*PI))
190 IF X1>PI THEN X0=-1
200 IF X1>PI THEN X1=X1-PI
210 IF X1>PI/2 THEN X0=-1
220 IF X1>PI/2 THEN X1=PI-X1
230 X2=X1*X1
239 REM - COEFICIENTES DEL DESARROLLO DE LA SERIE COS(X) -
240 C(1)=1
250 C(2)=-.4999999999982
260 C(3)= .041666666664651
270 C(4)=-.0013888888805755
280 C(5)= .000024801428034
290 C(6)=-.00000002754213324
300 C(7)= .00000000020189405
309 REM - SUMACION DE LOS TERMINOS DE LA SERIE -
```



```

310 N=7
320 Y=0
330 FOR J=N TO 1 STEP -1
340 Y=Y*X2+C(J)
350 NEXT J
360 Y=XO*Y
370 PRINT X,Y,COS(X),Y-COS(X)
380 NEXT X
390 END

```

## ejercicios resueltos:

1. Aproximar la función  $y = \cos x$  mediante desarrollo en serie, entre 0 y  $\pi$ , con un incremento igual a 0,1.

### SOLUCION :

X	serie: COS(X)	func: COS(X)	DIFERENCIA
*****			
0.0	1.000000000	1.000000000	4.7293724E-11
.1	.995004165	.995004165	1.0004442E-10
.2	.980066578	.980066578	3.5470293E-11
.3	.955336489	.955336489	1.6370905E-11
.4	.921060994	.921060994	3.6379788E-11
.5	.877582562	.877582562	-9.094947E-13
.6	.825335615	.825335615	2.6375346E-11
.7	.764842187	.764842187	5.3660187E-11
.8	.696706709	.696706709	4.0017767E-11
.9	.621609968	.621609968	2.4556357E-11
1.0	.540302306	.540302306	2.7284841E-12
1.1	.453596121	.453596121	-1.2732926E-11
1.2	.362357754	.362357754	1.8189894E-11
1.3	.267498828	.267498828	9.094947E-12
1.4	.169967142	.169967143	-6.8212103E-11
1.5	.070737201	.070737201	-7.6397555E-11
1.6	-.029199523	-.029199523	1.5552359E-10
1.7	-.128844495	-.128844495	3.1650416E-10
1.8	-.227202095	-.227202095	8.4583007E-11
1.9	-.323289568	-.323289568	2.5193003E-10
2.0	-.416146837	-.416146837	5.8207661E-11
2.1	-.504846105	-.504846106	2.1373125E-10
2.2	-.588501118	-.588501118	7.548806E-11
2.3	-.666276022	-.666276022	1.7826096E-10
2.4	-.737393716	-.737393716	2.0008883E-11
2.5	-.801143616	-.801143617	1.9554136E-10
2.6	-.856888754	-.856888754	1.0732037E-10
2.7	-.904072143	-.904072143	4.7293724E-11
2.8	-.942222341	-.942222341	5.0022209E-11
2.9	-.970958166	-.970958166	8.2764018E-11
3.0	-.989992497	-.989992497	-1.0459189E-10
3.1	-.999135150	-.999135150	4.7293724E-11

# APROXIMACION DE LA FUNCION $Y = \ln(X)$

## introducción:

Este programa permite obtener una aproximación de la función:  $y = \ln x$ , mediante el desarrollo en serie. Al mismo tiempo calcula el valor del neperiano, con arreglo a la función de librería  $\text{LOG}(X)$  que poseen los microordenadores, por último calcula la diferencia entre las dos funciones.

## listado del programa:

```
10 REM *****
20 REM
30 REM      A P R O X I M A C I O N      D E      L A      F U N C I O N
40 REM
50 REM      Y=LN(X)      P O R      D E S A R R O L L O      E N      S E R I E
60 REM
70 REM *****
100 PRINT
110 PRINT "      X      serie: LN (X)      func: LN (X)      DIFERENCIA"
120 PRINT "*****"
130 PRINT
140 E= 2.718281828459045
150 FOR X=0.5 TO 10 STEP .5
160 X0=1
170 X1=-1
180 X2=X
190 IF X>=1 THEN GOTO 220
200 X=1/X
210 X0=-1
220 X=E*X
230 X1=X1+1
240 X=X/E
250 IF X>E THEN GOTO 230
260 X4=(X-1.6487212707)/(X+1.6487212707)
270 X5=X4*X4
279 REM - COEFICIENTES DEL DESARROLLO DE LA SERIE Y=LN (X) -
280 C(1)=2
290 C(2)=-.66666672443
300 C(3)=.3999895288
310 C(4)=-.286436047
```

```

320 C(5)=.197959107
330 C(6)=.6283533
340 C(7)=-4.54692
350 C(8)=28.117
360 C(9)=-86.42
370 C(10)=106.1
379 REM - SUMACION DE LOS TERMINOS DE LA SERIE -
380 N=10
390 Y=0
400 FOR J=N TO 1 STEP -1
410 Y=Y*X5+C(J)
420 NEXT J
430 Y=X0*(Y*X4+X1+.5)
440 X=X2
450 PRINT X,Y,LOG(X),Y-LOG(X)
460 NEXT X
470 END

```

## ejercicios resueltos:

1. Aproximar la función:  $y = \ln x$  mediante desarrollo en serie, entre 0.5 y 10, con un incremento igual a .5.

### SOLUCION :

X	serie: LN (X)	func: LN (X)	DIFERENCIA
*****			
.5	-.693147181	-.693147181	-1.7280399E-11
1.0	0.000000000	0.000000000	1.0913936E-11
1.5	.405465108	.405465108	-1.546141E-11
2.0	.693147181	.693147181	-9.276846E-11
2.5	.916290732	.916290732	-6.002665E-11
3.0	1.098612289	1.098612289	0
3.5	1.252762968	1.252762969	-2.3283064E-10
4.0	1.386294361	1.386294361	-2.3283064E-10
4.5	1.504077397	1.504077397	-2.3283064E-10
5.0	1.609437912	1.609437912	-2.3283064E-10
5.5	1.704748092	1.704748092	-2.3283064E-10
6.0	1.791759469	1.791759469	-2.3283064E-10
6.5	1.871802177	1.871802177	-2.3283064E-10
7.0	1.945910149	1.945910149	-2.3283064E-10
7.5	2.014903020	2.014903021	-2.3283064E-10
8.0	2.079441541	2.079441542	-2.3283064E-10
8.5	2.140066163	2.140066163	-2.3283064E-10
9.0	2.197224577	2.197224577	-2.3283064E-10
9.5	2.251291798	2.251291798	0
10.0	2.302585093	2.302585093	-2.3283064E-10

# APROXIMACION DE LA FUNCION $Y = \exp(X)$

## introducción:

Este programa permite obtener una aproximación de la función:  $y = e^x$ , mediante desarrollo en serie. Al mismo tiempo calcula el valor de la función exponencial, con arreglo a la función de librería  $\exp(X)$  que poseen los microordenadores, por último calcula la diferencia entre las dos funciones.

## listado del programa:

```
10 REM *****
20 REM
30 REM      A P R O X I M A C I O N      D E      L A      F U N C I O N
40 REM
50 REM      Y=EXP(X)      P O R      D E S A R R O L L O      E N      S E R I E
60 REM
70 REM *****
100 PRINT
110 PRINT "      X      serie: EXP(X)      func: EXP(X)      DIFERENCIA"
120 PRINT "*****"
130 PRINT
140 FOR X=-5 TO 5 STEP .5
150 X0=1
160 X1=X
170 IF X<0 THEN X0=-1
180 X=ABS(X)
190 X3=INT(X)
200 X=X-X3
209 REM - COEFICIENTES DEL DESARROLLO DE Y=EXP(X) -
210 C(1)=1
220 C(2)=.99999999668
230 C(3)=.499999995173
240 C(4)=.16666704243
250 C(5)=.04166685027
260 C(6)=.00832672635
270 C(7)=.00140836136
280 C(8)=.00017358267
290 C(9)=.00003931683
300 X2=X
```

```

309 REM - SUMACION DE LOS TERMINOS DE LA SERIE -
310 N=9
320 Y=0
330 FOR J=N TO 1 STEP -1
340 Y=Y*X2+C(J)
350 NEXT J
360 FOR K=1 TO X3
370 E=2.718281828459045
380 Y=Y*E
390 NEXT K
400 IF X<0 THEN Y=1/Y
410 X=X1
420 PRINT X,Y, EXP(X), Y-EXP(X)
430 NEXT X
440 END

```

## ejercicios resueltos:

1. Aproximar la función  $y = e^x$  mediante desarrollo en serie, entre -5 y 5, con un incremento de 0.5.

### SOLUCION :

X	serie: EXP(X)	func: EXP(X)	DIFERENCIA
*****			
-5.0	.0067379	.0067379	-2.7284841E-12
-4.5	.0111090	.0111090	1.8189894E-12
-4.0	.0183156	.0183156	-5.4569682E-12
-3.5	.0301974	.0301974	7.2759576E-12
-3.0	.0497871	.0497871	-9.094947E-12
-2.5	.0820850	.0820850	2.7284841E-11
-2.0	.1353353	.1353353	-2.0918378E-11
-1.5	.2231302	.2231302	8.0945028E-11
-1.0	.3678794	.3678794	-6.2755134E-11
-.5	.6065307	.6065307	1.70985E-10
0.0	1.0000000	1.0000000	4.0017767E-11
.5	1.6487213	1.6487213	-4.6566129E-10
1.0	2.7182818	2.7182818	4.6566129E-10
1.5	4.4816891	4.4816891	-1.6298145E-09
2.0	7.3890561	7.3890561	1.1641532E-09
2.5	12.1824940	12.1824940	-3.9581209E-09
3.0	20.0855369	20.0855369	3.7252903E-09
3.5	33.1154520	33.1154520	-7.6834112E-09
4.0	54.5981501	54.5981500	1.5832484E-08
4.5	90.0171313	90.0171313	-9.0803951E-09
5.0	148.4131592	148.4131591	6.3562766E-08

# REGLA DE SIMPSON (FUNCION CONOCIDA)

## introducción:

Este programa permite calcular integrales definidas mediante la regla de Simpson. Este procedimiento de cálculo numérico obtiene una aproximación al verdadero valor de la integral. Para que la aproximación sea mejor basta con hacer más fina la partición del intervalo, es decir, aumentar el número de intervalos (siempre pares).

La función a integrar se introduce en la línea 170 del siguiente modo:

```
170 DEF FN F(X) = .....
```

## listado del programa:

```
10 REM *****
20 REM
30 REM             I N T E G R A C I O N             N U M E R I C A
40 REM
50 REM             R E G L A             D E             S I M P S O N
60 REM
70 REM *****
100 PRINT
109 REM - ENTRADA DE DATOS -
110 PRINT "DEFINA LA FUNCION QUE DESEA INTEGRAR EN LA LINEA 170"
120 PRINT "DEL SIGUIENTE MODO: 170 DEF FN F(X) = ....."
130 PRINT "A CONTINUACION PULSAR <CONT>"
140 STOP
150 PRINT "INTRODUZCA LOS LIMITES DE INTEGRACION: INFERIOR, SUPERIOR";
160 INPUT A,B
170 DEF FN F(X)=
180 PRINT "INTRODUZCA EL NUMERO DE INTERVALOS (PARES)";
190 INPUT N
200 IF N/2<>INT(N/2) GOTO 200
210 H=(B-A)/(2*N)
220 N1=2*N-1
230 N2=2*(N-1)
240 SI=0
250 SP=0
259 REM - SUMA DE LOS PUNTOS IMPARES -
260 FOR J=1 TO N1 STEP 2
270 X=A+H*J
280 SI=SI+FNF(X)
```

```

290 NEXT J
299 REM - SUMA DE LOS PUNTOS PARES -
300 FOR J=2 TO N2 STEP 2
310 X=A+H*J
320 SP=SP+FN F(X)
330 NEXT J
339 REM - SALIDA DEL RESULTADO -
340 PRINT
350 PRINT
360 IS=H*(FN F(A)+4*SI+2*SP+FN F(B))/3
370 PRINT "LA INTEGRAL PEDIDA ES =";IS;
380 PRINT "(MEDIANTE LA REGLA DE SIMPSON)"
390 END

```

## ejercicios resueltos:

1. Calcular  $\int_0^{1.2} \frac{dx}{1+x}$

**SOLUCION :**

```

DEFINA LA FUNCION QUE DESEA INTEGRAR EN LA LINEA 170
DEL SIGUIENTE MODO: 170 DEF FN F(X) = .....
A CONTINUACION PULSAR <CONT>
170 DEF FN F(X) =1/(1+X)
INTRODUZCA LOS LIMITES DE INTEGRACION: INFERIOR, SUPERIOR? 0      1.2
INTRODUZCA EL NUMERO DE INTERVALOS (PARES)? 10

```

LA INTEGRAL PEDIDA ES = .78845777 (MEDIANTE LA REGLA DE SIMPSON)

2. Calcular el valor de  $\int_0^3 \frac{dx}{\sqrt{1+x}}$

**SOLUCION :**

```

DEFINA LA FUNCION QUE DESEA INTEGRAR EN LA LINEA 170
DEL SIGUIENTE MODO: 170 DEF FN F(X) = .....
A CONTINUACION PULSAR <CONT>
170 DEF FN F(X) = 1/SQR(1+X)
INTRODUZCA LOS LIMITES DE INTEGRACION: INFERIOR, SUPERIOR? 0      3
INTRODUZCA EL NUMERO DE INTERVALOS (PARES)? 50

```

LA INTEGRAL PEDIDA ES = 2 (MEDIANTE LA REGLA DE SIMPSON)

3. Calcular el valor de  $\int_4^8 \frac{x dx}{\sqrt{x^2-15}}$

**SOLUCION :**

DEFINA LA FUNCION QUE DESEA INTEGRAR EN LA LINEA 170  
 DEL SIGUIENTE MODO: 170 DEF FN F(X) = .....  
 A CONTINUACION PULSAR <CONT>  
 170 DEF FN F(X) = X/SQR(X\*X-15)  
 INTRODUZCA LOS LIMITES DE INTEGRACION: INFERIOR, SUPERIOR? 4 8  
 INTRODUZCA EL NUMERO DE INTERVALOS (PARES)? 30

LA INTEGRAL PEDIDA ES = 6.000277 (MEDIANTE LA REGLA DE SIMPSON)

4. Calcular el valor de 
$$\int_{-.5}^0 \frac{x^3 dx}{x^2 + x + 1}$$

**S O L U C I O N :**

DEFINA LA FUNCION QUE DESEA INTEGRAR EN LA LINEA 170  
 DEL SIGUIENTE MODO: 170 DEF FN F(X) = .....  
 A CONTINUACION PULSAR <CONT>  
 170 DEF FN F(X) = X^3/(X\*X+X+1)  
 INTRODUZCA LOS LIMITES DE INTEGRACION: INFERIOR, SUPERIOR? -.5 0  
 INTRODUZCA EL NUMERO DE INTERVALOS (PARES)? 60

LA INTEGRAL PEDIDA ES = -.020400212 (MEDIANTE LA REGLA DE SIMPSON)



# REGLA DE SIMPSON (FUNCION DESCONOCIDA)

## introducción:

Este programa sirve para calcular el valor aproximado de una integral de una función desconocida, sabiendo el valor que toma la función en cada uno de los puntos de la partición del intervalo. El cálculo se realiza mediante la regla de integración numérica de Simpson.

## listado del programa:

```
10 REM *****
20 REM
30 REM             I N T E G R A C I O N             N U M E R I C A
40 REM
50 REM             R E G L A       D E       S I M P S O N       (FUNCION DESCONOCIDA)
60 REM
70 REM *****
110 PRINT "INTRODUZCA LOS LIMITES DEL INTERVALO: INFERIOR Y SUPERIOR";
120 INPUT A,B
130 PRINT "INTRODUZCA EL NUMERO DE INTERVALOS (PARES)";
140 INPUT M
150 N=M/2
160 PRINT "VALORES DE LA FUNCION EN LOS EXTREMOS INFERIOR Y SUPERIOR";
170 INPUT Y1, Y2
180 H=(B-A)/(2*N)
190 N1=2*N-1
200 N2=2*(N-1)
210 SI=0
220 SP=0
229 REM - SUMA DE LOS PUNTOS IMPARES -
230 FOR J=1 TO N1 STEP 2
240 PRINT "VALOR DE F(X) PARA X = ";A+H*J;
250 INPUT Y
260 SI=SI+Y
270 NEXT J
279 REM - SUMA DE LOS PUNTOS PARES -
280 FOR J=2 TO N2 STEP 2
290 PRINT "VALOR DE F(X) PARA X = ";A+H*J;
300 INPUT Y
310 SP=SP+Y
320 NEXT J
```

```

329 REM -SALIDA DEL RESULTADO -
330 PRINT
340 PRINT
350 IS=H*(Y1+4*SI+2*SP+Y2)/3
360 PRINT "LA INTEGRAL PEDIDA ES =";IS;
370 PRINT "(MEDIANTE LA REGLA DE SIMPSON)"
380 END

```

## ejercicios resueltos:

1. Hallar la integral de una función desconocida entre 1 y 3, sabiendo los siguientes valores de  $x$  y de  $f(x)$ .

x	1	1.25	1.5	1.75	2	2.25	2.5	2.75	3
f(x)	1.2	1.6	1.9	2.5	2.8	1.7	2	2.6	2.9

### S O L U C I O N :

```

INTRODUZCA LOS LIMITES DEL INTERVALO:INFERIOR Y SUPERIOR? 1      3
INTRODUZCA EL NUMERO DE INTERVALOS (PARES)? 8
VALORES DE LA FUNCION EN LOS EXTREMOS INFERIOR Y SUPERIOR? 1.2    2.9
VALOR DE F(X) PARA X = 1.25 ? 1.6
VALOR DE F(X) PARA X = 1.75 ? 1.9
VALOR DE F(X) PARA X = 2.25 ? 2.5
VALOR DE F(X) PARA X = 2.75 ? 2.8
VALOR DE F(X) PARA X = 1.5 ? 1.7
VALOR DE F(X) PARA X = 2 ? 2
VALOR DE F(X) PARA X = 2.5 ? 2.6

```

LA INTEGRAL PEDIDA ES = 4.325 (MEDIANTE LA REGLA DE SIMPSON)

2. Hallar la integral de una función desconocida entre 4 y 5.5, sabiendo los siguientes valores de  $x$  y de  $f(x)$ .

x	4	4.25	4.75	5.25	4.5	5	5.5
f(x)	1	1.3	1.7	1.3	1.5	1.5	1

### S O L U C I O N :

```

INTRODUZCA LOS LIMITES DEL INTERVALO:INFERIOR Y SUPERIOR? 4      5.5
INTRODUZCA EL NUMERO DE INTERVALOS (PARES)? 6
VALORES DE LA FUNCION EN LOS EXTREMOS INFERIOR Y SUPERIOR? 1      1
VALOR DE F(X) PARA X = 4.25 ? 1.3
VALOR DE F(X) PARA X = 4.75 ? 1.7
VALOR DE F(X) PARA X = 5.25 ? 1.3
VALOR DE F(X) PARA X = 4.5 ? 1.5
VALOR DE F(X) PARA X = 5 ? 1.5

```

LA INTEGRAL PEDIDA ES = 2.1 (MEDIANTE LA REGLA DE SIMPSON)

**3.** Hallar la integral de una función desconocida entre 1 y 1.3 sabiendo los valores de  $x$  y de  $f(x)$  siguientes:

**SOLUCION :**

$x$	$f(x)$
1	1.
1.05	1.02470
1.1	1.04881
1.15	1.07238
1.2	1.09544
1.25	1.11803
1.3	1.14017

INTRODUZCA LOS LIMITES DEL INTERVALO: INFERIOR Y SUPERIOR? 1      1.3  
 INTRODUCZA EL NUMERO DE INTERVALOS (PARES)? 6  
 VALORES DE LA FUNCION EN LOS EXTREMOS INFERIOR Y SUPERIOR? 1      1.14017  
 VALOR DE F(X) PARA X = 1.05 ? 1.0247  
 VALOR DE F(X) PARA X = 1.15 ? 1.07238  
 VALOR DE F(X) PARA X = 1.25 ? 1.11803  
 VALOR DE F(X) PARA X = 1.1 ? 1.04881  
 VALOR DE F(X) PARA X = 1.2 ? 1.09544

LA INTEGRAL PEDIDA ES = .32148517 (MEDIANTE LA REGLA DE SIMPSON)

# REGLA DEL TRAPECIO

## introducción:

Este programa permite calcular integrales definidas mediante la regla de los trapecios. Este procedimiento de cálculo numérico, obtiene una aproximación al verdadero valor de la integral. Cuanto más fina sea la partición del intervalo, mejor será la aproximación. Para hacer más fina la partición del intervalo basta con aumentar el número de intervalos en el que subdividimos el intervalo inicial.

La función a integrar se define en la línea 170 del siguiente modo:

170 DEF FN F(X) = .....

## listado del programa:

```
10 REM *****
20 REM
30 REM           I N T E G R A C I O N           N U M E R I C A
40 REM
50 REM           R E G L A       D E       L O S       T R A P E C I O S
60 REM
70 REM *****
100 PRINT
109 REM - ENTRADA DE DATOS -
110 PRINT "DEFINA LA FUNCION QUE DESEA INTEGRAR EN LA LINEA 170"
120 PRINT "DEL SIGUIENTE MODO: 170 DEF FN F(X) = ....."
130 PRINT "A CONTINUACION PULSAR <CONT>"
140 STOP
150 PRINT "INTRODUZCA LOS LIMITES DE INTEGRACION: INFERIOR, SUPERIOR";
160 INPUT A,B
170 DEF FN F(X)=
180 PRINT "INTRODUZCA EL NUMERO DE INTERVALOS";
190 INPUT N
200 S=0
210 H=(B-A)/N
220 N1=N-1
230 FOR J=1 TO N1
240 S=S+FN F(A+J*H)
250 NEXT J
259 REM - SALIDA DEL RESULTADO -
260 PRINT
```

```

270 PRINT
280 IT=H*(S+(FN F(A)+FN F(B))/2)
290 PRINT "LA INTEGRAL PEDIDA ES =";IT;
300 PRINT "(MEDIANTE LA REGLA DEL TRAPECIO)"
310 END

```

## ejercicios resueltos:

1. Calcular el valor de la integral:  $\int_0^1 \frac{\cos(x-1)}{x^2+1} dx$

**SOLUCION :**

```

DEFINA LA FUNCION QUE DESEA INTEGRAR EN LA LINEA 170
DEL SIGUIENTE MODO: 170 DEF FN F(X) = .....
A CONTINUACION PULSAR <CONT>
170 DEF FN F(X) = COS(X-1)/(X^2+1)
INTRODUZCA LOS LIMITES DE INTEGRACION: INFERIOR, SUPERIOR? 0 1
INTRODUZCA EL NUMERO DE INTERVALOS ? 100

```

LA INTEGRAL PEDIDA ES = .63975904 (MEDIANTE LA REGLA DEL TRAPECIO)

2. Calcular el valor de la integral:  $\int_0^1 \arcsin x dx$

**SOLUCION :**

```

DEFINA LA FUNCION QUE DESEA INTEGRAR EN LA LINEA 170
DEL SIGUIENTE MODO: 170 DEF FN F(X) = .....
A CONTINUACION PULSAR <CONT>
170 DEF FN F(X) = ASN(X)
INTRODUZCA LOS LIMITES DE INTEGRACION: INFERIOR, SUPERIOR? 0 1
INTRODUZCA EL NUMERO DE INTERVALOS ? 60

```

LA INTEGRAL PEDIDA ES = .57140586 (MEDIANTE LA REGLA DEL TRAPECIO)

3. Calcular el valor de la integral:  $\int_1^3 (x^3 + 3x^2 + \ln x) dx$

**SOLUCION :**

```

DEFINA LA FUNCION QUE DESEA INTEGRAR EN LA LINEA 170
DEL SIGUIENTE MODO: 170 DEF FN F(X) = .....
A CONTINUACION PULSAR <CONT>
170 DEF FN F(X) = X^3+3*X^2+LOG(X)
INTRODUZCA LOS LIMITES DE INTEGRACION: INFERIOR, SUPERIOR? 1 3
INTRODUZCA EL NUMERO DE INTERVALOS ? 100

```

LA INTEGRAL PEDIDA ES = 47.297015 (MEDIANTE LA REGLA DEL TRAPECIO)

4. Calcular el valor de la integral:  $\int_{.5}^1 \text{Ln}(\text{sen } x) dx$

**SOLUCION :**

DEFINA LA FUNCION QUE DESEA INTEGRAR EN LA LINEA 170

DEL SIGUIENTE MODO: 170 DEF FN F(X) = .....


A CONTINUACION PULSAR <CONT>

170 DEF FN F(X) = LOG(SIN(X))

INTRODUZCA LOS LIMITES DE INTEGRACION: INFERIOR, SUPERIOR .5

INTRODUZCA EL NUMERO DE INTERVALOS ? 100

LA INTEGRAL PEDIDA ES = -.20316953 (MEDIANTE LA REGLA DEL TRAPECIO)



# AREA ENCERRADA POR UNA CURVA EN CARTESIANAS

## introducción:

Este programa permite calcular el area encerrada por una curva de la forma  $y = f(x)$ , el eje  $x$ , y las rectas  $x = a$ ,  $x = b$ . El cálculo del area se realiza por dos procedimientos: regla de Simpson y regla de los trapezios.

La función a integrar se define en la línea 170 del siguiente modo:

```
170 DEF FN F(X) = .....
```

## listado del programa:

```
10 REM *****
20 REM
30 REM          A R E A      E N C E R R A D A      P O R      U N A      C U R V A
40 REM
50 REM          D A D A      E N      F. C A R T E S I A N A      Y=F(X)
60 REM
70 REM          M E T O D O S : S I M P S O N      Y      T R A P E C I O S
80 REM
90 REM *****
100 PRINT
109 REM - ENTRADA DE DATOS -
110 PRINT "DEFINA LA FUNCION QUE DESEA INTEGRAR EN LA LINEA 170"
120 PRINT "DEL SIGUIENTE MODO: 170 DEF FN F(X) = ....."
130 PRINT "A CONTINUACION PULSAR <CONT>"
140 STOP
150 PRINT "INTRODUZCA LOS LIMITES DE INTEGRACION: INFERIOR, SUPERIOR";
160 INPUT A,B
170 DEF FN F(X)=
180 PRINT "INTRODUZCA EL NUMERO DE INTERVALOS (PARES)";
190 INPUT N
200 IF N/2<>INT(N/2) GOTO 200
210 H=(B-A)/(2*N)
220 N1=2*N-1
230 N2=2*(N-1)
240 SI=0
250 SP=0
259 REM - SUMA DE LOS PUNTOS IMPARES -
260 FOR J=1 TO N1 STEP 2
270 X=A+H*J
```

```

280 SI=SI+FNF(X)
290 NEXT J
299 REM - SUMA DE LOS PUNTOS PARES -
300 FOR J=2 TO N2 STEP 2
310 X=A+H*J
320 SP=SP+FNF(X)
330 NEXT J
339 REM - SALIDA DE RESULTADOS -
340 PRINT
350 PRINT
360 IS=H*(FNF(A)+4*SI+2*SP+FNF(B))/3
370 PRINT "LA INTEGRAL PEDIDA ES =";IS;
380 PRINT "(MEDIANTE LA REGLA DE SIMPSON)"
390 PRINT
400 IT=H*(SI+SP+(FNF(A)+FNF(B))/2)
410 PRINT "LA INTEGRAL PEDIDA ES =";IT;
420 PRINT "(MEDIANTE LA REGLA DE LOS TRAPECIOS)"
430 END

```

## ejercicios resueltos:

1. Hallar el area encerrada por la curva  $f(x) = \frac{1}{1+x}$ , el eje x y las rectas:  $x = 0$  y  $x = 1.2$ .

### SOLUCION :

```

DEFINA LA FUNCION QUE DESEA INTEGRAR EN LA LINEA 170
DEL SIGUIENTE MODO: 170 DEF FN F(X) = .....
A CONTINUACION PULSAR <CONT>
170 DEF FN F(X) = 1/(1+X)

```

INTRODUZCA LOS LIMITES DE INTEGRACION: INFERIOR, SUPERIOR? O 1.2  
 INTRODUZCA EL NUMERO DE INTERVALOS (PARES)? 40

LA INTEGRAL PEDIDA ES = .78845736 (MEDIANTE LA REGLA DE SIMPSON)

LA INTEGRAL PEDIDA ES = .78847224 (MEDIANTE LA REGLA DE LOS TRAPECIOS)

2. Hallar el area encerrada por la curva  $f(x) = \frac{1}{\sqrt{1+x}}$ , el eje x y las rectas:  $x = 0$  y  $x = 3$ .

### SOLUCION :

```

DEFINA LA FUNCION QUE DESEA INTEGRAR EN LA LINEA 170
DEL SIGUIENTE MODO: 170 DEF FN F(X) = .....
A CONTINUACION PULSAR <CONT>
170 DEF FN F(X) = 1/SQR(1+X)

```

INTRODUZCA LOS LIMITES DE INTEGRACION: INFERIOR, SUPERIOR? O 3  
 INTRODUZCA EL NUMERO DE INTERVALOS (PARES)? 40



LA INTEGRAL PEDIDA ES = 2 (MEDIANTE LA REGLA DE SIMPSON)

LA INTEGRAL PEDIDA ES = 2.0000228 (MEDIANTE LA REGLA DE LOS TRAPECIOS)

**3.** Hallar el area del recinto encerrado por la curva  $y = \frac{x}{\sqrt{x^2 - 15}}$  el eje x, y las rectas:  $x = 4$  y  $x = 8$ .


**SOLUCION :**

DEFINA LA FUNCION QUE DESEA INTEGRAR EN LA LINEA 170  
DEL SIGUIENTE MODO: 170 DEF FN F(X) = .....  
A CONTINUACION PULSAR <CONT>  
170 DEF FN F(X) = X/SQR(X\*X-15)

INTRODUZCA LOS LIMITES DE INTEGRACION: INFERIOR, SUPERIOR? 4 8  
INTRODUZCA EL NUMERO DE INTERVALOS (PARES)? 100

LA INTEGRAL PEDIDA ES = 6.000003 (MEDIANTE LA REGLA DE SIMPSON)

LA INTEGRAL PEDIDA ES = 6.0004978 (MEDIANTE LA REGLA DE LOS TRAPECIOS)



# AREA ENCERRADA POR UNA CURVA EN POLARES

## introducción:

Este programa permite calcular el area encerrada por una curva dada en coordenadas polares de la forma:  $r = r(w)$  y las semirrectas  $w = w_1$  y  $w = w_2$ . El cálculo del area se realiza por dos procedimientos: regla de Simpson y regla de los trapecios.

La función a integrar se define en la linea 170 del siguiente modo:

```
170 DEF FN R(W) = .....
```

## listado del programa:

```
10 REM *****
20 REM
30 REM      AREA      ENCERRADA      POR      UNA      CURVA
40 REM
50 REM      DADA      EN      F. POLAR      R=R(W)
60 REM
70 REM      METODOS: S I M P S O N      Y      T R A P E C I O S
80 REM
90 REM *****
100 PRINT
109 REM - ENTRADA DE DATOS -
110 PRINT "DEFINA LA FUNCION QUE DESEA INTEGRAR EN LA LINEA 170"
120 PRINT "DEL SIGUIENTE MODO: 170 DEF FN R(W) = ....."
130 PRINT "A CONTINUACION PULSAR <CONT>"
140 STOP
150 PRINT "LIMITES DE INTEGRACION: INFERIOR, SUPERIOR (EN RADIANES)";
160 INPUT W1,W2
170 DEF FN R(W) = .....
180 PRINT "INTRODUZCA EL NUMERO DE INTERVALOS (PARES)";
190 INPUT N
200 IF N/2<>INT(N/2) GOTO 200
210 H=(W2-W1)/(2*N)
220 N1=2*N-1
230 N2=2*(N-1)
240 SI=0
250 SP=0
259 REM - SUMA DE LOS PUNTOS IMPARES -
260 FOR J=1 TO N1 STEP 2
270 W=W1+H*J
```

```

280 SI=SI+(FNR(W)^2)/2
290 NEXT J
299 REM - SUMA DE LOS PUNTOS PARES -
300 FOR J=2 TO N2 STEP 2
310 W=W1+H*J
320 SP=SP+(FNR(W)^2)/2
330 NEXT J
339 REM - SALIDA DE RESULTADOS -
340 PRINT
350 PRINT
360 IS=H*(FNR(W1)+4*SI+2*SP+FNR(W2))/3
370 PRINT "LA INTEGRAL PEDIDA ES=";IS;
380 PRINT "(MEDIANTE LA REGLA DE SIMPSON)"
390 PRINT
400 IT=H*(SI+SP+(FNR(W1)+FNR(W2))/2)
410 PRINT "LA INTEGRAL PEDIDA ES=";IT;
420 PRINT "(MEDIANTE LA REGLA DE LOS TRAPECIOS)"
430 END

```

## ejercicios resueltos:

1. Hallar el área encerrada por la cardioide de ecuación:  $r = 1 + \cos w$

### SOLUCION :

El área pedida será igual al doble del área entre 0 y  $\pi$ . Por tanto calculemos esta última.

```

DEFINA LA FUNCION QUE DESEA INTEGRAR EN LA LINEA 170
DEL SIGUIENTE MODO: 170 DEF FN R W) = .....
A CONTINUACION PULSAR <CONT>
170 DEF FN R(W) = 1+COS(W)

```

LIMITES DE INTEGRACION: INFERIOR, SUPERIOR (EN RADIANES)? 0 3.1415927  
 INTRODUZCA EL NUMERO DE INTERVALOS (PARES)? 60

LA INTEGRAL PEDIDA ES = 2.3561945 (MEDIANTE LA REGLA DE SIMPSON)

LA INTEGRAL PEDIDA ES = 2.3561945 (MEDIANTE LA REGLA DE LOS TRAPECIOS)

El área encerrada será: 4.71238

2. Hallar el área encerrada por la lemniscata de Bernouilli de ecuación:  $r^2 = 9 \cos 2 w$

### SOLUCION :

El área pedida será igual a 4 veces el area encerrada entre 0 y  $\pi/4$ .  
 Por tanto calculemos esta última.

```

DEFINA LA FUNCION QUE DESEA INTEGRAR EN LA LINEA 170
DEL SIGUIENTE MODO: 170 DEF FN R W) = .....
A CONTINUACION PULSAR <CONT>
170 DEF FN R(W) = 3*SQR(COS(2*W))

```

LIMITES DE INTEGRACION: INFERIOR, SUPERIOR (EN RADIANTES)? 0 .7853981  
 INTRODUZCA EL NUMERO DE INTERVALOS (PARES)? 60

LA INTEGRAL PEDIDA ES = 2.2467298 (MEDIANTE LA REGLA DE SIMPSON)

LA INTEGRAL PEDIDA ES = 2.2450626 (MEDIANTE LA REGLA DE LOS TRAPECIOS)

El área encerrada será:  $4(2.2467298) = 8.96..$

**3.** Hallar el área encerrada por la rosa de tres pétalos de ecuación:

$$r = 2 \cos 3 w$$

**S O L U C I O N :**

El área pedida será igual a 6 veces el área encerrada entre 0 y  $\pi/6$ .

Por tanto, calculemos esta última.

DEFINA LA FUNCION QUE DESEA INTEGRAR EN LA LINEA 170

DEL SIGUIENTE MODO: 170 DEF FN R W) = .....

A CONTINUACION PULSAR <CONT>

170 DEF FNR(W) = .2 \* COS(3 \* W)


LIMITES DE INTEGRACION: INFERIOR, SUPERIOR (EN RADIANTES)? 0 .5235988

INTRODUZCA EL NUMERO DE INTERVALOS (PARES)? 40

LA INTEGRAL PEDIDA ES = .52359878 (MEDIANTE LA REGLA DE SIMPSON)

LA INTEGRAL PEDIDA ES = .52359878 (MEDIANTE LA REGLA DE LOS TRAPECIOS)

El área encerrada será:  $6(.52359878) = 3.12...$



# AREA ENCERRADA POR UNA CURVA EN PARAMETRICAS

## introducción:

Este programa permite calcular el area encerrada por una curva dada en forma paramétrica:  $x = x(t)$ ,  $y = y(t)$  y las semirrectas  $t = t_1$  y  $t = t_2$ . El cálculo del área se realiza por dos procedimientos : regla de Simpson y regla de los trapecios.

En la linea 210 se define la función  $y = y(t)$  del siguiente modo:

210 DEF FN Y(T) = .....

y en la linea 220 se define la derivada de la función  $x = x(t)$  del siguiente modo:

220 DEF FN D(T) = .....

## listado del programa:

```
10 REM *****
20 REM
30 REM      A R E A      E N C E R R A D A      P O R      U N A      C U R V A
40 REM
50 REM      D A D A      E N      F . P A R A M E T R I C A      X=X(T);      Y=Y(T)
60 REM
70 REM      M E T O D O S : S I M P S O N      Y      T R A P E C I O S
80 REM
90 REM *****
100 PRINT
109 REM - ENTRADA DE DATOS -
110 PRINT "DEFINA LA FUNCION QUE DESEA INTEGRAR EN LA LINEA 210"
120 PRINT "DEL SIGUIENTE MODO: 210 DEF FN Y(T) = ....."
130 PRINT "A CONTINUACION PULSAR <CONT>"
140 STOP
150 PRINT "DEFINA LA DERIVADA DE X=X(T) RESPECTO DE T, EN LA LINEA 220"
160 PRINT "DEL SIGUIENTE MODO: 220 DEF FN D(T) = ....."
170 PRINT "A CONTINUACION PULSAR <CONT>"
180 STOP
190 PRINT "LIMITES DE INTEGRACION: INFERIOR, SUPERIOR ";
200 INPUT T1,T2
210 DEF FN Y(T)=
220 DEF FN D(T)=
230 PRINT "INTRODUZCA EL NUMERO DE INTERVALOS (PARES)";
240 INPUT N
```

```

250 IF N/2<>INT(N/2) GOTO 250
260 H=(T2-T1)/(2*N)
270 N1=2*N-1
280 N2=2*(N-1)
290 SI=0
300 SP=0
310 FOR J=1 TO N1 STEP 2
320 T=T1+H*J
330 SI=SI+FNY(T)*FND(T)
340 NEXT J
350 FOR J=2 TO N2 STEP 2
360 T=T1+H*J
370 SP=SP+FNY(T)*FND(T)
380 NEXT J
389 REM - SALIDA DE RESULTADOS -
390 PRINT
400 PRINT
410 Y1=FNY(T1)*FND(T1)
420 Y2=FNY(T2)*FND(T2)
430 IS =H*(Y1+4*SI+2*SP+Y2)/3
440 PRINT "EL AREA PEDIDA ES = ";IS ; "UNIDADES CUADRADAS";
450 PRINT "(MEDIANTE LA REGLA DE SIMPSON)"
460 PRINT
470 IT=H*(SI+SP+(Y1+Y2)/2)
480 PRINT "EL AREA PEDIDA ES = ";IT ; "UNIDADES CUADRADAS";
490 PRINT "(MEDIANTE LA REGLA DE LOS TRAPECIOS)"
500 END

```

## ejercicios resueltos:

1. Hallar el área encerrada por la curva:  $\begin{cases} x(t) = 3 + \cos t \\ y(t) = 4 \sin t \end{cases}$

### SOLUCION :

El área pedida es 4 veces el area desde 0 a  $\pi/2$ , y además como la función es negativa resulta: Area pedida = -4 Area desde 0 a  $\pi/2$ .

Calculemos esta última:

```

DEFINA LA FUNCION QUE DESEA INTEGRAR EN LA LINEA 210
DEL SIGUIENTE MODO: 210 DEF FN Y(T) = .....
A CONTINUACION PULSAR <CONT>
210 DEF FN Y(T) = 4*SIN(T)

```

```

DEFINA LA DERIVADA DE X=X(T) RESPECTO DE T, EN LA LINEA 220
DEL SIGUIENTE MODO: 220 DEF FN D(T) = .....
A CONTINUACION PULSAR <CONT>
220 DEF FN D(T)=-SIN(T)

```

LIMITES DE INTEGRACION: INFERIOR, SUPERIOR ? 0 1.57079  
 INTRODUZCA EL NUMERO DE INTERVALOS (PARES)? 60

EL AREA PEDIDA ES = -3.1415673 UNIDADES CUADRADAS (MEDIANTE LA REGLA DE SIMPSON)

EL AREA PEDIDA ES = -3.1415673 UNIDADES CUADRADAS (MEDIANTE LA REGLA DE LOS TRAPECIOS)

El área encerrada por la curva será:  $4\pi$ .

2. Hallar el área encerrada por la elipse de ecuación: 
$$\begin{cases} x = 3 \cos t \\ y = 2 \sin t \end{cases}$$

**SOLUCION :**

El área pedida es en valor absoluto igual a 4 veces el área encerrada en el primer cuadrante, desde  $t=0$  hasta  $t = \pi/2$ . Calculemos esta cuarta parte.

```
DEFINA LA FUNCION QUE DESEA INTEGRAR EN LA LINEA 210
DEL SIGUIENTE MODO: 210 DEF FN Y(T) = .....
A CONTINUACION PULSAR <CONT>
210 DEF FN Y(T) = 2*SIN(T)
```

```
DEFINA LA DERIVADA DE X=X(T) RESPECTO DE T, EN LA LINEA 220
DEL SIGUIENTE MODO: 220 DEF FN D(T) = .....
A CONTINUACION PULSAR <CONT>
220 DEF FN D(T) = -3*SIN(T)
```

```
LIMITES DE INTEGRACION: INFERIOR, SUPERIOR ? 0    1.5707963
INTRODUZCA EL NUMERO DE INTERVALOS (PARES)? 80
```

EL AREA PEDIDA ES = -4.7123888 UNIDADES CUADRADAS (MEDIANTE LA REGLA DE SIMPSON)

EL AREA PEDIDA ES = -4.7123888 UNIDADES CUADRADAS (MEDIANTE LA REGLA DE LOS TRAPEZIOS) El área encerrada por la curva será:  $4 \cdot (4.7123888) = 18.84..$

3. Hallar el área encerrada por el arco de cicloide de ecuación:

$$\begin{cases} x(t) = t - \sin t \\ y(t) = 1 - \cos t \end{cases}$$

**SOLUCION :**

Un arco completo de cicloide se describe al variar  $t$  desde  $t = 0$  hasta  $t = 2\pi$ . Por tanto el área será:

```
DEFINA LA FUNCION QUE DESEA INTEGRAR EN LA LINEA 210
DEL SIGUIENTE MODO: 210 DEF FN Y(T) = .....
A CONTINUACION PULSAR <CONT>
210 DEF FN Y(T) = 1-COS(T)
```

```
DEFINA LA DERIVADA DE X=X(T) RESPECTO DE T, EN LA LINEA 220
DEL SIGUIENTE MODO: 220 DEF FN D(T) = .....
A CONTINUACION PULSAR <CONT>
220 DEF FN D(T) = 1-COS(T)
```

```
LIMITES DE INTEGRACION: INFERIOR, SUPERIOR ? 0    6.28318
INTRODUZCA EL NUMERO DE INTERVALOS (PARES)? 100
```

EL AREA PEDIDA ES = 9.424778 UNIDADES CUADRADAS (MEDIANTE LA REGLA DE SIMPSON)

EL AREA PEDIDA ES = 9.424778 UNIDADES CUADRADAS (MEDIANTE LA REGLA DE LOS TRAPEZIOS)

# LONGITUD DE UN ARCO DE CURVA DADA EN CARTESIANAS

## introducción:

Este programa sirve para calcular la longitud de un arco de curva de la forma  $y = f(x)$  desde  $x = a$  hasta  $x = b$ . El cálculo de la longitud de la curva se realiza por dos procedimientos: regla de Simpson y regla de los trapecios.

Para poder ejecutar el programa, es preciso definir la función en la línea 210 del siguiente modo: 210 DEF FN F(X) = .....

Y también, definir en la línea 220 la derivada de la función  $y = f(x)$  respecto de  $x$ , del siguiente modo:

220 DEF FN D(X) = .....

## listado del programa:

```
10 REM *****
20 REM
30 REM      L O N G I T U D      D E      U N      A R C O      D E      C U R V A
40 REM
50 REM      D A D A      E N      F. C A R T E S I A N A      Y=F(X)
60 REM
70 REM      M E T O D O S : S I M P S O N      Y      T R A P E C I O S
80 REM
90 REM *****
100 PRINT
109 REM - ENTRADA DE DATOS -
110 PRINT "DEFINA LA FUNCION QUE DESEA CALCULAR SU LONGITUD EN LA LINEA 210"
120 PRINT "DEL SIGUIENTE MODO: 210 DEF FN F(X) = ....."
130 PRINT "A CONTINUACION PULSAR <CONT>"
140 STOP
150 PRINT "DEFINA LA DERIVADA DE F(X) RESPECTO DE X EN LA LINEA 220"
160 PRINT "DEL SIGUIENTE MODO: 220 DEF FN D(X) = ....."
170 PRINT "A CONTINUACION PULSAR <CONT>"
180 STOP
190 PRINT "LIMITES DE INTEGRACION: INFERIOR, SUPERIOR ";
200 INPUT A,B
210 DEF FN F(X) =
220 DEF FN D(X) =
230 PRINT "INTRODUZCA EL NUMERO DE INTERVALOS (PARES)";
240 INPUT N
```



```

250 IF N/2<>INT(N/2) GOTO 250
260 H=(B-A)/(2*N)
270 N1=2*N-1
280 N2=2*(N-1)
290 SI=0
300 SP=0
309 REM - SUMA DE LOS PUNTOS IMPARES -
310 FOR J=1 TO N1 STEP 2
320 X=A+H*J
330 SI=SI+SQR(1+FN D(X)^2)
340 NEXT J
349 REM - SUMA DE LOS PUNTOS PARES -
350 FOR J=2 TO N2 STEP 2
360 X=A+H*J
370 SP=SP+SQR(1+FN D(X)^2)
380 NEXT J
389 REM - SALIDA DE RESULTADOS -
390 PRINT
400 PRINT
410 Y1=SQR(1+FN D(A)^2)
420 Y2=SQR(1+FN D(B)^2)
430 IS=H*(Y1+4*SI+2*SP+Y2)/3
440 PRINT "LA INTEGRAL PEDIDA ES =";IS;
450 PRINT "(MEDIANTE LA REGLA DE SIMPSON)"
460 PRINT
470 IT=H*(SI+SP+(Y1+Y2)/2)
480 PRINT "LA INTEGRAL PEDIDA ES =";IT;
490 PRINT "(MEDIANTE LA REGLA DE LOS TRAPECIOS)"
500 END

```

## ejercicios resueltos:

1. Hallar la longitud de un arco de catenaria de ecuación:

$$y = \frac{3}{2} (e^{x/3} + e^{-x/3}) \quad \text{desde } x = 0 \quad \text{hasta } x = 3.$$

### S O L U C I O N :

DEFINA LA FUNCION QUE DESEA CALCULAR SU LONGITUD EN LA LINEA 210  
DEL SIGUIENTE MODO: 210 DEF FN F(X) = .....

A CONTINUACION PULSAR <CONT>

210 DEF FN F(X) = 3\*(EXP(X/3)+EXP(-X/3))/2

DEFINA LA DERIVADA DE F(X) RESPECTO DE X EN LA LINEA 220


DEL SIGUIENTE MODO: 220 DEF FN D(X) = .....

A CONTINUACION PULSAR <CONT>

220 DEF FN D(X) = (EXP(X/3)-EXP(-X/3))/2

LIMITES DE INTEGRACION: INFERIOR, SUPERIOR ? 0 3

INTRODUZCA EL NUMERO DE INTERVALOS (PARES)? 40



LA INTEGRAL PEDIDA ES = 3.5256036 (MEDIANTE LA REGLA DE SIMPSON)

LA INTEGRAL PEDIDA ES = 3.5256495 (MEDIANTE LA REGLA DE LOS TRAPECIOS)

**2.** Hallar la longitud de un arco de la curva de ecuación:  $y = \ln x$  entre  $x = 1$  y  $x = 3$ .

**SOLUCION :**

DEFINA LA FUNCION QUE DESEA CALCULAR SU LONGITUD EN LA LINEA 210  
DEL SIGUIENTE MODO: 210 DEF FN F(X) = .....  
A CONTINUACION PULSAR <CONT>  
210 DEF FN F(X) = LOG(X)

DEFINA LA DERIVADA DE F(X) RESPECTO DE X EN LA LINEA 220  
DEL SIGUIENTE MODO: 220 DEF FN D(X) = .....  
A CONTINUACION PULSAR <CONT>  
220 DEF FN D(X) = 1/X

LIMITES DE INTEGRACION: INFERIOR, SUPERIOR ? 1 3  
INTRODUZCA EL NUMERO DE INTERVALOS (PARES)? 40

LA INTEGRAL PEDIDA ES = 2.3019875 (MEDIANTE LA REGLA DE SIMPSON)

LA INTEGRAL PEDIDA ES = 2.3020225 (MEDIANTE LA REGLA DE LOS TRAPECIOS)

**3.** Hallar la longitud de un arco de la curva de ecuación:  $y = e^{x/2}$  entre  $x = -2$  hasta  $x = 2$ .

**SOLUCION :**


DEFINA LA FUNCION QUE DESEA CALCULAR SU LONGITUD EN LA LINEA 210  
DEL SIGUIENTE MODO: 210 DEF FN F(X) = .....  
A CONTINUACION PULSAR <CONT>  
210 DEF FN F(X) = EXP(X/2)

DEFINA LA DERIVADA DE F(X) RESPECTO DE X EN LA LINEA 220  
DEL SIGUIENTE MODO: 220 DEF FN D(X) = .....  
A CONTINUACION PULSAR <CONT>  
220 DEF FN D(X) = EXP(X/2)\*.5

LIMITES DE INTEGRACION: INFERIOR, SUPERIOR ? -2 2  
INTRODUZCA EL NUMERO DE INTERVALOS (PARES)? 60

LA INTEGRAL PEDIDA ES = 4.7670795 (MEDIANTE LA REGLA DE SIMPSON)

LA INTEGRAL PEDIDA ES = 4.7671286 (MEDIANTE LA REGLA DE LOS TRAPECIOS)



# LONGITUD DE UN ARCO DE CURVA DADA EN POLARES

## introducción:

Este programa sirve para calcular la longitud de un arco de curva dada en forma polar  $r = r(w)$ , desde un valor particular de  $w = w_1$  hasta  $w = w_2$ . El cálculo de la longitud se realiza por dos procedimientos: regla de Simpson y regla de los trapecios.

Para poder ejecutar el programa, es preciso definir la función en la línea 210, del siguiente modo: 210 DEF FN R(W) = .....  
A continuación, se ha de definir en la línea 220, la derivada de  $r(w)$  respecto de  $w$ , del siguiente modo: 220 DEF FN D(W) = .....

## listado del programa:

```
10 REM *****
20 REM
30 REM      L O N G I T U D      D E      U N      A R C O      D E      C U R V A
40 REM
50 REM      D A D A      E N      F. P O L A R      R=R(W)
60 REM
70 REM      M E T O D O S : S I M P S O N      Y      T R A P E C I O S
80 REM
90 REM *****
100 PRINT
109 REM - ENTRADA DE DATOS -
110 PRINT "DEFINA LA FUNCION QUE DESEA CALCULAR SU LONGITUD EN LA LINEA 210"
120 PRINT "DEL SIGUIENTE MODO: 210 DEF FN R(W) = ....."
130 PRINT "A CONTINUACION PULSAR <CONT>"
140 STOP
150 PRINT "DEFINA LA DERIVADA DE R(W) RESPECTO DE W EN LA LINEA 220"
160 PRINT "DEL SIGUIENTE MODO: 220 DEF FN D(W) = ....."
170 PRINT "A CONTINUACION PULSAR <CONT>"
180 STOP
190 PRINT "LIMITES DE INTEGRACION: INFERIOR, SUPERIOR (EN RADIANES)";
200 INPUT W1,W2
210 DEF FN R(W)=
220 DEF FN D(W)=
230 PRINT "INTRODUZCA EL NUMERO DE INTERVALOS (PARES)";
240 INPUT N
250 IF N/2<>INT(N/2) GOTO 250
```

```

260 H=(W2-W1)/(2*N)
270 N1=2*N-1
280 N2=2*(N-1)
290 SI=0
300 SP=0
309 REM - SUMA DE LOS PUNTOS IMPARES -
310 FOR J=1 TO N1 STEP 2
320 W=W1+H*J
330 SI=SI+SQR((FN R(W))^2+(FN D(W))^2)
340 NEXT J
349 REM - SUMA DE LOS PUNTOS PARES -
350 FOR J=2 TO N2 STEP 2
360 W=W1+H*J
370 SP=SP+SQR((FN R(W))^2+(FN D(W))^2)
380 NEXT J
389 REM - SALIDA DE RESULTADOS -
390 PRINT
400 PRINT
410 R1=SQR(FNR(W1)^2+FND(W1)^2)
420 R2=SQR(FNR(W2)^2+FND(W2)^2)
430 IS=H*(R1+4*SI+2*SP+R2)/3
440 PRINT "LA INTEGRAL PEDIDA ES =";IS;
450 PRINT "(MEDIANTE LA REGLA DE SIMPSON)"
460 PRINT
470 IT=H*(SI+SP+(R1+R2)/2)
480 PRINT "LA INTEGRAL PEDIDA ES =";IT;
490 PRINT "(MEDIANTE LA REGLA DE LOS TRAPECIOS)"
500 END

```

## ejercicios resueltos:

1. Hallar la longitud de la cardioide de ecuación:  $r = 3(1 - \cos w)$

### SOLUCION :

La longitud pedida es el doble de la longitud entre 0 y  $\pi$ . Por tanto, calcularemos esta última:

```

DEFINA LA FUNCION QUE DESEA CALCULAR SU LONGITUD EN LA LINEA 210
DEL SIGUIENTE MODO: 210 DEF FN R (W) = .....
A CONTINUACION PULSAR <CONT>
210 DEF FN R(W) = 3*(1-COS(W))

```

```

DEFINA LA DERIVADA DE R(W) RESPECTO DE W EN LA LINEA 220
DEL SIGUIENTE MODO: 220 DEF FN D(W) = .....
A CONTINUACION PULSAR <CONT>
220 DEF FN D(W) = 3*SIN(W)
LIMITES DE INTEGRACION: INFERIOR, SUPERIOR (EN RADIANES)? 0      3.1415
INTRODUZCA EL NUMERO DE INTERVALOS (PARES)? 60

```

LA INTEGRAL PEDIDA ES = 11.999444 (MEDIANTE LA REGLA DE SIMPSON)

LA INTEGRAL PEDIDA ES = 11.999273 (MEDIANTE LA REGLA DE LOS TRAPECIOS)

La longitud de la cardioide será: 24 unidades.

2. Hallar la longitud de la rosa de tres pétalos de ecuación:

$r = 2 \cos 3 w$ .

### SOLUCION :

La longitud pedida es 6 veces la longitud entre 0 y  $\pi/6$ . Por tanto, calcu  
laremos esta última:

```
DEFINA LA FUNCION QUE DESEA CALCULAR SU LONGITUD EN LA LINEA 210
DEL SIGUIENTE MODO: 210 DEF FN R (W) = .....
A CONTINUACION PULSAR <CONT>
210 DEF FN R(W) = 2*COS(3*W)
```

```
DEFINA LA DERIVADA DE R(W) RESPECTO DE W EN LA LINEA 220
DEL SIGUIENTE MODO: 220 DEF FN D(W) = .....
A CONTINUACION PULSAR <CONT>
220 DEF FN D(W) = -6*SIN(W)
```

```
LIMITES DE INTEGRACION: INFERIOR, SUPERIOR (EN RADIANES)? 0 .7853982
INTRODUZCA EL NUMERO DE INTERVALOS (PARES)? 100
```

LA INTEGRAL PEDIDA ES = 2.1515775 (MEDIANTE LA REGLA DE SIMPSON)

LA INTEGRAL PEDIDA ES = 2.1515844 (MEDIANTE LA REGLA DE LOS TRAPECIOS)

La longitud de la rosa será: 12.9

### 3. Hallar la longitud de la lemniscata de Bernoulli de ecuación:

$$r^2 = 9 \cos 2 w.$$

#### S O L U C I O N :

La longitud pedida es 4 veces la longitud entre 0 y  $\pi/4$ . Por tanto, calcu  
laremos esta última:

```
DEFINA LA FUNCION QUE DESEA CALCULAR SU LONGITUD EN LA LINEA 210
DEL SIGUIENTE MODO: 210 DEF FN R (W) = .....
A CONTINUACION PULSAR <CONT>
210 DEF FN R(W) = 3*SQR(COS(2*W))
```

```
DEFINA LA DERIVADA DE R(W) RESPECTO DE W EN LA LINEA 220
DEL SIGUIENTE MODO: 220 DEF FN D(W) = .....
A CONTINUACION PULSAR <CONT>
220 DEF FN D(W) = -3*SIN(2*W)/SQR(COS(2*W))
```

```
LIMITES DE INTEGRACION: INFERIOR, SUPERIOR (EN RADIANES)? 0 .78
INTRODUZCA EL NUMERO DE INTERVALOS (PARES)? 200
```

LA INTEGRAL PEDIDA ES = 3.6213924 (MEDIANTE LA REGLA DE SIMPSON)

LA INTEGRAL PEDIDA ES = 3.6222104 (MEDIANTE LA REGLA DE LOS TRAPECIOS)

La longitud de la lemniscata será: 14.484

# LONGITUD DE UN ARCO DE CURVA DADA EN PARAMETRICAS

## introducción:

Este programa sirve para calcular la longitud de un arco de curva dada en forma paramétrica:  $x = x(t)$ ,  $y = y(t)$ , desde  $t = t_1$  hasta  $t = t_2$ . El cálculo de la longitud se realiza por dos procedimientos: regla de Simpson y regla de los trapecios.

Para poder ejecutar el programa, es preciso definir en la línea 210 la derivada de la función  $x = x(t)$  respecto de  $t$ , del siguiente modo:

210 DEF FN F(T) = .....

A continuación, se ha de definir en la línea 220, la derivada de la función  $y = y(t)$  respecto de  $t$ , del siguiente modo:

220 DEF FN G(T) = .....

## listado del programa:

```
10 REM *****
20 REM
30 REM     L O N G I T U D       D E       U N       A R C O       D E       C U R V A
40 REM
50 REM     D A D A       E N       F. P A R A M E T R I C A  X=X(T); Y=Y(T)
60 REM
70 REM     M E T O D O S : S I M P S O N       Y       T R A P E C I O S
80 REM
90 REM *****
100 PRINT
109 REM - ENTRADA DE DATOS -
110 PRINT "DEFINA LA DERIVADA DE X(T) RESPECTO DE T EN LA LINEA 210"
120 PRINT "DEL SIGUIENTE MODO: 210 DEF FN F(T) = ....."
130 PRINT "A CONTINUACION PULSAR <CONT>"
140 STOP
150 PRINT "DEFINA LA DERIVADA DE Y(T) RESPECTO DE T EN LA LINEA 220"
160 PRINT "DEL SIGUIENTE MODO: 220 DEF FN G(T) = ....."
170 PRINT "A CONTINUACION PULSAR <CONT>"
180 STOP
190 PRINT "LIMITES DE INTEGRACION: INFERIOR, SUPERIOR ";
200 INPUT T1,T2
210 DEF FN F(T) =
220 DEF FN G(T) =
230 PRINT "INTRODUZCA EL NUMERO DE INTERVALOS (PARES)";
```

```

240 INPUT N
250 IF N/2<>INT(N/2) GOTO 250
260 H=(T2-T1)/(2*N)
270 N1=2*N-1
280 N2=2*(N-1)
290 SI=0
300 SP=0
309 REM - SUMA DE LOS PUNTOS IMPARES -
310 FOR J=1 TO N1 STEP 2
320 T=T1+H*J
330 SI=SI+SQR(FN F(T)^2 + FN G(T)^2)
340 NEXT J
349 REM - SUMA DE LOS PUNTOS PARES -
350 FOR J=2 TO N2 STEP 2
360 T=T1+H*J
370 SP=SP+SQR(FN F(T)^2 + FN G(T)^2)
380 NEXT J
389 REM - SALIDA DE RESULTADOS -
390 PRINT
400 PRINT
410 Y1=SQR(FN F(T1)^2+FN G(T1)^2)
420 Y2=SQR(FN F(T2)^2+FN G(T2)^2)
430 IS=H*(Y1+4*SI+2*SP+Y2)/3
440 PRINT "LA INTEGRAL PEDIDA ES =";IS;
450 PRINT "(MEDIANTE LA REGLA DE SIMPSON)"
460 PRINT
470 IT=H*(SI+SP+(Y1+Y2)/2)
480 PRINT "LA INTEGRAL PEDIDA ES =";IT;
490 PRINT "(MEDIANTE LA REGLA DE LOS TRAPECIOS)"
500 END

```

## ejercicios resueltos:

1. Hallar la longitud de un arco de la curva:  $x = t^2$ ,  $y = t^3$ , desde  $t = 0$  a  $t = 4$ .

### S O L U C I O N :

```

DEFINA LA DERIVADA DE X(T) RESPECTO DE T EN LA LINEA 210
DEL SIGUIENTE MODO: 210 DEF FN F(T) = .....
A CONTINUACION PULSAR <CONT>
210 DEF FN F(T) = 2*T

```

```

DEFINA LA DERIVADA DE Y(T) RESPECTO DE T EN LA LINEA 220
DEL SIGUIENTE MODO: 220 DEF FN G(T) = .....
A CONTINUACION PULSAR <CONT>
220 DEF FN G(T) = 3*T^2

```

LÍMITES DE INTEGRACION: INFERIOR, SUPERIOR ? 0 4  
 INTRODUZCA EL NUMERO DE INTERVALOS (PARES) ? 50

LA INTEGRAL PEDIDA ES = 66.388804 (MEDIANTE LA REGLA DE SIMPSON)

LA INTEGRAL PEDIDA ES = 66.391738 (MEDIANTE LA REGLA DE LOS TRAPECIOS)

**2.** Hallar la longitud del arco completo descrito por una cicloide de ecuación:  $x = t - \sin t$ ,  $y = 1 - \cos t$

**S O L U C I O N :**

Un arco completo de cicloide se describe desde  $t = 0$  hasta  $t = 2\pi$ . Así pues la longitud pedida será:

```
DEFINA LA DERIVADA DE X(T) RESPECTO DE T EN LA LINEA 210
DEL SIGUIENTE MODO: 210 DEF FN F(T) = .....
A CONTINUACION PULSAR <CONT>
210 DEF FN F(T) = 1-COS(T)
```

```
DEFINA LA DERIVADA DE Y(T) RESPECTO DE T EN LA LINEA 220
DEL SIGUIENTE MODO: 220 DEF FN G(T) = .....
A CONTINUACION PULSAR <CONT>
220 DEF FN G(T) = SIN(T)
```

```
LIMITES DE INTEGRACION: INFERIOR, SUPERIOR ? 0 6.28
INTRODUZCA EL NUMERO DE INTERVALOS (FARES)? 60
```

LA INTEGRAL PEDIDA ES = 7.9999949 (MEDIANTE LA REGLA DE SIMPSON)

LA INTEGRAL PEDIDA ES = 7.9995385 (MEDIANTE LA REGLA DE LOS TRAPECIOS)

**3.** La posición de un punto en un instante viene dada por las siguientes ecuaciones:  $x = t^2/2$ ,  $y = \frac{1}{9} (6t + 9)^{3/2}$

Hallar el espacio recorrido por el punto desde  $t = 0$  hasta  $t = 4$ .

**S O L U C I O N :**

```
DEFINA LA DERIVADA DE X(T) RESPECTO DE T EN LA LINEA 210
DEL SIGUIENTE MODO: 210 DEF FN F(T) = .....
A CONTINUACION PULSAR <CONT>
210 DEF FN F(T) = T
```

```
DEFINA LA DERIVADA DE Y(T) RESPECTO DE T EN LA LINEA 220
DEL SIGUIENTE MODO: 220 DEF FN G(T) = .....
A CONTINUACION PULSAR <CONT>
220 DEF FN G(T) = SQR(6*T+9)
```

```
LIMITES DE INTEGRACION: INFERIOR, SUPERIOR ? 0 4
INTRODUZCA EL NUMERO DE INTERVALOS (PARES)? 100
```

LA INTEGRAL PEDIDA ES = 20 (MEDIANTE LA REGLA DE SIMPSON)

LA INTEGRAL PEDIDA ES = 20 (MEDIANTE LA REGLA DE LOS TRAPECIOS)



# AREA DE REVOLUCION

## introducción:

Este programa sirve para calcular el área de un sólido de revolución engendrado al girar la curva  $y = f(x)$  alrededor del eje  $x$ , desde  $x = a$  hasta  $x = b$ . El cálculo del área se realiza por dos procedimientos: regla de Simpson y regla de los trapecios.

Para ejecutar este programa, es preciso definir en la línea 210 la función  $y = f(x)$ , del siguiente modo:

210 DEF FN F(X) = .....

A continuación se ha de definir la derivada de la función  $y = f(x)$  respecto de  $x$  en la línea 220 del siguiente modo:

220 DEF FN D(X) = .....

## listado del programa:

```
10 REM *****
20 REM
30 REM      AREA   DEL   SOLIDO   DE   REVOLUCION
40 REM
50 REM      ENGENDRADO   AL   GIRAR   LA   CURVA   Y=F(X)
60 REM
70 REM      ALREDEDOR   DEL   EJE   X   [SIMPSON Y TRAPECIOS]
80 REM
90 REM *****
100 PRINT
109 REM - ENTRADA DE DATOS -
110 PRINT "DEFINA LA FUNCION QUE DESEA INTEGRAR EN LA LINEA 210"
120 PRINT "DEL SIGUIENTE MODO: 210 DEF FN F(X) = ....."
130 PRINT "A CONTINUACION PULSAR <CONT>"
140 STOP
150 PRINT "DEFINA LA DERIVADA DE Y=F(X) RESPECTO DE X, EN LA LINEA 220"
160 PRINT "DEL SIGUIENTE MODO: 220 DEF FN D(X) = ....."
170 PRINT "A CONTINUACION PULSAR <CONT>"
180 STOP
190 PRINT "LIMITES DE INTEGRACION: INFERIOR, SUPERIOR ";
200 INPUT A,B
210 DEF FN F(X) =
220 DEF FN D(X) =
230 PRINT "INTRODUZCA EL NUMERO DE INTERVALOS (PARES)";
```

```

240 INPUT N
250 IF N/2<>INT(N/2) GOTO 250
260 H=(B-A)/(2*N)
270 N1=2*N-1
280 N2=2*(N-1)
290 SI=0
300 SP=0
310 FOR J=1 TO N1 STEP 2
320 X=A+H*J
330 SI=SI+FN F(X)*SQR(1+FND(X)^2)
340 NEXT J
350 FOR J=2 TO N2 STEP 2
360 X=A+H*J
370 SP=SP+FN F(X)*SQR(1+FND(X)^2)
380 NEXT J
389 REM - SALIDA DE RESULTADOS -
390 PRINT
400 PRINT
410 Y1=FN F(A)*SQR(1+FND(A)^2)
420 Y2=FN F(B)*SQR(1+FND(B)^2)
430 IS =6.2831853*H*(Y1+4*SI+2*SP+Y2)/3
440 PRINT "EL AREA PEDIDA ES = ";IS ; "UNIDADES CUADRADAS ";
450 PRINT "(MEDIANTE LA REGLA DE SIMPSON)"
460 PRINT
470 IT=6.2831853*H*(SI+SP+(Y1+Y2)/2)
480 PRINT "EL AREA PEDIDA ES = ";IT ; "UNIDADES CUADRADAS ";
490 PRINT "(MEDIANTE LA REGLA DE LOS TRAPECIOS)"
500 END

```

## ejercicios resueltos:

**1.** Hallar el área del sólido de revolución engendrado por la curva  $y = L(x + 1)$  al girar alrededor del eje  $x$ , desde  $x = 1$  hasta  $x = 2$ .

### SOLUCION :

```

DEFINA LA FUNCION QUE DESEA INTEGRAR EN LA LINEA 210
DEL SIGUIENTE MODO: 210 DEF FN F(X) = .....
A CONTINUACION PULSAR <CONT>
210 DEF FN F(X) =LOG(X+1)

```

```

DEFINA LA DERIVADA DE Y=F(X) RESPECTO DE X, EN LA LINEA 220
DEL SIGUIENTE MODO: 220 DEF FN D(X) = .....
A CONTINUACION PULSAR <CONT>
220 DEF FN D(X) = 1/(X+1)

```

LIMITES DE INTEGRACION: INFERIOR, SUPERIOR ? 1 2  
 INTRODUZCA EL NUMERO DE INTERVALOS (PARES)? 40

EL AREA PEDIDA ES = 6.1586332 UNIDADES CUADRADAS (MEDIANTE LA REGLA DE SIMPSON)  
 EL AREA PEDIDA ES = 6.1586194 UNIDADES CUADRADAS (MEDIANTE LA REGLA DE LOS TRAPECIOS)

**2.** Hallar el área del sólido de revolución engendrado por la curva  $y = e^x$  al girar alrededor del eje  $x$ , desde  $x = -1$  hasta  $x = 1$ .

**SOLUCION :**

DEFINA LA FUNCION QUE DESEA INTEGRAR EN LA LINEA 210  
DEL SIGUIENTE MODO: 210 DEF FN F(X) = .....  
A CONTINUACION PULSAR <CONT>  
210 DEF FN F(X)= EXP(X)

DEFINA LA DERIVADA DE  $Y=F(X)$  RESPECTO DE  $X$ , EN LA LINEA 220  
DEL SIGUIENTE MODO: 220 DEF FN D(X) = .....  
A CONTINUACION PULSAR <CONT>  
220 DEF FN D(X)= EXP(X)

LIMITES DE INTEGRACION: INFERIOR, SUPERIOR ? -1 1  
INTRODUZCA EL NUMERO DE INTERVALOS (PARES)? 60

EL AREA PEDIDA ES = 27.792241 UNIDADES CUADRADAS (MEDIANTE LA REGLA DE SIMPSON)

EL AREA PEDIDA ES = 27.794331 UNIDADES CUADRADAS (MEDIANTE LA REGLA DE LOS TRAPECIOS)

**3.** Hallar el área descrita por una semionda de la senoide  $y = \sin x$  al girar alrededor del eje  $x$ .

**SOLUCION :**

Al ser una semionda los extremos del intervalo de integración son: 0 y  $\pi$ .

DEFINA LA FUNCION QUE DESEA INTEGRAR EN LA LINEA 210  
DEL SIGUIENTE MODO: 210 DEF FN F(X) = .....  
A CONTINUACION PULSAR <CONT>  
210 DEF FN F(X) = SIN(X)

DEFINA LA DERIVADA DE  $Y=F(X)$  RESPECTO DE  $X$ , EN LA LINEA 220  
DEL SIGUIENTE MODO: 220 DEF FN D(X) = .....  
A CONTINUACION PULSAR <CONT>  
220 DEF FN D(X) = COS(X)

LIMITES DE INTEGRACION: INFERIOR, SUPERIOR ? 0 3.14159  
INTRODUZCA EL NUMERO DE INTERVALOS (PARES)? 40

EL AREA PEDIDA ES = 14.4236 UNIDADES CUADRADAS (MEDIANTE LA REGLA DE SIMPSON)

EL AREA PEDIDA ES = 14.421315 UNIDADES CUADRADAS (MEDIANTE LA REGLA DE LOS TRAPECIOS)

# VOLUMEN DE REVOLUCION

## introducción:

Este programa sirve para calcular el volumen de un sólido de revolución engendrado al girar la curva  $y = f(x)$  alrededor del eje  $x$ , desde  $x = a$  hasta  $x = b$ . El cálculo del volumen se realiza por dos procedimientos: regla de Simpson y regla de los trapecios.

Para ejecutar este programa, es preciso definir en la línea 170 la función  $y = f(x)$  del siguiente modo:

```
170 DEF FN F(X) = .....
```

## listado del programa:

```
10 REM *****
20 REM
30 REM   V O L U M E N   D E L   S O L I D O   D E   R E V O L U C I O N
40 REM
50 REM   E N G E N D R A D O   A L   G I R A R   L A   C U R V A   Y=F(X)
60 REM
70 REM   A L R E D E D O R   D E L   E J E   X   [SIMPSON Y TRAPÉCIOS]
80 REM
90 REM *****
100 PRINT
109 REM - ENTRADA DE DATOS -
110 PRINT "DEFINA LA FUNCION DE LA CURVA Y=F(X), EN LA LINEA 170"
120 PRINT "DEL SIGUIENTE MODO: 170 DEF FN F(X) = ....."
130 PRINT "A CONTINUACION PULSAR <CONT>"
140 STOP
150 PRINT "LIMITES DE INTEGRACION: INFERIOR, SUPERIOR ";
160 INPUT A,B
170 DEF FN F(X)=
180 PRINT "INTRODUZCA EL NUMERO DE INTERVALOS (PARES)";
190 INPUT N
200 IF N/2<>INT(N/2) GOTO 200
210 H=(B-A)/(2*N)
220 N1=2*N-1
230 N2=2*(N-1)
240 SI=0
250 SP=0
260 FOR J=1 TO N1 STEP 2
```

```

270 X=A+H*J
280 SI=SI+FN F(X)^2
290 NEXT J
300 FOR J=2 TO N2 STEP 2
310 X=A+H*J
320 SP=SP+FN F(X)^2
330 NEXT J
339 REM - SALIDA DE RESULTADOS -
340 PRINT
350 PRINT
360 Y1=FN F(A)^2
370 Y2=FN F(B)^2
380 IS =3.1415927*H*(Y1+4*SI+2*SP+Y2)/3
390 PRINT "EL VOLUMEN PEDIDO ES = ";IS ; "UNIDADES CUBICAS";
400 PRINT "(MEDIANTE LA REGLA DE SIMPSON)"
410 PRINT
420 IT=3.1415927*H*(SI+SP+(Y1+Y2)/2)
430 PRINT "EL VOLUMEN PEDIDO ES = ";IT ; "UNIDADES CUBICAS";
440 PRINT "(MEDIANTE LA REGLA DE LOS TRAPECIOS)"
450 END

```

## ejercicios resueltos:

1. Hallar el volumen del sólido de revolución engendrado al girar la curva  $y = \operatorname{cosec} x$  alrededor del eje  $x$  entre  $\pi/4$  y  $\pi/2$ .

### SOLUCION :

DEFINA LA FUNCION DE LA CURVA  $Y=F(X)$ , EN LA LINEA 170 DEL SIGUIENTE MODO: 170 DEF FN F(X) = .....  
A CONTINUACION PULSAR <CONT>

```
170 DEF FN F(X)=1/SIN(X)
```

LIMITES DE INTEGRACION: INFERIOR, SUPERIOR ? .78539 1.57079  
INTRODUZCA EL NUMERO DE INTERVALOS (PARES)? 60

EL VOLUMEN PEDIDO ES = 3.1416241 UNIDADES CUBICAS  
(MEDIANTE LA REGLA DE SIMPSON)

EL VOLUMEN PEDIDO ES = 3.141669 UNIDADES CUBICAS  
(MEDIANTE LA REGLA DE LOS TRAPECIOS)

2. Hallar el volumen del sólido de revolución engendrado al girar la curva  $y = 4x - x^2$  alrededor del eje  $x$ , entre  $x = 0$  y  $x = 2$ .

### SOLUCION :

DEFINA LA FUNCION DE LA CURVA  $Y=F(X)$ , EN LA LINEA 170 DEL SIGUIENTE MODO: 170 DEF FN F(X) = .....  
A CONTINUACION PULSAR <CONT>

```
170 DEF FN F(X)=4*X-X*X
```

LIMITES DE INTEGRACION: INFERIOR, SUPERIOR ? 0 2  
 INTRODUZCA EL NUMERO DE INTERVALOS (PARES)? 30

EL VOLUMEN PEDIDO ES = 53.616516 UNIDADES CUBICAS  
 (MEDIANTE LA REGLA DE SIMPSON)

EL VOLUMEN PEDIDO ES = 53.616515 UNIDADES CUBICAS  
 (MEDIANTE LA REGLA DE LOS TRAPECIOS)

**3.** Hallar el volúmen del sólido de revolución engendrado al girar la curva  $y = e^x \sin x$  alrededor del eje  $x$ , entre  $x = 0$  y  $x = \pi$ .

**S O L U C I O N :**

DEFINA LA FUNCION DE LA CURVA  $Y=F(X)$ , EN LA LINEA 170  
 DEL SIGUIENTE MODO: 170 DEF FN F(X) = .....  
 A CONTINUACION PULSAR <CONT>

170 DEF FN F(X)=EXP(X)\*SIN(X)

LIMITES DE INTEGRACION: INFERIOR, SUPERIOR ? 0 3.14159  
 INTRODUZCA EL NUMERO DE INTERVALOS (PARES)? 60

EL VOLUMEN PEDIDO ES = 209.89444 UNIDADES CUBICAS  
 (MEDIANTE LA REGLA DE SIMPSON)

EL VOLUMEN PEDIDO ES = 209.89437 UNIDADES CUBICAS  
 (MEDIANTE LA REGLA DE LOS TRAPECIOS)

# REGLA DE TCHEBICHEFF

## introducción:

El presente programa permite obtener mediante aproximación integra les definidas, por la regla de integración numérica de Tchebicheff.

Para la ejecución del programa, es preciso definir en la línea 170 la función a integrar del siguiente modo:

```
170 DEF FN F(X) = .....
```

## listado del programa:

```
10 REM *****
20 REM
30 REM           I N T E G R A C I O N       N U M E R I C A
40 REM
50 REM           R E G L A       D E       T C H E B I C H E F F
60 REM
70 REM *****
100 PRINT
109 REM - ENTRADA DE DATOS -
110 PRINT "DEFINA LA FUNCION QUE DESEA INTEGRAR EN LA LINEA 170"
120 PRINT "DEL SIGUIENTE MODO: 170 DEF FN F(X) = ....."
130 PRINT "A CONTINUACION PULSAR <CONT>"
140 STOP
150 PRINT "INTRODUZCA LOS LIMITES DE INTEGRACION : INFERIOR Y SUPERIOR";
160 INPUT A,B
170 DEF FN F(X)=
180 DIM X(9)
190 I=0
199 REM - VALORES DE LOS COEFICIENTES X(J) -
200 X(1)=0
210 X(2)=.1679061842
220 X(3)=.5287617831
230 X(4)=.6010186554
240 X(5)=.9115893077
250 X(6)=-.1679061842
260 X(7)=-.5287617831
270 X(8)=-.6010186554
280 X(9)=-.9115893077
290 FOR J=1 TO 9
300 Y=.5*(B-A)*X(J)
310 Y=Y+.5*(B+A)
320 I=I+FN F(Y)
```

```

330 NEXT J
340 I=I*(B-A)/9
349 REM - SALIDA DEL RESULTADO -
350 PRINT
360 PRINT "LA INTEGRAL PEDIDA ES =";I;
370 PRINT "(REGLA DE TCHEBICHEFF)"
380 END

```

## ejercicios resueltos:

**1.** Calcular el valor de la siguiente integral:

$$\int_0^{1.2} \frac{dx}{1+x}$$

**SOLUCION :**

```

DEFINA LA FUNCION QUE DESEA INTEGRAR EN LA LINEA 170
DEL SIGUIENTE MODO: 170 DEF FN F(X) = .....
A CONTINUACION PULSAR <CONT>
170 DEF FN F(X) = 1/(1+X)

```

INTRODUZCA LOS LIMITES DE INTEGRACION : INFERIOR Y SUPERIOR? 0 1.2

LA INTEGRAL PEDIDA ES = .78845729 (REGLA DE TCHEBICHEFF)

**2.** Calcular el valor de la siguiente integral:

$$\int_0^3 \frac{dx}{\sqrt{1+x}}$$

**SOLUCION :**

```

DEFINA LA FUNCION QUE DESEA INTEGRAR EN LA LINEA 170
DEL SIGUIENTE MODO: 170 DEF FN F(X) = .....
A CONTINUACION PULSAR <CONT>
170 DEF FN F(X) = 1/SQR(1+X)

```

INTRODUZCA LOS LIMITES DE INTEGRACION : INFERIOR Y SUPERIOR? 0 3

LA INTEGRAL PEDIDA ES = 1.9999934 (REGLA DE TCHEBICHEFF)

**3.** Calcular el valor de la siguiente integral:

$$\int_4^8 \frac{x \, dx}{\sqrt{x^2 - 15}}$$

**SOLUCION :**

```

DEFINA LA FUNCION QUE DESEA INTEGRAR EN LA LINEA 170
DEL SIGUIENTE MODO: 170 DEF FN F(X) = .....
A CONTINUACION PULSAR <CONT>
170 DEF FN F(X) = X/SQR(X*X-15)

```

INTRODUZCA LOS LIMITES DE INTEGRACION : INFERIOR Y SUPERIOR? 4 8

LA INTEGRAL PEDIDA ES = 5.9782786 (REGLA DE TCHEBICHEFF)



# REGLA DE BODE

## introducción:

Este programa sirve para calcular mediante aproximación integrales definidas de funciones de la forma  $y = f(x)$ . El procedimiento empleado es la regla de integración numérica de Bode.

Para la ejecución del programa, es preciso definir en la línea 170 la función a integrar del siguiente modo:

```
170 DEF FN F(X) = .....
```

## listado del programa:

```
10 REM *****
20 REM
30 REM           I N T E G R A C I O N       N U M E R I C A
40 REM
50 REM           R E G L A       D E       B O D E
60 REM
70 REM *****
100 PRINT
109 REM - ENTRADA DE DATOS -
110 PRINT "DEFINA LA FUNCION QUE DESEA INTEGRAR EN LA LINEA 170"
120 PRINT "DEL SIGUIENTE MODO: 170 DEF FN F(X) = ....."
130 PRINT "A CONTINUACION PULSAR <CONT>"
140 STOP
150 PRINT "INTRODUZCA LOS LIMITES DE INTEGRACION : INFERIOR Y SUPERIOR";
160 INPUT A,B
170 DEF FN F(X) =
180 DIM W(6)
190 I=0
200 H=(B-A)/10
209 REM - VALORES DE LOS COEFICIENTES W(J) -
210 W(1)= .2683414836
220 W(2)= 1.775359414
230 W(3)= -.8104357063
240 W(4)= 4.549462883
250 W(5)=-4.351551227
260 W(6)= 3.568823152
270 FOR J=1 TO 6
280 K=11-J
290 X=A+H*(J-1)
300 S=FN F(X)
310 X=A+H*K
```

```

320 S=S+FN F(X)
330 I=I+S*W(J)
340 NEXTJ
350 I=I*H
359 REM - SALIDA DEL RESULTADO -
360 PRINT
370 PRINT "LA INTEGRAL PEDIDA ES =";I;
380 PRINT "(MEDIANTE LA REGLA DE BODE)"
390 END

```

## **ejercicios resueltos:**

**1.** Calcular el valor de la siguiente integral:

$$\int_0^{1.2} \frac{dx}{1+x}$$

**SOLUCION :**

```

DEFINA LA FUNCION QUE DESEA INTEGRAR EN LA LINEA 170
DEL SIGUIENTE MODO: 170 DEF FN F(X) = .....
A CONTINUACION PULSAR <CONT>
170 DEF FN F(X) = 1/(1+X)

```

```

INTRODUZCA LOS LIMITES DE INTEGRACION : INFERIOR Y SUPERIOR? 0 1.2
LA INTEGRAL PEDIDA ES = .78845737 (MEDIANTE LA REGLA DE BODE)

```

**2.** Calcular el valor de la siguiente integral:

$$\int_0^3 \frac{dx}{\sqrt{1+x}}$$

**SOLUCION :**

```

DEFINA LA FUNCION QUE DESEA INTEGRAR EN LA LINEA 170
DEL SIGUIENTE MODO: 170 DEF FN F(X) = .....
A CONTINUACION PULSAR <CONT>
170 DEF FN F(X) = 1/SQR(1+X)

```

```

INTRODUZCA LOS LIMITES DE INTEGRACION : INFERIOR Y SUPERIOR? 0 3
LA INTEGRAL PEDIDA ES = 2.00000018 (MEDIANTE LA REGLA DE BODE)

```

**3.** Calcular el valor de la siguiente integral:

$$\int_4^8 \frac{x \, dx}{\sqrt{x^2 - 15}}$$

**SOLUCION :**

```

DEFINA LA FUNCION QUE DESEA INTEGRAR EN LA LINEA 170
DEL SIGUIENTE MODO: 170 DEF FN F(X) = .....
A CONTINUACION PULSAR <CONT>
170 DEF FN F(X) = X/SQR(X*X-15)

```

```

INTRODUZCA LOS LIMITES DE INTEGRACION : INFERIOR Y SUPERIOR? 4 8
LA INTEGRAL PEDIDA ES = 6.0234465 (MEDIANTE LA REGLA DE BODE)

```

# REGLA DE GAUSS

## introducción:

Este programa sirve para calcular mediante aproximación integrales definidas de funciones de la forma  $y = f(x)$ . El procedimiento empleado es la regla de integración numérica de Gauss.

Para la ejecución del programa, es preciso definir en la línea 170 la función a integrar del siguiente modo:

```
170 DEF FN F(X) = .....
```

## listado del programa:

```
10 REM *****
20 REM
30 REM             I N T E G R A C I O N             N U M E R I C A
40 REM
50 REM             R E G L A       D E       G A U S S
60 REM
70 REM *****
100 PRINT
109 REM - ENTRADA DE DATOS -
110 PRINT "DEFINA LA FUNCION QUE DESEA INTEGRAR EN LA LINEA 170"
120 PRINT "DEL SIGUIENTE MODO: 170 DEF FN F(X) = ....."
130 PRINT "A CONTINUACION PULSAR <CONT>"
140 STOP
150 PRINT "INTRODUZCA LOS LIMITES DE INTEGRACION : INFERIOR Y SUPERIOR";
160 INPUT A,B
170 DEF FN F(X) =
180 DIM X(12),V(12)
190 I=0
199 REM - VALORES DE LOS COEFICIENTES X(J)   Y   V(J) -
200 X(1) = .1252334085 : V(1) = .2491470458
210 X(2) = .3678314989 : V(2) = .2334925365
220 X(3) = .5873179542 : V(3) = .2031674267
230 X(4) = .7699026741 : V(4) = .1600783285
240 X(5) = .9041172563 : V(5) = .1069393259
250 X(6) = .9815606342 : V(6) = .0471753363
260 X(7) = -X(1)       : V(7) = V(1)
270 X(8) = -X(2)       : V(8) = V(2)
280 X(9) = -X(3)       : V(9) = V(3)
290 X(10) = -X(4)      : V(10)=V(4)
300 X(11) = -X(5)      : V(11)=V(5)
```

```

310 X(12)=-X(6)           :V(12)=V(6)
320 FOR J=1 TO 12
330 Y=(B-A)*X(J)*.5
340 Y=Y+(B+A)*.5
349 REM - SALIDA DEL RESULTADO -
350 I=I+V(J)*FN F(Y)
360 NEXT J
370 I=I*(B-A)*.5
380 PRINT
390 PRINT "LA INTEGRAL PEDIDA ES = "; I;
400 PRINT "(MEDIANTE LA REGLA DE GAUSS)"
410 END

```

## ejercicios resueltos:

1. Calcular el valor de la siguiente integral:

$$\int_0^{1.2} \frac{dx}{1+x}$$

**SOLUCION :**

```

DEFINA LA FUNCION QUE DESEA INTEGRAR EN LA LINEA 170
DEL SIGUIENTE MODO: 170 DEF FN F(X) = .....
A CONTINUACION PULSAR <CONT>
170 DEF FN F(X) =1/(1+X)

```

INTRODUZCA LOS LIMITES DE INTEGRACION : INFERIOR Y SUPERIOR? 0 1.2

LA INTEGRAL PEDIDA ES = .78845736 (MEDIANTE LA REGLA DE GAUSS)

2. Calcular el valor de la siguiente integral:

$$\int_0^3 \frac{dx}{\sqrt{1+x}}$$

**SOLUCION :**

```

DEFINA LA FUNCION QUE DESEA INTEGRAR EN LA LINEA 170
DEL SIGUIENTE MODO: 170 DEF FN F(X) = .....
A CONTINUACION PULSAR <CONT>
170 DEF FN F(X) =1/SQR(1+X)

```

INTRODUZCA LOS LIMITES DE INTEGRACION : INFERIOR Y SUPERIOR? 0 3

LA INTEGRAL PEDIDA ES = 2 (MEDIANTE LA REGLA DE GAUSS)

3. Calcular el valor de la siguiente integral:

$$\int_1^8 \frac{x dx}{\sqrt{x^2 - 15}}$$

**SOLUCION :**

```

DEFINA LA FUNCION QUE DESEA INTEGRAR EN LA LINEA 170
DEL SIGUIENTE MODO: 170 DEF FN F(X) = .....
A CONTINUACION PULSAR <CONT>
170 DEF FN F(X) =X/SQR(X*X-15)

```

INTRODUZCA LOS LIMITES DE INTEGRACION : INFERIOR Y SUPERIOR? 4 8

LA INTEGRAL PEDIDA ES = 5.9998834 (MEDIANTE LA REGLA DE GAUSS)

# INTEGRALES IMPROPIAS CON EXTREMO SUPERIOR INFINITO

## introducción:

Este programa sirve para calcular integrales impropias con extremo superior infinito de funciones de la forma  $y = f(x)$ . El cálculo se realiza mediante la regla de Gauss.

Para la ejecución de este programa, es necesario definir en la línea 170 la función a integrar, del siguiente modo:

```
170 DEF FN F(X) = .....
```

Como el extremo superior es siempre infinito, el programa solo requiere la introducción del extremo inferior.

## listado del programa:

```
10 REM *****
20 REM
30 REM           I N T E G R A L E S       I M P R O P I A S
40 REM
50 REM       C O N       E X T R E M O       S U P E R I O R       I N F I N I T O
60 REM
70 REM *****
100 PRINT
109 REM - ENTRADA DE DATOS -
110 PRINT "DEFINA LA FUNCION QUE DESEA INTEGRAR EN LA LINEA 170"
120 PRINT "DEL SIGUIENTE MODO: 170 DEF FN F(X) = ....."
130 PRINT "A CONTINUACION PULSAR <CONT>"
140 STOP
150 PRINT "INTRODUZCA EL LIMITE INFERIOR";
160 INPUT A
170 DEF FN F(X)=
180 DIM X(12),V(12)
190 I=0
199 REM - VALORES DE LOS COEFICIENTES X(J)   Y   V(J) -
200 X(1) = .1252334085 : V(1) = .2491470458
210 X(2) = .3678314989 : V(2) = .2334925365
220 X(3) = .5873179542 : V(3) = .2031674267
230 X(4) = .7699026741 : V(4) = .1600783285
240 X(5) = .9041172563 : V(5) = .1069393259
250 X(6) = .9815606342 : V(6) = .0471753363
260 X(7) = -X(1)      : V(7) = V(1)
270 X(8) = -X(2)      : V(8) = V(2)
280 X(9) = -X(3)      : V(9) = V(3)
290 X(10) = -X(4)     : V(10) = V(4)
```

```

300 X(11)= -X(5)           : V(11)=V(5)
310 X(12)=-X(6)           : V(12)=V(6)
320 FOR J=1 TO 12
330 Y=A+(2/(X(J)+1))-1
340 YE=FN F(Y)
350 YE=YE/((X(J)+1)^2)
360 I=I+V(J)*YE
370 NEXT J
380 I=2*I
389 REM - SALIDA DEL RESULTADO -
390 PRINT
400 PRINT "LA INTEGRAL IMPROPIA PEDIDA ES = "; I;
410 PRINT "(MEDIANTE LA REGLA DE GAUSS CON EXTREMO SUPERIOR INFINITO)"
420 END

```

## ejercicios resueltos:

**1.** Calcular el valor de la siguiente integral:  $\int_0^{\infty} \frac{dx}{x^2 + 4}$

**SOLUCION :**

```

DEFINA LA FUNCION QUE DESEA INTEGRAR EN LA LINEA 170
DEL SIGUIENTE MODO: 170 DEF FN F(X) = .....
A CONTINUACION PULSAR <CONT>
170 DEF FN F(X)=1/(X*X+4)

```

INTRODUZCA EL LIMITE INFERIOR? 0

LA INTEGRAL IMPROPIA PEDIDA ES = .78539816 (MEDIANTE LA REGLA DE GAUSS CON EXTREMO SUPERIOR INFINITO)

**2.** Calcular el valor de la siguiente integral:  $\int_0^{\infty} e^{-x} \sin x \, dx$

**SOLUCION :**

```

DEFINA LA FUNCION QUE DESEA INTEGRAR EN LA LINEA 170
DEL SIGUIENTE MODO: 170 DEF FN F(X) = .....
A CONTINUACION PULSAR <CONT>
170 DEF FN F(X)=EXP(-X)*SIN(X)

```

INTRODUZCA EL LIMITE INFERIOR? 0

LA INTEGRAL IMPROPIA PEDIDA ES = .49735825 (MEDIANTE LA REGLA DE GAUSS CON EXTREMO SUPERIOR INFINITO)

**3.** Calcular el valor de la siguiente integral:  $\int_1^{\infty} \frac{dx}{x^2}$

**SOLUCION :**

```

DEFINA LA FUNCION QUE DESEA INTEGRAR EN LA LINEA 170
DEL SIGUIENTE MODO: 170 DEF FN F(X) = .....
A CONTINUACION PULSAR <CONT>
170 DEF FN F(X)=1/(X*X)

```

INTRODUZCA EL LIMITE INFERIOR? 1

LA INTEGRAL IMPROPIA PEDIDA ES = 1 (MEDIANTE LA REGLA DE GAUSS CON EXTREMO SUPERIOR INFINITO)

# INTERPOLACION LINEAL

## introducción:

Este programa sirve para, a partir de dos puntos conocidos A, B , construir la recta determinada por ellos, y así poder interpolar para valores intermedios de x la y esperada o reciprocamente.

## listado del programa:

```
10 REM *****
20 REM
30 REM
40 REM             I N T E R P O L A C I O N             L I N E A L
50 REM
60 REM
70 REM *****
100 PRINT
109 REM - ENTRADA DE DATOS -
110 PRINT "INTRODUZCA COORDENADAS XO,YO DEL PUNTO A(XO,YO)";
120 INPUT XO,YO
130 PRINT "INTRODUZCA COORDENADAS X1,Y1 DEL PUNTO B(X1,Y1)";
140 INPUT X1,Y1
150 PRINT
160 PRINT "SI DESEA HALLAR X A PARTIR DE Y CONOCIDO, PULSE 1"
170 PRINT "SI DESEA HALLAR Y A PARTIR DE X CONOCIDO, PULSE 2"
180 INPUT C
190 PRINT
200 IF C=1 THEN 270
210 PRINT "INTERPOLACION PARA X =";
220 INPUT X
230 Y=Y1+(X-X1)*(Y1-YO)/(X1-XO)
240 PRINT "CORRESPONDE          Y =";Y
250 PRINT
260 GOTO 210
270 PRINT
280 PRINT "INTERPOLACION PARA Y =";
290 INPUT Y
300 X=X1+(X1-XO)*(Y-Y1)/(Y1-YO)
310 PRINT "CORRESPONDE          X =";X
320 PRINT
330 GOTO 280
340 END
```

## ejercicios resueltos:

1. Dados los puntos A(2.35, 7.98) y B(12.39, -3.56), calcular los valores de y para los siguientes valores de x: 5.6, 3.9, 12.

### SOLUCION :

INTRODUZCA COORDENADAS X0,Y0 DEL PUNTO A(X0,Y0)? 2.35 7.98  
INTRODUZCA COORDENADAS X1,Y1 DEL PUNTO B(X1,Y1)? 12.39 -3.56

SI DESEA HALLAR X A PARTIR DE Y CONOCIDO, PULSE 1  
SI DESEA HALLAR Y A PARTIR DE X CONOCIDO, PULSE 2  
? 2

INTERPOLACION PARA X =? 5.6  
CORRESPONDE Y = 4.2444422

INTERPOLACION PARA X =? 3.9  
CORRESPONDE Y = 6.1984263

INTERPOLACION PARA X =? 12  
CORRESPONDE Y = -3.1117331

2. Dados los puntos A(1, 43.75) y B(5, 117.98), calcular los valores de x para los siguientes valores de y: 10, 6, 15.

### SOLUCION :

INTRODUZCA COORDENADAS X0,Y0 DEL PUNTO A(X0,Y0)? 1 43.75  
INTRODUZCA COORDENADAS X1,Y1 DEL PUNTO B(X1,Y1)? 5 117.78

SI DESEA HALLAR X A PARTIR DE Y CONOCIDO, PULSE 1  
SI DESEA HALLAR Y A PARTIR DE X CONOCIDO, PULSE 2  
? 1

INTERPOLACION PARA Y =? 10  
CORRESPONDE X = -.82358503

INTERPOLACION PARA Y =? 6  
CORRESPONDE X = -1.0397136

INTERPOLACION PARA Y =? 15  
CORRESPONDE X = -.55342429



# INTERPOLACION DE LAGRANGE

## introducción:

Este programa sirve para ajustar un polinomio a un conjunto de puntos. El método del ajuste se conoce con el nombre de interpolación de Lagrange.

Gracias a este ajuste es posible calcular las ordenadas de puntos cuyas abscisas son conocidos.

## listado del programa:

```
10 REM *****
20 REM
30 REM
40 REM          I N T E R P O L A C I O N          D E          L A G R A N G E
50 REM
60 REM
70 REM *****
100 PRINT
109 REM - ENTRADA DE DATOS -
110 PRINT "NUMERO DE PUNTOS CONOCIDOS";
120 INPUT N
130 DIM X(N),Y(N)
140 PRINT "INTRODUZCA LAS COORDENADAS DE LOS";N;"PUNTOS CONOCIDOS"
150 PRINT
160 FOR J=1 TO N
170 PRINT "COORDENADAS X,Y DEL PUNTO";J ;
180 INPUT X(J),Y(J)
190 NEXT J
200 PRINT
210 PRINT "INTERPOLACION PARA X =";
220 INPUT X0
230 Y=0
240 FOR I=1 TO N
250 S=1
260 FOR J=1 TO N
270 IF J=I THEN 290
280 S=S*(X0-X(J))/(X(I)-X(J))
290 NEXT J
300 Y=Y+S*Y(I)
310 NEXT I
```

```

320 PRINT "CORRESPONDE      Y =" ; Y
330 PRINT
340 GOTO 210
350 END

```

## ejercicios resueltos:

**1.** Dados los siguientes puntos: (1, 4.5), (2, 5), (3, 5.7), (4, 6), (5, 6.3), (6, 6.8), (7, 7.3), (8, 7.9), hallar los valores correspondientes para  $x = 1.7$ ,  $x = 3.4$ ,  $x = 7.2$ ,  $x = 4.8$ .

### SOLUCION :

NUMERO DE PUNTOS CONOCIDOS? 8  
INTRODUZCA LAS COORDENADAS DE LOS 8 PUNTOS CONOCIDOS

COORDENADAS X,Y DEL PUNTO 1 ? 1	4.5
COORDENADAS X,Y DEL PUNTO 2 ? 2	5
COORDENADAS X,Y DEL PUNTO 3 ? 3	5.7
COORDENADAS X,Y DEL PUNTO 4 ? 4	6
COORDENADAS X,Y DEL PUNTO 5 ? 5	6.3
COORDENADAS X,Y DEL PUNTO 6 ? 6	6.8
COORDENADAS X,Y DEL PUNTO 7 ? 7	7.3
COORDENADAS X,Y DEL PUNTO 8 ? 8	7.9

INTERPOLACION PARA X =? 1.7  
CORRESPONDE Y = 4.741505

INTERPOLACION PARA X =? 3.4  
CORRESPONDE Y = 5.8501988

INTERPOLACION PARA X =? 7.2  
CORRESPONDE Y = 7.3891246

INTERPOLACION PARA X =? 4.8  
CORRESPONDE Y = 6.2231109

**2.** Dados los puntos: (0, 4.57), (3, 8.23), (6, 12.06), (8, 15), hallar los valores de y correspondientes a los siguientes valores de x: 4, 5, 7, 1.

### SOLUCION :

NUMERO DE PUNTOS CONOCIDOS? 4  
INTRODUZCA LAS COORDENADAS DE LOS 4 PUNTOS CONOCIDOS

COORDENADAS X,Y DEL PUNTO 1 ? 0	4.57,
COORDENADAS X,Y DEL PUNTO 2 ? 3	8.23
COORDENADAS X,Y DEL PUNTO 3 ? 6	12.06
COORDENADAS X,Y DEL PUNTO 4 ? 8	15

INTERPOLACION PARA X =? 4  
CORRESPONDE Y = 9.4585556

INTERPOLACION PARA X =? 5  
CORRESPONDE Y = 10.727917

INTERPOLACION PARA X =? 7  
CORRESPONDE Y = 13.476722

INTERPOLACION PARA X =? 1  
CORRESPONDE Y = 5.8076389

**3.** Dados los puntos:  $(-6, 7.25)$ ,  $(-4, 8)$ ,  $(-2, 7.75)$ ,  $(0, 8.25)$ ,  $(2, 8.3)$ ,  $(4, 9)$ ,  $(5, 9.2)$ ,  $(7, 10)$ ,  $(9, 10.5)$ , hallar los valores de y correspondientes a los siguientes valores de x: 3, 8.5, -5, 1.

**S O L U C I O N :**

INTERPOLACION PARA X =NUMERO DE PUNTOS CONOCIDOS? 9  
INTRODUZCA LAS COORDENADAS DE LOS 9 PUNTOS CONOCIDOS

NUMERO DE PUNTOS CONOCIDOS? 9  
INTRODUZCA LAS COORDENADAS DE LOS 9 PUNTOS CONOCIDOS

COORDENADAS X,Y DEL PUNTO 1 ?-6	7.25
COORDENADAS X,Y DEL PUNTO 2 ?-4	8
COORDENADAS X,Y DEL PUNTO 3 ?-2	7.75
COORDENADAS X,Y DEL PUNTO 4 ? 0	8.25
COORDENADAS X,Y DEL PUNTO 5 ? 2	8.3
COORDENADAS X,Y DEL PUNTO 6 ? 4	9
COORDENADAS X,Y DEL PUNTO 7 ? 5	9.2
COORDENADAS X,Y DEL PUNTO 8 ? 7	10
COORDENADAS X,Y DEL PUNTO 9 ? 9	10.5

INTERPOLACION PARA X =? 3  
CORRESPONDE Y = 8.6272072

INTERPOLACION PARA X =? 8.5  
CORRESPONDE Y = 11.593759

INTERPOLACION PARA X =?-5  
CORRESPONDE Y = 9.3737544

INTERPOLACION PARA X =? 1  
CORRESPONDE Y = 8.2041084



**PROGRAMAS  
DE  
  
CALCULO  
DE  
PROBABILIDADES  
Y  
ESTADISTICA**

# NUMEROS ALEATORIOS ENTRE 0 Y 1

## introducción:

Este breve programa sirve para generar numeros aleatorios entre 0 y 1, para ello se utiliza la función de librería que poseen los microordenadores RND.

En algunos micros para obtener números aleatorios entre 0 y 1 se ha de escribir en la linea 160 `X = RND(1)`.

La sentencia de la linea 110 `RANDOMIZE` sirve para restablecer la generación de números aleatorios, ya que si no se hiciera al ejecutar la función `RND` se produce la misma serie de números. No todos los micros precisan de la sentencia `RANDOMIZE`.

## listado del programa:

```
10 REM *****
20 REM
30 REM          N U M E R O S      A L E A T O R I O S
40 REM
50 REM          E N T R E      O      Y      1
60 REM
70 REM *****
100 PRINT
110 RANDOMIZE
120 PRINT "CUANTOS NUMEROS ALEATORIOS QUIERE GENERAR " ;
130 INPUT N
140 PRINT
150 FOR I=1 TO N
160 X=RND
170 PRINT X,
180 NEXT I
190 END
```

## ejercicios resueltos:

1. Generar 100 números aleatorios comprendidos entre 0 y 1.

### SOLUCION :

CUANTOS NUMEROS ALEATORIOS QUIERE GENERAR ? 100

.66111636	.99669448	.74986924	.73111219	.65541254
.30280804	.34766493	.40634861	.25326	.28743388
.8733116	.47250194	.2871809	.55212093	.82923229
.46287042	.92966703	.15479702	.18181261	.43484974
.19922229	.45345791	.31910335	.5641103	.75644424
.44692652	.9782497	.25928223	.91201971	.47205755
.689522	.78152328	.018509465	.36676237	.53382305
.6033445	.66779691	.45822923	.26054705	.99106874
.36607594	.98571379	.062664971	.54529696	.60825273
.81727499	.55591164	.68375735	.5236367	.56983113
.75894927	.30091226	.51479547	.49191142	.12138261
.088697927	.56144892	.71847771	.89650064	.25277209
.11127729	.75719156	.40561985	.60864429	.72951833
.31550171	.99090403	.29206412	.92922857	.82792673
.7035357	.56804217	.85408357	.29290808	.02411402
.47287886	.73559229	.27860843	.16215432	.27357606
.29660054	.20703035	.73156418	.97900652	.52295841
.27583021	.84897105	.3872161	.47522683	.62720254
.95177646	.40323387	.062298208	.62320955	.8828118
.61176495	.94880605	.33693107	.03847059	.33914117

2. Generar 40 números aleatorios comprendidos entre 0 y 1.

### SOLUCION :

CUANTOS NUMEROS ALEATORIOS QUIERE GENERAR ? 40

.22396731	.68521682	.43491258	.036455817	.48510919
.29040264	.16478835	.61818638	.15839611	.9295695
.16235884	.1025526	.88885022	.26676632	.24254711
.58535272	.6614633	.42242092	.68423876	.68648243
.30723926	.81702908	.18798458	.14411213	.29927566
.51191371	.80342976	.92937195	.76075934	.9064295
.37334463	.86591964	.85675104	.90098784	.29113201
.98876625	.99163419	.88527696	.61868608	.12673463

3. Generar 25 números aleatorios comprendidos entre 0 y 1.

### SOLUCION :

CUANTOS NUMEROS ALEATORIOS QUIERE GENERAR ? 25

.85267519	.54384414	.15895738	.65106458	.54179559
.61314129	.89298354	.98401585	.18602978	.99376323
.093923174	.27819652	.4928999	.66225887	.13919933
.44190484	.50989774	.051556056	.87019762	.90753651

# NUMEROS ALEATORIOS ENTRE 0 Y 100

## introducción:

Este breve programa sirve para generar números aleatorios entre 0 y 100 (solo parte entera), para ello escribiremos en la línea 160, los siguientes:

$$X = \text{INT} (\text{RND} * 100 + .5)$$

Todo lo dicho en el programa anterior sobre RND( ) y RANDOMIZE es igualmente válido en este y en los siguientes programas.

## listado del programa:

```
10 REM *****
20 REM
30 REM          N U M E R O S   A L E A T O R I O S
40 REM
50 REM          P A R T E   E N T E R A   E N T R E   0   Y   100
60 REM
70 REM *****
100 PRINT
110 RANDOMIZE
120 PRINT
130 PRINT "CUANTOS NUMEROS ALEATORIOS QUIERE GENERAR";
140 INPUT N
150 FOR I=1 TO N
160 X=INT(RND*100+.5)
170 PRINT X,
180 NEXT I
190 END
```

## ejercicios resueltos:

1. Generar 100 números aleatorios (parte entera) entre 0 y 100.

S O L U C I O N :



CUANTOS NUMEROS ALEATORIOS QUIERE CALCULAR ? 100

88	60	26	83	56	22	66	89	78	52
21	69	77	100	74	54	36	38	67	74
41	18	40	44	58	70	64	87	28	44
33	20	96	93	85	6	26	41	51	30
66	94	82	76	44	82	89	90	27	37
85	64	82	52	30	93	99	74	14	59
71	23	25	61	55	42	92	6	50	19
94	8	56	40	74	73	15	56	20	59
46	85	68	49	15	66	62	76	8	52
75	69	7	100	64	33	70	93	28	73

**2.** Generar 30 números aleatorios (parte entera) entre 0 y 100.

S O L U C I O N :

CUANTOS NUMEROS ALEATORIOS QUIERE CALCULAR ? 30

91	81	8	38	83	32	74	17	21	55
21	48	76	49	82	11	81	36	48	88
73	18	10	43	69	76	88	49	57	8

**3.** Generar 50 números aleatorios (parte entera) entre 0 y 100.

S O L U C I O N :

CUANTOS NUMEROS ALEATORIOS QUIERE CALCULAR ? 50

25	12	94	48	3	91	60	29	45	95
80	96	84	33	52	42	16	43	28	60
41	38	60	27	37	10	34	30	57	73
57	46	58	40	75	1	18	100	36	45
96	20	1	21	32	93	96	40	8	13

# NUMEROS ALEATORIOS QUE SE DISTRIBUYEN SEGUN UNA DISTRIBUCION BINOMIAL B (N, P)

## introducción:

Este programa sirve para generar números aleatorios que siguen una distribución binomial de parámetros conocidos n y p.

La ley de probabilidad de la distribución binomial viene dada por la siguiente expresión:

$$p(X = r) = \binom{n}{r} p^r q^{n-r} \quad \text{siendo } q = 1-p$$

Obsérvese que al ser la media de la distribución binomial igual a np, los números aleatorios generados se concentran entorno a np en cada caso.

## listado del programa:

```
10 REM *****
20 REM
30 REM          GENERACION DE NUMEROS ALEATORIOS
40 REM
50 REM          QUE SE DISTRIBUYEN SEGUN UNA BINOMIAL B(N,P)
60 REM
70 REM *****
100 PRINT
110 PRINT "CUANTOS NUMEROS QUIERE GENERAR M =";
120 INPUT M
130 PRINT "NUMERO DE PRUEBAS REALIZADAS N =";
140 INPUT N
150 PRINT "PROBABILIDAD DEL SUCESO EXITO P =";
160 INPUT P
170 PRINT
180 FOR J=1 TO M
190 Z=0
200 FOR I=1 TO N
210 IF RND < P THEN Z=Z+1
220 NEXT I
230 PRINT Z,
240 NEXT J
250 END
```

## ejercicios resueltos:

**1.** Generar 100 números aleatorios que se distribuyan según una binomial de parámetros  $B(15, 0.4)$ .

### SOLUCION :

CUANTOS NUMEROS QUIERE GENERAR M =? 100  
 NUMERO DE PRUEBAS REALIZADAS N =? 15  
 PROBABILIDAD DEL SUCESO EXITO P =? .4

6	8	4	5	5	7	4	9	7	6
5	8	7	6	9	4	7	5	8	8
7	8	4	4	1	6	7	1	6	8
7	4	5	5	5	7	8	3	4	4
7	7	5	8	5	8	5	4	7	7
8	7	6	5	5	8	8	5	6	3
10	4	6	6	8	5	6	10	6	6
6	6	8	7	7	6	4	7	10	4
8	5	6	7	5	6	4	7	4	5
5	4	3	5	5	6	5	4	6	6

**2.** Generar 50 números aleatorios que se distribuyan según una binomial de parámetros  $B(12, 0.3)$ .

### SOLUCION :

CUANTOS NUMEROS QUIERE GENERAR M =? 50  
 NUMERO DE PRUEBAS REALIZADAS N =? 12  
 PROBABILIDAD DEL SUCESO EXITO P =? .3

4	6	3	2	4	4	6	2	4	6
5	4	4	5	5	4	2	3	6	2
5	3	4	2	6	4	4	4	0	3
1	2	5	1	1	4	6	5	4	2
3	3	5	2	2	4	3	0	2	1

**3.** Generar 30 números aleatorios que se distribuyan según una binomial de parámetros  $B(8, 0.65)$ .

### SOLUCION :

CUANTOS NUMEROS QUIERE GENERAR M =? 30  
 NUMERO DE PRUEBAS REALIZADAS N =? 8  
 PROBABILIDAD DEL SUCESO EXITO P =? .65

6	6	6	7	4	5	5	4	5	5
6	3	5	6	5	5	6	6	4	5
5	5	6	8	7	5	6	8	3	5

# NUMEROS ALEATORIOS QUE SE DISTRIBUYEN SEGUN UNA DISTRIBUCION DE POISSON P (L)

## introducción:

Este programa sirve para generar números aleatorios que se distribuyen según una ley de Poisson de parámetro conocido L.

La ley de probabilidad de esta distribución viene dada por la siguiente expresión:

$$p(X = r) = \frac{L^r}{r!} e^{-L}$$

El parámetro L es la media de la distribución.

## listado del programa:

```
10 REM *****
20 REM
30 REM GENERACION DE NUMEROS ALEATORIOS
40 REM
50 REM QUE SE DISTRIBUYEN SEGUN UNA DISTRIBUCION DE POISSON P(L)
60 REM
70 REM *****
100 PRINT
110 PRINT "CUANTOS NUMEROS QUIERE GENERAR ";
120 INPUT N
130 PRINT "INTRODUZCA EL PARAMETRO L DE LA DISTRIBUCION DE POISSON";
140 INPUT L
150 PRINT
160 FOR J=1 TO N
170 X=RND*EXP(L)
180 A1=1
190 B1=1
200 B=0
210 IF A1>X THEN GOTO 260
220 B=B+1
230 B1=B1*L/B
240 A1=A1+B1
250 GOTO 210
260 IF B>0 THEN B=B-(A1-X)/B1
270 Z=B
280 PRINT Z,
290 NEXT J
300 END
```

## ejercicios resueltos:

1. Generar 100 números aleatorios que se distribuyan según una ley de Poisson de parámetro 4.5.

### SOLUCION :

CUANTOS NUMEROS QUIERE GENERAR ? 100

INTRODUZCA EL PARAMETRO L DE LA DISTRIBUCION DE POISSON? 4.5

4.4790046	7.8768798	1.1382678	2.2340479	.441786
4.3899337	4.1750698	8.7034091	1.62817	3.9125009
2.9691246	3.5307637	6.6998425	2.4097152	4.9753847
1.0481329	1.9980574	1.0643043	7.3704396	4.6607317
3.389521	4.992263	1.4101242	1.5416204	3.7250827
2.0376018	3.8150252	5.2728204	2.9852433	2.7965388
2.7354873	3.5958806	3.4807801	2.0143558	4.9225839
6.0145394	5.6717572	4.1284886	3.5207337	5.4562648
2.6883133	2.2100816	4.0966309	5.3790696	4.910408
3.5224978	4.684139	4.8034561	2.2877596	3.2461986
2.0717965	1.4880771	7.8964168	6.7554964	4.190031
6.3337475	2.0394883	8.7444758	5.6254214	9.6360966
3.9325078	2.3260118	2.5575084	5.2563448	4.1579961
4.6356179	6.8796726	1.2419169	1.7395451	4.9809084
3.7837442	4.8407635	4.87616	5.5472892	1.4661649
4.1187839	1.0565397	1.2710927	1.7913829	5.9421862
1.2954799	6.5097622	7.1190892	2.6292761	3.5752319
3.2212459	2.0637678	3.6230742	4.825986	9.5821035
.033714553	5.9346837	4.6771683	3.710739	5.0690967
7.3808591	4.076205	2.3014229	6.0611323	6.1918552

2. Generar 50 números aleatorios que se distribuyan según una ley de Poisson de parámetro 5.7.

### SOLUCION :

NUMEROS QUIERE GENERAR ? 50

INTRODUZCA EL PARAMETRO L DE LA DISTRIBUCION DE POISSON? 5.7

5.7462943	9.5648654	1.9977642	3.2243582	1.1982538
5.6508272	5.4205337	10.432007	2.5323847	5.1286894
4.058842	4.6903744	8.2061556	3.4257503	6.3417963
1.8112457	2.9352359	1.8447096	8.9213368	5.9410714
4.5305786	6.3640124	2.294907	2.438122	4.9102183
2.9987808	5.0126054	6.643523	4.0750517	3.8692205
3.7992287	4.7640447	4.6338251	2.9608044	6.2722967
7.5201647	7.0524224	5.3706075	4.6790269	6.8246182
3.7451464	3.1968822	5.3364621	6.7484116	6.25627
4.6810226	5.9661595	6.1154937	3.2859355	4.36843

# NUMEROS ALEATORIOS QUE SE DISTRIBUYEN SEGUN UNA DISTRIBUCION NORMAL N (M, S)

## introducción:

Este programa sirve para generar números aleatorios que siguen una distribución normal de parámetros conocidos M y S.

El cálculo de estos números está basado en el teorema central del límite.

La función de densidad de una distribución normal N(M,S) viene dada por la expresión:

$$f(x) = \frac{1}{S \sqrt{2\pi}} \exp\left(-\frac{1}{2} \left(\frac{x-M}{S}\right)^2\right)$$

Obsérvese que para valores de N suficientemente grandes se verifica que el 68,2% de los números obtenidos se encuentran en el intervalo: (M - S, M + S).

## listado del programa:

```
10 REM *****
20 REM
30 REM          GENERACION DE NUMEROS ALEATORIOS
40 REM
50 REM          QUE SE DISTRIBUYEN SEGUN UNA NORMAL N(M,S)
60 REM
70 REM *****
100 PRINT
110 PRINT "CUANTOS NUMEROS QUIERE GENERAR";
120 INPUT N
130 PRINT "INTRODUZCA LA MEDIA Y LA DESVIACION TIPICA"
140 PRINT "MEDIA  M =";
150 INPUT M
160 PRINT "DESVIACION TIPICA  S =";
170 INPUT S
180 PRINT
190 FOR J=1 TO N
200 Z=0
210 FOR I=1 TO 48
220 Z=Z+RND-.5
230 NEXT I
240 Z=M+S*Z/2
250 PRINT Z,
260 NEXT J
270 END
```

## ejercicios resueltos:

1. Generar 100 números aleatorios que se distribuyan normalmente siendo la media  $M = 7$ , y la desviación típica  $S = 0,5$ .

### S O L U C I O N :

CUANTOS NUMEROS QUIERE GENERAR? 100  
INTRODUZCA LA MEDIA Y LA DESVIACION TIPICA  
MEDIA M=? 7  
DESVIACION TIPICA S=? .5

6.6837221	7.3083412	6.5928658	6.5808139	6.9133598
6.579334	6.1652235	8.1051712	7.1809766	7.522095
7.1056385	7.5063749	7.6467287	6.5467805	6.8242669
6.9445812	7.2207726	6.8135466	6.7312651	7.0799463
7.0632642	6.4825497	6.7367894	6.5726264	6.5843601
6.7139459	7.0009958	7.3327778	7.6942162	7.1678916
7.1840408	7.7705568	6.8029891	7.004543	6.9460805
6.7961197	7.8208347	6.8840563	5.9472712	7.3196226
6.4079096	7.4665878	6.8477692	6.0012218	7.7243698
6.9122938	6.0577307	7.5010735	6.9303716	7.6313304
6.2373119	6.4793341	7.1860713	6.533854	7.0466691
7.5961597	7.4016251	7.0300208	7.1459586	7.5117067
6.7371893	6.279987	7.1953366	5.8861312	7.1029201
7.4439089	7.3549593	7.3795895	6.6589738	8.1819427
7.0349831	7.9022379	6.0655066	6.9042447	6.395564
6.6142326	6.4826748	8.2709712	7.0968585	6.6757299
7.0706347	7.6922783	7.0490228	7.246886	6.7395424
7.5783224	7.662213	6.7378571	6.8995542	7.0892597
7.5965855	6.8087996	7.4608264	7.1352464	6.2622965
6.6198696	7.0835152	6.8764388	7.8195022	6.5812235

2. Generar 50 números aleatorios que se distribuyan según una  $N(0,1)$

### S O L U C I O N :

CUANTOS NUMEROS QUIERE GENERAR? 50  
INTRODUZCA LA MEDIA Y LA DESVIACION TIPICA  
MEDIA M=? 0  
DESVIACION TIPICA S=? 1

-.63255586	.61668242	-.81426843	-.83837228	-.1732805
-.84133195	-1.669553	2.2103425	.36195312	1.0441901
.21127693	1.0127499	1.2934575	-.90643906	-.35146614
-.11083763	.44154511	-.3729068	-.53746971	.15989251
.12652848	-1.0349007	-.52642127	-.85474723	-.8312799
-.57210814	.0019916678	.66555566	1.3884325	.33578325
.36808161	1.5411137	-.39402189	.0090860212	-.10783895
-.40776067	1.6416695	-.23188733	-2.1054575	.6392451
-1.1841809	.93317569	-.30446156	-1.9975565	1.4487395
-.17541241	-1.8845386	1.002147	-.13925686	1.2626609

# LANZAMIENTO DE UN DADO

## introducción:

Este programa sirve para simular el experimento aleatorio consistente en el lanzamiento N veces de un dado anotando el resultado obtenido en la cara superior. Posteriormente realiza un recuento para expresar las frecuencias absolutas y relativas de cada uno de los resultados posibles.

Si el número de lanzamientos realizados fuese suficientemente grande la frecuencia relativa de cada suceso tendería a  $1/6$  que es la probabilidad de cada uno de los sucesos.

## listado del programa:

```
10 REM *****
20 REM
30 REM          SIMULACION DEL LANZAMIENTO DE UN DADO
40 REM
50 REM          CON RECuento DE FRECUENCIAS ABSOLUTAS Y RELATIVAS
60 REM
70 REM *****
100 PRINT
110 PRINT "CUANTOS LANZAMIENTOS DESEA REALIZAR N =";
120 INPUT N
130 PRINT
140 RANDOMIZE
150 FOR I=1 TO N
160 A=INT(RND*6+1)
170 PRINT A,
180 ON A GOTO 190,200,210,220,230,240
190 F(1)=F(1)+1:GOTO 250
200 F(2)=F(2)+1:GOTO 250
210 F(3)=F(3)+1:GOTO 250
220 F(4)=F(4)+1:GOTO 250
230 F(5)=F(5)+1:GOTO 250
240 F(6)=F(6)+1:GOTO 250
250 NEXT I
260 PRINT
270 PRINT "SUCESO","FREC.ABSOLUT","FREC.RELAT"
280 PRINT "*****","*****","*****"
290 FOR J=1 TO 6
300 PRINT "CARA";J, F(J),F(J)/N
310 NEXT J
320 END
```



## ejercicios resueltos:

1. Se lanza 200 veces un dado equilibrado, calcular las frecuencias absolutas y relativas de cada uno de los seis resultados posibles.

### S O L U C I O N :

CUANTOS LANZAMIENTOS DESEA REALIZAR N =? 200

1	3	1	2	2	3	4	5	2	6
1	3	3	1	5	5	5	2	1	5
5	4	6	6	6	4	4	2	1	4
6	6	2	4	4	2	1	5	1	4
1	6	4	2	6	1	1	3	6	6
6	2	6	5	2	2	1	2	6	4
3	2	5	6	2	3	4	5	1	3
6	5	4	1	5	5	6	6	2	1
4	4	5	1	2	4	2	1	4	5
6	5	6	5	6	2	4	6	6	3
2	3	5	1	4	6	4	3	5	5
6	4	5	3	1	5	2	6	3	4
3	5	2	1	6	3	1	4	1	6
3	6	3	6	5	4	3	2	5	5
3	1	6	4	2	3	4	3	5	1
6	3	5	6	5	6	1	2	1	6
2	1	5	3	2	4	2	3	1	5
6	4	6	3	4	5	3	4	2	4
3	1	3	1	5	3	6	2	3	3
4	3	1	5	5	4	1	1	1	3

SUCESO	FREC.ABSOLUT	FREC.RELAT
*****	*****	*****
CARA 1	34	.17
CARA 2	29	.145
CARA 3	33	.165
CARA 4	32	.16
CARA 5	35	.175
CARA 6	37	.185

2. Se lanza 100 veces un dado equilibrado, calcular las frecuencias absolutas y relativas de cada uno de los seis resultados posibles.

### S O L U C I O N :

CUANTOS LANZAMIENTOS DESEA REALIZAR N =? 100

1	4	6	1	1	6	5	3	3	1
1	2	2	6	3	5	2	5	3	1
1	5	4	2	6	1	1	4	5	6
6	6	5	2	1	1	3	1	3	2
3	4	1	2	2	3	3	3	3	6
4	5	1	2	1	1	6	4	5	3
2	3	1	4	2	6	4	1	6	5
2	3	2	2	6	5	5	2	2	5
6	6	1	2	2	4	4	6	3	5
6	1	1	6	1	4	6	6	4	4

SUCESO	FREC.ABSOLUT	FREC.RELAT
*****	*****	*****
CARA 1	22	.22
CARA 2	18	.18
CARA 3	15	.15
CARA 4	13	.13
CARA 5	13	.13
CARA 6	19	.19

# LANZAMIENTO DE DOS DADOS (SUMA DE PUNTOS)

## introducción:

Este programa sirve para simular el experimento aleatorio consistente en el lanzamiento N veces de dos dados, anotando los resultados obtenidos en las dos caras superiores, este resultado se expresa como un par ordenado de manera que el primer elemento del par es el resultado del primer dado y el segundo elemento del par es el resultado del segundo dado.

El programa además efectúa el recuento para expresar las frecuencias absolutas y relativas de cada uno de los valores de la suma de los puntos obtenidos en las caras de los dados.

Obsérvese que si el número de lanzamientos es muy grande la mayor frecuencia relativa corresponderá a la suma 7 que se puede obtener de las siguientes formas: (1,6), (2,5), (3,4), (4,3), (5,2), (6,1), en cambio las menores frecuencias relativas corresponderán a la suma 2 y a la suma 12 que solo se pueden obtener de una forma (1,1), (6,6).

## listado del programa:

```
10 REM *****
20 REM
30 REM          SIMULACION DEL LANZAMIENTO DE DOS DADOS
40 REM
50 REM          CON RECuento DE FRECUENCIAS ABSOLUTAS Y RELATIVAS
60 REM
70 REM          DE LA SUMA DE LOS PUNTOS DE LAS CARAS
80 REM
90 REM *****
100 PRINT
105 DIM F(12)
110 RANDOMIZE
120 PRINT "CUANTOS LANZAMIENTOS DESEA REALIZAR N = ";
130 INPUT N
140 PRINT
150 FOR I=1 TO N
160 A=INT(RND*6+1)
170 B=INT(RND*6+1)
```

```

180 PRINT " (;A;",";B;")",
190 S=A+B
200 FOR J=2 TO 12
210 IF S=J THEN GOTO 230
220 NEXT J
230 F(J)=F(J)+1
240 GOTO 250
250 NEXT I
260 PRINT
270 PRINT "SUMA","FREC.ABSOLUT","FREC.RELAT"
280 PRINT "****","*****","*****"
290 FOR J=2 TO 12
300 PRINT J,F(J),F(J)/N
310 NEXT J
320 END

```

## ejercicios resueltos:

1. Se lanzan 100 veces dos dados y se suman los valores obtenidos en cada una de sus caras. Calcular las frecuencias absolutas y relativas de cada uno de los resultados posibles.

### SOLUCION :

CUANTOS LANZAMIENTOS DESEA REALIZAR N =? 100

( 6 , 6 )	( 6 , 5 )	( 6 , 4 )	( 4 , 4 )	( 4 , 2 )
( 2 , 2 )	( 4 , 6 )	( 1 , 2 )	( 5 , 6 )	( 6 , 5 )
( 5 , 5 )	( 5 , 2 )	( 1 , 3 )	( 5 , 6 )	( 2 , 3 )
( 6 , 4 )	( 6 , 1 )	( 2 , 4 )	( 2 , 3 )	( 1 , 6 )
( 4 , 3 )	( 5 , 5 )	( 5 , 5 )	( 3 , 6 )	( 4 , 4 )
( 1 , 4 )	( 5 , 4 )	( 3 , 1 )	( 2 , 3 )	( 2 , 6 )
( 3 , 2 )	( 6 , 6 )	( 6 , 4 )	( 6 , 2 )	( 1 , 5 )
( 3 , 3 )	( 3 , 1 )	( 1 , 3 )	( 3 , 1 )	( 5 , 1 )
( 4 , 2 )	( 3 , 3 )	( 5 , 5 )	( 5 , 1 )	( 1 , 2 )
( 2 , 4 )	( 2 , 3 )	( 1 , 6 )	( 5 , 3 )	( 1 , 5 )
( 3 , 2 )	( 1 , 1 )	( 6 , 5 )	( 4 , 4 )	( 1 , 2 )
( 4 , 1 )	( 1 , 2 )	( 1 , 3 )	( 1 , 5 )	( 1 , 1 )
( 1 , 3 )	( 3 , 1 )	( 1 , 5 )	( 5 , 1 )	( 1 , 2 )
( 3 , 4 )	( 5 , 2 )	( 2 , 6 )	( 2 , 3 )	( 4 , 4 )
( 4 , 5 )	( 2 , 6 )	( 5 , 4 )	( 1 , 3 )	( 4 , 1 )
( 4 , 5 )	( 1 , 4 )	( 4 , 1 )	( 4 , 4 )	( 5 , 3 )
( 4 , 6 )	( 5 , 1 )	( 2 , 4 )	( 4 , 1 )	( 4 , 4 )
( 5 , 4 )	( 4 , 5 )	( 3 , 6 )	( 6 , 1 )	( 3 , 1 )
( 1 , 5 )	( 2 , 5 )	( 3 , 1 )	( 6 , 2 )	( 5 , 2 )
( 1 , 1 )	( 6 , 5 )	( 3 , 6 )	( 5 , 5 )	( 3 , 2 )

SUMA	FREC.ABSOLUT	FREC.RELAT
****	*****	*****
2	3	.03
3	5	.05
4	12	.12
5	14	.14
6	16	.16
7	10	.1
8	13	.13
9	9	.09
10	10	.1
11	6	.06
12	2	.02

# LANZAMIENTO DE TRES DADOS (SUMA DE PUNTOS)

## introducción:

Este programa sirve para simular el experimento aleatorio consistente en el lanzamiento de N veces 3 dados, el resultado se expresa en forma de terna ordenada. Además se efectúa el cálculo de las frecuencias absolutas y relativas del suceso suma de los puntos obtenidos en las caras de los tres dados, esta suma oscila de 3 (1,1,1) a 18 (6,6,6).

## listado del programa:

```
10 REM *****
20 REM
30 REM             SIMULACION DEL LANZAMIENTO DE TRES DADOS
40 REM
50 REM             CON RECUESTO DE FRECUENCIAS ABSOLUTAS Y RELATIVAS
60 REM
70 REM             DE LA SUMA DE LOS PUNTOS DE LAS CARAS
80 REM
90 REM *****
100 PRINT
105 DIM F(18)
110 RANDOMIZE
120 PRINT "CUANTOS LANZAMIENTOS DESEA REALIZAR N =";
130 INPUT N
140 PRINT
150 FOR I=1 TO N
160 A=INT(RND*6+1)
170 B=INT(RND*6+1)
175 C=INT(RND*6+1)
180 PRINT "(";A;",";B;",";C;")",
190 S=A+B+C
200 FOR J=3 TO 18
210 IF S=J THEN GOTO 230
220 NEXT J
230 F(J)=F(J)+1
240 GOTO 250
250 NEXT I
260 PRINT
270 PRINT "SUMA","FREC.ABSOLUT","FREC.RELAT"
280 PRINT "****","*****","*****"
```

```

290 FOR J=3 TO 18
300 PRINT J,F(J),F(J)/N
310 NEXT J
320 END

```

## ejercicios resueltos:

1. Se lanzan 60 veces tres dados equilibrados. Se pide: expresar el resultado obtenido, y calcular las frecuencias absolutas y relativas de cada uno de los sucesos consistentes en la suma de los puntos obtenidos en las caras de los tres dados.

### SOLUCION :

CUANTOS LANZAMIENTOS DESEA REALIZAR N =? 60

```

( 4 , 1 , 1 ) ( 5 , 3 , 6 ) ( 6 , 1 , 1 ) ( 1 , 1 , 2 ) ( 6 , 6 , 3 )
( 5 , 6 , 3 ) ( 4 , 3 , 4 ) ( 1 , 6 , 5 ) ( 2 , 2 , 6 ) ( 5 , 4 , 2 )
( 4 , 5 , 5 ) ( 2 , 2 , 3 ) ( 5 , 4 , 3 ) ( 6 , 4 , 4 ) ( 1 , 6 , 5 )
( 4 , 3 , 6 ) ( 5 , 5 , 4 ) ( 6 , 1 , 2 ) ( 1 , 4 , 6 ) ( 5 , 3 , 5 )
( 3 , 2 , 2 ) ( 2 , 6 , 5 ) ( 1 , 2 , 1 ) ( 4 , 2 , 6 ) ( 1 , 3 , 1 )
( 2 , 5 , 1 ) ( 1 , 1 , 4 ) ( 6 , 4 , 2 ) ( 3 , 1 , 5 ) ( 4 , 3 , 5 )
( 5 , 2 , 5 ) ( 3 , 1 , 2 ) ( 2 , 6 , 4 ) ( 6 , 4 , 1 ) ( 6 , 3 , 1 )
( 4 , 1 , 6 ) ( 3 , 1 , 6 ) ( 6 , 2 , 4 ) ( 3 , 4 , 3 ) ( 1 , 2 , 1 )
( 1 , 1 , 5 ) ( 4 , 4 , 1 ) ( 5 , 2 , 3 ) ( 6 , 2 , 1 ) ( 6 , 4 , 3 )
( 4 , 1 , 3 ) ( 2 , 6 , 2 ) ( 2 , 1 , 6 ) ( 4 , 2 , 1 ) ( 3 , 2 , 4 )
( 5 , 1 , 3 ) ( 5 , 5 , 6 ) ( 4 , 6 , 4 ) ( 6 , 1 , 1 ) ( 6 , 3 , 3 )
( 3 , 1 , 1 ) ( 4 , 5 , 3 ) ( 2 , 2 , 3 ) ( 2 , 3 , 3 ) ( 5 , 3 , 4 )

```

SUMA ****	FREC.ABSOLUT *****	FREC.RELAT *****
3	0	0
4	3	.05
5	2	.033333333
6	3	.05
7	5	.083333333
8	5	.083333333
9	7	.116666667
10	6	.1
11	5	.083333333
12	12	.2
13	4	.066666667
14	6	.1
15	1	.016666667
16	1	.016666667
17	0	0
18	0	0

# LANZAMIENTO DE UNA MONEDA

## introducción:

Este programa sirve para simular el lanzamiento de una moneda N veces expresando los resultados de CARA y CRUZ según corresponda, posteriormente realiza un recuento para calcular las frecuencias absolutas y relativas de cada uno de los dos sucesos.

Si el número de lanzamientos es suficientemente grande, la frecuencia de cada suceso tiende a 0,5 que es su probabilidad.

## listado del programa:

```
10 REM *****
20 REM
30 REM             SIMULACION DEL LANZAMIENTO DE UNA MONEDA
40 REM
50 REM             CON RECUESTO DE FRECUENCIAS ABSOLUTAS Y RELATIVAS
60 REM
70 REM *****
100 PRINT
110 RANDOMIZE
120 C=0
130 X=0
140 PRINT
150 PRINT "CUANTOS LANZAMIENTOS DESEA REALIZAR  N=";
160 INPUT N
170 PRINT
180 FOR I=1 TO N
190 A=INT(RND+.5)
200 IF A=0 THEN 240
210 PRINT "CARA",
220 C=C+1
230 GOTO 260
240 PRINT "CRUZ",
250 X=X+1
260 NEXT I
270 PRINT
280 PRINT "SUCEO", "FREC.ABSOL", "FREC.RELAT"
290 PRINT "*****", "*****", "*****"
300 PRINT "CARA", C, C/N
310 PRINT "CRUZ", X, X/N
320 END
```

## ejercicios resueltos:

**1.** Simular 200 lanzamientos de una moneda y calcular las frecuencias absolutas y relativas de cada uno de los dos resultados posibles.

### S O L U C I O N :

CUANTOS LANZAMIENTOS DESEA REALIZAR N=? 200

CRUZ	CRUZ	CRUZ	CRUZ	CRUZ	CRUZ	CARA	CARA	CARA	CARA
CRUZ	CARA	CRUZ	CRUZ	CARA	CRUZ	CRUZ	CARA	CARA	CARA
CARA	CARA	CARA	CRUZ	CRUZ	CARA	CARA	CRUZ	CARA	CRUZ
CRUZ	CARA	CRUZ	CRUZ	CARA	CRUZ	CARA	CARA	CRUZ	CARA
CARA	CARA	CARA	CRUZ	CARA	CARA	CARA	CARA	CRUZ	CRUZ
CARA	CRUZ	CARA	CRUZ	CRUZ	CARA	CRUZ	CARA	CARA	CRUZ
CRUZ	CARA	CARA	CRUZ	CARA	CARA	CARA	CRUZ	CRUZ	CARA
CARA	CRUZ	CARA	CRUZ	CARA	CRUZ	CARA	CARA	CARA	CARA
CRUZ	CARA	CARA	CARA	CARA	CRUZ	CRUZ	CARA	CARA	CARA
CRUZ	CRUZ	CRUZ	CRUZ	CRUZ	CRUZ	CRUZ	CRUZ	CARA	CARA
CRUZ	CRUZ	CRUZ	CRUZ	CRUZ	CRUZ	CRUZ	CARA	CRUZ	CRUZ
CARA	CRUZ	CRUZ	CARA	CARA	CARA	CRUZ	CARA	CRUZ	CRUZ
CARA	CARA	CRUZ	CRUZ	CRUZ	CARA	CARA	CRUZ	CRUZ	CARA
CARA	CRUZ	CRUZ	CRUZ	CARA	CRUZ	CARA	CRUZ	CRUZ	CARA
CRUZ	CRUZ	CRUZ	CARA	CARA	CARA	CRUZ	CARA	CARA	CARA
CARA	CRUZ	CRUZ	CARA	CARA	CRUZ	CARA	CARA	CRUZ	CRUZ
CRUZ	CRUZ	CRUZ	CARA	CARA	CRUZ	CARA	CARA	CRUZ	CARA
CRUZ	CRUZ	CRUZ	CRUZ	CARA	CRUZ	CARA	CARA	CARA	CARA
CRUZ	CRUZ	CRUZ	CRUZ	CARA	CRUZ	CRUZ	CRUZ	CARA	CRUZ

SUCESO	FREC.ABSOL	FREC.RELAT
*****	*****	*****
CARA	96	.48
CRUZ	104	.52

**2.** Simular 100 lanzamientos de una moneda y calcular las frecuencias absolutas y relativas de cada uno de los dos resultados posibles.

### S O L U C I O N :

CUANTOS LANZAMIENTOS DESEA REALIZAR N=? 100

CRUZ	CARA	CARA	CARA	CARA	CARA	CARA	CRUZ	CRUZ	CRUZ
CARA	CRUZ	CARA	CARA	CARA	CRUZ	CRUZ	CRUZ	CRUZ	CRUZ
CARA	CRUZ	CARA	CRUZ	CRUZ	CRUZ	CRUZ	CRUZ	CRUZ	CARA
CRUZ	CRUZ	CARA	CRUZ	CRUZ	CARA	CARA	CARA	CARA	CARA
CARA	CRUZ	CRUZ	CRUZ	CRUZ	CARA	CRUZ	CARA	CARA	CARA
CARA	CARA	CRUZ	CARA	CRUZ	CARA	CRUZ	CARA	CRUZ	CRUZ
CRUZ	CARA	CARA	CARA	CRUZ	CARA	CRUZ	CRUZ	CARA	CARA
CARA	CRUZ	CARA	CRUZ	CARA	CRUZ	CRUZ	CARA	CARA	CRUZ
CARA	CARA	CARA	CRUZ	CARA	CARA	CARA	CRUZ	CARA	CARA
CRUZ	CRUZ	CARA	CARA	CARA	CRUZ	CARA	CRUZ	CRUZ	CRUZ

SUCESO	FREC.ABSOL	FREC.RELAT
*****	*****	*****
CARA	52	.52
CRUZ	48	.48



# LANZAMIENTO DE DOS MONEDAS

## introducción:

Este programa sirve para simular el lanzamiento de N veces dos monedas expresando los resultados en forma de par ordenado. Posteriormente calcula las frecuencias absolutas y relativas de cada uno de los resultados posibles.

## listado del programa:

```
10 REM *****
20 REM
30 REM             SIMULACION DEL LANZAMIENTO DE DOS MONEDAS
40 REM
50 REM             CON RECUENTO DE FRECUENCIAS ABSOLUTAS Y RELATIVAS
60 REM
70 REM *****
100 PRINT
110 RANDOMIZE
120 C(J)=0
130 PRINT
140 PRINT "CUANTOS LANZAMIENTOS DESEA REALIZAR  N=";
150 INPUT N
160 PRINT
170 FOR I=1 TO N
180 A=INT(RND*4+1)
190 IF A=1 PRINT "(C,C)",
200 IF A=2 PRINT "(C,X)",
210 IF A=3 PRINT "(X,C)",
220 IF A=4 PRINT "(X,X)",
230 FOR J=1 TO 4
240 IF A=J THEN GOTO 260
250 NEXT J
260 C(J)=C(J)+1
270 NEXT I
280 PRINT
290 PRINT "SUCESO", "FREC.ABSOL", "FREC.RELAT"
300 PRINT " *****", "*****", "*****"
310 PRINT "CARA-CARA" ,C(1), C(1)/N
320 PRINT "CARA-CRUZ" ,C(2), C(2)/N
330 PRINT "CRUZ-CARA" ,C(3), C(3)/N
340 PRINT "CRUZ-CRUZ" ,C(4), C(4)/N
350 END
```

## ejercicios resueltos:

1. Simular 200 lanzamientos de dos monedas, calculando las frecuencias absolutas y relativas de cada uno de los resultados posibles.

### SOLUCION :

CUANTOS LANZAMIENTOS DESEA REALIZAR N=? 200

(C,X)	(X,C)	(C,X)	(X,C)	(X,C)	(X,X)	(C,C)	(C,C)	(X,X)	(C,X)
(C,X)	(X,C)	(C,C)	(C,X)	(C,C)	(X,C)	(X,C)	(X,X)	(X,C)	(X,C)
(C,X)	(X,X)	(C,C)	(X,C)	(C,C)	(X,C)	(C,C)	(X,X)	(C,X)	(C,X)
(X,C)	(C,X)	(X,X)	(X,C)	(C,X)	(C,X)	(C,X)	(C,X)	(X,X)	(X,X)
(X,C)	(X,C)	(C,X)	(C,C)	(X,C)	(C,C)	(C,C)	(C,X)	(X,X)	(X,C)
(X,X)	(X,C)	(X,X)	(C,C)	(X,C)	(C,X)	(C,C)	(X,X)	(X,C)	(C,C)
(X,X)	(X,X)	(C,X)	(C,C)	(C,C)	(C,X)	(X,X)	(C,C)	(X,X)	(X,X)
(C,C)	(C,C)	(X,C)	(X,C)	(X,C)	(C,X)	(X,X)	(C,C)	(X,C)	(X,X)
(X,X)	(X,X)	(C,X)	(X,C)	(X,X)	(C,C)	(X,C)	(C,C)	(C,C)	(C,C)
(C,X)	(C,X)	(C,C)	(X,X)	(C,C)	(C,X)	(C,X)	(C,X)	(X,C)	(X,C)
(X,C)	(C,X)	(C,C)	(C,C)	(X,X)	(C,C)	(C,C)	(C,X)	(C,C)	(X,X)
(X,C)	(C,X)	(C,C)	(X,X)	(C,C)	(C,C)	(C,C)	(X,C)	(X,C)	(C,C)
(X,X)	(X,C)	(X,X)	(X,C)	(X,X)	(X,X)	(X,X)	(C,C)	(X,X)	(C,C)
(C,C)	(X,X)	(C,C)	(C,C)	(C,C)	(C,X)	(X,C)	(C,C)	(C,X)	(X,X)
(C,X)	(C,C)	(C,C)	(C,C)	(X,X)	(C,X)	(X,C)	(C,C)	(C,C)	(C,C)
(C,C)	(X,X)	(C,X)	(C,C)	(X,C)	(X,X)	(C,X)	(C,C)	(X,C)	(X,X)
(C,X)	(C,C)	(C,C)	(X,X)	(X,C)	(C,C)	(C,C)	(X,X)	(C,C)	(C,C)
(X,C)	(X,X)	(X,X)	(X,X)	(C,X)	(X,C)	(C,C)	(C,X)	(C,X)	(X,C)
(C,C)	(X,C)	(X,X)	(X,X)	(C,C)	(X,X)	(X,C)	(X,C)	(C,C)	(C,X)
(X,X)	(C,X)	(C,X)	(X,X)	(X,C)	(X,X)	(C,X)	(C,X)	(X,C)	(X,X)

SUCESO	FREC.ABSOL	FREC.RELAT
*****	*****	*****
CARA-CARA	61	.305
CARA-CRUZ	43	.215
CRUZ-CARA	47	.235
CRUZ-CRUZ	49	.245

2. Simular 40 lanzamientos de dos monedas, calculando las frecuencias absolutas y relativas de cada uno de los resultados posibles.

### SOLUCION :

CUANTOS LANZAMIENTOS DESEA REALIZAR N=? 40

(C,X)	(X,X)	(X,C)	(C,C)	(C,X)	(C,C)	(X,X)	(C,C)	(X,C)	(C,C)
(C,C)	(X,X)	(C,C)	(C,C)	(C,X)	(X,C)	(X,X)	(X,X)	(C,C)	(C,X)
(C,C)	(X,X)	(C,X)	(C,C)	(X,C)	(X,X)	(X,X)	(C,X)	(C,C)	(X,X)
(X,C)	(C,X)	(X,X)	(X,C)	(X,X)	(C,X)	(X,C)	(X,C)	(C,X)	(C,C)

SUCESO	FREC.ABSOL	FREC.RELAT
*****	*****	*****
CARA-CARA	12	.3
CARA-CRUZ	10	.25
CRUZ-CARA	8	.2
CRUZ-CRUZ	10	.25

# EXTRACCIONES SUCESIVAS CON REEMPLAZAMIENTO DE CARTAS DE UNA BARAJA

## introducción:

Este programa sirve para simular extracciones sucesivas con reemplazamiento de cartas de una baraja española. Entre las líneas 119 y 250 se realiza la composición de la baraja, siendo las líneas siguientes para simular la extracción con reemplazamiento. Obsérvese que al tratarse de extracciones con reemplazamiento pueden obtenerse cartas repetidas.

## listado del programa:

```
10 REM *****
20 REM
30 REM             SIMULACION DE EXTRACCIONES SUCESIVAS
40 REM
50 REM             CON REEMPLAZAMIENTO DE CARTAS DE UNA BARAJA ESPANOLA
60 REM
70 REM *****
100 PRINT
110 RANDOMIZE
119 REM - COMPOSICION DE LA BARAJA -
120 DIM N$(10),P$(4),C$(10,4)
130 FOR I=1 TO 10
140 READ N$(I)
150 NEXT I
160 FOR J=1 TO 4
170 READ P$(J)
180 NEXT J
190 FOR I=1 TO 10
200 FOR J=1 TO 4
210 C$(I,J)=N$(I)+" DE "+P$(J)
220 NEXT J
230 NEXT I
240 DATA AS,DOS,TRES,CUATRO,CINCO,SEIS,SIETE,SOTA,CABALLO,REY
250 DATA OROS,COPAS,ESPADAS,BASTOS
259 REM - EXTRACCIONES SUCESIVAS CON REEMPLAZAMIENTO -
260 PRINT "CUANTAS CARTAS DESEA EXTRAER L =";
270 INPUT L
280 PRINT
290 FOR K=1 TO L
300 I=INT(RND*10+1)
```

```
310 J=INT(RND*4+1)
320 PRINT C$(I,J)
330 NEXT K
340 END
```

## ejercicios resueltos:

1. Simúlese la extracción con reemplazamiento de 15 cartas de una baraja española.

### S O L U C I O N :

CUANTAS CARTAS DESEA EXTRAER L =? 15

CABALLO DE COPAS  
DOS DE BASTOS  
REY DE OROS  
TRES DE BASTOS  
DOS DE ESPADAS  
SOTA DE OROS  
TRES DE ESPADAS  
TRES DE OROS  
CINCO DE ESPADAS  
TRES DE COPAS  
TRES DE COPAS  
SIETE DE BASTOS  
REY DE COPAS  
SEIS DE OROS  
TRES DE OROS

2. Simúlese la extracción con reemplazamiento de 7 cartas de una baraja española.

### S O L U C I O N :

CUANTAS CARTAS DESEA EXTRAER L =? 7

DOS DE BASTOS  
SEIS DE BASTOS  
SEIS DE BASTOS  
TRES DE BASTOS  
TRES DE COPAS  
DOS DE COPAS  
REY DE OROS

3. Simúlese la extracción con reemplazamiento de 5 cartas de una baraja española.

### S O L U C I O N :

CUANTAS CARTAS DESEA EXTRAER L =? 5

TRES DE OROS  
AS DE BASTOS  
SOTA DE COPAS  
SOTA DE BASTOS  
SEIS DE ESPADAS

# EXTRACCIONES SUCESIVAS SIN REEMPLAZAMIENTO DE CARTAS DE UNA BARAJA

## introducción:

Este programa sirve para simular extracciones sucesivas sin reemplazamiento de cartas de una baraja. En esencia es idéntico al anterior, únicamente hay que tener en cuenta en la línea 320 que si una carta ha sido ya extraída ha de ser apartada pues ya no puede ser extraída nuevamente.

## listado del programa:

```
10 REM *****
20 REM
30 REM          SIMULACION DE EXTRACCIONES SUCESIVAS
40 REM
50 REM          SIN REEMPLAZAMIENTO DE CARTAS DE UNA BARAJA ESPANOLA
60 REM
70 REM *****
100 PRINT
110 RANDOMIZE
119 REM - COMPOSICION DE LA BARAJA -
120 DIM N$(10),P$(4),C$(10,4)
130 FOR I=1 TO 10
140 READ N$(I)
150 NEXT I
160 FOR J=1 TO 4
170 READ P$(J)
180 NEXT J
190 FOR I=1 TO 10
200 FOR J=1 TO 4
210 C$(I,J)=N$(I)+" DE "+P$(J)
220 NEXT J
230 NEXT I
240 DATA AS,DOS,TRES,CUATRO,CINCO,SEIS,SIETE,SOTA,CABALLO,REY
250 DATA OROS,COPAS,ESPADAS,BASTOS
259 REM - EXTRACCIONES SUCESIVAS CON REEMPLAZAMIENTO -
260 PRINT "CUANTAS CARTAS DESEA EXTRAER L = ";
270 INPUT L
280 PRINT
290 FOR K=1 TO L
300 I=INT(RND*10+1)
310 J=INT(RND*4+1)
```

```
320 IF C$(I,J)="" THEN GOTO 300
330 PRINT C$(I,J)
340 C$(I,J)=""
350 NEXT K
360 END
```

## **ejercicios resueltos:**

**1.** Simúlese la extracción sin reemplazamiento de 18 cartas de una baraja española.

### **SOLUCION :**

CUANTAS CARTAS DESEA EXTRAER L =? 18

CABALLO DE OROS  
REY DE ESPADAS  
CUATRO DE OROS  
SOTA DE COPAS  
AS DE COPAS  
SOTA DE ESPADAS  
CABALLO DE ESPADAS  
SIETE DE OROS  
TRES DE BASTOS  
CINCO DE ESPADAS  
AS DE ESPADAS  
SIETE DE COPAS  
TRES DE OROS  
SEIS DE BASTOS  
SEIS DE OROS  
SOTA DE OROS  
DOS DE ESPADAS  
DOS DE OROS

**2.** Simúlese la extracción sin reemplazamiento de 10 cartas de una baraja española.

### **SOLUCION :**

CUANTAS CARTAS DESEA EXTRAER L =? 10

CINCO DE COPAS  
AS DE ESPADAS  
SOTA DE OROS  
REY DE ESPADAS  
DOS DE OROS  
CINCO DE BASTOS  
CUATRO DE OROS  
AS DE BASTOS  
CINCO DE ESPADAS  
DOS DE COPAS

# MEDIDAS DE CENTRALIZACION, DISPERSION Y FORMA DE DISTRIBUCIONES UNIDIMENSIONALES DE DATOS NO AGRUPADOS

## introducción:

Este programa sirve para calcular medidas de centralización, dispersión y de forma de una distribución unidimensional de datos no agrupados.

En la línea 210 se introducen en una variable subindicada X(I) cada uno de los valores de la variable estadística. Si algún X(I) es igual a ce ro entonces la media armónica no está definida y la media geométrica es nu la.

Como el coeficiente de variación de una distribución estadística unidimensional está definido como  $s/\bar{x}$ , entonces cuando  $\bar{x}$  tiende a cero el coeficiente de variación crece indefinidamente, de manera que se trata, en ese caso de una medida poco fiable.

Para valores muy próximos a cero del coeficiente de asimetría de Fisher, el resultado del programa dará sesgada hacia la derecha o hacia la izquierda según que sea positivo o negativo, si bien se trata de distribuciones practicamente insesgadas.

## listado del programa:

```
10 REM *****
20 REM
30 REM          M E D I D A S      D E      C E N T R A L I Z A C I O N
40 REM
50 REM          D E      D I S P E R S I O N      Y      D E      F O R M A
60 REM
70 REM          D I S T R I B U C I O N E S U N I D I M E N S I O N A L E S D E D A T O S N O A G R U P A D O S
80 REM
90 REM *****
100 PRINT
110 PRINT
120 PRINT "NUMERO DE DATOS";
130 INPUT N
135 PRINT
140 DIM X(N),S(10)
150 FOR J=1 TO 9
160 S(J)=0
170 NEXT J
```

```

180 S(10)=1
189 REM-ENTRADA DE DATOS-
190 FOR I=1 TO N
200 PRINT "VALOR DE LA VARIABLE NUMERO      ";I;
210 INPUT X(I)
220 S(1)=S(1)+X(I)
230 S(2)=S(2)+X(I)^2
240 S(3)=S(3)+X(I)^3
250 S(4)=S(4)+X(I)^4
260 NEXT I
265 M=S(1)/N
270 FOR I=1 TO N
280 S(6)=S(6)+(X(I)-M)^2
290 S(7)=S(7)+(X(I)-M)^3
300 S(8)=S(8)+(X(I)-M)^4
310 S(5)=S(5)+1/X(I)
320 S(9)=S(9)+ABS(X(I)-M)
330 S(10)=S(10)*X(I)
340 NEXT I
349 REM - CALCULO DE PARAMETROS -
350 G=S(10)^(1/N)
360 H=N/S(5)
370 Q=SQR(S(2)/N)
380 V=S(6)/N
390 D=S(9)/N
400 A1=M
410 A2=S(2)/N
420 A3=S(3)/N
430 A4=S(4)/N
440 M2=S(6)/N
450 M3=S(7)/N
460 M4=S(8)/N
470 C=SQR(V)/M
480 Y1=M3/V^(3/2)
490 Y2=M4/V^2
499 REM - SALIDA DE RESULTADOS -
500 PRINT
510 PRINT "***** MEDIDAS DE CENTRALIZACION *****"
520 PRINT "MEDIA ARITMETICA          ";M
530 PRINT "MEDIA GEOMETRICA          ";G
540 PRINT "MEDIA ARMONICA            ";H
550 PRINT "MEDIA CUADRATICA           ";Q
560 PRINT
570 PRINT "***** MEDIDAS DE DISPERSION *****"
580 PRINT "VARIANZA                      ";V
590 PRINT "DESVIACION TIPICA            ";SQR(V)
600 PRINT "DESVIACION MEDIA            ";D
610 PRINT "MOMENTO DE PRIMER ORDEN RESPECTO AL ORIGEN ";A1
620 PRINT "MOMENTO DE SEGUNDO ORDEN RESPECTO AL ORIGEN ";A2
630 PRINT "MOMENTO DE TERCER ORDEN RESPECTO AL ORIGEN ";A3
640 PRINT "MOMENTO DE CUARTO ORDEN RESPECTO AL ORIGEN ";A4
650 PRINT "MOMENTO DE PRIMER ORDEN RESPECTO A LA MEDIA ";O
660 PRINT "MOMENTO DE SEGUNDO ORDEN RESPECTO A LA MEDIA ";M2
670 PRINT "MOMENTO DE TERCER ORDEN RESPECTO A LA MEDIA ";M3
680 PRINT "MOMENTO DE CUARTO ORDEN RESPECTO A LA MEDIA ";M4
690 PRINT "COEFICIENTE DE VARIACION      ";C
700 PRINT
710 PRINT "***** MEDIDAS DE FORMA *****"
720 PRINT "COEFICIENTE DE ASIMETRIA DE FISHER ";Y1
730 IF Y1>0 THEN PRINT "(DISTRIBUCION SESGADA A LA DERECHA)"
740 IF Y1<0 THEN PRINT "(DISTRIBUCION SESGADA A LA IZQUIERDA)"
750 IF Y1=0 THEN PRINT "(DISTRIBUCION INSESGADA)"
760 PRINT "COEFICIENTE DE CURTOSIS O APLASTAMIENTO ";Y2

```



```

770 IF Y2 >3 THEN PRINT "(DISTRIBUCION LEPTOCURTICA)"
780 IF Y2 <3 THEN PRINT "(DISTRIBUCION PLANICURTICA)"
790 IF Y2 =3 THEN PRINT "(DISTRIBUCION MESOCURTICA)"
800 END

```

## ejercicios resueltos:

1. El número de individuos muertos por cólera en un determinado país por año en el transcurso de 11 meses, han sido:

2, 17, 5, 8, 12, 3, 2, 8, 12, 12, 3.

Calcular:

- Medias aritmética, geométrica, armónica y cuadrática.
- Varianza, desviación media, momentos de 1°, 2°, 3° y 4° orden respecto al origen y respecto a la media y coeficiente de variación.
- Coefficientes de asimetría y curtosis.

### S O L U C I O N :

NUMERO DE DATOS? 11

VALOR DE LA VARIABLE NUMERO	1 ? 2
VALOR DE LA VARIABLE NUMERO	2 ? 17
VALOR DE LA VARIABLE NUMERO	3 ? 5
VALOR DE LA VARIABLE NUMERO	4 ? 8
VALOR DE LA VARIABLE NUMERO	5 ? 12
VALOR DE LA VARIABLE NUMERO	6 ? 3
VALOR DE LA VARIABLE NUMERO	7 ? 2
VALOR DE LA VARIABLE NUMERO	8 ? 8
VALOR DE LA VARIABLE NUMERO	9 ? 12
VALOR DE LA VARIABLE NUMERO	10 ? 12
VALOR DE LA VARIABLE NUMERO	11 ? 3

#### \*\*\*\*\* MEDIDAS DE CENTRALIZACION \*\*\*\*\*

MEDIA ARITMETICA	= 7.6363636
MEDIA GEOMETRICA	= 5.9621665
MEDIA ARMONICA	= 4.5351657
MEDIA CUADRATICA	= 9.0453403

#### \*\*\*\*\* MEDIDAS DE DISPERSION \*\*\*\*\*

VARIANZA	= 23.504132
DESVIACION MEDIA	= 4.214876
MOMENTO DE PRIMER ORDEN RESPECTO AL ORIGEN	= 7.6363636
MOMENTO DE SEGUNDO ORDEN RESPECTO AL ORIGEN	= 81.818182
MOMENTO DE TERCER ORDEN RESPECTO AL ORIGEN	= 1028.7273
MOMENTO DE CUARTO ORDEN RESPECTO AL ORIGEN	= 14067.273
MOMENTO DE PRIMER ORDEN RESPECTO A LA MEDIA	= 0
MOMENTO DE SEGUNDO ORDEN RESPECTO A LA MEDIA	= 23.504132
MOMENTO DE TERCER ORDEN RESPECTO A LA MEDIA	= 44.961683
MOMENTO DE CUARTO ORDEN RESPECTO A LA MEDIA	= 1069.6425
COEFICIENTE DE VARIACION	= .63487103

#### \*\*\*\*\* MEDIDAS DE FORMA \*\*\*\*\*

COEFICIENTE DE ASIMETRIA DE FISHER	= .39457197
(DISTRIBUCION SESGADA A LA DERECHA)	
COEFICIENTE DE CURTOSIS O APLASTAMIENTO	= 1.9362
(DISTRIBUCION PLANICURTICA)	

2. Se ha realizado la determinación del contenido de Calcio (Ca) en la sangre de 25 pacientes, obteniéndose los siguientes resultados:

8.7    9.3    10.1    9.2    9.1    9.3    9.4    8.7    8.8    8.7    9.2    8.3  
 10.2    9.5    9.6    9.7    9.2    9.3    8.8    9.5    9.8    9.1    9.2    9.6  
 8.4

Calcular:

- Medias aritmética, geométrica, armónica y cuadrática.
- Varianza, desviación media, momentos de 1°, 2°, 3° y 4° orden respecto al origen y respecto a la media y coeficiente de variación.
- Coeficiente de asimetría y de curtosis.

### S O L U C I O N :

NUMERO DE DATOS? 25

VALOR DE LA VARIABLE	NUMERO	1 ? 8.7
VALOR DE LA VARIABLE	NUMERO	2 ? 9.3
VALOR DE LA VARIABLE	NUMERO	3 ? 10.1
VALOR DE LA VARIABLE	NUMERO	4 ? 9.2
VALOR DE LA VARIABLE	NUMERO	5 ? 9.1
VALOR DE LA VARIABLE	NUMERO	6 ? 9.3
VALOR DE LA VARIABLE	NUMERO	7 ? 9.4
VALOR DE LA VARIABLE	NUMERO	8 ? 8.7
VALOR DE LA VARIABLE	NUMERO	9 ? 8.8
VALOR DE LA VARIABLE	NUMERO	10 ? 8.7
VALOR DE LA VARIABLE	NUMERO	11 ? 9.2
VALOR DE LA VARIABLE	NUMERO	12 ? 8.3
VALOR DE LA VARIABLE	NUMERO	13 ? 10.2
VALOR DE LA VARIABLE	NUMERO	14 ? 9.5
VALOR DE LA VARIABLE	NUMERO	15 ? 9.6
VALOR DE LA VARIABLE	NUMERO	16 ? 9.7
VALOR DE LA VARIABLE	NUMERO	17 ? 9.2
VALOR DE LA VARIABLE	NUMERO	18 ? 9.3
VALOR DE LA VARIABLE	NUMERO	19 ? 8.8
VALOR DE LA VARIABLE	NUMERO	20 ? 9.5
VALOR DE LA VARIABLE	NUMERO	21 ? 9.8
VALOR DE LA VARIABLE	NUMERO	22 ? 9.1
VALOR DE LA VARIABLE	NUMERO	23 ? 9.2
VALOR DE LA VARIABLE	NUMERO	24 ? 9.6
VALOR DE LA VARIABLE	NUMERO	25 ? 8.4

#### \*\*\*\*\* MEDIDAS DE CENTRALIZACION \*\*\*\*\*

MEDIA ARITMETICA	= 9.228
MEDIA GEOMETRICA	= 9.2160328
MEDIA ARMONICA	= 9.2040395
MEDIA CUADRATICA	= 9.2399351

#### \*\*\*\*\* MEDIDAS DE DISPERSION \*\*\*\*\*

VARIANZA	= .220416
DESVIACION MEDIA	= .36512
MOMENTO DE PRIMER ORDEN RESPECTO AL ORIGEN	= 9.228
MOMENTO DE SEGUNDO ORDEN RESPECTO AL ORIGEN	= 85.3764
MOMENTO DE TERCER ORDEN RESPECTO AL ORIGEN	= 791.92356
MOMENTO DE CUARTO ORDEN RESPECTO AL ORIGEN	= 7364.3668
MOMENTO DE PRIMER ORDEN RESPECTO A LA MEDIA	= 0
MOMENTO DE SEGUNDO ORDEN RESPECTO A LA MEDIA	= .220416
MOMENTO DE TERCER ORDEN RESPECTO A LA MEDIA	= .0021431039
MOMENTO DE CUARTO ORDEN RESPECTO A LA MEDIA	= .12760652
COEFICIENTE DE VARIACION	= .050876119

```
***** MEDIDAS DE FORMA *****
COEFICIENTE DE ASIMETRIA DE FISHER      = .020709926
(DISTRIBUCION SESGADA A LA DERECHA)
COEFICIENTE DE CURTOSIS O APLASTAMIENTO = 2.6265558
(DISTRIBUCION PLANICURTICA)
```

**3.** El número de accidentes mortales diarios en una gran ciudad en nueve días han sido: 6, 4, 8, 1, 5, 3, 3, 7, 2. Calcular:

- Medias aritmética, geométrica, armónica y cuadrática.
- Varianza, desviación media, momentos de 1°, 2°, 3° y 4° orden respecto al origen y respecto a la media y coeficiente de variación.
- Coeficientes de asimetría y de curtosis.

### S O L U C I O N :

NUMERO DE DATOS? 9

VALOR DE LA VARIABLE NUMERO	1 ? 6
VALOR DE LA VARIABLE NUMERO	2 ? 4
VALOR DE LA VARIABLE NUMERO	3 ? 8
VALOR DE LA VARIABLE NUMERO	4 ? 1
VALOR DE LA VARIABLE NUMERO	5 ? 5
VALOR DE LA VARIABLE NUMERO	6 ? 3
VALOR DE LA VARIABLE NUMERO	7 ? 3
VALOR DE LA VARIABLE NUMERO	8 ? 7
VALOR DE LA VARIABLE NUMERO	9 ? 2

```
***** MEDIDAS DE CENTRALIZACION *****
MEDIA ARITMETICA      = 4.3333333
MEDIA GEOMETRICA      = 3.6706077
MEDIA ARMONICA        = 2.9496684
MEDIA CUADRATICA      = 4.8648398
```

```
***** MEDIDAS DE DISPERSION *****
VARIANZA              = 4.8888889
DESVIACION MEDIA      = 1.9259259
MOMENTO DE PRIMER ORDEN RESPECTO AL ORIGEN = 4.3333333
MOMENTO DE SEGUNDO ORDEN RESPECTO AL ORIGEN = 23.666667
MOMENTO DE TERCER ORDEN RESPECTO AL ORIGEN = 147
MOMENTO DE CUARTO ORDEN RESPECTO AL ORIGEN = 983.66667
MOMENTO DE PRIMER ORDEN RESPECTO A LA MEDIA = 0
MOMENTO DE SEGUNDO ORDEN RESPECTO A LA MEDIA = 4.8888889
MOMENTO DE TERCER ORDEN RESPECTO A LA MEDIA = 2.0740741
MOMENTO DE CUARTO ORDEN RESPECTO A LA MEDIA = 44.296296
COEFICIENTE DE VARIACION = .51024997
```

```
***** MEDIDAS DE FORMA *****
COEFICIENTE DE ASIMETRIA DE FISHER      = .19187086
(DISTRIBUCION SESGADA A LA DERECHA)
COEFICIENTE DE CURTOSIS O APLASTAMIENTO = 1.8533058
(DISTRIBUCION PLANICURTICA)
```

# MEDIDAS DE CENTRALIZACION, DISPERSION Y FORMA DE DISTRIBUCIONES UNIDIMENSIONALES DE DATOS AGRUPADOS

## introducción:

Este programa calcula las medidas de centralización, dispersión y de forma de una distribución unidimensional de datos agrupados. Los valores de la variable estadística se introducen en una variable subindicada  $X(I)$  y los valores de las frecuencias absolutas se introducen en otra  $F(I)$  ambas en la línea 210.

En todo lo demás el programa es análogo al anterior.

## listado del programa:

```
10 REM *****
20 REM
30 REM          M E D I A S   D E   C E N T R A L I Z A C I O N
40 REM
50 REM          D E   D I S P E R S I O N   Y   D E   F O R M A
60 REM
70 REM          D I S T R I B U C I O N E S   U N I D I M E N S I O N A L E S   D E   D A T O S   A G R U P A D O S
80 REM
90 REM *****
100 PRINT
110 PRINT
120 PRINT "NUMERO DE DATOS";
130 INPUT L
135 PRINT
140 DIM X(L),F(L),S(10)
150 FOR J=0 TO 9
160 S(J)=0
170 NEXT J
180 S(10)=1
185 REM-ENTRADA DE DATOS-
190 FOR I=1 TO L
200 PRINT "VARIABLE X   Y   FRECUENCIA ABSOLUTA F DEL DATO";I;
210 INPUT X(I),F(I)
215 S(0)=S(0)+F(I)
220 S(1)=S(1)+F(I)*X(I)
230 S(2)=S(2)+F(I)*X(I)^2
240 S(3)=S(3)+F(I)*X(I)^3
250 S(4)=S(4)+F(I)*X(I)^4
260 NEXT I
263 N=S(0)
```

```

265 M=S(1)/N
270 FOR I=1 TO L
280 S(6)=S(6)+F(I)*(X(I)-M)^2
290 S(7)=S(7)+F(I)*(X(I)-M)^3
300 S(8)=S(8)+F(I)*(X(I)-M)^4
310 S(5)=S(5)+F(I)/X(I)
320 S(9)=S(9)+ABS(F(I)*(X(I)-M))
330 S(10)=S(10)*X(I)^F(I)
340 NEXT I
349 REM - CALCULO DE PARAMETROS -
350 G=S(10)^(1/N)
360 H=N/S(5)
370 Q=SQR(S(2)/N)
380 V=S(6)/N
390 D=S(9)/N
400 A1=M
410 A2=S(2)/N
420 A3=S(3)/N
430 A4=S(4)/N
440 M2=S(6)/N
450 M3=S(7)/N
460 M4=S(8)/N
470 C=SQR(V)/M
480 Y1=M3/V^(3/2)
490 Y2=M4/V^2
499 REM - SALIDA DE RESULTADOS -
500 PRINT
510 PRINT "***** MEDIDAS DE CENTRALIZACION *****"
520 PRINT "MEDIA ARITMETICA                      ";M
530 PRINT "MEDIA GEOMETRICA                      ";G
540 PRINT "MEDIA ARMONICA                          ";H
550 PRINT "MEDIA CUADRATICA                          ";Q
560 PRINT
570 PRINT "***** MEDIDAS DE DISPERSION *****"
580 PRINT "VARIANZA                      ";V
590 PRINT "DESVIACION TIPICA            ";SQR(V)
600 PRINT "DESVIACION MEDIA            ";D
610 PRINT "MOMENTO DE PRIMER ORDEN RESPECTO AL ORIGEN ";A1
620 PRINT "MOMENTO DE SEGUNDO ORDEN RESPECTO AL ORIGEN ";A2
630 PRINT "MOMENTO DE TERCER ORDEN RESPECTO AL ORIGEN ";A3
640 PRINT "MOMENTO DE CUARTO ORDEN RESPECTO AL ORIGEN ";A4
650 PRINT "MOMENTO DE PRIMER ORDEN RESPECTO A LA MEDIA ";O
660 PRINT "MOMENTO DE SEGUNDO ORDEN RESPECTO A LA MEDIA ";M2
670 PRINT "MOMENTO DE TERCER ORDEN RESPECTO A LA MEDIA ";M3
680 PRINT "MOMENTO DE CUARTO ORDEN RESPECTO A LA MEDIA ";M4
690 PRINT "COEFICIENTE DE VARIACION      ";C
700 PRINT
710 PRINT "***** MEDIDAS DE FORMA *****"
720 PRINT "COEFICIENTE DE ASIMETRIA DE FISHER ";Y1
730 IF Y1>0 THEN PRINT "(DISTRIBUCION SESGADA A LA DERECHA)"
740 IF Y1<0 THEN PRINT "(DISTRIBUCION SESGADA A LA IZQUIERDA)"
750 IF Y1=0 THEN PRINT "(DISTRIBUCION INSESGADA)"
760 PRINT "COEFICIENTE DE CURTOSIS O APLASTAMIENTO ";Y2
770 IF Y2>3 THEN PRINT "(DISTRIBUCION LEPTOCURTICA)"
780 IF Y2<3 THEN PRINT "(DISTRIBUCION PLANICURTICA)"
790 IF Y2=3 THEN PRINT "(DISTRIBUCION MESOCURTICA)"
800 END

```

## ejercicios resueltos:

1. Examinadas 150 camadas de 8 ratones, respecto del número de ellos con pelo liso, se obtuvo la siguiente tabla:

Calcular:

- Medias aritmética, geométrica armónica y cuadrática.
- Varianza, desviación típica, desviación media, momentos de 1°, 2° 3° y 4° orden respecto al origen y respecto a la media y coeficiente de variación.
- Coefficientes de asimetría y de curtosis.

n° de ratones con pelo liso en cada camada	n° de camadas
1	1
2	6
3	13
4	50
5	33
6	30
7	17
8	0
	150

### SOLUCION :

NUMERO DE DATOS? 8

VARIABLE X	Y	FRECUENCIA ABSOLUTA F DEL DATO 1 ? 1	1
VARIABLE X	Y	FRECUENCIA ABSOLUTA F DEL DATO 2 ? 2	6
VARIABLE X	Y	FRECUENCIA ABSOLUTA F DEL DATO 3 ? 3	13
VARIABLE X	Y	FRECUENCIA ABSOLUTA F DEL DATO 4 ? 4	50
VARIABLE X	Y	FRECUENCIA ABSOLUTA F DEL DATO 5 ? 5	33
VARIABLE X	Y	FRECUENCIA ABSOLUTA F DEL DATO 6 ? 6	30
VARIABLE X	Y	FRECUENCIA ABSOLUTA F DEL DATO 7 ? 7	17
VARIABLE X	Y	FRECUENCIA ABSOLUTA F DEL DATO 8 ? 8	0

#### \*\*\*\*\* MEDIDAS DE CENTRALIZACION \*\*\*\*\*

MEDIA ARITMETICA	= 4.7733333
MEDIA GEOMETRICA	= 4.5630981
MEDIA ARMONICA	= 4.3026909
MEDIA CUADRATICA	= 4.9531135

#### \*\*\*\*\* MEDIDAS DE DISPERSION \*\*\*\*\*

VARIANZA	= 1.7486222
DESVIACION TIPICA	= 1.3223548
DESVIACION MEDIA	= 1.0951111
MOMENTO DE PRIMER ORDEN RESPECTO AL ORIGEN	= 4.7733333
MOMENTO DE SEGUNDO ORDEN RESPECTO AL ORIGEN	= 24.533333
MOMENTO DE TERCER ORDEN RESPECTO AL ORIGEN	= 133.57333
MOMENTO DE CUARTO ORDEN RESPECTO AL ORIGEN	= 761.81333
MOMENTO DE PRIMER ORDEN RESPECTO A LA MEDIA	= 0
MOMENTO DE SEGUNDO ORDEN RESPECTO A LA MEDIA	= 1.7486222
MOMENTO DE TERCER ORDEN RESPECTO A LA MEDIA	= -.22595793
MOMENTO DE CUARTO ORDEN RESPECTO A LA MEDIA	= 7.9334499
COEFICIENTE DE VARIACION	= .27702964

#### \*\*\*\*\* MEDIDAS DE FORMA \*\*\*\*\*

COEFICIENTE DE ASIMETRIA DE FISHER	= -.097720029
(DISTRIBUCION SEGGADA A LA IZQUIERDA)	
COEFICIENTE DE CURTOSIS O APLASTAMIENTO	= 2.5945981
(DISTRIBUCION PLANICURTICA)	

**2.** Las calificaciones en Matemáticas de 41 alumnos de una determinada clase han sido las siguientes:

Calcular:

- a) Medias aritmética, geométrica, armónica y cuadrática.
- b) Varianza, desviación típica, desviación media, momentos de 1°, 2°, 3° y 4° orden respecto al origen y respecto a la media y coeficiente de variación.
- c) Coeficientes de asimetría y de curtosis.

calif.	n°de alumnos
2	5
3	8
4	7
5	10
6	4
7	3
8	2
9	1
1	1
	41

**S O L U C I O N :**

NUMERO DE DATOS? 9

VARIABLE X	y	FRECUENCIA ABSOLUTA F DEL DATO 1	? 2	5
VARIABLE X	y	FRECUENCIA ABSOLUTA F DEL DATO 2	? 3	8
VARIABLE X	y	FRECUENCIA ABSOLUTA F DEL DATO 3	? 4	7
VARIABLE X	y	FRECUENCIA ABSOLUTA F DEL DATO 4	? 5	10
VARIABLE X	y	FRECUENCIA ABSOLUTA F DEL DATO 5	? 6	4
VARIABLE X	y	FRECUENCIA ABSOLUTA F DEL DATO 6	? 7	3
VARIABLE X	y	FRECUENCIA ABSOLUTA F DEL DATO 7	? 8	2
VARIABLE X	y	FRECUENCIA ABSOLUTA F DEL DATO 8	? 9	1
VARIABLE X	y	FRECUENCIA ABSOLUTA F DEL DATO 9	? 1	1

\*\*\*\*\* MEDIDAS DE CENTRALIZACION \*\*\*\*\*

MEDIA ARITMETICA	= 4.4634146
MEDIA GEOMETRICA	= 4.0565799
MEDIA ARMONICA	= 3.6050244
MEDIA CUADRATICA	= 4.8262494

\*\*\*\*\* MEDIDAS DE DISPERSION \*\*\*\*\*

VARIANZA	= 3.3706127
DESVIACION TIPICA	= 1.8359229
DESVIACION MEDIA	= 1.4991077
MOMENTO DE PRIMER ORDEN RESPECTO AL ORIGEN	= 4.4634146
MOMENTO DE SEGUNDO ORDEN RESPECTO AL ORIGEN	= 23.292683
MOMENTO DE TERCER ORDEN RESPECTO AL ORIGEN	= 136.60976
MOMENTO DE CUARTO ORDEN RESPECTO AL ORIGEN	= 875.87805
MOMENTO DE PRIMER ORDEN RESPECTO A LA MEDIA	= 0
MOMENTO DE SEGUNDO ORDEN RESPECTO A LA MEDIA	= 3.3706127
MOMENTO DE TERCER ORDEN RESPECTO A LA MEDIA	= 2.5559699
MOMENTO DE CUARTO ORDEN RESPECTO A LA MEDIA	= 30.458254
COEFICIENTE DE VARIACION	= .41132698

\*\*\*\*\* MEDIDAS DE FORMA \*\*\*\*\*

COEFICIENTE DE ASIMETRIA DE FISHER	= .41304032
(DISTRIBUCION SESGADA A LA DERECHA)	
COEFICIENTE DE CURTOSIS O APLASTAMIENTO	= 2.6809412
(DISTRIBUCION PLANICURTICA)	

**3.** La tabla adjunta muestra el número de semillas germinadas por placa y la frecuencia observada en cada caso.

n°de semillas germinadas p/plac	5	4	3	2	1
frecuencia observad	30	81	57	24	7

Calcular:

- Medias aritmética, geométrica, armónica y cuadrática.
- Varianza, desviación típica, desviación media, momentos de 1°, 2°, 3° y 4° orden respecto a la media y respecto al origen y coeficiente de variación.
- Coeficientes de asimetría y curtosis.

### **S O L U C I O N :**

NUMERO DE DATOS? 5

```
VARIABLE X y FRECUENCIA ABSOLUTA F DEL DATO 1 ? 5      30
VARIABLE X y FRECUENCIA ABSOLUTA F DEL DATO 2 ? 4      81
VARIABLE X y FRECUENCIA ABSOLUTA F DEL DATO 3 ? 3      57
VARIABLE X y FRECUENCIA ABSOLUTA F DEL DATO 4 ? 2      24
VARIABLE X y FRECUENCIA ABSOLUTA F DEL DATO 5 ? 1       7
```

#### \*\*\*\*\* MEDIDAS DE CENTRALIZACION \*\*\*\*\*

```
MEDIA ARITMETICA      = 3.5175879
MEDIA GEOMETRICA      = 3.3373362
MEDIA ARMONICA        = 3.0972763
MEDIA CUADRATICA      = 3.6574423
```

#### \*\*\*\*\* MEDIDAS DE DISPERSION \*\*\*\*\*

```
VARIANZA              = 1.0034595
DESVIACION TIPICA      = 1.0017283
DESVIACION MEDIA      = .83967577
MOMENTO DE PRIMER ORDEN RESPECTO AL ORIGEN = 3.5175879
MOMENTO DE SEGUNDO ORDEN RESPECTO AL ORIGEN = 13.376884
MOMENTO DE TERCER ORDEN RESPECTO AL ORIGEN = 53.628141
MOMENTO DE CUARTO ORDEN RESPECTO AL ORIGEN = 223.58794
MOMENTO DE PRIMER ORDEN RESPECTO A LA MEDIA = 0
MOMENTO DE SEGUNDO ORDEN RESPECTO A LA MEDIA = 1.0034595
MOMENTO DE TERCER ORDEN RESPECTO A LA MEDIA = -.48574074
MOMENTO DE CUARTO ORDEN RESPECTO A LA MEDIA = 2.8234534
COEFICIENTE DE VARIACION = .28477703
```

#### \*\*\*\*\* MEDIDAS DE FORMA \*\*\*\*\*

```
COEFICIENTE DE ASIMETRIA DE FISHER      = -.48323096
(DISTRIBUCION SESGADA A LA IZQUIERDA)
COEFICIENTE DE CURTOSIS O APLASTAMIENTO = 2.8040188
(DISTRIBUCION PLANICURTICA)
```



# PARAMETROS ESTADISTICOS DE DISTRIBUCIONES BIDIMENSIONALES DE DATOS NO AGRUPADOS

## introducción:

Este programa calcula parámetros estadísticos de distribuciones bidimensionales de datos no agrupados. La variable bidimensional se introduce en la línea 220 bajo dos variables subindicadas  $X(I)$ ,  $Y(I)$ .

El coeficiente de correlación lineal da la medida de la dependencia que existe entre las dos variables unidimensionales que componen la variable bidimensional.

## listado del programa:

```
10 REM *****
20 REM
30 REM          C A L C U L O   D E   P A R A M E T R O S
40 REM
50 REM          E S T A D I S T I C O S
60 REM
70 REM          DISTRIBUCIONES BIDIMENSIONALES DE DATOS NO AGRUPADOS
80 REM
90 REM *****
100 PRINT
110 PRINT
120 PRINT "NUMERO DE PARES DE DATOS";
130 INPUT N
140 DIM X(N),Y(N),S(5)
150 FOR J=1 TO 5
160 S(J)=0
170 NEXT J
180 PRINT
189 REM-ENTRADA DE DATOS-
190 PRINT "INTRODUZCA LA X   y   LA Y DE CADA PAR"
195 PRINT
200 FOR I=1 TO N
210 PRINT "X,Y DEL PUNTO";I;
220 INPUT X(I),Y(I)
230 S(1)=S(1)+X(I)
240 S(2)=S(2)+Y(I)
250 NEXT I
260 M1=S(1)/N
270 M2=S(2)/N
280 FOR I=1 TO N
```

```

290 S(3)=S(3)+(X(I)-M1)^2
300 S(4)=S(4)+(Y(I)-M2)^2
310 S(5)=S(5)+(X(I)-M1)*(Y(I)-M2)
320 NEXT I
330 V1=S(3)/N
340 V2=S(4)/N
350 W1=S(3)/(N-1)
360 W2=S(4)/(N-1)
370 C1=S(5)/N
380 C2=S(5)/(N-1)
390 R=C1/SQR(V1*V2)
399 REM /- SALIDA DE RESULTADOS -
400 PRINT
410 PRINT "MEDIA ARITMETICA DE LA VARIABLE X      =" ; M1
420 PRINT "MEDIA ARITMETICA DE LA VARIABLE Y      =" ; M2
430 PRINT
440 PRINT "VARIANZA DE LA VARIABLE X              =" ; V1
450 PRINT "VARIANZA DE LA VARIABLE Y              =" ; V2
460 PRINT
470 PRINT "CUASIVARIANZA DE LA VARIABLE X            =" ; W1
480 PRINT "CUASIVARIANZA DE LA VARIABLE Y            =" ; W2
490 PRINT
500 PRINT "COVARIANZA DE X,Y                          =" ; C1
510 PRINT "CUASICOVARIANZA DE X,Y                     =" ; C2
520 PRINT
530 PRINT "COEFICIENTE DE CORRELACION LINEAL           =" ; R
540 PRINT "COEFICIENTE DE DETERMINACION              =" ; R^2
550 END

```

## ejercicios resueltos:

1. Las calificaciones en las asignaturas de Física y Matemáticas de 9 alumnos han sido las siguientes:

MATEMATICAS	7.5	4.5	3.7	2.5	8.5	1.5	4.5	5	5.5
FISICA	8.5	6	4	2.5	9	3	6	5.5	6

Calcular los parámetros estadísticos más importantes.

### S O L U C I O N :

NUMERO DE PARES DE DATOS? 9

INTRODUZCA LA X Y LA Y DE CADA PAR

```

X,Y DEL PUNTO 1 ? 7.5      8.5
X,Y DEL PUNTO 2 ? 4.5      6
X,Y DEL PUNTO 3 ? 3.7      4
X,Y DEL PUNTO 4 ? 2.5      2.5
X,Y DEL PUNTO 5 ? 8.5      9
X,Y DEL PUNTO 6 ? 1.5      3
X,Y DEL PUNTO 7 ? 4.5      6
X,Y DEL PUNTO 8 ? 5        5.5
X,Y DEL PUNTO 9 ? 5.5      6

```

```

MEDIA ARITMETICA DE LA VARIABLE X      = 4.8
MEDIA ARITMETICA DE LA VARIABLE Y      = 5.6111111

```

VARIANZA DE LA VARIABLE X	= 4.3422222
VARIANZA DE LA VARIABLE Y	= 4.3765432
CUASIVARIANZA DE LA VARIABLE X	= 4.885
CUASIVARIANZA DE LA VARIABLE Y	= 4.9236111
COVARIANZA DE X,Y	= 4.2111111
CUASICOVARIANZA DE X,Y	= 4.7375
COEFICIENTE DE CORRELACION LINEAL	= .96599542
COEFICIENTE DE DETERMINACION	= .93314715

**2.** Los datos siguientes representan los valores de dos constantes biológicas de 12 individuos de una cierta población:

X	65	50	55	65	55	70	65	70	55	70	50	50
Y	85	79	76	90	85	87	94	98	81	95	76	74

Calcular los parámetros estadísticos más importantes.

### **SOLUCION :**

NUMERO DE PARES DE DATOS? 12

INTRODUZCA LA X y LA Y DE CADA PAR

X,Y DEL PUNTO 1 ?	65	85
X,Y DEL PUNTO 2 ?	50	79
X,Y DEL PUNTO 3 ?	55	76
X,Y DEL PUNTO 4 ?	65	90
X,Y DEL PUNTO 5 ?	55	85
X,Y DEL PUNTO 6 ?	70	87
X,Y DEL PUNTO 7 ?	65	94
X,Y DEL PUNTO 8 ?	70	98
X,Y DEL PUNTO 9 ?	55	81
X,Y DEL PUNTO 10 ?	70	95
X,Y DEL PUNTO 11 ?	50	76
X,Y DEL PUNTO 12 ?	50	74

MEDIA ARITMETICA DE LA VARIABLE X	= 60
MEDIA ARITMETICA DE LA VARIABLE Y	= 85

VARIANZA DE LA VARIABLE X	= 62.5
VARIANZA DE LA VARIABLE Y	= 59.5

CUASIVARIANZA DE LA VARIABLE X	= 68.181818
CUASIVARIANZA DE LA VARIABLE Y	= 64.909091

COVARIANZA DE X,Y	= 53.75
CUASICOVARIANZA DE X,Y	= 58.636364

COEFICIENTE DE CORRELACION LINEAL	= .88141407
COEFICIENTE DE DETERMINACION	= .77689076

**3.** La temperatura que ha marcado el termómetro los diferentes días de la semana, ha sido:

D I A S	MINIMA	MAXIMA
Lunes	4	19
Martes	-2	18
Miercoles	-3	21
Jueves	1	13
Viernes	4	12
Sábado	0	14
Domingo	3	22

Calcular los parámetros estadísticos bidimensionales más importantes.

**S O L U C I O N :**

NUMERO DE PARES DE DATOS? 7

INTRODUZCA LA X y LA Y DE CADA PAR

X,Y DEL PUNTO 1 ? 4 19

X,Y DEL PUNTO 2 ? -2 18

X,Y DEL PUNTO 3 ? -3 21

X,Y DEL PUNTO 4 ? 1 13

X,Y DEL PUNTO 5 ? 4 12

X,Y DEL PUNTO 6 ? 0 14

X,Y DEL PUNTO 7 ? 3 22

MEDIA ARITMETICA DE LA VARIABLE X = 1

MEDIA ARITMETICA DE LA VARIABLE Y = 17

VARIANZA DE LA VARIABLE X = 6.8571429

VARIANZA DE LA VARIABLE Y = 13.714286

CUASIVARIANZA DE LA VARIABLE X = 8

CUASIVARIANZA DE LA VARIABLE Y = 16

COVARIANZA DE X,Y = -2.1428571

CUASICOVARIANZA DE X,Y = -2.5

COEFICIENTE DE CORRELACION LINEAL = -.22097087

COEFICIENTE DE DETERMINACION = .048828125

# PARAMETROS ESTADISTICOS DE DISTRIBUCIONES BIDIMENSIONALES DE DATOS AGRUPADOS

## introducción:

Este programa sirve para calcular parámetros estadísticos de distribuciones bidimensionales de datos agrupados. Tanto los valores de X e Y como la frecuencia absoluta del par (X, Y) se introducen bajo variables subindicadas en la línea 220. El número de pares de datos N con el que podremos operar y que dimensiona a las variables anteriores en la línea 140 dependerá en cada caso de la capacidad de la memoria de su micro.

## listado del programa:

```
10 REM *****
20 REM
30 REM          C A L C U L O   D E   P A R A M E T R O S
40 REM
50 REM          E S T A D I S T I C O S
60 REM
70 REM          D I S T R I B U C I O N E S   B I D I M E N S I O N A L E S   D E   D A T O S   A G R U P A D O S
80 REM
90 REM *****
100 PRINT
110 PRINT
120 PRINT "NUMERO DE PARES DE DATOS";
130 INPUT L
140 DIM X(L),Y(L),F(L),S(6)
150 FOR J=1 TO 6
160 S(J)=0
170 NEXT J
180 PRINT
189 REM-ENTRADA DE DATOS-
190 PRINT "INTRODUZCA LA X , LA Y, Y LA FRECUENCIA ABSOLUTA DE CADA PAR"
200 FOR I=1 TO L
210 PRINT "X,Y,F DEL PAR";I;
220 INPUT X(I),Y(I),F(I)
225 S(6)=S(6)+F(I)
230 S(1)=S(1)+F(I)*X(I)
240 S(2)=S(2)+F(I)*Y(I)
250 NEXT I
260 M1=S(1)/S(6)
```

```

270 M2=S(2)/S(6)
280 FOR I=1 TO L
290 S(3)=S(3)+F(I)*(X(I)-M1)^2
300 S(4)=S(4)+F(I)*(Y(I)-M2)^2
310 S(5)=S(5)+F(I)*(X(I)-M1)*(Y(I)-M2)
320 NEXT I
330 V1=S(3)/S(6)
340 V2=S(4)/S(6)
350 W1=S(3)/(S(6)-1)
360 W2=S(4)/(S(6)-1)
370 C1=S(5)/S(6)
380 C2=S(5)/(S(6)-1)
390 R=C1/SQR(V1*V2)
399 REM - SALIDA DE RESULTADOS -
400 PRINT
410 PRINT "MEDIA ARITMETICA DE LA VARIABLE X      =" ; M1
420 PRINT "MEDIA ARITMETICA DE LA VARIABLE Y      =" ; M2
430 PRINT
440 PRINT "VARIANZA DE LA VARIABLE X              =" ; V1
450 PRINT "VARIANZA DE LA VARIABLE Y              =" ; V2
460 PRINT
470 PRINT "CUASIVARIANZA DE LA VARIABLE X            =" ; W1
480 PRINT "CUASIVARIANZA DE LA VARIABLE Y            =" ; W2
490 PRINT
500 PRINT "COVARIANZA DE X,Y                            =" ; C1
510 PRINT "CUASICOVARIANZA DE X,Y                       =" ; C2
520 PRINT
530 PRINT "COEFICIENTE DE CORRELACION LINEAL             =" ; R
540 PRINT "COEFICIENTE DE DETERMINACION                 =" ; R^2
550 END

```

## ejercicios resueltos:

1. Se han estudiado 35 familias respecto al número de hijos y de hijas que tienen, obteniéndose los siguientes resultados:

n°de hijos	0	1	0	1	2	1	2	3	2	3
n°de hijas	0	0	1	1	1	2	2	2	3	3
n°de famil.	5	3	4	5	4	6	4	2	1	1

Calcular los parámetros estadísticos más importantes.

### SOLUCION :

NUMERO DE PARES DE DATOS? 10

INTRODUZCA LA X , LA Y. Y LA FRECUENCIA ABSOLUTA DE CADA PAR

X,Y,F DEL PAR 1 ? 0	0	5
X,Y,F DEL PAR 2 ? 1	0	3
X,Y,F DEL PAR 3 ? 0	1	4
X,Y,F DEL PAR 4 ? 1	1	5
X,Y,F DEL PAR 5 ? 2	1	4
X,Y,F DEL PAR 6 ? 1	2	6
X,Y,F DEL PAR 7 ? 2	2	4
X,Y,F DEL PAR 8 ? 3	2	2

X,Y,F DEL PAR 9 ? 2      3      1  
 X,Y,F DEL PAR 10 ? 3      3      1  
  
 MEDIA ARITMETICA DE LA VARIABLE X      = '1.1714286  
 MEDIA ARITMETICA DE LA VARIABLE Y      = 1.2285714  
  
 VARIANZA DE LA VARIABLE X      = .8277551  
 VARIANZA DE LA VARIABLE Y      = .7477551  
  
 CUASIVARIANZA DE LA VARIABLE X      = .85210084  
 CUASIVARIANZA DE LA VARIABLE Y      = .7697479  
  
 COVARIANZA DE X,Y      = .50367347  
 CUASICOVARIANZA DE X,Y      = .51848739  
  
 COEFICIENTE DE CORRELACION LINEAL      = .6402041  
 COEFICIENTE DE DETERMINACION      = .40986129

**2.** Las calificaciones de 22 alumnos en Matemáticas y Física han sido las siguientes:

MATEMAT.	5	3	3	5	7	9	1
FISICA	5	5	3	7	7	9	3
nºalumnos	6	4	4	3	2	1	2

Calcular los parámetros estadísticos más importantes.

### **S O L U C I O N :**

NUMERO DE PARES DE DATOS? 7

INTRODUZCA LA X , LA Y, Y LA FRECUENCIA ABSOLUTA DE CADA PAR

X,Y,F DEL PAR 1 ? 5      5      6  
 X,Y,F DEL PAR 2 ? 3      5      4  
 X,Y,F DEL PAR 3 ? 3      3      4  
 X,Y,F DEL PAR 4 ? 5      7      3  
 X,Y,F DEL PAR 5 ? 7      7      2  
 X,Y,F DEL PAR 6 ? 9      9      1  
 X,Y,F DEL PAR 7 ? 1      3      2

MEDIA ARITMETICA DE LA VARIABLE X      = 4.2727273  
 MEDIA ARITMETICA DE LA VARIABLE Y      = 5.0909091

VARIANZA DE LA VARIABLE X      = 3.4710744  
 VARIANZA DE LA VARIABLE Y      = 2.7190083

CUASIVARIANZA DE LA VARIABLE X      = 3.6363636  
 CUASIVARIANZA DE LA VARIABLE Y      = 2.8484848

COVARIANZA DE X,Y      = 2.6115702  
 CUASICOVARIANZA DE X,Y      = 2.7359307

COEFICIENTE DE CORRELACION LINEAL      = .85008918  
 COEFICIENTE DE DETERMINACION      = .72265161

# PROBABILIDAD TOTAL

## introducción:

Este programa calcula la probabilidad total de un suceso B conocidas las probabilidades a priori  $p(A_i)$  de cada uno de los sucesos que componen el sistema completo de sucesos del experimento y las verosimilitudes  $p(B/A_i)$ .

El teorema de la probabilidad total establece que:

$$p(B) = \sum_{i=1}^n p(A_i) \cdot p(B/A_i)$$

## listado del programa:

```
10 REM *****
20 REM
30 REM               P R O B A B I L I D A D       T O T A L
40 REM
50 REM *****
100 PRINT
110 PRINT "NUMERO DE SUCECOS QUE COMPONEN EL SISTEMA COMPLETO";
120 INPUT N
130 PRINT
140 DIM P(N),B(N)
150 S=0
159 REM - ENTRADA DE DATOS -
160 PRINT "INTRODUZCA LA PROBABILIDAD A PRIORI P(Ai), Y LA PROBABILIDAD DEL SU-
CESO B CONDICIONADA A Ai, P(B/Ai)"
169 REM
170 PRINT
180 FOR I=1 TO N
190 PRINT "P(A";I;")   =";
200 INPUT P(I)
209 REM - SALIDA DE RESULTADOS -
210 PRINT "P(B/A";I;") =";
220 INPUT B(I)
230 PRINT
240 S=S+P(I)*B(I)
250 NEXT I
259 REM - SALIDA DE RESULTADOS -
260 PRINT
```



```
270 PRINT "LA PROBABILIDAD TOTAL DEL SUCESO B ES =";S
280 END
```

## ejercicios resueltos:

**1.** Un sistema completo de sucesos de un experimento aleatorio está formado por 6 sucesos de los que se sabe:

$$\begin{array}{ll|ll} p(A_1) = 0,2 & p(B/A_1) = 0,24 & p(A_4) = 0,1 & p(B/A_4) = 0,2 \\ p(A_2) = 0,3 & p(B/A_2) = 0,65 & p(A_5) = 0,1 & p(B/A_5) = 0,4 \\ p(A_3) = 0,2 & p(B/A_3) = 0,35 & p(A_6) = 0,1 & p(B/A_6) = 0,25 \end{array}$$

Se desea conocer la probabilidad total del suceso B.

### SOLUCION :

NUMERO DE SUCECOS QUE COMPONEN EL SISTEMA COMPLETO? 6

INTRODUZCA LA PROBABILIDAD A PRIORI  $P(A_i)$ , Y LA PROBABILIDAD DEL SUCESO B CONDICIONADA A  $A_i$ ,  $P(B/A_i)$

$P(A_1) = ? .2$   
 $P(B/A_1) = ? .24$

$P(A_2) = ? .3$   
 $P(B/A_2) = ? .65$

$P(A_3) = ? .2$   
 $P(B/A_3) = ? .35$

$P(A_4) = ? .1$   
 $P(B/A_4) = ? .2$

$P(A_5) = ? .1$   
 $P(B/A_5) = ? .4$

$P(A_6) = ? .1$   
 $P(B/A_6) = ? .25$

LA PROBABILIDAD TOTAL DEL SUCESO B ES = .398

**2.** En una población y en un determinado momento, se conocen las probabilidades de que un individuo tenga las enfermedades A, B y C y de que esté sano, siendo estas las siguientes:

$$p(A) = 0,02 \quad p(B) = 0,05 \quad p(C) = 0,03 \quad p(S) = 0,9$$

Consideremos un síntoma determinado representado por el símbolo D. Las probabilidades de que se presente dicho síntoma en cada caso anterior son:  $p(D/A) = 0,99$   $p(D/B) = 0,1$   $p(D/C) = 0,4$   $p(D/S) = 0,01$

Hallar la probabilidad de que se presente el síntoma D

### SOLUCION :

NUMERO DE SUCESOS QUE COMPONEN EL SISTEMA COMPLETO? 4

INTRODUZCA LA PROBABILIDAD A PRIORI  $P(A_i)$ , Y LA PROBABILIDAD DEL SUCESO B CONDICIONADA A  $A_i$ ,  $P(B/A_i)$

$P(A_1) = ? .02$   
 $P(D/A_1) = ? .99$

$P(A_2) = ? .05$   
 $P(D/A_2) = ? .1$

$P(A_3) = ? .03$   
 $P(D/A_3) = ? .4$

$P(A_4) = ? .9$   
 $P(D/A_4) = ? .01$

LA PROBABILIDAD TOTAL DEL SUCESO D ES = .0458

**3.** Un sistema completo de un experimento aleatorio está formado por los sucesos  $A_1$ ,  $A_2$ , y  $A_3$  tales que sus probabilidades son:  $p(A_1) = 0,3$ ,  $p(A_2) = 0,2$  y  $p(A_3) = 0,5$ . Sea B un suceso cualquiera para el que se conocen las probabilidades:  $p(B/A_1) = 0,25$ ,  $p(B/A_2) = 0,5$ ,  $p(B/A_3) = 0,5$ . Hallar la probabilidad total del suceso B.

**S O L U C I O N :**

NUMERO DE SUCESOS QUE COMPONEN EL SISTEMA COMPLETO? 3

INTRODUZCA LA PROBABILIDAD A PRIORI  $P(A_i)$ , Y LA PROBABILIDAD DEL SUCESO B CONDICIONADA A  $A_i$ ,  $P(B/A_i)$

$P(A_1) = ? .3$   
 $P(B/A_1) = ? .25$

$P(A_2) = ? .2$   
 $P(B/A_2) = ? .5$

$P(A_3) = ? .5$   
 $P(B/A_3) = ? .5$

LA PROBABILIDAD TOTAL DEL SUCESO B ES = .425

# PROBABILIDAD BAYESIANA

## introducción:

Este programa calcula las probabilidades a posteriori que se obtienen como aplicación del teorema de Bayes, para ello es necesario introducir en la línea 200 las probabilidades a priori  $p(A_i)$  y en la línea 220 las verosimilitudes  $p(B/A_i)$ .

El teorema de Bayes establece que:

$$p(A_i/B) = \frac{p(A_i) p(B/A_i)}{\sum_{i=1}^n p(A_i) p(B/A_i)}$$

## listado del programa:

```
10 REM *****
20 REM
30 REM          P R O B A B I L I D A D          B A Y E S I A N A
40 REM
50 REM *****
100 PRINT
110 PRINT "NUMERO DE SUCEOS QUE COMPONEN EL SISTEMA COMPLETO";
120 INPUT N
130 PRINT
140 DIM P(N),B(N)
150 S=0
159 REM - ENTRADA DE DATOS -
160 PRINT "INTRODUZCA LA PROBABILIDAD A PRIORI P(Ai), Y LA PROBAEILIDAD DEL SU-
CESO B CONDICIONADA A Ai, P(B/Ai)"
169 REM
170 PRINT
180 FOR I=1 TO N
190 PRINT "P(A";I;")  =";
200 INPUT P(I)
210 PRINT "P(B/A";I;") = " ;
220 INPUT B(I)
230 PRINT
240 S=S+P(I)*B(I)
250 NEXT I
259 REM - SALIDA DE RESULTADOS -
260 PRINT
```

```

270 PRINT "PROBABILIDADES A POSTERIORI"
280 PRINT
290 REM - OBTENIDAS MEDIANTE EL TEOREMA DE BAYES -
300 FOR I=1 TO N
310 PRINT "LA PROBABILIDAD DE (A";I;"/B) =" ;P(I)*B(I)/S
320 NEXT I
330 END

```

## ejercicios resueltos:

1. En una población se conocen las probabilidades de que un individuo tenga las enfermedades A, B, y C y de que esté sano. siendo estas las siguientes:  $p(A) = 0,02$ ;  $p(B) = 0,05$ ;  $p(C) = 0,03$ ;  $p(S) = 0,9$

Consideremos un síntoma, representado por el símbolo D. Las probabilidades de que se presente dicho síntoma en cada caso son:

$p(D/A) = 0,99$ ;  $p(D/B) = 0,4$ ;  $p(D/C) = 0,1$ ;  $p(D/S) = 0,01$

Examinado un individuo, se observa que presenta el citado síntoma. Hallar las probabilidades de que padezcan cada una de las tres enfermedades o de que esté sano.

### SOLUCION :

NUMERO DE SUCECOS QUE COMPONEN EL SISTEMA COMPLETO? 4

INTRODUZCA LA PROBABILIDAD A PRIORI  $P(A_i)$ , Y LA PROBABILIDAD DEL SUCECO B CONDICIONADA A  $A_i$ ,  $P(B/A_i)$

$P(A_1) = ? .02$   
 $P(B/A_1) = ? .99$

$P(A_2) = ? .05$   
 $P(B/A_2) = ? .4$

$P(A_3) = ? .03$   
 $P(B/A_3) = ? .1$

$P(A_4) = ? .9$   
 $P(B/A_4) = ? .01$

PROBABILIDADES A POSTERIORI

LA PROBABILIDAD DE  $(A_1/B) = .38223938$   
 LA PROBABILIDAD DE  $(A_2/B) = .38610039$   
 LA PROBABILIDAD DE  $(A_3/B) = .057915058$   
 LA PROBABILIDAD DE  $(A_4/B) = .17374517$

2. En un hospital especializado en enfermedades del torax, ingresan un promedio de 50% de enfermos con bronquitis, 30% con neumonía y 20% con gripe. La probabilidad de curación completa de cada una de estas enfermedades

es respectivamente 0,7; 0,8; 0,9.

Un enfermo ingresado en el hospital ha sido dado de alta completamente sano. Hallar la probabilidad de que el enfermo ingresara con bronquitis.

### SOLUCION :

NUMERO DE SUCESOS QUE COMPONEN EL SISTEMA COMPLETO? 3

INTRODUZCA LA PROBABILIDAD A PRIORI  $P(A_i)$ , Y LA PROBABILIDAD DEL SUCESO B CONDICIONADA A  $A_i, P(B/A_i)$

$P(A_1) = ? .5$   
 $P(B/A_1) = ? .7$

$P(A_2) = ? .3$   
 $P(B/A_2) = ? .8$

$P(A_3) = ? .2$   
 $P(B/A_3) = ? .9$

PROBABILIDADES A POSTERIORI

LA PROBABILIDAD DE  $(A_1/B) = .45454545$   
 LA PROBABILIDAD DE  $(A_2/B) = .31168831$   
 LA PROBABILIDAD DE  $(A_3/B) = .23376623$

**3.** Se cree que un individuo ha podido contraer las enfermedades A o B con probabilidades 0,25 y 0,4 respectivamente.

Para establecer un diagnostico diferencial, se somete al paciente a un análisis clínico del que solo se pueden obtener dos resultados positivo o negativo. Por la experiencia acumulada en otros casos análogos, se sabe que la probabilidad de que el análisis resulte positivo para los que padecen la enfermedad A es 0,04 y 0,07 para los que padecen la enfermedad B.

Al enfermo se le hace el análisis y da resultado positivo. ¿Cuál es la probabilidad de que el paciente tuviera la enfermedad A y cuál es la probabilidad de que tuviera la enfermedad B.

### SOLUCION :

NUMERO DE SUCESOS QUE COMPONEN EL SISTEMA COMPLETO? 2

INTRODUZCA LA PROBABILIDAD A PRIORI  $P(A_i)$ , Y LA PROBABILIDAD DEL SUCESO B CONDICIONADA A  $A_i, P(B/A_i)$

$P(A_1) = ? .25$   
 $P(B/A_1) = ? .04$

$P(A_2) = ? .4$   
 $P(B/A_2) = ? .07$

## PROBABILIDADES A POSTERIORI

LA PROBABILIDAD DE  $(A_1 / B) = .26315789$ LA PROBABILIDAD DE  $(A_2 / B) = .73684211$ 

4. Un producto consta de 6 partes del siguiente modo: 15% de la parte  $A_1$ , 25% de la parte  $A_2$ , 20% de la parte  $A_3$ , 25% de la parte  $A_4$ , 0,05% de la parte  $A_5$ , y 0,1% de la parte  $A_6$ . El laboratorio de control de calidad informa que, después de muchos análisis, se ha determinado que la probabilidad de que se produzca un fallo en la parte  $A_1$  es 0,3, en la parte  $A_2$  es 0,4, en la parte  $A_3$  es 0,5, en la parte  $A_4$  es 0,3, en la parte  $A_5$  es 0,25 y en la parte  $A_6$  es 0,1.

Elegida una pieza al azar se analiza y se observa que está defectuosa. Hallar la probabilidad de que el defecto proceda de cada una de las partes.

**SOLUCION :**

NUMERO DE SUCEOS QUE COMPONEN EL SISTEMA COMPLETO? 6

INTRODUZCA LA PROBABILIDAD A PRIORI  $P(A_i)$ , Y LA PROBABILIDAD DEL SUCESO B CONDICIONADA A  $A_i$ ,  $P(B/A_i)$

$P(A_1) = ? .15$   
 $P(B/A_1) = ? .3$

$P(A_2) = ? .25$   
 $P(B/A_2) = ? .4$

$P(A_3) = ? .2$   
 $P(B/A_3) = ? .5$

$P(A_4) = ? .25$   
 $P(B/A_4) = ? .3$

$P(A_5) = ? .05$   
 $P(B/A_5) = ? .25$

$P(A_6) = ? .1$   
 $P(B/A_6) = ? .1$

## PROBABILIDADES A POSTERIORI

LA PROBABILIDAD DE  $(A_1 / B) = .13138686$   
 LA PROBABILIDAD DE  $(A_2 / B) = .2919708$   
 LA PROBABILIDAD DE  $(A_3 / B) = .2919708$   
 LA PROBABILIDAD DE  $(A_4 / B) = .2189781$   
 LA PROBABILIDAD DE  $(A_5 / B) = .03649635$   
 LA PROBABILIDAD DE  $(A_6 / B) = .02919708$

# DISTRIBUCION BINOMIAL

## introducción:

Este programa permite calcular las probabilidades de la distribución binomial, para lo cual es necesario conocer el número de pruebas realizadas  $n$ , el número de exitos deseados  $r$ , ( $0 \leq r \leq n$ ) y la probabilidad del suceso exito  $p$ , ( $0 \leq p \leq 1$ ).

La ley de probabilidad de la distribución binomial viene dada por la expresión:

$$p(X = r) = \binom{n}{r} p^r q^{n-r} \quad \text{siendo } q = 1 - p$$

El programa obtiene:

- La probabilidad puntual de obtener  $r$  exitos en  $n$  pruebas.
- La probabilidad de obtener un número de exitos  $\leq r$  en  $n$  pruebas.
- La probabilidad de obtener un número de exitos  $\geq r$  en  $n$  pruebas.

## listado del programa:

```
10 REM *****
20 REM
30 REM          D I S T R I B U C I O N          B I N O M I A L
40 REM
50 REM *****
100 PRINT
109 REM - ENTRADA DE DATOS -
110 PRINT "NUMERO DE PRUEBAS REALIZADAS  =";
120 INPUT N
130 PRINT "NUMERO DE EXITOS DESEADOS    =";
140 INPUT X
150 IF X>N THEN GOTO 400
160 PRINT "PROBABILIDAD DEL SUCEO EXITO =";
170 INPUT P
179 REM - CALCULO DE LAS PROBABILIDADES -
180 Q=1-P
190 FX=Q^N
200 LX=FX
210 U=P/Q
220 M=N+1
230 IF X=0 THEN GOTO 280
```

```

240 FOR I=1 TO X
250 FX=(M-I)*U*FX/I
260 LX=LX+FX
270 NEXT I
280 RX=1+FX-LX
289 REM -SALIDA DE RESULTADOS -
290 PRINT
300 PRINT "LA PROBABILIDAD DE OBTENER";X;"EXITOS EN";N;"PRUEBAS ES" =";FX
310 PRINT "LA PROBABILIDAD DE OBTENER <=";X;"EXITOS EN";N;"PRUEBAS ES" =";LX
320 PRINT "LA PROBABILIDAD DE OBTENER >=";X;"EXITOS EN";N;"PRUEBAS ES" =";RX
330 PRINT
339 REM - MAS CALCULOS -
340 PRINT "DESEA REALIZAR MAS CALCULOS PARA OTRO VALOR DE X? S/N";
350 INPUT C$
360 IF C$="N" THEN GOTO 410
370 PRINT "X =";
380 INPUT X
390 GOTO 180
400 PRINT "IMPOSIBLE N HA DE SER MAYOR O IGUAL QUE X"
410 END

```

## ejercicios resueltos:

1. La aplicación de un determinado tratamiento a enfermos de cirrosis produce una cierta mejoría en el 70% de los casos. Si se aplica el tratamiento a 10 enfermos cirróticos, calcular:

- ¿Cuál es la probabilidad de que mejoren 4?
- ¿Cuál es la probabilidad de que mejoren 5?

### S O L U C I O N :

```

NUMERO DE PRUEBAS REALIZADAS  =? 10
NUMERO DE EXITOS DESEADOS      =? 4
PROBABILIDAD DEL SUCESO EXITO  =? .7

```

```

LA PROBABILIDAD DE OBTENER 4 EXITOS EN 10 PRUEBAS ES  = .036756909
LA PROBABILIDAD DE OBTENER <= 4 EXITOS EN 10 PRUEBAS ES = .047348987
LA PROBABILIDAD DE OBTENER >= 4 EXITOS EN 10 PRUEBAS ES = .98940792

```

```

DESEA REALIZAR MAS CALCULOS PARA OTRO VALOR DE X? S/N?S
X =? 5

```

```

LA PROBABILIDAD DE OBTENER 5 EXITOS EN 10 PRUEBAS ES  = .10291935
LA PROBABILIDAD DE OBTENER <= 5 EXITOS EN 10 PRUEBAS ES = .15026833
LA PROBABILIDAD DE OBTENER >= 5 EXITOS EN 10 PRUEBAS ES = .95265101

```

2. La probabilidad de que un estudiante obtenga el título de Licenciado en Ciencias Biológicas es 0,3.

Calcular la probabilidad de que de un grupo de 7 estudiantes:

- Finalicen los 7.
- Finalicen un número menor o igual a 7
- Finalicen un número mayor o igual a 7



Análogo para 0

### **SOLUCION :**

NUMERO DE PRUEBAS REALIZADAS =? 7  
 NUMERO DE EXITOS DESEADOS =? 7  
 PROBABILIDAD DEL SUCESO EXITO =? .3

LA PROBABILIDAD DE OBTENER 7 EXITOS EN 7 PRUEBAS ES = .0002187  
 LA PROBABILIDAD DE OBTENER <= 7 EXITOS EN 7 PRUEBAS ES = 1  
 LA PROBABILIDAD DE OBTENER >= 7 EXITOS EN 7 PRUEBAS ES = .00021869952

DESEA REALIZAR MAS CALCULOS PARA OTRO VALOR DE X? S/N?S  
 X =? 0

LA PROBABILIDAD DE OBTENER 0 EXITOS EN 7 PRUEBAS ES = .0823543  
 LA PROBABILIDAD DE OBTENER <= 0 EXITOS EN 7 PRUEBAS ES = .0823543  
 LA PROBABILIDAD DE OBTENER >= 0 EXITOS EN 7 PRUEBAS ES = 1

DESEA REALIZAR MAS CALCULOS PARA OTRO VALOR DE X? S/N?N

**3.** La probabilidad de que un individuo vacunado contra una cierta enfermedad la contraiga es 0,2.

De un grupo de 8 individuos vacunados, ¿cuál es la probabilidad de que:

- Contraiga la enfermedad un solo individuo.
- Contraiga la enfermedad a lo sumo uno.
- Contraiga la enfermedad por lo menos uno.

Análogo para 2 individuos y para 8 individuos.

### **SOLUCION :**

NUMERO DE PRUEBAS REALIZADAS =? 8  
 NUMERO DE EXITOS DESEADOS =? 1  
 PROBABILIDAD DEL SUCESO EXITO =? .2

LA PROBABILIDAD DE OBTENER 1 EXITOS EN 8 PRUEBAS ES = .33554432  
 LA PROBABILIDAD DE OBTENER <= 1 EXITOS EN 8 PRUEBAS ES = .50331648  
 LA PROBABILIDAD DE OBTENER >= 1 EXITOS EN 8 PRUEBAS ES = .83222784

DESEA REALIZAR MAS CALCULOS PARA OTRO VALOR DE X? S/N?S  
 X =? 2

LA PROBABILIDAD DE OBTENER 2 EXITOS EN 8 PRUEBAS ES = .29360128  
 LA PROBABILIDAD DE OBTENER <= 2 EXITOS EN 8 PRUEBAS ES = .79691776  
 LA PROBABILIDAD DE OBTENER >= 2 EXITOS EN 8 PRUEBAS ES = .49668352

DESEA REALIZAR MAS CALCULOS PARA OTRO VALOR DE X? S/N?S  
 X =? 8

LA PROBABILIDAD DE OBTENER 8 EXITOS EN 8 PRUEBAS ES = 2.56E-06  
 LA PROBABILIDAD DE OBTENER <= 8 EXITOS EN 8 PRUEBAS ES = 1  
 LA PROBABILIDAD DE OBTENER >= 8 EXITOS EN 8 PRUEBAS ES = 2.5599729E-06

DESEA REALIZAR MAS CALCULOS PARA OTRO VALOR DE X? S/N?N

# DISTRIBUCION DE POISSON

## introducción:

Este programa permite calcular las probabilidades de la distribución de Poisson, para lo cual es necesario conocer el parámetro  $m$  de la distribución de Poisson, así como el valor de la variable aleatoria discreta para la que se han de efectuar los cálculos.

La ley de probabilidad de la distribución de Poisson viene dada por la siguiente expresión:

$$p(X = r) = \frac{m^r}{r!} e^{-m}$$

El programa permite hallar:

- La probabilidad puntual de que la variable aleatoria tome el valor  $r$
- La probabilidad de que la variable aleatoria tome a lo sumo el valor  $r$ .
- La probabilidad de que la variable aleatoria tome por lo menos el valor  $r$ .

## listado del programa:

```
10 REM *****
20 REM
30 REM          D I S T R I B U C I O N          D E          P O I S S O N
40 REM
50 REM *****
100 PRINT
109 REM - ENTRADA DE DATOS -
110 PRINT "INTRODUZCA EL PARAMETRO M DE LA DISTRIBUCION DE POISSON ";
120 INPUT M
130 PRINT "VALOR DE LA VARIABLE ALEATORIA X PARA EL QUE SE DESEA EL CALCULO ";
140 INPUT X
149 REM - CALCULO DE LAS PROBABILIDADES -
150 FX=EXP(-M)
160 LX=FX
170 IF X=0 THEN 220
180 FOR I=1 TO X
190 FX=FX*M/I
200 LX=LX+FX
210 NEXT I
220 RX=1+FX-LX
```

```

230 PRINT
239 REM - SALIDA DE RESULTADOS -
240 PRINT
250 PRINT "PROBABILIDAD DE QUE LA VARIABLE ALEATORIA SEA =" ; X ; "ES " ; FX
260 PRINT "PROBABILIDAD DE QUE LA VARIABLE ALEATORIA SEA <=" ; X ; "ES " ; LX
270 PRINT "PROBABILIDAD DE QUE LA VARIABLE ALEATORIA SEA >=" ; X ; "ES " ; RX
280 PRINT
289 REM - MAS CALCULOS -
290 PRINT "DESEA HACER CALCULOS PARA OTRO VALOR DE X ? S/N";
300 INPUT C$
310 IF C$="N" THEN GOTO 360
320 PRINT
330 PRINT "X =";
340 INPUT X
350 GOTO 150
360 END

```

## ejercicios resueltos:

1. Se supone que en un determinado país la proporción de individuos albinos es de 0,005. Calcular la probabilidad de que elegida una muestra de la citada población de tamaño 1000, se presenten los siguientes casos:

- Ningun individuo sea albino.
- Tres individuos sean albinos.
- A lo sumo haya tres individuos albinos.
- Por lo menos haya tres individuos albinos.

### S O L U C I O N :

Al ser  $p = 0,005$  y  $n = 1000$  se cumple que  $m = n.p = 5$ . Luego el parámetro de la distribución es 5.

INTRODUZCA EL PARAMETRO M DE LA DISTRIBUCION DE POISSON =? 5  
 VALOR DE LA VARIABLE ALEATORIA X PARA EL QUE SE DESEA EL CALCULO =? 0

```

PROBABILIDAD DE QUE LA VARIABLE ALEATORIA SEA = 0 ES .006737947
PROBABILIDAD DE QUE LA VARIABLE ALEATORIA SEA <= 0 ES .006737947
PROBABILIDAD DE QUE LA VARIABLE ALEATORIA SEA >= 0 ES 1

```

DESEA HACER CALCULOS PARA OTRO VALOR DE X ? S/N?S

X =? 3

```

PROBABILIDAD DE QUE LA VARIABLE ALEATORIA SEA = 3 ES .1403739
PROBABILIDAD DE QUE LA VARIABLE ALEATORIA SEA <= 3 ES .26502592
PROBABILIDAD DE QUE LA VARIABLE ALEATORIA SEA >= 3 ES .87534798

```

DESEA HACER CALCULOS PARA OTRO VALOR DE X ? S/N?N

2. La probabilidad de que se produzca un choque anafiláctico al suministro de un suero en un individuo es de 0,002. Sea X la variable aleato-

ría que representa el número de choques anafilácticos en un grupo de 1200 enfermos a los que se les ha suministrado el suero. Calcular:

- Probabilidad de que se produzcan 5 choques anafilácticos.
  - Probabilidad de que se produzcan a lo sumo 5 choques anafilácticos.
  - Probabilidad de que se produzcan por lo menos 5 choques anafilácticos.
- Análogo para 7 choques.

### **S O L U C I O N :**

La variable X sigue trivialmente una distribución binomial. Ahora bien, al ser  $n = 1200$  muy grande y  $p = 0,02$  muy pequeño, obtenemos una buena aproximación de la distribución binomial mediante la distribución de Poisson de parámetro:  $m = n.p = 1200.0,02 = 2,4$

INTRODUZCA EL PARAMETRO M DE LA D

ISTRIBUCION DE POISSON =? 2.4

VALOR DE LA VARIABLE ALEATORIA X PARA EL QUE SE DESEA EL CALCULO =? 5

PROBABILIDAD DE QUE LA VARIABLE ALEATORIA SEA = 5 ES .060196079

PROBABILIDAD DE QUE LA VARIABLE ALEATORIA SEA <= 5 ES .96432749

PROBABILIDAD DE QUE LA VARIABLE ALEATORIA SEA >= 5 ES .09586859

DESEA HACER CALCULOS PARA OTRO VALOR DE X ? S/N?S

X =? 7

PROBABILIDAD DE QUE LA VARIABLE ALEATORIA SEA = 7 ES .0082554623

PROBABILIDAD DE QUE LA VARIABLE ALEATORIA SEA <= 7 ES .99666138

PROBABILIDAD DE QUE LA VARIABLE ALEATORIA SEA >= 7 ES .011594079

DESEA HACER CALCULOS PARA OTRO VALOR DE X ? S/N?N

**3.** Un almacén farmacéutico recibió 1000 botellines de suero fisiológico. La probabilidad de que al transportar los botellines se rompa uno es igual a 0,003. Hallar la probabilidad de que al desembalar los botellines, resulten:

- Haya exactamente 2 rotos.
- Haya a lo sumo 2 rotos.
- Haya por lo menos 2 rotos.

### **S O L U C I O N :**

El parametro de la distribución de Poisson es:  $m = n.p = 1000.0,003 = 3$

INTRODUZCA EL PARAMETRO M DE LA DISTRIBUCION DE POISSON =? 3

VALOR DE LA VARIABLE ALEATORIA X PARA EL QUE SE DESEA EL CALCULO =? 2

PROBABILIDAD DE QUE LA VARIABLE ALEATORIA SEA = 2 ES .22404181

PROBABILIDAD DE QUE LA VARIABLE ALEATORIA SEA <= 2 ES .42319008

PROBABILIDAD DE QUE LA VARIABLE ALEATORIA SEA >= 2 ES .80085173

DESEA HACER CALCULOS PARA OTRO VALOR DE X ? S/N?N

# DISTRIBUCION HIPERGEOMETRICA

## introducción:

Este programa calcula las probabilidades de la distribución hipergeométrica, para ello es necesario conocer los tres parámetros que caracterizan a esta distribución: el tamaño de la población  $N$ , el tamaño de la muestra  $s$  y la probabilidad del suceso exito  $p$ .

La ley de probabilidad de la distribución hipergeométrica viene dada por la siguiente expresión:

$$p(X = r) = \frac{\binom{Np}{r} \binom{Nq}{s-r}}{\binom{N}{s}}$$

siendo  $Np$  y  $Nq=N(1-p)$  números enteros.

## listado del programa:

```
10 REM *****
20 REM
30 REM          D I S T R I B U C I O N      H I P E R G E O M E T R I C A
40 REM
50 REM *****
100 PRINT
109 REM - ENTRADA DE DATOS -
110 PRINT "TAMANO DE LA POBLACION          =";
120 INPUT N
130 PRINT "TAMANO DE LA MUESTRA            =";
140 INPUT S
150 PRINT "PROBABILIDAD DEL SUCESO EXITO =";
160 INPUT P
170 PRINT "NUMERO X DE EXITOS DESEADOS    =";
180 INPUT X
190 A=N*P
200 B=X
210 GOSUB 400
220 U=C
230 A=N*(1-P)
240 B=S-X
250 GOSUB 400
260 V=C
```

```

270 A=N
280 B=S
290 GOSUB 400
300 W=C
310 T=U*V/W
319 REM - SALIDA DE RESULTADOS -
320 PRINT
330 PRINT "LA PROBABILIDAD DE QUE OCURRAN"; X ;"EXITOS ES =" ;T
333 PRINT
339 REM - MAS CALCULOS -
340 PRINT "DESEA REALIZAR MAS CALCULOS PARA OTRO VALOR DE X? S/N";
350 INPUT C$
360 IF C$="N" THEN GOTO 460
370 PRINT "X =";
380 INPUT X
390 GOTO 190
400 REM - SUBROUTINA CALCULO DE NUMEROS COMBINATORIOS -
410 IF B=0 THEN C=1:GOTO 450
420 FOR J=1 TO B
430 C=C*(A-J+1)/J
440 NEXT J
450 RETURN
460 END

```

## ejercicios resueltos:

1. Por prescripción facultativa un enfermo debe hacer una toma de tres píldoras de un determinado medicamento. De las 25 píldoras que contiene el envase 5 están en malas condiciones. ¿Cuál es la probabilidad de que de las tres píldoras de la toma, una se encuentre en malas condiciones?.

### SOLUCION :

TAMANO DE LA POBLACION	=? 25
TAMANO DE LA MUESTRA	=? 3
PROBABILIDAD DEL SUCESO EXITO	=? .2
NUMERO X DE EXITOS DESEADOS	=? 1

LA PROBABILIDAD DE QUE OCURRAN 1 EXITOS ES = .41304348

DESEA REALIZAR MAS CALCULOS PARA OTRO VALOR DE X? S/N?N

2. Una urna de votación contiene 200 papeletas, de las que 160 son para el candidato A y 40 son para el candidato B, Se efectuan tres extracciones consecutivas sin reemplazamiento de la urna, calcular, ¿cuál es la probabilidad de que las tres papeletas sean para el candidato A?.

### SOLUCION :

La variable que representa el número de votos conseguidos por el candidato A sigue una distribución hipergeométrica de parámetros:

$N = 200$ ,  $n = 3$ ,  $p = \frac{160}{200} = 0.8$ ,  $Np = 160$

TAMAÑO DE LA POBLACION? 200  
TAMAÑO DE LA MUESTRA? 3  
PROBABILIDAD DEL SUCESO EXITO? .8  
NUMERO DE EXITOS DESEADOS? 3  
LA PROBABILIDAD DE QUE OCURRAN 3 EXITOS ES = .51006548

**3.** Una jaula de laboratorio contiene 15 cobayas, de las que 6 son blancas y 9 pardas. Un ayudante de laboratorio con los ojos vendados extrae sin reemplazamiento 4 cobayas de la jaula.

Calcular la probabilidad de que ninguna sea blanca.

**S O L U C I O N :**

TAMAÑO DE LA POBLACION? 15  
TAMAÑO DE LA MUESTRA? 4  
PROBABILIDAD DEL SUCESO EXITO? .4  
NUMERO DE EXITOS DESEADOS? 0  
LA PROBABILIDAD DE QUE OCURRAN 0 EXITOS ES = .092307692

# DISTRIBUCION NORMAL

## introducción:

Este programa sirve para calcular las probabilidades de la distribución normal de Gauss. Para poder aplicarlo es necesario conocer los parámetros que la caracterizan: media y desviación típica. En el supuesto de que no se trate de la normal standard  $N(0,1)$  en la línea 270 se realiza la tipificación de la variable.

La función de densidad de la distribución normal  $N(m,s)$ , viene dada por la expresión :

$$f(x) = \frac{1}{s \sqrt{2\pi}} \exp\left(-\frac{1}{2} \left(\frac{x-m}{s}\right)^2\right)$$

El programa permite calcular:

- a) La ordenada de la función de densidad en el punto  $x$ .
- b) La probabilidad de que la variable aleatoria tome a lo sumo el valor  $x$ .
- c) La probabilidad de que la variable aleatoria tome por lo menos el valor  $x$ .

## listado del programa:


```
10 REM *****
20 REM
30 REM           D I S T R I B U C I O N       N O R M A L
40 REM
50 REM *****
100 PRINT
110 DIM A(4)
120 REM - ENTRADA DE DATOS -
130 PRINT "MEDIA M =";
140 INPUT M
150 PRINT "DESVIACION TIPICA =";
160 INPUT S
170 PRINT "VALOR DE LA VARIABLE X PARA LA QUE SE DESEA EL CALCULO, X =";
180 INPUT X
190 REM -COEFICIENTES A(I)-
200 A(0)=1.330274429
210 A(1)=-1.821255978
220 A(2)=1.781477937
```



```

230 A(3)=-.356563782
240 A(4)=.319381530
250 B=.2316419
260 REM - TIPIFICACION DE LA VARIABLE -
270 Y=ABS((X-M)/S)
280 Z=EXP(-Y*Y/2)/2.506628275
290 FX=Z/S
300 C=1+B*Y
310 D=1/C
320 HO =0
330 FOR I=0 TO 4
340 HO=HO*D+A(I)
350 NEXT I
360 HO=HO*D
370 RX=Z*HO
380 LX=1-RX
389 REM - SALIDA DE RESULTADOS -
390 PRINT
400 PRINT "ORDENADA DE LA FUNCION DE DENSIDAD EN EL PUNTO =" ;X;"ES " ;FX
410 PRINT "PROBABILIDAD DE QUE LA VARIABLE ALEATORIA SEA <=" ;X;"ES " ;LX
420 PRINT "PROBABILIDAD DE QUE LA VARIABLE ALEATORIA SEA >=" ;X;"ES " ;RX
430 PRINT
440 PRINT
450 PRINT "DESEA REALIZAR MAS CALCULOS PARA OTRO VALOR DE X? S/N";
460 INPUT C$
470 IF C$="N" THEN GOTO 520
480 PRINT
490 PRINT "X = ";
500 INPUT X
510 GOTO 190
520 END

```



## ejercicios resueltos:

1. La duración media, sin ninguna rotura, de las sábanas de un Sanatorio Psiquiátrico de 300 camas, es de 50 días con una desviación típica de 8 días. Si suponemos que la duración de las sábanas sigue una distribución normal, calcular:

- ¿Cuál es la probabilidad de que una sábana dure a lo sumo 35 días?.
  - ¿Cuál es la probabilidad de que una sábana dure por lo menos 35 días?.
- Análogo para 60 días.

### SOLUCION :

```

MEDIA M =? 50
DESVIACION TIPICA =? 8
VALOR DE LA VARIABLE X PARA LA QUE SE DESEA EL CALCULO, X =? 35

ORDENADA DE LA FUNCION DE DENSIDAD EN EL PUNTO    35 ES    .0085962845
PROBABILIDAD DE QUE LA VARIABLE ALEATORIA SEA <= 35 ES    .0044729692
PROBABILIDAD DE QUE LA VARIABLE ALEATORIA SEA >= 35 ES    .99552703

```

DESEA REALIZAR MAS CALCULOS PARA OTRO VALOR DE X? S/N ?S

X =? 60

ORDENADA DE LA FUNCION DE DENSIDAD EN EL PUNTO 60 ES .022831136  
 PROBABILIDAD DE QUE LA VARIABLE ALEATORIA SEA  $\leq$  60 ES .89435016  
 PROBABILIDAD DE QUE LA VARIABLE ALEATORIA SEA  $\geq$  60 ES .10564984

DESEA REALIZAR MAS CALCULOS PARA OTRO VALOR DE X? S/N ?N

**2.** La medida de un cerrojo de precisión se distribuye según una normal  $N(10; 0,1)$ . Tomado un cerrojo al azar se desea saber:

- La probabilidad de que a lo sumo mida 10,01 cms.
- La probabilidad de que por lo menos mida 10,01 cms.

**SOLUCION :**

MEDIA  $M = ?$  10  
 DESVIACION TIPICA  $= ?$  .01  
 VALOR DE LA VARIABLE X PARA LA QUE SE DESEA EL CALCULO,  $X = ?$  10.01

ORDENADA DE LA FUNCION DE DENSIDAD EN EL PUNTO 10.01 ES 24.197072  
 PROBABILIDAD DE QUE LA VARIABLE ALEATORIA SEA  $\leq$  10.01 ES .84134474  
 PROBABILIDAD DE QUE LA VARIABLE ALEATORIA SEA  $\geq$  10.01 ES .15865526

**3.** Las notas de los alumnos de una clase se distribuyen según una normal de parámetros  $N(6.5; 1.3)$ . Tomado un alumno al azar se desea saber:

- La probabilidad de que a lo sumo tenga una calificación igual a 7.2.
- La probabilidad de que por lo menos tenga una calificación igual a 7.2.

**SOLUCION :**

MEDIA  $M = ?$  6.5  
 DESVIACION TIPICA  $= ?$  1.3  
 VALOR DE LA VARIABLE X PARA LA QUE SE DESEA EL CALCULO,  $X = ?$  7.2

ORDENADA DE LA FUNCION DE DENSIDAD EN EL PUNTO 7.2 ES .26546475  
 PROBABILIDAD DE QUE LA VARIABLE ALEATORIA SEA  $\leq$  7.2 ES .7048708  
 PROBABILIDAD DE QUE LA VARIABLE ALEATORIA SEA  $\geq$  7.2 ES .2951292

# DISTRIBUCION CHI-CUADRADO

## introducción:

Este programa calcula las probabilidades de la distribución chi-cuadrado de Pearson, para ello es preciso conocer unicamente los grados de libertad de la distribución.

La función de densidad de la distribución chi-cuadrado viene dada por la siguiente expresión:

$$f(x) = \frac{1}{2^{n/2} \Gamma(\frac{n}{2})} e^{-\frac{x}{2}} x^{\frac{(n}{2}-1)}$$

siendo  $\Gamma$  la función gamma.

El programa permite calcular:

- a) La probabilidad de que la variable aleatoria tome el valor  $x$ .
- b) La probabilidad de que la variable aleatoria tome a lo sumo el valor  $x$ .
- c) La probabilidad de que la variable aleatoria tome por lo menos el valor  $x$ .


## listado del programa:

```
10 REM *****
20 REM
30 REM          D I S T R I B U C I O N      C H I - C U A D R A D O
40 REM
50 REM                      D E      P E A R S O N
60 REM
70 REM *****
100 PRINT
109 REM - ENTRADA DE DATOS -
110 PRINT "NUMERO DE GRADOS DE LIBERTAD";
120 INPUT V
130 PRINT "VALOR DE LA VARIABLE X PARA EL QUE SE DESEA EL CALCULO, X=";
140 INPUT X
149 REM - CALCULO DE LA FUNCION GAMMA -
150 G=1
160 IF V=2*INT(V/2) THEN GOTO 219
169 REM - CALCULO DE G PARA V IMPAR -
```

```

170 IF V=1 THEN GOTO 210
180 FOR I=1 TO 2*INT(V/2)-1 STEP 2
190 G=G*I/2
200 NEXT I
210 G=G*1.772452851
219 REM - CALCULO DE G PARA V PAR -
220 IF V=2 THEN GOTO 260
230 FOR I=1 TO INT(V/2-1)
240 G=G*I
250 NEXT I
259 REM - CALCULO DE LAS PROBABILIDADES -
260 FX=X^(V/2-1)
270 C=EXP(-X/2)
280 D=G*(2^(V/2))
290 FX=FX*C/D
300 FC=(X/2)^(V/2)
310 FC=FC*C/((V/2)*G)
320 FD=1
330 LX=1
340 I=2
350 FD=FD*FX/(V+I)
360 LX=LX+FD
370 IF (FD*FC)>1E-8 THEN 410
380 LX=LX*FC
390 RX=1-LX
400 GOTO 430
410 I=I+2
420 GOTO 350
429 REM - SALIDA DE RESULTADOS -
430 PRINT
440 PRINT
450 PRINT "LA PROBABILIDAD DE QUE LA VARIABLE SEA =" ; X ; "ES " ; FX
460 PRINT "LA PROBABILIDAD DE QUE LA VARIABLE SEA <=" ; X ; "ES " ; LX
470 PRINT "LA PROBABILIDAD DE QUE LA VARIABLE SEA >=" ; X ; "ES " ; RX
480 PRINT
489 REM - MAS CALCULOS -
490 PRINT "DESEA HACER MAS CALCULOS PARA OTRO VALOR DE X? S/N ";
500 INPUT C$
510 IF C$="N" THEN GOTO 560
520 PRINT
530 PRINT "X =";
540 INPUT X
550 GOTO 150
560 END

```



## ejercicios resueltos:

**1.** Para una  $\chi^2$ -cuadrado con 10 grados de libertad ejecutar el programa para  $x=20.4$  y para  $x=15.3$ .

### S O L U C I O N :

NUMERO DE GRADOS DE LIBERTAD? 10

VALOR DE LA VARIABLE X PARA EL QUE SE DESEA EL CALCULO, X=? 20.4

LA PROBABILIDAD DE QUE LA VARIABLE SEA = 20.4 ES .0083821559

LA PROBABILIDAD DE QUE LA VARIABLE SEA <= 20.4 ES .97431154

LA PROBABILIDAD DE QUE LA VARIABLE SEA >= 20.4 ES .025688463

DESEA HACER MAS CALCULOS PARA OTRO VALOR DE X? S/N ?S

X =? 15.3

LA PROBABILIDAD DE QUE LA VARIABLE SEA = 15.3 ES .033966572  
 LA PROBABILIDAD DE QUE LA VARIABLE SEA <= 15.3 ES .87849883  
 LA PROBABILIDAD DE QUE LA VARIABLE SEA >= 15.3 ES .12150117

DESEA HACER MAS CALCULOS PARA OTRO VALOR DE X? S/N ?N

**2.** Ejecutar el programa anterior para una chi-cuadrado con 18 grados de libertad, siendo los valores de x: 36.5; 12.6.

### S O L U C I O N :

NUMERO DE GRADOS DE LIBERTAD? 18

VALOR DE LA VARIABLE X PARA EL QUE SE DESEA EL CALCULO, X=? 36.5

LA PROBABILIDAD DE QUE LA VARIABLE SEA = 36.5 ES .0018099982  
 LA PROBABILIDAD DE QUE LA VARIABLE SEA <= 36.5 ES .99391535  
 LA PROBABILIDAD DE QUE LA VARIABLE SEA >= 36.5 ES .0060846528

DESEA HACER MAS CALCULOS PARA OTRO VALOR DE X? S/N ?S

X =? 12.6

LA PROBABILIDAD DE QUE LA VARIABLE SEA = 12.6 ES .056509133  
 LA PROBABILIDAD DE QUE LA VARIABLE SEA <= 12.6 ES .18523335  
 LA PROBABILIDAD DE QUE LA VARIABLE SEA >= 12.6 ES .81476665

DESEA HACER MAS CALCULOS PARA OTRO VALOR DE X? S/N ?N

**3.** Para una chi-cuadrado con 9 grados de libertad, ejecutar el programa para los valores de x = 3, y de x=7.

### S O L U C I O N :

NUMERO DE GRADOS DE LIBERTAD? 9

VALOR DE LA VARIABLE X PARA EL QUE SE DESEA EL CALCULO, X=? 7

LA PROBABILIDAD DE QUE LA VARIABLE SEA = 7 ES .017353306  
 LA PROBABILIDAD DE QUE LA VARIABLE SEA <= 7 ES .060480133  
 LA PROBABILIDAD DE QUE LA VARIABLE SEA >= 7 ES .93951987

DESEA HACER MAS CALCULOS PARA OTRO VALOR DE X? S/N ?S

X =? 3

LA PROBABILIDAD DE QUE LA VARIABLE SEA = 3 ES .0066077321  
 LA PROBABILIDAD DE QUE LA VARIABLE SEA <= 3 ES .0059508408  
 LA PROBABILIDAD DE QUE LA VARIABLE SEA >= 3 ES .99404916

DESEA HACER MAS CALCULOS PARA OTRO VALOR DE X? S/N ?N

# DISTRIBUCION t DE STUDENT

## introducción:

Este programa sirve para calcular las probabilidades de la distribución t de Student, para lo cual es necesario conocer el número de grados de libertad de la distribución.

La función de densidad de la distribución t de Student viene dada por la siguiente expresión:

$$t_n(x) = \frac{1}{\sqrt{n\pi}} \frac{\Gamma(\frac{n+1}{2})}{\Gamma(\frac{n}{2})} \left(1 + \frac{x^2}{n}\right)^{-(n+1)/2}$$

El programa permite calcular:

- a) La probabilidad de que la variable aleatoria tome el valor x.
- b) La probabilidad de que la variable aleatoria tome a lo sumo el valor x.
- c) La probabilidad de que la variable aleatoria tome por lo menos el valor x.

El programa permite calcular:

- a) La probabilidad de que el valor absoluto de la variable aleatoria sea menor o igual que un valor x.
- b) La probabilidad de que la variable aleatoria sea menor o igual que un valor x.
- c) La probabilidad de que la variable aleatoria sea mayor o igual que un valor x.

## listado del programa:

```
10 REM *****
20 REM
30 REM          D I S T R I B U C I O N      D E      S T U D E N T
40 REM
50 REM *****
100 PRINT
109 REM - ENTRADA DE DATOS -
110 PRINT "NUMERO DE GRADOS DE LIBERTAD";
120 INPUT V
```

```
130 PRINT "VALOR DE LA VARIABLE X PARA EL QUE SE DESEA EL CALCULO, X=";
140 INPUT X
150 X=X/SQR(V)
160 Z=ATN(X)
170 IF V=1 THEN GOTO 430
180 NZ=2*INT(V/2)
190 IF NZ=V THEN GOTO 310
199 REM - CALCULOS PARA V IMPAR -
200 AZ=COS(Z)
210 TZ=AZ^2
220 VZ=AZ
230 IF V=3 THEN GOTO 280
240 FOR I=2 TO V-3 STEP 2
250 VZ=VZ*I*TZ/(I+1)
260 AZ=AZ+VZ
270 NEXT I
280 AZ=AZ*SIN(Z)
290 AZ=(AZ+Z)*.6366197724
300 GOTO 400
309 REM - CALCULOS PARA V PAR -
310 TZ=COS(Z)^2
320 VZ=1
330 AZ=1
340 IF V=2 THEN GOTO 390
350 FOR I=1 TO V-3 STEP 2
360 VZ=VZ*I*TZ/(I+1)
370 AZ=AZ+VZ
380 NEXT I
390 AZ=AZ*SIN(Z)
400 LX=(1+AZ)/2
410 RX=1-LX
420 GOTO 450
430 AZ=Z*.6366197724
440 GOTO 400
449 REM - SALIDA DE RESULTADOS -
450 PRINT
460 PRINT "LA PROBABILIDAD DE (-X <= VARIABLE ALEATORIA <= X) ES ";AZ
470 PRINT "LA PROBABILIDAD DE QUE LA VARIABLE ALEATORIA SEA <= X ES ";LX
480 PRINT "LA PROBABILIDAD DE QUE LA VARIABLE ALEATORIA SEA >= X ES ";RX
490 PRINT
499 REM - MAS CALCULOS -
500 PRINT "DESEA HACER CALCULOS PARA OTRO VALOR DE LA VARIABLE X ? S/N";
510 INPUT C$
520 IF C$="N" THEN GOTO 570
530 PRINT
540 PRINT "X =";
550 INPUT X
560 GOTO 150
570 END
```

## ejercicios resueltos:

1. Para una distribución t de Student con 25 grados de libertad, ejecutar el programa siendo  $x = 1.316$ .

**SOLUCION :**

NUMERO DE GRADOS DE LIBERTAD? 25  
? 1.316

LA PROBABILIDAD DE  $(-X \leq \text{VARIABLE ALEATORIA} \leq X)$  ES = .79988591  
LA PROBABILIDAD DE QUE LA VARIABLE ALEATORIA SEA  $\leq X$  ES = .89994295  
LA PROBABILIDAD DE QUE LA VARIABLE ALEATORIA SEA  $\geq X$  ES = .10005705

DESEA HACER CALCULOS PARA OTRO VALOR DE LA VARIABLE X ? S/N?N

**2.** Ejecutar el programa para una t de Student con 15 grados de libertad, siendo los valores de x:  $x = 1.35$  y  $x = 2.316$ .

**SOLUCION :**

NUMERO DE GRADOS DE LIBERTAD? 15  
? 1.35

LA PROBABILIDAD DE  $(-X \leq \text{VARIABLE ALEATORIA} \leq X)$  ES = .80296346  
LA PROBABILIDAD DE QUE LA VARIABLE ALEATORIA SEA  $\leq X$  ES = .90148173  
LA PROBABILIDAD DE QUE LA VARIABLE ALEATORIA SEA  $\geq X$  ES = .098518271

DESEA HACER CALCULOS PARA OTRO VALOR DE LA VARIABLE X ? S/N?S

X =? 2.316

LA PROBABILIDAD DE  $(-X \leq \text{VARIABLE ALEATORIA} \leq X)$  ES = .9648798  
LA PROBABILIDAD DE QUE LA VARIABLE ALEATORIA SEA  $\leq X$  ES = .9824399  
LA PROBABILIDAD DE QUE LA VARIABLE ALEATORIA SEA  $\geq X$  ES = .017560099

DESEA HACER CALCULOS PARA OTRO VALOR DE LA VARIABLE X ? S/N?N

**3.** Sea una distribución t de Student con 7 grados de libertad, ejecutar el programa anterior para los valores de  $x = 0.025$  y  $x = 0.01$

**SOLUCION :**

NUMERO DE GRADOS DE LIBERTAD? 7  
? .025

LA PROBABILIDAD DE  $(-X \leq \text{VARIABLE ALEATORIA} \leq X)$  ES = .019247281  
LA PROBABILIDAD DE QUE LA VARIABLE ALEATORIA SEA  $\leq X$  ES = .50962364  
LA PROBABILIDAD DE QUE LA VARIABLE ALEATORIA SEA  $\geq X$  ES = .49037636

DESEA HACER CALCULOS PARA OTRO VALOR DE LA VARIABLE X ? S/N?S

X =? .01

LA PROBABILIDAD DE  $(-X \leq \text{VARIABLE ALEATORIA} \leq X)$  ES = .0076996824  
LA PROBABILIDAD DE QUE LA VARIABLE ALEATORIA SEA  $\leq X$  ES = .50384984  
LA PROBABILIDAD DE QUE LA VARIABLE ALEATORIA SEA  $\geq X$  ES = .49615016

DESEA HACER CALCULOS PARA OTRO VALOR DE LA VARIABLE X ? S/N?N



# DISTRIBUCION F DE SNEDECOR

## introducción:

Este programa calcula las probabilidades de la distribución F de Fisher, para lo cual es necesario conocer los grados de libertad de la chi cuadrado que se sitúa en el numerador  $n_1$  y los de la chi-cuadrado del denominador  $n_2$ .

La función de densidad de la distribución F de Fisher viene dada por la siguiente expresión:

$$f_{n_1, n_2}(x) = n_1^{n_1/2} n_2^{n_2/2} \frac{\Gamma(\frac{n_1+n_2}{2})}{\Gamma(\frac{n_1}{2}) \Gamma(\frac{n_2}{2})} \frac{x^{\frac{n_1}{2}-1}}{(n_2+n_1x)^{(n_1+n_2)/2}}$$

El programa permite calcular:

- El area de la región encerrada bajo curva de la función de densidad desde 0 hasta el valor del estadístico F que corresponda en cada caso.
- El area de la región encerrada bajo la curva de la función de densidad desde el valor del estadístico F hasta infinito.

## listado del programa:

```
10 REM *****
20 REM
30 REM          D I S T R I B U C I O N      F      D E      F I S H E R
40 REM
50 REM *****
100 PRINT
110 REM - ENTRADA DE DATOS -
120 PRINT "NUMERO DE GRADOS DE LIBERTAD DEL NUMERADOR      N1 =";
130 INPUT N1
140 PRINT "NUMERO DE GRADOS DE LIBERTAD DEL DENOMINADOR      N2 =";
150 INPUT N2
160 PRINT "VALOR DEL ESTADISTICO F =";
170 INPUT F
180 IF F=0 THEN GOTO 470
189 REM - CALCULO DE LAS PROBABILIDADES POR APROXIMACION -
190 X=1
```

```

200 IF F<1 THEN GOTO 250
210 A=N1
220 B=N2
230 C=F
240 GOTO 280
250 A=N2
260 B=N1
270 C=1/F
280 G1=2/(9*A)
290 G2=2/(9*B)
300 Y=ABS((1-G2)*C^(1/3)-1+G1)/SQR(G2*C^(2/3)+G1)
310 IF B<4 THEN 350
320 X=.5/(1+.196854*Y+.115194*Y^2+.000344*Y^3+.019527*Y^4)^4
330 X= INT(10000*X+.5)/10000
340 GOTO 370
350 Y=Y*(.08*Y^4+1 /B^3)
360 GOTO 320
370 IF F>=1 THEN 390
380 X=1-X
390 PRINT
390 PRINT "EL AREA DE LA REGION ENCERRADA DESDE 0 HASTA";F;"ES      =" ;1-X
400 PRINT "EL AREA DE LA REGION ENCERRADA DESDE";F;"HASTA + INFINITO ES =" ;X
410 PRINT
419 REM - MAS CALCULOS -
420 PRINT "DESEA HACER MAS CALCULOS PARA OTRO VALOR DE F? S/N ";
430 INPUT C$
440 IF C$="N" THEN GOTO 470
450 PRINT
459 REM - MAS CALCULOS -
460 GOTO 120
470 END

```

## ejercicios resueltos:

**1.** Para una F de Fisher de 12 y 16 grados de libertad, ejecutar el programa para el valor del estadístico  $F = 0.845$ .

### SOLUCION :

```

NUMERO DE GRADOS DE LIBERTAD DEL NUMERADOR      N1 =? 12
NUMERO DE GRADOS DE LIBERTAD DEL DENOMINADOR    N2 =? 16
VALOR DEL ESTADISTICO F =? .845

```

```

EL AREA DE LA REGION ENCERRADA DESDE 0 HASTA .845 ES      = .3898
EL AREA DE LA REGION ENCERRADA DESDE .845 HASTA + INFINITO ES = .6102

```

**2.** Para una F de Fisher de 13 y 7 grados de libertad, ejecutar el programa para el valor del estadístico  $F = 4.78$ .

### SOLUCION :

```

NUMERO DE GRADOS DE LIBERTAD DEL NUMERADOR      N1 =? 13
NUMERO DE GRADOS DE LIBERTAD DEL DENOMINADOR    N2 =? 7
VALOR DEL ESTADISTICO F =? 4.78

```

EL AREA DE LA REGION ENCERRADA DESDE 0 HASTA 4.78 ES = .9763  
EL AREA DE LA REGION ENCERRADA DESDE 4.78 HASTA + INFINITO ES = .0237

DESEA HACER MAS CALCULOS PARA OTRO VALOR DE F? S/N ?N

**3.** Ejecutar el programa de la distribución F de Fisher para una que tiene 14 y 9 grados de libertad, siendo  $F = 1.89$ .

**S O L U C I O N :**

NUMERO DE GRADOS DE LIBERTAD DEL NUMERADOR N1 =? 14

NUMERO DE GRADOS DE LIBERTAD DEL DENOMINADOR N2 =? 9

VALOR DEL ESTADISTICO F =? 1.89

EL AREA DE LA REGION ENCERRADA DESDE 0 HASTA 1.89 ES = .8307

EL AREA DE LA REGION ENCERRADA DESDE 1.89 HASTA + INFINITO ES = .1693

# DISTRIBUCION NORMAL DE DOS VARIABLES

## introducción:

Este programa calcula la función de densidad para los valores  $x$  e  $y$  de las dos variables aleatorias, para ello es preciso conocer las medias de ambas variables, las desviaciones típicas de ambas variables y el coeficiente de correlación lineal entre las dos variables.

La función de densidad de la distribución normal con dos variables viene dada por la siguiente expresión:

$$f(x,y) = \frac{1}{2\pi S_X S_Y \sqrt{1-R^2}} \exp - \left[ \frac{\left(\frac{X-MX}{SX}\right)^2 - 2R\left(\frac{X-MX}{SX}\right)\left(\frac{Y-MY}{SY}\right) + \left(\frac{Y-MY}{SY}\right)^2}{2(1-R^2)} \right]$$

## listado del programa:

```
10 REM *****
20 REM
30 REM           D I S T R I B U C I O N           N O R M A L
40 REM
50 REM           D E       D O S       V A R I A B L E S
60 REM
70 REM *****
100 PRINT
109 REM - ENTRADA DE DATOS -
110 PRINT "MEDIA DE LA VARIABLE X,      MX =";
120 INPUT MX
130 PRINT "MEDIA DE LA VARIABLE Y,      MY =";
140 INPUT MY
150 PRINT "DESVIACION TIPICA DE LA VARIABLE X,  SX =";
160 INPUT SX
170 PRINT "DESVIACION TIPICA DE LA VARIABLE Y,  SY =";
180 INPUT SY
190 PRINT "COEFICIENTE DE CORRELACION, R =";
200 INPUT R
210 PRINT
220 PRINT "VALOR DE LA VARIABLE X PARA EL QUE SE DESEA EL CALCULO,  X =";
230 INPUT X
240 PRINT "VALOR DE LA VARIABLE Y PARA EL QUE SE DESEA EL CALCULO,  Y =";
250 INPUT Y
259 REM - TIPIFICACION DE LAS VARIABLES -
260 ZX=(X-MX)/SX
```

```

270 ZY=(Y-MY)/SY
280 RF=2*(1-R^2)
290 VF=ZX^2+ZY^2-2*R*ZX*ZY
300 VF=EXP(-VF/RF)
310 ZX=SX*SY*SQR(2*RF)*PI
320 F=VF/ZX
330 PRINT
339 REM - SALIDA DE RESULTADOS -
340 PRINT "PROBABILIDAD DE QUE X =" ; X ; "Y QUE Y =" ; Y ; "ES =" ; F
350 PRINT
359 REM - MAS CALCULOS -
360 PRINT "DESEA HACER CALCULOS PARA OTROS VALORES DE X E Y ? S/N";
370 INPUT C$
380 IF C$="N" THEN GOTO 400
390 GOTO 210
400 END

```

## ejercicios resueltos:

1. Sea X una variable aleatoria de media 4,5 y desviación típica 1,6 e Y otra variable aleatoria de media 6,8 y desviación típica 2,3 y sea el coeficiente de correlación lineal 0,89. Hallar la probabilidad de que la variable X tome el valor 2 y la variable Y el valor 3.

Idem para X = 2,5 e Y = 3,8

### S O L U C I O N :

```

MEDIA DE LA VARIABLE X,      MX =? 4.5
MEDIA DE LA VARIABLE Y,      MY =? 6.8
DESVIACION TIPICA DE LA VARIABLE X,  SX =? 1.6
DESVIACION TIPICA DE LA VARIABLE Y,  SY =? 2.3
COEFICIENTE DE CORRELACION, R =? .89

VALOR DE LA VARIABLE X PARA EL QUE SE DESEA EL CALCULO,  X =? 2
VALOR DE LA VARIABLE Y PARA EL QUE SE DESEA EL CALCULO,  Y =? 3

PROBABILIDAD DE QUE X = 2 Y QUE Y = 3 ES = .023738314

DESEA HACER CALCULOS PARA OTROS VALORES DE X E Y ? S/N?S

VALOR DE LA VARIABLE X PARA EL QUE SE DESEA EL CALCULO,  X =? 2.5
VALOR DE LA VARIABLE Y PARA EL QUE SE DESEA EL CALCULO,  Y =? 3.8

PROBABILIDAD DE QUE X = 2.5 Y QUE Y = 3.8 ES = .039747501

```

# RECTAS DE REGRESION

## introducción:

Este programa calcula parámetros estadísticos de distribuciones bi-dimensionales, así como la recta de regresión de x sobre y y de y sobre x permitiendo hacer estimaciones de y conocido x y de x conocido y.

Las expresiones de ambas rectas son las siguientes:

Recta de regresión de y sobre x:  $y - \bar{y} = \frac{\text{covarianza}}{\text{varianza } x}(x - \bar{x})$

Recta de regresión de x sobre y:  $x - \bar{x} = \frac{\text{covarianza}}{\text{varianza } y}(y - \bar{y})$

A medida que el coeficiente de determinación  $R^2$  tiende a 1, ambas rectas tienden a confundirse.

## listado del programa:

```
10 REM *****
20 REM
30 REM          D I S T R I B U C I O N E S          B I D I M E N S I O N A L E S
40 REM
50 REM          R E C T A S          D E          R E G R E S I O N
60 REM
70 REM *****
100 PRINT
110 PRINT "NUMERO DE DATOS CONOCIDOS" ;
120 INPUT N
130 DIM S(5)
140 FOR J=1 TO 5
150 S(J)=0
160 NEXT J
170 PRINT
179 REM --ENTRADA DE DATOS--
180 PRINT "INTRODUZCA LA X Y LA Y DE CADA PUNTO"
190 FOR I=1 TO N
200 PRINT "X,Y DEL PUNTO"; I;
210 INPUT X,Y
219 REM --CALCULO DE LAS SUMAS ACUMULADAS--
220 S(1)=S(1)+X
230 S(2)=S(2)+Y
240 S(3)=S(3)+X^2
```

```

250 S(4)=S(4)+Y^2
260 S(5)=S(5)+X*Y
270 NEXT I
280 M1=S(1)/N
290 PRINT
300 PRINT "MEDIA DE LA VARIABLE X =" ; M1
310 M2=S(2)/N
320 PRINT "MEDIA DE LA VARIABLE Y =" ; M2.
330 V1=(S(3)/N)-M1^2
340 PRINT
350 PRINT "VARIANZA DE LA VARIABLE X =" ; V1
360 V2=(S(4)/N)-M2^2
370 PRINT "VARIANZA DE LA VARIABLE Y =" ; V2
380 PRINT
390 C=(S(5)/N)-M1*M2
399 REM -- ESTIMACION DE VALORES --
400 PRINT "COVARIANZA =" ; C
410 A1=C/V1
420 PRINT
430 PRINT "COEFICIENTE DE REGRESION DE Y SOBRE X =" ; A1
440 A2=C/V2
450 PRINT "COEFICIENTE DE REGRESION DE X SOBRE Y =" ; A2
460 PRINT
470 B1=M2-A1*M1
480 B2=M1-A2*M2
490 PRINT "LA RECTA DE REGRESION DE Y SOBRE X ES: Y=" ; A1 ; "X+" ; B1
500 PRINT "LA RECTA DE REGRESION DE X SOBRE Y ES: X=" ; A2 ; "Y+" ; B2
510 PRINT
520 PRINT "EL COEFICIENTE DE CORRELACION LINEAL ES =" ; SQR(A1*A2)
530 PRINT "EL COEFICIENTE DE DETERMINACION ES =" ; A1*A2
539 REM -- ESTIMACION DE VALORES --
540 PRINT
550 PRINT "SI QUIERE ESTIMAR Y CONOCIDO X: PULSE 1"
560 PRINT "SI QUIERE ESTIMAR X CONOCIDO Y: PULSE 2"
570 INPUT L
580 PRINT
590 PRINT "VALORES ESTIMADOS (para acabar el programa pulse 9999)"
600 IF L=1 THEN 670
610 PRINT
620 PRINT "Y=" ;
630 INPUT Y
640 IF Y=9999 THEN 720
650 PRINT "X=" ; A2*Y+B2
660 GOTO 620
670 PRINT "X=" ;
680 INPUT X
690 IF X=9999 THEN 720
700 PRINT "Y=" ; A1*X+B1
710 GOTO 620
720 END

```

## ejercicios resueltos:

1. Se determina la pérdida de actividad de un preparado hormonal en el curso del tiempo y se obtiene el siguiente resultado:

Tiempo (en meses)	1	2	3	4	5
% Activ. restante	90	75	42	30	21

Se desea saber cuando tiempo en meses tendrán que transcurrir para que la actividad restante sea nula.

### **S O L U C I O N :**

NUMERO DE DATOS CONOCIDOS? 5

INTRODUZCA LA X Y LA Y DE CADA PUNTO

X,Y DEL PUNTO 1 ? 1	90
X,Y DEL PUNTO 2 ? 2	75
X,Y DEL PUNTO 3 ? 3	42
X,Y DEL PUNTO 4 ? 4	30
X,Y DEL PUNTO 5 ? 5	21

MEDIA DE LA VARIABLE X = 3

MEDIA DE LA VARIABLE Y = 51.6

VARIANZA DE LA VARIABLE X = 2

VARIANZA DE LA VARIABLE Y = 703.44

COVARIANZA = -36.6

COEFICIENTE DE REGRESION DE Y SOBRE X = -18.3

COEFICIENTE DE REGRESION DE X SOBRE Y = -.052030024

LA RECTA DE REGRESION DE Y SOBRE X ES:  $Y = -18.3 * X + 106.5$

LA RECTA DE REGRESION DE X SOBRE Y ES:  $X = -.052030024 * Y + 5.6847492$

EL COEFICIENTE DE CORRELACION LINEAL ES = .97578145

EL COEFICIENTE DE DETERMINACION ES = .95214944

SI QUIERE ESTIMAR Y CONOCIDO X: PULSE 1

SI QUIERE ESTIMAR X CONOCIDO Y: PULSE 2

? 2

VALORES ESTIMADOS (para acabar el programa pulse 9999)

Y=? 0

X= 5.6847492 meses

**2.** En la siguiente tabla se recogen las medidas de la presión sanguínea máxima y la edad de 12 mujeres.

EDAD (años)	56	42	70	35	64	47	53	49	38	42	68	60
P.máx. (Mm Hg)	147	122	160	118	150	130	146	145	113	145	152	155

Calcular:

- Parámetros estadísticos más importantes de la distribución bidimensional
- Rectas de regresión de y sobre x, y de x sobre y.
- Calcular el valor la presión máxima para mujeres de 30, 45, 70 años de edad.

### **S O L U C I O N :**



NUMERO DE DATOS CONOCIDOS? 12

INTRODUZCA LA X Y LA Y DE CADA PUNTO

X,Y DEL PUNTO 1 ?	56	147
X,Y DEL PUNTO 2 ?	42	122
X,Y DEL PUNTO 3 ?	70	160
X,Y DEL PUNTO 4 ?	35	118
X,Y DEL PUNTO 5 ?	64	150
X,Y DEL PUNTO 6 ?	47	130
X,Y DEL PUNTO 7 ?	53	146
X,Y DEL PUNTO 8 ?	49	145
X,Y DEL PUNTO 9 ?	38	113
X,Y DEL PUNTO 10 ?	42	145
X,Y DEL PUNTO 11 ?	68	152
X,Y DEL PUNTO 12 ?	60	155

MEDIA DE LA VARIABLE X = 52

MEDIA DE LA VARIABLE Y = 140.25

VARIANZA DE LA VARIABLE X = 127

VARIANZA DE LA VARIABLE Y = 220.02083

COVARIANZA = 145.25

COEFICIENTE DE REGRESION DE Y SOBRE X = 1.1437008

COEFICIENTE DE REGRESION DE X SOBRE Y = .66016476

LA RECTA DE REGRESION DE Y SOBRE X ES:  $Y = 1.1437008 * X + 80.777559$

LA RECTA DE REGRESION DE X SOBRE Y ES:  $X = .66016476 * Y - 40.588107$

EL COEFICIENTE DE CORRELACION LINEAL ES = .86892517

EL COEFICIENTE DE DETERMINACION ES = .75503095

SI QUIERE ESTIMAR Y CONOCIDO X: PULSE 1

SI QUIERE ESTIMAR X CONOCIDO Y: PULSE 2

? 1

VALORES ESTIMADOS (para acabar el programa pulse 9999)

X=? 30

Y= 115.08858

X=? 45

Y= 132.24409

X=? 70

Y= 160.83661

X=? 9999

**3.** Los datos siguientes representan los valores de dos constantes biológicas de 12 individuos de una cierta población:

65	50	65	55	70	65	70	55	70	50	50	55	VARIABLE X
85	79	90	85	87	94	98	81	95	76	74	76	VARIABLE Y

Calcular los parámetros estadísticos de una distribución bidimensional. Hallar las rectas de regresión de y sobre x, y de x sobre y.

Estimar que valor de X corresponderá para los valores de  $Y = 80$ ,  $Y = 90$   
 $Y = 100$ .

**SOLUCION :**

NUMERO DE DATOS CONOCIDOS? 12

INTRODUZCA LA X Y LA Y DE CADA PUNTO

X,Y DEL PUNTO 1 ?	65	85
X,Y DEL PUNTO 2 ?	50	79
X,Y DEL PUNTO 3 ?	55	76
X,Y DEL PUNTO 4 ?	65	90
X,Y DEL PUNTO 5 ?	55	85
X,Y DEL PUNTO 6 ?	70	87
X,Y DEL PUNTO 7 ?	65	94
X,Y DEL PUNTO 8 ?	70	98
X,Y DEL PUNTO 9 ?	55	81
X,Y DEL PUNTO 10 ?	70	95
X,Y DEL PUNTO 11 ?	50	76
X,Y DEL PUNTO 12 ?	50	74

MEDIA DE LA VARIABLE X = 60

MEDIA DE LA VARIABLE Y = 85

VARIANZA DE LA VARIABLE X = 62.5

VARIANZA DE LA VARIABLE Y = 59.5

COVARIANZA = 53.75

COEFICIENTE DE REGRESION DE Y SOBRE X = .86

COEFICIENTE DE REGRESION DE X SOBRE Y = .90336134

LA RECTA DE REGRESION DE Y SOBRE X ES:  $Y = .86 * X + 33.4$

LA RECTA DE REGRESION DE X SOBRE Y ES:  $X = .90336134 * Y - 16.785714$

EL COEFICIENTE DE CORRELACION LINEAL ES = .88141407

EL COEFICIENTE DE DETERMINACION ES = .77689076

SI QUIERE ESTIMAR Y CONOCIDO X: PULSE 1

SI QUIERE ESTIMAR X CONOCIDO Y: PULSE 2

? 2

VALORES ESTIMADOS (para acabar el programa pulse 9999)

Y=? 80

X= 55.483193

Y=? 90

X= 64.516807

Y=? 100

X= 73.55042

Y=? 9999

# REGRESION LINEAL

## introducción:

Este programa calcula la recta de regresión de  $y$  sobre  $x$  por el método de los mínimos cuadrados, también permite conocer el coeficiente de correlación lineal  $R$  y el coeficiente de determinación  $R^2$ , así como la varianza y la desviación típica de la estima. Posteriormente se pueden realizar estimas de  $X$  para valores conocidos de  $Y$ , y estimas de  $Y$  para valores conocidos de  $X$ .

## listado del programa:

```
10 REM *****
20 REM
30 REM                      R E G R E S I O N      L I N E A L
40 REM
50 REM *****
100 PRINT
110 PRINT "NUMERO DE DATOS CONOCIDOS";
120 INPUT N
130 DIM S(5)
140 FOR J=1 TO 5
150 S(J)=0
160 NEXT J
170 PRINT
179 REM - ENTRADA DE DATOS -
180 PRINT "INTRODUZCA LAS COORDENADAS X,Y DE CADA PUNTO"
190 PRINT
200 FOR I=1 TO N
210 PRINT "X,Y DEL DATO";I;
220 INPUT X,Y
240 S(1)=S(1)+X
250 S(2)=S(2)+Y
260 S(3)=S(3)+X^2
270 S(4)=S(4)+Y^2
280 S(5)=S(5)+X*Y
290 NEXT I
299 REM - CALCULO DE LOS COEFICIENTES DE LA ECUACION DE REGRESION -
300 C=N*S(5)-S(2)*S(1)
310 D=N*S(3)-S(1)^2
320 B=C/D
330 A=(S(2)-B*S(1))/N
340 PRINT
```

```

350 PRINT "MODELO DE LINEA AJUSTADA"          Y ="; A; "+" ;B; "*X"
360 S(4)=S(4)-S(2)^2/N
370 S(1)=B*(S(5)-S(1)*S(2)/N)
380 S(2)=S(4)-S(1)
390 S(5)=S(1)/S(4)
400 PRINT "COEFICIENTE DE CORRELACION"        =";SQR(S(5))
410 PRINT "COEFICIENTE DE DETERMINACION"      =";S(5)
420 PRINT "VARIANZA DE LA ESTIMACION"        =";S(2)/(N-2)
430 PRINT "DESVIACION TIPICA DE LA ESTIMACION" =";SQR(S(2)/(N-2))
440 PRINT
450 REM -I N T E R P O L A C I O N -
460 PRINT
470 PRINT "ESTIMACION DE Y CONOCIDA X : pulse 1"
480 PRINT "ESTIMACION DE X CONOCIDA Y : pulse 2"
490 INPUT P
500 IF P=2 THEN 580
510 PRINT "VALORES ESTIMADOS: (entre 9999 para acabar el programa)"
520 PRINT
530 PRINT "X =";
540 INPUT X
550 IF X=9999 THEN 700
560 PRINT "Y =";A+B*X
570 GOTO 520
580 PRINT "VALORES ESTIMADOS: (entre 9999 para acabar el programa)"
590 PRINT
600 PRINT "Y =";
610 INPUT Y
620 IF Y= 9999 THEN 700
630 PRINT "X =";(Y-A)/B
640 GOTO 590
700 END

```

## ejercicios resueltos:

1. El mismo enunciado que el numerado con 1 del programa anterior.

### SOLUCION :

NUMERO DE DATOS CONOCIDOS? 12  
INTRODUZCA LAS COORDENADAS X,Y DE CADA PUNTO

X,Y DEL DATO 1 ?	65	85
X,Y DEL DATO 2 ?	50	79
X,Y DEL DATO 3 ?	55	76
X,Y DEL DATO 4 ?	65	90
X,Y DEL DATO 5 ?	55	85
X,Y DEL DATO 6 ?	70	87
X,Y DEL DATO 7 ?	65	94
X,Y DEL DATO 8 ?	70	98
X,Y DEL DATO 9 ?	55	81
X,Y DEL DATO 10 ?	70	95
X,Y DEL DATO 11 ?	50	76
X,Y DEL DATO 12 ?	50	74

MODELO DE LINEA AJUSTADA	Y = 33.4 + .86 *X
COEFICIENTE DE CORRELACION	= .88141407
COEFICIENTE DE DETERMINACION	= .77689076
VARIANZA DE LA ESTIMACION	= 15.93
DESVIACION TIPICA DE LA ESTIMACION	= 3.9912404

ESTIMACION DE Y CONOCIDA X : pulse 1  
 ESTIMACION DE X CONOCIDA Y : pulse 2  
 ? 2  
 VALORES ESTIMADOS: (entre 9999 para acabar el programa)

Y =? 80  
 X = 54.186047

Y =? 90  
 X = 65.813953

Y =? 9999

**2.** Las calificaciones en Matemáticas y Lengua de 6 alumnos viene dada por la siguiente tabla:

MATEMATICAS	2	3	3	4	5	4
LENGUA	6	7	8	8	9	8

Ajustar a esta distribución bidimensional una recta de regresión. Estimar que calificación le correspondería en Lengua a tres alumnos que hubiesen sacado 7, 9, 15 puntos en Matemáticas.

### S O L U C I O N

NUMERO DE DATOS CONOCIDOS? 6  
 INTRODUCZA LAS COORDENADAS X,Y DE CADA PUNTO

X,Y DEL DATO 1 ? 2      6  
 X,Y DEL DATO 2 ? 3      7  
 X,Y DEL DATO 3 ? 3      8  
 X,Y DEL DATO 4 ? 4      8  
 X,Y DEL DATO 5 ? 5      9  
 X,Y DEL DATO 6 ? 4      8

MODELO DE LINEA AJUSTADA       $Y = 4.4848485 + .90909091 * X$   
 COEFICIENTE DE CORRELACION      = .92318618  
 COEFICIENTE DE DETERMINACION      = .85227273  
 VARIANZA DE LA ESTIMACION      = .19696969  
 DESVIACION TIPICA DE LA ESTIMACION = .44381268

ESTIMACION DE Y CONOCIDA X : pulse 1  
 ESTIMACION DE X CONOCIDA Y : pulse 2  
 ? 1  
 VALORES ESTIMADOS: (entre 9999 para acabar el programa)

X =? 7  
 Y = 10.848485

X =? 9  
 Y = 12.666667

X =? 15  
 Y = 18.121212

X =? 9999

**3.** La tabla adjunta muestra el peso en Kgrs y la talla en metros de 8 individuos.

Peso	65	68	69	70	71	72	73	76
Talla	1.67	1.65	1.70	1.69	1.72	1.75	1.78	1.75

Hallar que tallas podemos estimar para cuatro personas cuyos pesos son: 80, 90, 45 y 50 Kgrs. respectivamente.

### S O L U C I O N :

NUMERO DE DATOS CONOCIDOS? 8  
INTRODUZCA LAS COORDENADAS X,Y DE CADA PUNTO

X,Y DEL DATO 1 ? 65	1.67
X,Y DEL DATO 2 ? 68	1.65
X,Y DEL DATO 3 ? 69	1.7
X,Y DEL DATO 4 ? 70	1.69
X,Y DEL DATO 5 ? 71	1.72
X,Y DEL DATO 6 ? 72	1.75
X,Y DEL DATO 7 ? 73	1.78
X,Y DEL DATO 8 ? 76	1.75

MODELO DE LINEA AJUSTADA	Y = .94096149 + .010961539 *X
COEFICIENTE DE CORRELACION	= .82447229
COEFICIENTE DE DETERMINACION	= .67975457
VARIANZA DE LA ESTIMACION	= .00073589733
DESVIACION TIPICA DE LA ESTIMACION	= .027127428

ESTIMACION DE Y CONOCIDA X : pulse 1  
ESTIMACION DE X CONOCIDA Y : pulse 2  
? 1

VALORES ESTIMADOS: (entre 9999 para acabar el programa)

X =? 80  
Y = 1.8178846

X =? 90  
Y = 1.9275

X =? 45  
Y = 1.4342308

X =? 50  
Y = 1.4890384

X =? 9999

# REGRESION EXPONENCIAL

## introducción:

Este programa permite ajustar mediante el método de los mínimos cuadrados, una curva exponencial de la forma  $y = A \cdot e^{Bx}$  a una nube de puntos, que viene dada por las coordenadas (x,y) de los puntos conocidos.

Además de la ecuación de la línea ajustada permite conocer el coeficiente de correlación y el de determinación, así como la varianza y la desviación típica de la estima. Posteriormente permite hacer estimaciones de y conocido el valor de x, o de x conocido el valor de y

## listado del programa:

```
10 REM *****
20 REM
30 REM             REGRESION EXPONENCIAL
40 REM
50 REM *****
100 PRINT
110 PRINT "NUMERO DE DATOS CONOCIDOS";
120 INPUT N
130 DIM S(5)
140 FOR J=1 TO 5
150 S(J)=0
160 NEXT J
170 PRINT
179 REM - ENTRADA DE DATOS -
180 PRINT "INTRODUZCA LAS COORDENADAS X,Y DE CADA PUNTO"
190 PRINT
200 FOR I=1 TO N
210 PRINT "X,Y DEL DATO";I;
220 INPUT X,Y
230 Y=LOG(Y)
240 S(1)=S(1)+X
250 S(2)=S(2)+Y
260 S(3)=S(3)+X^2
270 S(4)=S(4)+Y^2
280 S(5)=S(5)+X*Y
290 NEXT I
299 REM - CALCULO DE LOS COEFICIENTES DE LA ECUACION DE REGRESION -
300 C=N*S(5)-S(2)*S(1)
```

```

310 D=N*S(3)-S(1)^2
320 B=C/D
330 A=(S(2)-B*S(1))/N
340 PRINT
349 REM - SALIDA DE RESULTADOS -
350 PRINT "MODELO DE LINEA AJUSTADA          Y ="; EXP(A); "*EXP(";B;"*X)"
360 S(4)=S(4)-S(2)^2/N
370 S(1)=B*(S(5)-S(1)*S(2)/N)
380 S(2)=S(4)-S(1)
390 S(5)=S(1)/S(4)
400 PRINT "COEFICIENTE DE CORRELACION          ="; SQR(S(5))
410 PRINT "COEFICIENTE DE DETERMINACION        ="; S(5)
420 PRINT "VARIANZA DE LA ESTIMACION          ="; S(2)/(N-2)
430 PRINT "DESVIACION TIPICA DE LA ESTIMACION ="; SQR(S(2)/(N-2))
440 PRINT
450 REM - I N T E R P O L A C I O N -
460 PRINT
470 PRINT "ESTIMACION DE Y CONOCIDA X : pulse 1"
480 PRINT "ESTIMACION DE X CONOCIDA Y : pulse 2"
490 INPUT P
500 IF P=2 THEN 580
510 PRINT "VALORES ESTIMADOS: (entre 9999 para acabar el programa)"
520 PRINT
530 PRINT "X =";
540 INPUT X
550 IF X=9999 THEN 700
560 PRINT "Y ="; EXP(A)*EXP(B*X)
570 GOTO 520
580 PRINT "VALORES ESTIMADOS: (entre 9999 para acabar el programa)"
590 PRINT
600 PRINT "Y =";
610 INPUT Y
620 IF Y= 9999 THEN 700
630 PRINT "X ="; LOG(Y/EXP(A))/B
640 GOTO 590
700 END

```

## ejercicios resueltos:

1. Dados los puntos (5,7; 700); (4,1; 80); (2,5; 15,6); (1,7; 6,45) ajústese una curva exponencial. ¿Que valores de y le corresponderán a cuatro puntos que tienen  $x = 1$ ;  $x = 6$ ;  $x = 9$ ;  $x = 15$ .

### SOLUCION :

NUMERO DE DATOS CONOCIDOS? 4

INTRODUZCA LAS COORDENADAS X,Y DE CADA PUNTO

X,Y DEL DATO 1 ?	5.7	700
X,Y DEL DATO 2 ?	4.1	80
X,Y DEL DATO 3 ?	2.5	15.6
X,Y DEL DATO 4 ?	1.7	6.45

MODELO DE LINEA AJUSTADA	Y = .84395974 *EXP( 1.15879 *X)
COEFICIENTE DE CORRELACION	= .99773642
COEFICIENTE DE DETERMINACION	= .99547796
VARIANZA DE LA ESTIMACION	= .028790835
DESVIACION TIPICA DE LA ESTIMACION	= .16967862



ESTIMACION DE Y CONOCIDA X : pulse 1  
 ESTIMACION DE X CONOCIDA Y : pulse 2  
 ? 1  
 VALORES ESTIMADOS: (entre 9999 para acabar el programa)

X = ? 1  
 Y = 2.6889197

X = ? 6  
 Y = 882.7919

X = ? 9  
 Y = 28551.35

X = ? 15  
 Y = 29865050

X = ? 9999

**2.** El numero de bacterias por  $\text{mm}^2$  crece en función del tiempo con arreglo a la siguiente tabla:

TIEMPO (horas)	1.5	2	3	4	5	6	7	8
Nº de BACTERIAS	24	67	124	256	478	789	1024	1689

Ajústese una curva exponencial a los datos observados. Calcúlese cuantas horas habrán de transcurrir para que el número de bacterias sea:  $y = 100$ ;  $y = 200$ ;  $y = 500$ ;  $y = 1000$ ;  $y = 2000$ .

### S O L U C I O N :

NUMERO DE DATOS CONOCIDOS? 8

INTRODUZCA LAS COORDENADAS X,Y DE CADA PUNTO

X,Y DEL DATO 1 ? 1.5      24  
 X,Y DEL DATO 2 ? 2        67  
 X,Y DEL DATO 3 ? 3        124  
 X,Y DEL DATO 4 ? 4        256  
 X,Y DEL DATO 5 ? 5        478  
 X,Y DEL DATO 6 ? 6        789  
 X,Y DEL DATO 7 ? 7        1024  
 X,Y DEL DATO 8 ? 8        1689

MODELO DE LINEA AJUSTADA       $Y = 17.136994 * \text{EXP}( .60882274 * X )$   
 COEFICIENTE DE CORRELACION    = .9774856  
 COEFICIENTE DE DETERMINACION   = .95547809  
 VARIANZA DE LA ESTIMACION     = .11145635  
 DESVIACION TIPICA DE LA ESTIMACION = .3338508

ESTIMACION DE Y CONOCIDA X : pulse 1  
 ESTIMACION DE X CONOCIDA Y : pulse 2  
 ? 2  
 VALORES ESTIMADOS: (entre 9999 para acabar el programa)

Y =? 100  
X = 2.8972812

Y =? 200  
X = 4.0357853

Y =? 500  
X = 5.5408058

Y =? 1000  
X = 6.6793099

Y =? 2000  
X = 7.817814

# REGRESION LOGARITMICA

## introducción:

Este programa permite ajustar, mediante el método de los mínimos cuadrados, una curva logarítmica de la forma:  $y = A + B \log(x)$ , a una nube de puntos que viene dada por las coordenadas (x,y) de los puntos conocidos.

Además de la ecuación de la línea ajustada, permite conocer el coeficiente de correlación y el de determinación, así como la varianza y la desviación típica de la estima. Posteriormente permite hacer estimaciones de y conocido x, o de x conocido y.

## listado del programa:

```
10 REM *****
20 REM
30 REM           R E G R E S I O N   L O G A R I T M I C A
40 REM
50 REM *****
100 PRINT
110 PRINT "NUMERO DE DATOS CONOCIDOS";
120 INPUT N
130 DIM S(5)
140 FOR J=1 TO 5
150 S(J)=0
160 NEXT J
170 R=0
179 REM - ENTRADA DE DATOS -
180 PRINT "INTRODUZCA LAS COORDENADAS X,Y DE CADA PUNTO"
190 PRINT
200 FOR I=1 TO N
210 PRINT "X,Y DEL DATO";I;
220 INPUT X,Y
230 X=LOG(X)
240 S(1)=S(1)+X
250 S(2)=S(2)+Y
260 S(3)=S(3)+X^2
270 S(4)=S(4)+Y^2
280 S(5)=S(5)+X*Y
```

```

290 NEXT I
299 REM - CALCULO DE LOS COEFICIENTES DE LA ECUACION DE REGRESION -
300 C=N*S(5)-S(2)*S(1)
310 D=N*S(3)-S(1)^2
320 B=C/D
330 A=(S(2)-B*S(1))/N
340 PRINT
349 REM -SALDA DE RESULTADOS -
350 PRINT "MODELO DE LINEA AJUSTADA          Y ="; A;"+";B;"*LOG(X)"
360 S(4)=S(4)-S(2)^2/N
370 S(1)=B*(S(5)-S(1)*S(2)/N)
380 S(2)=S(4)-S(1)
390 S(5)=S(1)/S(4)
400 PRINT "COEFICIENTE DE CORRELACION          =";SQR(S(5))
410 PRINT "COEFICIENTE DE DETERMINACION        =";S(5)
420 PRINT "VARIANZA DE LA ESTIMACION          =";S(2)/(N-2)
430 PRINT "DESVIACION TIPICA DE LA ESTIMACION =";SQR(S(2)/(N-2))
440 PRINT
450 REM -I N T E R P O L A C I O N -
460 PRINT
470 PRINT "ESTIMACION DE Y CONOCIDA X : pulse 1"
480 PRINT "ESTIMACION DE X CONOCIDA Y : pulse 2"
490 INPUT P
500 IF P=2 THEN 580
510 PRINT "VALORES ESTIMADOS: (entre 9999 para acabar el programa)"
520 PRINT
530 PRINT "X =";
540 INPUT X
550 IF X=9999 THEN 700
560 PRINT "Y =";A+B*LOG(X)
570 GOTO 520
580 PRINT "VALORES ESTIMADOS: (entre 9999 para acabar el programa)"
590 PRINT
600 PRINT "Y =";
610 INPUT Y
620 IF Y= 9999 THEN 700
630 PRINT "X =";EXP((Y-A)/B)
640 GOTO 590
700 END

```

## ejercicios resueltos:

1. Como consecuencia de la observación de un experimento se han obtenido los siguientes puntos: (6; 4,5); (7; 4,7); (3; 4); (14; 5,3); (28; 5,9). Ajustese una curva logarítmica y estímesese que valores de y corresponderán a los valores de x siguientes:  $x = 10$ ;  $x = 20$ ;  $x = 50$ .

### S O L U C I O N :

NUMERO DE DATOS CONOCIDOS? 5

X,Y DEL DATO 1 ? 6	4.5
X,Y DEL DATO 2 ? 7	4.7
X,Y DEL DATO 3 ? 3	4
X,Y DEL DATO 4 ? 14	5.3
X,Y DEL DATO 5 ? 28	5.9

MODELO DE LINEA AJUSTADA  
COEFICIENTE DE CORRELACION

$Y = 3.0170728 + .86186428 * \text{LOG}(X)$   
 $= .99877964$

COEFICIENTE DE DETERMINACION = .99756077  
 VARIANZA DE LA ESTIMACION = .0017627493  
 DESVIACION TIPICA DE LA ESTIMACION = .041985108

ESTIMACION DE Y CONOCIDA X : pulse 1  
 ESTIMACION DE X CONOCIDA Y : pulse 2  
 ? 1  
 VALORES ESTIMADOS: (entre 9999 para acabar el programa)

X =? 10  
 Y = 5.0015887

X =? 20  
 Y = 5.5989875

X =? 50  
 Y = 6.3887057

X =? 9999

**2.** La tabla ajunta muestra los valores correspondientes a dos variables X e Y:

X	15	25	35	75	95	125	145	215
Y	12.1	13.4	14.3	16.3	16.8	17.6	17.9	18.9

Ajustese una curva logarítmica y estílese que valores de y corresponderán a los valores de x siguientes: x = 100; x = 10; x = 200.

### **SOLUCION :**

NUMERO DE DATOS CONOCIDOS? 8

X,Y DEL DATO 1 ? 15 12.1  
 X,Y DEL DATO 2 ? 25 13.4  
 X,Y DEL DATO 3 ? 35 14.3  
 X,Y DEL DATO 4 ? 75 16.3  
 X,Y DEL DATO 5 ? 95 16.8  
 X,Y DEL DATO 6 ? 125 17.6  
 X,Y DEL DATO 7 ? 145 17.9  
 X,Y DEL DATO 8 ? 215 18.9

MODELO DE LINEA AJUSTADA  $Y = 5.1766611 + 2.5615401 * \text{LOG}(X)$   
 COEFICIENTE DE CORRELACION = .99985582  
 COEFICIENTE DE DETERMINACION = .99971167  
 VARIANZA DE LA ESTIMACION = .0019274477  
 DESVIACION TIPICA DE LA ESTIMACION = .043902708

ESTIMACION DE Y CONOCIDA X : pulse 1  
 ESTIMACION DE X CONOCIDA Y : pulse 2  
 ? 1  
 VALORES ESTIMADOS: (entre 9999 para acabar el programa)

X =? 100  
 Y = 16.972989

X =? 10  
 Y = 11.074825

X =? 200  
 Y = 18.748514

X =? 9999

# REGRESION POTENCIAL

## introducción:

Este programa permite ajustar, mediante el método de los mínimos cuadrados, una curva potencial de la forma:  $y = A x^B$ , a una nube de puntos que viene dada por las coordenadas (x,y) de los puntos conocidos.

Además de la ecuación de la línea ajustada, permite conocer el coeficiente de correlación y de determinación, así como la varianza y la desviación típica de la estima. Posteriormente permite hacer estimaciones de y conocido x, o de x conocido y.

## listado del programa:

```
10 REM *****
20 REM
30 REM             R E G R E S I O N   P O T E N C I A L
40 REM
50 REM *****
100 PRINT
110 PRINT "NUMERO DE DATOS CONOCIDOS";
120 INPUT N
130 DIM S(5)
140 FOR J=1 TO 5
150 S(J)=0
160 NEXT J
170 PRINT
179 REM - ENTRADA DE DATOS -
180 PRINT "INTRODUZCA LAS COORDENADAS X,Y DE CADA PUNTO"
190 PRINT
200 FOR I=1 TO N
210 PRINT "X,Y DEL DATO";I;
220 INPUT X,Y
225 X=LOG(X)
230 Y=LOG(Y)
240 S(1)=S(1)+X
250 S(2)=S(2)+Y
260 S(3)=S(3)+X^2
270 S(4)=S(4)+Y^2
280 S(5)=S(5)+X*Y
290 NEXT I
299 REM - CALCULO DE LOS COEFICIENTES DE LA ECUACION DE REGRESION -
```

```

300 C=N*S(5)-S(2)*S(1)
310 D=N*S(3)-S(1)^2
320 B=C/D
330 A=(S(2)-B*S(1))/N
340 PRINT
349 REM -SALIDA DE RESULTADOS -
350 PRINT "MODELO DE LINEA AJUSTADA          Y ="; EXP(A); "* X ^"; B
360 S(4)=S(4)-S(2)^2/N
370 S(1)=B*(S(5)-S(1)*S(2)/N)
380 S(2)=S(4)-S(1)
390 S(5)=S(1)/S(4)
400 PRINT "COEFICIENTE DE CORRELACION          ="; SQR(S(5))
410 PRINT "COEFICIENTE DE DETERMINACION        ="; S(5)
420 PRINT "VARIANZA DE LA ESTIMACION           ="; S(2)/(N-2)
430 PRINT "DESVIACION TIPICA DE LA ESTIMACION  ="; SQR(S(2)/(N-2))
440 PRINT
450 REM -I N T E R P O L A C I O N -
460 PRINT
470 PRINT "ESTIMACION DE Y CONOCIDA X : pulse 1"
480 PRINT "ESTIMACION DE X CONOCIDA Y : pulse 2"
490 INPUT P
500 IF P=2 THEN 580
510 PRINT "VALORES ESTIMADOS: (entre 9999 para acabar el programa)"
520 PRINT
530 PRINT "X =";
540 INPUT X
550 IF X=9999 THEN 700
560 PRINT "Y ="; EXP(A)*X^B
570 GOTO 520
580 PRINT "VALORES ESTIMADOS: (entre 9999 para acabar el programa)"
590 PRINT
600 PRINT "Y =";
610 INPUT Y
620 IF Y= 9999 THEN 700
630 PRINT "X ="; (Y/EXP(A))^(1/B)
640 GOTO 590
700 END

```

## ejercicios resueltos:

1. Como consecuencia de la observación de un experimentos se han obtenido los siguientes puntos: (6; 34); (8; 90); (10; 160); (12; 450). Ajustese una curva potencial y estímesese que valores de y corresponderán a los siguientes valores de x:  $x = 2$ ,  $x = 4$ ,  $x = 20$ .

### SOLUCION :

NUMERO DE DATOS CONOCIDOS? 4

INTRODUZCA LAS COORDENADAS X,Y DE CADA PUNTO

X,Y DEL DATO 1 ? 6	34
X,Y DEL DATO 2 ? 8	90
X,Y DEL DATO 3 ? 10	160
X,Y DEL DATO 4 ? 12	450

MODELO DE LINEA AJUSTADA  
COEFICIENTE DE CORRELACION

$Y = .053065158 * X ^ 3.5750813$   
 $= .98879663$

COEFICIENTE DE DETERMINACION = .97771878  
 VARIANZA DE LA ESTIMACION = .039015466  
 DESVIACION TIPICA DE LA ESTIMACION = .19752333

ESTIMACION DE Y CONOCIDA X : pulse 1

ESTIMACION DE X CONOCIDA Y : pulse 2

? 1

VALORES ESTIMADOS: (entre 9999 para acabar el programa)

X =? 2

Y = .6324354

X =? 4

Y = 7.5374228

X =? 20

Y = 2377.3737

**2.** La tabla adjunta muestra los valores de dos variables X e Y.

VARIABLE X	2	3	4	5	6	7
VARIABLE Y	40.23	135.67	320.65	640.68	1234.34	1689.78

Ajustese una curva potencial y estílese que valores de x corresponderán a los siguientes valores de y:  $y = 100$ ,  $y = 500$ ,  $y = 1000$ .

### **SOLUCION :**

NUMERO DE DATOS CONOCIDOS? 6

INTRODUZCA LAS COORDENADAS X,Y DE CADA PUNTO

X,Y DEL DATO 1 ? 2      40.23  
 X,Y DEL DATO 2 ? 3      135.67  
 X,Y DEL DATO 3 ? 4      320.65  
 X,Y DEL DATO 4 ? 5      640.68  
 X,Y DEL DATO 5 ? 6      1234.34  
 X,Y DEL DATO 6 ? 7      1689.78

MODELO DE LINEA AJUSTADA       $Y = 4.8658196 * X ^ 3.0374737$   
 COEFICIENTE DE CORRELACION      = .99934303  
 COEFICIENTE DE DETERMINACION      = .9986865  
 VARIANZA DE LA ESTIMACION      = .0032866898  
 DESVIACION TIPICA DE LA ESTIMACION = .057329659

ESTIMACION DE Y CONOCIDA X : pulse 1

ESTIMACION DE X CONOCIDA Y : pulse 2

? 2

VALORES ESTIMADOS: (entre 9999 para acabar el programa)

Y =? 100

X = 2.7053021

Y =? 500

X = 4.5954849

Y =? 1000

X = 5.7734675



# REGRESION POLINOMIAL

## introducción:

Este programa permite ajustar, mediante el método de los mínimos cuadrados, un polinomio de grado G de la forma:

$$y = a_0 + a_1 x + a_2 x^2 + \dots + a_G x^G$$

a un conjunto de N puntos conocidos por sus coordenadas (x, y).

Al ejecutar el programa se imprimen cada uno de los coeficientes de la ecuación anterior, los coeficientes de correlación y de determinación, así como la varianza y la desviación típica de la estima.

Posteriormente y una vez que se conoce la ecuación del polinomio ajustado, el programa permite realizar estimaciones de y para valores conocidos de x.

El numero de puntos conocidos N ha de ser mayor estrictamente que G+1, ya que de lo contrario, en la linea 740 obtendríamos I negativo o nulo, provocando un error aritmético al sustituir en la linea 810.

## listado del programa:

```
10 REM *****
20 REM
30 REM           A J U S T E       P O L I N O M I A L
40 REM
50 REM *****
100 PRINT
109 REM - ENTRADA DE DATOS -
110 PRINT "NUMERO DE PARES CONOCIDOS";
120 INPUT N
130 PRINT "GRADO DEL POLINOMIO QUE DESEA AJUSTAR";
140 INPUT G
145 IF N<=G+1 THEN GOTO 950
150 DIM X(N),Y(N),A(G+2),B(2*G+1),C(G+1,G+2)
160 B(1)=N
170 PRINT
180 PRINT "INTRODUZCA LAS COORDENADAS X,Y DE CADA PUNTO";
190 PRINT
200 FOR I=1 TO N
```

```

210 PRINT "X,Y DEL DATO";I;
220 INPUT X(I),Y(I)
230 FOR J=2 TO 2*G+1
240 B(J)=B(J)+X(I)^(J-1)
250 NEXT J
260 FOR K=1 TO G+1
270 C(K,G+2)=A(K)+Y(I)*X(I)^(K-1)
280 A(K)=C(K,G+2)
290 NEXT K
300 A(G+2)=A(G+2)+Y(I)^2
310 NEXT I
320 FOR J=1 TO G+1
330 FOR K=1 TO G+1
340 C(J,K)=B(K+J-1)
350 NEXT K
360 NEXT J
370 FOR J=1 TO G+1
380 K=J
390 IF C(K,J)=0 THEN 580
400 FOR I=1 TO G+2
410 D=C(J,I)
420 C(J,I)=C(K,I)
430 C(K,I)=D
440 NEXT I
450 H=1/C(J,J)
460 FOR I=1 TO G+2
470 C(J,I)=H*C(J,I)
480 NEXT I
490 FOR K=1 TO G+1
500 IF K=J THEN 550
510 H=-C(K,J)
520 FOR I=1 TO G+2
530 C(K,I)=C(K,I)+H*C(J,I)
540 NEXT I
550 NEXT K
560 NEXT J
570 GOTO 620
580 K=K+1
590 IF K<=G+1 THEN 390
600 PRINT "LA SOLUCION NO ES UNICA"
610 GOTO 960
620 PRINT
629 REM - SALIDA DE RESULTADOS -
630 FOR J=0 TO G
640 PRINT "COEFICIENTE DE GRADO";J;"=";C(J+1,G+2)
650 NEXT J
660 PRINT
670 PRINT
679 REM - CALCULO DE LOS PARAMETROS DEL AJUSTE -
680 L=0
690 FOR J=2 TO G+1
700 M=A(J)-B(J)*A(1)/N
710 L=L+C(J,G+2)*M
720 NEXT J
730 P=A(G+2)-A(1)^2/N
740 I=N-G-1
750 H=P-L
760 R=L/P
770 PRINT
780 PRINT "COEFICIENTE DE CORRELACION" "=";SQR(R)
790 PRINT "COEFICIENTE DE DETERMINACION" "=";R
800 PRINT "VARIANZA DE LA ESTIMA" "=";H/I
810 PRINT "DESVIACION TIPICA DE LA ESTIMA" "=";SQR(H/I)

```

```

820 PRINT
829 REM - INTERPOLACION -
830 PRINT "VALORES ESTIMADOS: (ENTRE 9999 PARA ACABAR EL PROGRAMA)"
840 PRINT
850 PRINT "X=";
860 INPUT X
870 IF X=9999 THEN 960
880 L=C(1,G+2)
890 FOR J=1 TO G
900 L=L+C(J+1,G+2)*X^J
910 NEXT J
920 PRINT "Y =" ;L
930 PRINT
940 GOTO 840
950 PRINT "NO SE PUEDE AJUSTAR MEDIANTE UN POLINOMIO DE GRADO";G
960 END

```

## ejercicios resueltos:

1. Ajustar una parábola a los datos siguientes:

X	50	52	54	56	58	60	62
Y	31	50	77	96	112	130	162

Estimar que valor le corresponderá a y para los valores de x siguientes:  
 $x = 75$ ,  $x = 100$ .

### SOLUCION :

NUMERO DE PARES CONOCIDOS? 7

GRADO DEL POLINOMIO QUE DESEA AJUSTAR? 2

INTRODUZCA LAS COORDENADAS X,Y DE CADA PUNTO

X,Y DEL DATO 1 ?	50	31
X,Y DEL DATO 2 ?	52	50
X,Y DEL DATO 3 ?	54	77
X,Y DEL DATO 4 ?	56	96
X,Y DEL DATO 5 ?	58	112
X,Y DEL DATO 6 ?	60	130
X,Y DEL DATO 7 ?	62	162

COEFICIENTE DE GRADO 0 = -364.00004

COEFICIENTE DE GRADO 1 = 5.8333347

COEFICIENTE DE GRADO 2 = .041666655

COEFICIENTE DE CORRELACION = .99630943

COEFICIENTE DE DETERMINACION = .99263248

VARIANZA DE LA ESTIMA = 22.916667

DESVIACION TIPICA DE LA ESTIMA = 4.7871355

VALORES ESTIMADOS: (ENTRE 9999 PARA ACABAR EL PROGRAMA)

X=? 75

Y = 307.875

X=? 100  
Y = 635.99998

X=? 9999

**2.** Ajustar una cúbica a los datos siguientes: (4, 12), (5, 16), (6, 21), (7, 25), (8, 31). Posteriormente, estimar los valores de y correspondientes para  $x = 6$ ,  $x = 10$ ,  $x = 20$ .

**S O L U C I O N :**

NUMERO DE PARES CONOCIDOS? 5  
GRADO DEL POLINOMIO QUE DESEA AJUSTAR? 3

INTRODUZCA LAS COORDENADAS X,Y DE CADA PUNTO

X,Y DEL DATO 1 ? 4	12
X,Y DEL DATO 2 ? 5	16
X,Y DEL DATO 3 ? 6	21
X,Y DEL DATO 4 ? 7	25
X,Y DEL DATO 5 ? 8	31

COEFICIENTE DE GRADO 0 = -16.214296  
COEFICIENTE DE GRADO 1 = 10.845243  
COEFICIENTE DE GRADO 2 = -1.2857152  
COEFICIENTE DE GRADO 3 = .083333384

COEFICIENTE DE CORRELACION = .9991953  
COEFICIENTE DE DETERMINACION = .99839125  
VARIANZA DE LA ESTIMA = .35714281  
DESVIACION TIPICA DE LA ESTIMA = .59761426

VALORES ESTIMADOS: (ENTRE 9999 PARA ACABAR EL PROGRAMA)

X=? 6  
Y = 20.571429

X=? 10  
Y = 47.000003

X=? 20  
Y = 353.07157

X=? 9999

**3.** Ajústese un polinomio de grado 6° al conjunto siguiente de datos: (2, 7), (3, 9), (4, 12), (5, 12), (6, 12), (7, 14), (8, 17), (9, 18), (10, 21).

¿Que valores de y se estiman para puntos que tienen:  $x = 10$ ,  $x = 20$ ,  $x = 25$ ?

**S O L U C I O N :**

NUMERO DE PARES CONOCIDOS? 9

GRADO DEL POLINOMIO QUE DESEA AJUSTAR? 6

INTRODUZCA LAS COORDENADAS X,Y DE CADA PUNTO

X,Y DEL DATO 1 ? 2	7
X,Y DEL DATO 2 ? 3	9
X,Y DEL DATO 3 ? 4	12
X,Y DEL DATO 4 ? 5	12
X,Y DEL DATO 5 ? 6	12
X,Y DEL DATO 6 ? 7	14
X,Y DEL DATO 7 ? 8	17
X,Y DEL DATO 8 ? 9	18
X,Y DEL DATO 9 ? 10	21

COEFICIENTE DE GRADO 0 = 141.19584  
COEFICIENTE DE GRADO 1 = -189.70576  
COEFICIENTE DE GRADO 2 = 102.17999  
COEFICIENTE DE GRADO 3 = -26.894301  
COEFICIENTE DE GRADO 4 = 3.716685  
COEFICIENTE DE GRADO 5 = -.25849862  
COEFICIENTE DE GRADO 6 = .0071361728

COEFICIENTE DE CORRELACION = .99997832  
COEFICIENTE DE DETERMINACION = .99995665  
VARIANZA DE LA ESTIMA = .0034298301  
DESVIACION TIPICA DE LA ESTIMA = .058564751

VALORES ESTIMADOS: (ENTRE 9999 PARA ACABAR EL PROGRAMA)

X=? 10  
Y = 20.997296

X=? 20  
Y = 46253.754

X=? 25  
Y = 308696.8

X=? 9999



## **bibliografía**

- AGUADO, R. y otros., "BASIC básico, Curso de programación", 1983.
- AGUADO, R. y otros., "Programas comentados de BASIC básico", 1983.
- BAJPAI, A.C., "Statistical Methods for Scientists and Engineers", J.Wiley & Sons, 1978.
- BECKETT, R. & HURT, J., "Numerical Calculations and Algorithms", Mc Graw-Hill, New York, 1967.
- BEYER, W. H., "Standard Mathematical Tables", CRC Press, 1976.
- CARNAHAN, B. & WILKES, J.O., "Applied Numerical Methods", University of Michigan, 1978.
- CUTHBERT, D. & WOOD, F., "Fitting Equations to Data", J. Wiley & Sons, 1980.
- FARRANDO, R., "109 Programas para ordenadores personales y calculadoras", Marcombo, Barcelona, 1983.
- GALLEGO, A.E. y otros., "Informática BASIC. Aplicaciones", Ediciones S.M., Madrid, 1983.
- GOTTFRIED, B.S., "Programación BASIC", Mc Graw-Hill, Mexico, 1982.
- HACQUES, G. , "Mathematiques pour l'informatique", A. Colin, Paris 1971.
- HARNETT, "Introduction to Statistical Methods", Addison-Wesley, 1975.
- HAUT, H., "Mathematiques et statistiques", Editions P.S.I., 1983.
- HUBIN, W.N., "BASIC Programming for Scientists and Engineers", Prentice-Hall Inc., Englewood Cliffs, 1978.
- KEMENY, J.G. & KURTZ, T.E., "Programación BASIC", Compañía Editorial Continental, Mexico, 1983

- KUO, S., "Computer Applications of Numerical Methods", Addison-Wesley, Reading MA, 1972.
- NOWAKOWSKI, C., "Methodes de Calcul Numerique" tome 1, Editions P.S.I., 1984
- POOLE, L. & BORCHERS, M., "Some Common BASIC Programs", Adam Osborne & Associates, Berkeley CA, 1977.
- RUCKDESCHEL, F.R., "BASIC Scientific Subroutines, Vol. 1 y 2", BYTE and Mc Graw-Hill, Peterborougj, NH, 1981.
- SHOUP, T.E., "A Practical Guide to Computer Methods for Engineers", Prentice-Hall, 1979.
- STOER, J. & BULIRSCH., "Introduction to Numerical Analysis", Springer-Verlag, 1980.
- VIZMANOS, J.R. y ANZOLA, M., "Matematicas de C.O.U. Tomos 1,2", Madrid, 1978.
- VIZMANOS, J.R., ANZOLA, M., PRIMO, A., "FUNCIONES 1, 2 y 3", Ediciones S.M. Madrid, 1981.
- VIZMANOS, J.R. y ASENSIO, R., "Curso y Ejercicios de Bioestadística" Madrid, 1976.
- VIZMANOS, J.R. y CRESPO, D. , "Problemas resueltos de variable compleja", Madrid, 1977.



# índice de programas

## I) ANALISIS MATEMATICO

### 1.- MATEMATICA FINANCIERA

1.1 Interés simple .....	9
1.2 Interés compuesto .....	12
1.3 Tablas de interes compuesto .....	15
1.4 Anualidades de amortización .....	17
1.5 Anualidades de capitalización .....	20

### 2.- SUCESIONES DE NUMEROS REALES

2.1 Sucesiones (progresiones) aritméticas .....	23
2.2 Sucesiones (progresiones) geométricas .....	27
2.3 Formación de los términos de una sucesión .....	31
2.4 Sucesión del número e .....	33
2.5 Operaciones con sucesiones .....	35
2.6 Suma de términos de una sucesión .....	38
2.7 Sucesión de Fibonacci .....	42

### 3.- NUMEROS COMPLEJOS

3.1 Operaciones con números complejos en forma binómica .....	44
3.2 Conversión de coordenadas cartesianas a polares y viceversa .....	47
3.3 Potenciación y redicación de números complejos en forma binómica .....	50
3.4 Operaciones con números complejos en forma polar .....	54

### 4.- CALCULO DE RAICES DE UNA ECUACION

4.1 Ecuación algebraica de 2º grado .....	59
4.2 Búsqueda de raíces reales: Método de bisección .....	61
4.3 Búsqueda de raíces reales: Método de Newton .....	66
4.4 Búsqueda de raíces reales: Método de la secante .....	69
4.5 Búsqueda de raíces reales: Método de la regla falsi .....	72
4.6 Búsqueda de raíces reales: Método de Aitken .....	76

### 5.- FUNCIONES REALES DE VARIABLE REAL

5.1 Tabulación de una función real de variable real .....	80
5.2 Operaciones con funciones reales de variable real .....	83
5.3 Composición de funciones reales de variable real .....	86
5.4 Función exponencial .....	89
5.5 Cambio de base logarítmica .....	91
5.6 Función logarítmica .....	93
5.7 Funciones trigonométricas directas .....	95
5.8 Funciones trigonométricas inversas .....	99
5.9 Razones trigonométricas de un ángulo dado .....	103

5.10	Funciones trigonométricas recíprocas .....	105
5.11	Valor de las funciones trigonométricas recíprocas .....	109
5.12	Funciones hiperbólicas directas .....	112
5.13	Funciones hiperbólicas inversas .....	116
5.14	Razones hiperbólicas de un ángulo dado .....	119
5.15	Funciones hiperbólicas recíprocas .....	121

## 6.- CALCULO DIFERENCIAL

6.1	Valor numérico de la primera derivada de una función en un punto .....	125
6.2	Ecuaciones de la tangente y la normal a una curva en un punto, crecimiento y decrecimiento de una función en un punto. ....	127
6.3	Valor numérico de un polinomio y sus derivadas en un punto. ....	131
6.4	Aproximación de la función $y = \text{sen } x$ .....	135
6.5	Aproximación de la función $y = \text{cos } x$ .....	138
6.6	Aproximación de la función $y = \ln x$ .....	140
6.7	Aproximación de la función $y = e^x$ .....	142

## 7.- INTEGRACION NUMERICA

7.1	Regla de Simpson (función conocida) .....	144
7.2	Regla de Simpson (función desconocida) .....	147
7.3	Regla del trapecio .....	150
7.4	Area encerrada por una curva en cartesianas .....	153
7.5	Area encerrada por una curva en polares .....	156
7.6	Area encerrada por una curva en paramétricas .....	159
7.7	Longitud de un arco de curva dada en cartesianas .....	162
7.8	Longitud de un arco de curva dada en polares .....	165
7.9	Longitud de un arco de curva dada en paramétricas .....	168
7.10	Area de revolución .....	171
7.11	Volumen de revolución .....	174
7.12	Regla de Tchebicheff .....	177
7.13	Regla de Bode .....	179
7.14	Regla de Gauss .....	181
7.15	Integrales impropias con extremo superior infinito .....	183

## 8.- INPERPOLACION

8.1	Interpolación lineal .....	185
8.2	Interpolación de Lagrange .....	187

## II) CALCULO DE PROBABILIDADES Y ESTADISTICA

### 9.- GENERACION DE NUMEROS ALEATORIOS

9.1	Entre 0 y 1 .....	192
9.2	Entre 0 y 100 (parte entera) .....	194
9.3	Que se distribuyen según una binomial $B(n,p)$ .....	196
9.4	Que se distribuyen según una Poisson $P(L)$ .....	198
9.5	Que se distribuyen según una normal $N(m,s)$ .....	200

10.-	SIMULACION DE EXPERIMENTOS ALEATORIOS	
10.1	Lanzamiento de un dado con recuento de frecuencias .....	202
10.2	Lanzamiento de dos dados (suma de puntos) .....	205
10.3	Lanzamiento de tres dados (suma de puntos) .....	207
10.4	Lanzamiento de una moneda con recuento de frecuencias ....	209
10.5	Lanzamiento de dos monedas con recuento de frecuencias ...	211
10.6	Extracciones sucesivas con reemplazamiento de cartas de una baraja española .....	213
10.7	Extracciones sucesivas sin reemplazamiento de cartas de una baraja española .....	215
11.-	ESTADISTICA DESCRIPTIVA	
11.1	Medidas de centralización, dispersión y forma de distribu- ciones unidimensionales de datos no agrupados .....	217
11.2	Medidas de centralización, dispersión y forma de distribu- ciones unidimensionales de datos agrupados .....	222
11.3	Parámetros estadísticos de distribuciones bidimensionales de datos no agrupados .....	227
11.4	Parámetros estadísticos de distribuciones bidimensionales de datos agrupados .....	231
12.-	CALCULO DE PROBABILIDADES	
12.1	Probabilidad total .....	234
12.2	Probabilidad bayesiana .....	237
13.-	DISTRIBUCIONES UNIDIMENSIONALES DISCRETAS	
13.1	Distribución binomial .....	241
13.2	Distribución de Poisson .....	244
13.3	Distribución hipergeométrica .....	247
14.-	DISTRIBUCIONES UNIDIMENSIONALES CONTINUAS	
14.1	Distribución normal .....	250
14.2	Distribución chi-cuadrado .....	253
14.3	Distribución t de Student .....	256
14.4	Distribución F de Snedecor .....	259
15.-	DISTRIBUCIONES BIDIMENSIONALES CONTINUAS	
15.1	Distribución normal con dos variables .....	262
16.-	REGRESION Y CORRELACION	
16.1	Rectas de regresión .....	264
16.2	Regresión lineal .....	269
16.3	Regresión exponencial .....	273
16.4	Regresión logarítmica .....	277
16.5	Regresión potencial .....	280
16.6	Regresión polinomial .....	283

# PROGRAMAS EN BASIC PARA LA RESOLUCION DE PROBLEMAS

FP - BUP - COU - UNIVERSIDAD

Autor: J.R. VIZMANOS

## TOMO 1 - ALGEBRA Y GEOMETRIA

### ALGEBRA

- 1.- **Sistemas de numeración.** 1.1 Cambio de base de sistema de numeración. 1.2 Operaciones en cualquier sistema de numeración.
- 2.- **Ecuaciones diofánticas.** 2.1 Ecuaciones diofánticas del tipo:  $Ax + By = C$ . 2.2 Ecuaciones diofánticas del tipo:  $x^2 - y^2 = A$ . 2.3 Ecuaciones diofánticas del tipo:  $x^2 + y^2 = z^2$ . 2.3.1 Números pitagóricos consecutivos. 2.3.2 Ternas pitagóricas.
- 3.- **Divisibilidad.** 3.1. Descomposición de un número en factores primos. 3.2 M.C.D. y M.C.M. de dos números. 3.3 M.C.D. y M.C.M. de tres números. 3.4 Lista de números primos. 3.5 Formación ordenada de los divisores de un número. 3.6 Número de divisores de un número (suma y producto). 3.7 Números perfectos pares. 3.8 Números amigos. 3.9 Números de Mersenne. 3.10 Números de Fermat. 3.11 Múltiplos de un número K. 3.12 Múltiplos de K entre dos números dados. 3.13 Múltiplos comunes de dos números.
- 4.- **Números congruentes.** 4.1 Clases de restos módulo M. 4.2 Exponente de P en M! (regla de Legendre). 4.3 Restos potenciales. 4.4 Cálculo de la última cifra de una potencia. 4.5 Cálculo del resto de dividir una potencia por M. 4.6 Restos cuadráticos.
- 5.- **Combinatoria.** 5.1 Factorial de un número. 5.2 Número de cifras del factorial de un número. 5.3 Factorial de un número con todas sus cifras. 5.4 Combinaciones sin repetición. 5.5 Variaciones sin repetición. 5.6 Formación ordenada de las variaciones sin repetición. 5.7 Permutaciones sin repetición. 5.8 Formación ordenada de las permutaciones sin repetición. 5.9 Combinatoria sin repetición. 5.10 Combinaciones con repetición. 5.11 Variaciones con repetición. 5.12 Formación ordenada de las variaciones con repetición. 5.13 Permutaciones con repetición. 5.14 Triángulo de Tartaglia. 5.15 Binomio de Newton.
- 6.- **Polinomios.** 6.1 Valor numérico de un polinomio. 6.2 Suma y diferencia de polinomios. 6.3 Producto de un escalar por un polinomio. 6.4 Producto de un polinomio por el binomio  $(x + a)$ . 6.5 Producto de polinomios. 6.6 Regla de Ruffini. 6.7 Cociente de polinomios.
- 7.- **Espacios vectoriales.** 7.1 Coordenadas de un vector respecto de una base. 7.2 Cambio de base.
- 8.- **Matrices y Determinantes.** 8.1 Suma de las filas y las columnas de una matriz. 8.2 Matriz transpuesta. 8.3 Suma y diferencia de matrices. 8.4 Producto de un escalar por una matriz. 8.5 Producto de matrices. 8.6 Potencia de una matriz. 8.7 Determinante de una matriz cuadrada. 8.8 Matriz inversa. 8.9 Descomposición de una matriz cuadrada en suma de dos matrices una simétrica y otra hemisimétrica. 8.10 Resolución de sistemas de ecuaciones lineales. 8.11 Ecuación característica de una matriz. 8.12 Valores propios de una matriz. 8.13 Vectores propios de una matriz.

### GEOMETRIA

- 1.- **Geometría elemental.** 1.1 Áreas de triángulos y cuadriláteros. 1.2 Longitudes y áreas en la circunferencia. 1.3 Relaciones métricas en el triángulo. 1.4 Polígonos regulares: ángulos y diagonales. 1.5 Polígonos regulares: área en función del lado. 1.6 Polígonos regulares: área en función del radio de la circunferencia circunscrita. 1.7 Áreas y volúmenes de los poliedros regulares. 1.8 Áreas y volúmenes de cuerpos geométricos.
- 2.- **Plano afín.** 2.1 Operaciones con vectores en el plano afín. 2.2 Cambio de sistema de referencia en el plano afín. 2.3 Determinación de rectas en el plano afín. 2.4 Incidencia, paralelismo e intersección en el plano afín.
- 3.- **Plano afín euclideo.** 3.1 Conversión de ángulos. 3.2 Operaciones con vectores en el plano afín euclideo. 3.3 Ángulos y bisectrices de rectas en el plano afín euclideo. 3.4 Distancias en el plano afín euclideo. 3.5 Áreas en el plano afín euclideo. 3.6 Cálculo de puntos notables en un triángulo. 3.7 Resolución de triángulos rectángulos. 3.8 Resolución de triángulos cualesquiera.
- 4.- **Transformaciones en el plano.** 5.1 Movimientos en el plano. 5.2 Homotecia en el plano. 5.3 Semejanza en el plano.
- 5.- **Cónicas.** 5.1 Circunferencia. 5.2 Potencia de un punto respecto de una circunferencia. 5.3 Elipse. 5.4 Hipérbola. 5.5 Parábola. 5.6 Clasificación y reducción de cónicas.
- 6.- **Espacio afín.** 6.1 Operaciones con vectores en el espacio afín. 6.2 Cambio de sistema de referencia en el espacio afín. 6.3 Determinación de rectas en el espacio afín. 6.4 Determinación de planos en el espacio afín. 6.5 Incidencia, paralelismo e intersección en el espacio afín.
- 7.- **Espacio afín euclideo.** 7.1 Operaciones con vectores en el espacio afín euclideo. 7.2 Ángulos en el espacio afín euclideo. 7.3 Distancias en el espacio afín euclideo. 7.4 Áreas y volúmenes en el espacio afín euclideo.
- 8.- **Transformaciones en el espacio.** 8.1 Movimientos en el espacio. 8.2 Homotecia en el espacio. 8.3 Semejanza en el espacio.

ANEXO. I. Ordenación alfabética de nombres. II. Ordenación ascendente de números.

PEDIDOS A: EDICIONES VIZMANOS-ANZOLA

Príncipe de Vergara, 130 - 4.º A

Tel.: 262 12 47 - 279 70 47

28002-MADRID



